"Shuffling Around" by "Ned Yompus"

volume 1 - the main program

book debug modules - not for release

include Object Response Tests by Juhana Leinonen.

include Property Checking by Emily Short.

include Direction Response Testing by Andrew Schultz.

when play begins (this is the debug version info that should not be in the release rule):
	[repeat with X running through things:
		if X is not scenery and x is not fixed in place:
			say "[X] is [if X is not fixed in place]not [end if]fixed in place[if number of things containing X is 1]--it's part of something[end if].";
	repeat with X running through warpable things:
		say "[X] is warpable.";
	repeat with X running through things:
		if X is a person:
			say "[X] is a person.";]
	[end temporary tests]
	now debug-state is true;
	now debug-print is true;
	[now debug-state is false;
	now debug-print is false;]

book inform 6 stubs

chapter pronoun setting

[This allows us to refer to a plural noun as it/them. Thanks to Climbingstars!]

To set the/-- pronoun it to (O - an object): (- LanguagePronouns-->3 = {O}; -).
To set the/-- pronoun him to (O - an object): (- LanguagePronouns-->6 = {O}; -).
To set the/-- pronoun her to (O - an object): (- LanguagePronouns-->9 = {O}; -).
To set the/-- pronoun them to (O - an object): (- LanguagePronouns-->12 = {O}; -).

section setting specific pronouns

after doing something with a flower:
	set the pronoun them to the noun;
	continue the action;

chapter transcripting

[This makes a check for if the transcript is on. I use it to check if a person starts with * but transcripting is off. Thanks to Zarf!]

Include (-
[ CheckTranscriptStatus;
#ifdef TARGET_ZCODE;
return ((0-->8) & 1);
#ifnot;
return (gg_scriptstr ~= 0);
#endif;
];
-).

To decide whether currently transcripting: (- CheckTranscriptStatus() -)

book d = debug-say

to d (a - indexed text):
	if debug-print is true:
		say "DEBUG INFO: [a]"

book globals

the release number is 2.

release along with cover art and an interpreter. [and a website?]

use American dialect.

use MAX_STATIC_DATA of 300000.

use MAX_ACTIONS of 225.

use MAX_VERBS of 280.

include Conditional Undo by Jesse McGrew.

include Basic Screen Effects by Emily Short.

include Punctuation Removal by Emily Short.

include Glulx Status Window Control by Erik Temple.

include Flexible Windows by Jon Ingold.

the story description is "Yorpwald's a weirdly-named land. Perfect for your weird powers to save it."

the story headline is "Putzing around is unmagical. Shuffling around, though..."

the story genre is "Humor/wordplay"

[Index map with an EPS file.

index map with sorting area mapped east of Silent Rut.]

[release along with a solution.]

debug-state is a truth state that varies. debug-state is usually false. [!!!]

a thing has text called rgtext. rgtext of a thing is usually "[bug-report]".

a thing has a number called wordnum. the wordnum of a thing is usually 0.

a thing has a number called rpos. a thing has a number called gpos. a thing has a number called legth. a thing has a number called lgth.

a thing can be padded. a thing is usually not padded.

an undesc is a kind of thing. The description of an undesc is usually "BUG".

a portal is a kind of thing. a portal is usually fixed in place.

a portal can be fake-enterable. a portal is usually fake-enterable.

a portal can be checkedoff or available. a portal is usually available.

a portal can be enter-clued. a portal is usually not enter-clued.

a thing can be examined or unexamined. a thing is usually unexamined.

a thing can be flipped-yet. a thing is usually not flipped-yet.

a thing can be flippable, auxiliary or inflexible. a thing is usually inflexible.

a disguise-piece is a kind of thing. a disguise-piece has a number called elevation.

a flower is a kind of thing. understand "flower" and "flowers" as a flower.

when play begins (this is the status window rule):
	set the background of the status window to colored;
	set the background color of the status window to g-black;
	set the text color of the status window to g-white;
	open the status window;

a direction can be diagonal. a direction is usually not diagonal.

a direction can be ordinal. a direction is usually not ordinal.

northeast,southeast,northwest,southwest are diagonal.

north,east,south,west are ordinal.

before going (this is the reject diagonals rule):
	if location is roman manor:
		say "[one of]You don't feel like getting up. Suddenly you're in the middle of an exitstential crisis.[or]You've had it with doing and going[or]Sit, ex-adventurer.[or]Where would you go? Texis?[or]Seeing exits just makes you want to...[or]Come on, this one's just switching TWO WHOLE LETTERS. You had other tougher ones to MAKE it here![stopping]" instead;
	if noun is a diagonal:
		if player is in busiest subsite:
			say "You walk around the perimeter of the big hall. It's weird. The passage by the sign isn't stuck in a corner--it's just off to the side a bit, so it's not really north or whatever. You should be able to ENTER it or go INSIDE." instead;
		if trips strip is visited:
			say "[one of]Intermediate? Terminate! Die![or]Diagonals?! A sad lingo.[or]Diagonals?! So anal. Dig?[at random]" instead;

check dropping:
	if noun is shotgun:
		say "You[if wolves are visible] probably [otherwise]'re going to [end if] need that gun." instead;
	say "You shouldn't need to drop [if noun is plural-named]those[otherwise]that[end if]. You have enough hands and pockets." instead;

Procedural rule: ignore the print final score rule.

description of player is "[player-descrip][run paragraph on]"
 
to say player-descrip:
	if Trips Strip is unvisited:
		say "You're [tag-status]. And hey, your ex-company let you dress casually[one of]. It'd stink if the dress code STILL applied[or][stopping]";
	otherwise:
		if warts are part of player:
			say "Those warts are making you feel ugly right now";
		otherwise:
			say "You're wearing generic, comfortable and inoffensive clothes, which have enough pockets for a modest inventory";
	say "[if xray-vision is true]. You seem to be able to see under your own skin a bit--must be the saltine[end if]."

to say tag-status:
	if player is wearing the nametag:
		say "still wearing that nametag";
	otherwise:
		say "[if gateman is visible]at least [end if]not wearing that stupid nametag any more";

does the player mean examining the player: it is unlikely.

instead of dropping tomato:
	say "Much as you'd like to drop it, it [if player is in toh]is actually handy here[otherwise]might come in handy somewhere[end if]."
	
to say reject:
	let key be the hash of the player's command;
	d "The hash of the command is [key].[line break]";
	repeat through the table of anagrams:
		if the-from entry is visible:
			if key is the hashkey entry:
				say "You may be on the right track, there, with [the the-from entry]. Keep at it.";
				continue the action;
	say "That's not something you can say, do or see here."

to say bug-report:
	say "BUG. Contact me at schultz.andrew@sbcglobal.net with a transcript or description of where you are/what you did. Use up arrow to see previous commands. Or use UNDO several times and hit TRANSCRIPT to show me how you got here, what your inventory was, etc."

the air is a backdrop. the air is everywhere. understand "room" as air.

understand "nick" as air when player is in nick.

description of air is "[bug-report]";

instead of examining the air:
	if player is in busiest subsite:
		say "Reasonably busy here. Motivation is in the air! (Gag.)" instead;
	if player is in dy:
		say "You always hated the cheesy 'magic's in the air' phrase, but something's odd here." instead;
	if player has gadget:
		say "You've seen pretty much everything you need, so you look down at your gadget.";
		if player has gadget:
			try examining the gadget instead;
	say "You can't see much. It'd be nice if you had some sort of tool to gauge anything weird[if trips strip is visited], like back by Old Man Almond[end if].";

a room can be airscanned. a room is usually not airscanned.

the pockets are part of the player.

description of pockets is "This shouldn't be here since any action gives a default message."

instead of doing something with the pockets:
	say "[if location of player is busiest subsite]That would be unprofessional in future job interviews[else if location of player is dy]You fidget nervously with your pockets for a bit .There's not much else you can do with them[otherwise][one of]Your pockets have no holes. But r[or]R[stopping]eal adventurers don't futz with their pockets. Or pocket-equivalents[end if]."

section debugging flags and such

debug-print is a truth state that varies. debug-print is usually false.

two-of-three is a truth state that varies. two-of-three is true.

helpdebugflag is a truth state that varies. helpdebugflag is usually false.

hintfull is a truth state that varies. hintfull is usually false.

chapter glulx stuff

Table of Common Color Values (continued)
glulx color value	assigned number
g-pure-blue	255	[== $0000FF]
g-pure-green	40960	[== $00A000]
g-pure-red	16719904	[== $FF2020]

Table of User Styles (continued)
style name	justification	obliquity	indentation	first-line indentation	boldness	fixed width	relative size	glulx color
header-style	--	--	--	--	bold-weight	--	2	g-black
special-style-1	--	--	--	--	regular-weight	proportional-font	0	g-pure-red
special-style-2	--	--	--	--	regular-weight	proportional-font	0	g-pure-green
italic-style	--	--	--	--	--	--	0	g-black
fixed-letter-spacing-style	--	--	--	--	--	--	0	g-black
bold-style	--	--	--	--	--	--	0	g-black
alert-style	--	--	--	--	--	--	0	g-black
note-style	--	--	--	--	--	--	0	g-black
input-style	--	--	--	--	--	--	0	g-black
normal-style   	--	--	--	--	--	--	0	g-black
blockquote-style	left-justified	no-obliquity	0	0	regular-weight	proportional-font	0	g-pure-blue

The back-colour of the main-window is g-white.

to say rc:
	say "[first custom style]R[roman type]"

to say gc:
	say "[second custom style]G[roman type]"

to say bluetext:
	(- glk_set_style(style_BlockQuote); -)

to say bc:
	say "[bluetext]B[roman type]";

book final question options

Table of Final Question Options (continued)
final question wording	only if victorious	topic		final response rule		final response activity
"see what you may've MISSED (with minor spoilers for alternate routes) point-wise"	true	"missed"	--	showing what the player missed

showing what the player missed is an activity.

rule for showing what the player missed:
	let anything-missed be a truth state;
	now anything-missed is false;
	if static is not flipped-yet:
		now anything-missed is true;
		say "--the static in Notices Section was changeable.";
	if rock is in islet:
		now anything-missed is true;
		say "--the rock on the islet was changeable.";
	if swing is in islet:
		now anything-missed is true;
		say "--the swing on the islet was changeable.";
	if spore is in olw:
		now anything-missed is true;
		say "--the spore in the woodland was changeable. Though you got the tougher toolshed.";
	if sprig is in olw:
		now anything-missed is true;
		say "--the sprig in the woodland was changeable. Though you got the tougher toolshed.";
	if toolshed is in olw:
		now anything-missed is true;
		say "--the toolshed in the woodland was changeable.";
	if store f is in trips strip:
		now anything-missed is true;
		say "--store F in the trips strip led to a forest.";
	if store i is in trips strip:
		now anything-missed is true;
		say "--store I in the trips strip led to a sortie. Which is both an exit and a missile launch.";
	if store m is in trips strip:
		now anything-missed is true;
		say "--store M in the trips strip led to the Metros.";
	if cabinet is in trips strip:
		now anything-missed is true;
		say "--the cabinet could've become something really alive.";
	if forest is solved:
		if hasbeen is off-stage:
			say "--you missed a [if anything-missed is true]potential [end if]Last Lousy Point in the Forest.";
			now anything-missed is true;
	if sortie is solved:
		if pancake is off-stage and grits are off-stage:
			say "--you missed two [if anything-missed is true]potential [end if]Last Lousy Points in the Sortie. They're both together.";
			now anything-missed is true;
		otherwise if pancake is off-stage or grits are off-stage:
			say "--you missed a [if anything-missed is true]potential [end if]Last Lousy Points in the Sortie. But you got the other one near it.";
			now anything-missed is true;
	if metros is solved:
		if antlers are in cordoned red condo:
			say "--you missed a [if anything-missed is true]potential [end if]Last Lousy Point in the Metros.";
			now anything-missed is true;
		say "--there are two diverging ways through the Metros. You can also change the [if begonias are in elf row's flowers]begonias to a noise bag[otherwise]heaths to a sheath[end if], for a slightly different walkthrough.";
		say "--there are three ways to get the tulip. You can also [alt-sols].";
	if anything-missed is false:
		say "You made it through without missing anything[if number of solved regions < 5] in the regions you visited[end if]. Congratulations!";

to say alt-sols:
	let firsty be false;
	repeat through table of tulip-acq:
		if chosen entry is false:
			say "[sol entry]";
			if firsty is false:
				say " or ";
				now firsty is true;

table of tulip-acq
sol	chosen
"ASK NERDS ABOUT DARKNESS"	false
"OPEN NOISE BAG in Special Place after getting begonias"	false
"SHOW EMITTER TO DEADBEAT once he's friendly, then use it in Special Place"	false

book section-victory

to metro-victory:
	say "The beast scratches you as you draw your sword, which acts as a microphone, converting the beast's beats to equally horrendous lyrics and providing feedback that makes it wince. It gives an owly yowl and runs off, sobbing.[paragraph break]People can finally think again, so they figure how to find you quickly in the abyss! But you have to take the long way back. 'Subway Busy? Aw,' you moan, half-jokingly.[paragraph break]'You're quite a Smiter, Mister! You gave the monster no terms! You merit this mitre!' someone says. But all you want is to get on with your quest. The escalator in the undesired underside is rigged to go up. You turn your sore to a rose, lead a pared parade and, for slower fun, plant a sunflower before returning to the Trips Strip.";
	reg-inc;
	solve-region metros;

chapter commending

to commend (y - a region):
	if y is solved:
		say "[bug-report] already solved.";
		the rule succeeds;
	now y is solved;

chapter helping

helping is an action applying to nothing.

mainhelping is an action applying to nothing.

understand the command "hint/hints" as something new.

understand "hint" and "hints" as helping.

hintsoff is a truth state that varies. hintsoff is usually false.

just-print is a truth state that varies. just-print is usually true.

to say minus:
	say "[one of](-) [bracket][italic type]A minus sign means you've reached the end of a hint loop. You can cycle through them again, though.[roman type][close bracket][no line break][or](-)[stopping]";

to say plus:
	say "[one of](+) [italic type]Note: the plus sign means you can HINT again for something more spoilery. (-) means the end of a list of hints.[roman type][no line break][or](+)[stopping]";

carry out helping:
	if hintfull is false:
		try mainhelping;
	otherwise:
		say "[bug-report] Hints are already being traced. This is a debug-feature only.";

carry out mainhelping:
	if hintsoff is true:
		all-say "You elected to turn hints off for the remainder of the game. You'll need to restart to change that." instead;
	if player is in busiest:
		all-say "[one of]Look around. Most of the standard directions don't really seem to get you anywhere--going east with the crowd doesn't count. [plus][or]There's a passage that's not quite so prominent. [if sign above the passage is examined]You've already read the sign[otherwise]The sign has an odd message that's not quite stopping you entering[end if]. [plus][or]You can go IN, ENTER, or ENTER PASSAGE. [minus][cycling]" instead;
	if player is in dy:
		if toga is visible:
			all-say "[one of]There's a magic word that works here. The toga's useless but...the writing on it is a hint. [plus][or]As are the shrub and brush. [plus][or]You need to summon a GOAT. [minus][cycling]" instead;
		all-say "[one of]You got rid of the THORN, so which direction should you go? [plus][or]NORTH. You have to type the whole word in. [minus][cycling]" instead;
	if player is in notices section:
		if gateman is not visible:
			all-say "You need someone to help you with the gateway, and then you need to make the gateway into something else." instead;
		if getaway is not visible:
			all-say "The gateway doesn't help you get anywhere. Perhaps the tagged gadget can help take you away." instead;
		all-say "Just enter the gate!" instead;
	if location of player is trips strip:
		if forest is solved and sortie is solved and metros is solved:
			if r-p is visible:
				all-say "You've opened all the doors you can." instead;
			otherwise:
				all-say "Store R is your last challenge." instead;
		if you-can-advance:
			if store r is visible:
				if resort-known is true:
					all-say "You already figured what this is--a resort." instead;
				all-say "[one of]You need to get into store R. [plus][or]The roster provides a hint. Where's a place you can rest? [plus][or]Certifying or Rectifying tells the first letter, R, and the next must be a vowel. [plus][or]You want to go to a RESORT. [minus][cycling]" instead;
			all-say "The way is clear[if store m is visible or store f is visible or store i is visible]. You could try the remaining store, but it's not necessary[end if]." instead;
		if store f is visible:
			all-say "[one of]Store F has some real earth tones. [plus][or]Trees, leaves, mud. [plus][or]There's a FOREST behind it. [minus][cycling]" instead;
		if store i is visible:
			all-say "[one of]The outline of a trap door is what you need to get to. what word is a synonym for exit and missile launch? [plus][or][if player has gadget]If you scan the Tories, that and the store give you half the answer--more, since it eliminates some of the later letters. You can use both modes of the gadget[otherwise]Unfortunately, without any devices from the cabinet, this is tricky, if you draw a blank on synonyms[end if]. [plus][or]You need to create a SORTIE. [minus][cycling]" instead;
		if store m is visible:
			all-say "[one of]Examining the map, it is of Mt. Rose. Which is an anagram for Store M. It looks like you need to get to a subway station or something. [plus][or][if gadget is cert]The gadget tells you M----- if you scan both the store and the map of Mt. Rose[else if gadget is rect]M----S is only so helpful here, but remember a city and its transport system are being portrayed[otherwise]A city and its transport system are being portrayed[end if]. [plus][or]You need to discover the METROS. [minus][cycling]." instead;
		if store r is visible:
			all-say "You won't need to get into store R until you've solved another area, at least[if resort-known is true]. But you already guessed right--it's a resort[otherwise]. But you can still guess it. If you don't, this hint will tell you what it should be once you can advance[end if]." instead;
		otherwise:
			all-say "[bug-report] Somehow you flipped the resort before solving other areas you needed to." instead;
	if map region of the location of the player is forest:
		try forest-hinting instead;
	if map region of the location of the player is metros:
		try metros-hinting instead;
	if map region of the location of the player is sortie:
		try sortie-hinting instead;
	if map region of the location of the player is resort:
		try resort-hinting instead;
	all-say "Bug text. Should not appear.";
	the rule succeeds;
	
to all-say (b - text):
	let a be indexed text;
	now a is "[b]";
	if just-print is true:
		say "[a][line break]";
		continue the action;
	if helpdebugflag is true:
		append "[the player's command]: [a][line break]" to the file of shufhints;
	if hintfull is true:
		say "[if hintfull is true]THIS MOVE'S HINT: [end if][a][if hintfull is true] (turn full hints off with HF)[end if][line break]";
	[otherwise:
		say "[word number 1 in a] [word number 2 in a] [number of words in a] [number of characters in a].[line break]";]

outline-print is a truth state that varies. outline-print is usually true.

the file of shufhints is called "shufhints".

forest-hinting is an action applying to nothing.
metros-hinting is an action applying to nothing.
sortie-hinting is an action applying to nothing.
resort-hinting is an action applying to nothing.
nerd-hinting is an action applying to nothing.

coe-clue is a truth state that varies. coe-clue is usually false.

carry out forest-hinting:
	if location of player is sf:
		all-say "[one of]Follow your nose. [plus][or]SMELL then go in that direction.  [minus][cycling]" instead;
	if location of player is rf:
		if rut-door is not visible:
			all-say "[one of]What can the odor turn into? [plus][or]Twelve possibilities, really [plus][or]A door. [minus][cycling]" instead;
		if bulge is visible:
			all-say "[one of]You don't have any key, but that bulge shouldn't be part of the door [plus][or]The musical note is a clue [plus][or]You want to change the bulge into a BUGLE [minus][cycling]." instead;
		if player has bugle:
			all-say "You should PLAY the bugle." instead;
		all-say "You should be able to just enter the door now." instead;
	if location of player is silent rut:
		if sorting area is unvisited:
			all-say "Go east and look around a bit. There're only three rooms to start, and passing the turnstile needs stuff from each side room." instead;
		if storing area is unvisited:
			all-say "Go west and look around a bit. There're only three rooms to start, and passing the turnstile needs stuff from each side room." instead;
		if ones are off-stage or shotgun is off-stage:
			all-say "You need to solve a puzzle to the east." instead;
	if coe is unvisited:
		if coe-clue is false:
			now coe-clue is true;
			all-say "The men going through have certain things in common. If you can imitate them, you can get through, too." instead;
		if player has beard and shades are part of beard and nose is part of beard:
			all-say "You've got what you need for the next bit[if player wears beard]. You are even wearing it[otherwise]. Just wear it[end if]. Then go through the turnstile." instead;
		if nose is off-stage:
			all-say "[one of]The ones [if player is not in sorting area]in the sorting area[end if] can be changed. [plus][or]What part of the disguise? [plus][or]Ones become a NOSE. [minus][cycling]" instead;
		if bread is part of the sandwich:
			all-say "[one of]The sandwich [if player is not in storing area]in the storing area [end if] is useful for two puzzles. It has 'the usual two parts.' [plus][or]Take BREAD. [minus][cycling]" instead;
		if beard is off-stage:
			all-say "[one of]The bread [if player is not in storing area]in the storing area [end if]can be changed. [plus][or]It becomes a BEARD. [minus][cycling]" instead;
		if shades are off-stage:
			all-say "[one of]The dashes [if player is not in sorting area]in the sorting area [end if]can be changed to a disguise piece. [plus][or]You have the beard and nose, so you need something for your eyes. [plus][or]Impractical, but... SHADES. [minus][cycling]" instead;
		all-say "You have everything you need to enter the turnstile, now. You can worry about the other items later. PUT (item) ON (item) to create the full disguise." instead;
	if chisel is off-stage:
		all-say "[one of]The liches are a tool of some greater evil, and their descriptions suggest another word (+)[or]Make them into a CHISEL[cycling]." instead;
	if livers are off-stage:
		if player does not have chicken liver and player does not have cow liver:
			all-say "[one of]That chisel might be handy for some items[if player is in coe] you'll need to pass the drapes[end if]. [plus][or]What could it PEEL? [plus][or]Either of the livers in the Storing Area. [minus][cycling]" instead;
		if player does not have chicken liver or player does not have cow liver:
			all-say "You've thrown one in the canister. Now throw the other." instead;
		all-say "[one of]Examining the livers indicates they need to be processed. [plus][or]You need to PUT CHICKEN LIVER IN canister. Then the cow liver. [plus][or]The livers will be handy later. [minus][cycling]" instead; [what are good verbs?]
	if spread is off-stage:
		if player is not in coe:
			all-say "You can go to the Enclosure again and tackle the DRAPES." instead;
		all-say "[one of]The DRAPES can become something. [plus][or]You can't quite cut them as-is, but maybe you could make them less thick? [plus][or]'None will be SPARED.' is a hint message. [plus][or]You can figure out what the word is from there. [plus][or] DRAPES become a SPREAD. [minus][cycling]" instead;
	if spread is not cut:
		if sliver is off-stage:
			all-say "[one of]Transforming the livers is necessary. The gadget shows all reds, but the switch is not too bad. [plus][or]LIVERS become SLIVER. [minus][cycling]" instead;
		all-say "CUT the spread with the sliver." instead;
	if player does not have shotgun:
		if noughts are off-stage:
			all-say "[one of]Your last order of business is a weapon. There's still something in the sorting area. [plus][or]The zeroes aren't quite zeroes. They're NOUGHTS. [minus][cycling]" instead;
		if shotgun is off-stage:
			all-say "The NOUGHTS [if player is not in sorting area]in the sorting area [end if]can become a shotgun. Though it's not loaded yet." instead;
	if silver is off-stage:
		all-say "[one of]The sliver is still useful. [plus][or]It can become SILVER. [minus][cycling]" instead;
	if shotgun is not loaded:
		all-say "Put the silver in the shotgun.";
	if player does not have maps:
		all-say "[one of]The spam isn't particularly edible. [plus][or]It's more useful as MAPS. [minus][cycling]" instead;
	if player is in ghouls' slough:
		all-say "You need to READ the MAPS." instead;
	if player is not in frost forts:
		all-say "You should be able to go north [if player is not in coe]from the enclosure [end if]now." instead;
	if player is in frost forts:
		if vowels are visible:
			all-say "[one of]The vowels seem to be hiding a great evil, a Halloweeny menace. They seem to be saying 'We're Vowels.' [plus][or]Two solutions: WEREWOLVES or WOLVES. [minus][cycling]" instead;
		all-say "SHOOT WOLVES." instead;
	all-say "I can't find anything more to do in the forest. This is probably a [bug-report]" instead;

a thing can be hint-vulnerable. a thing is usually not hint-vulnerable.

carry out metros-hinting:
	if player is in toh:
		if night thing is visible:
			all-say "You can't kill the night thing on your own. Maybe something equally awful would take care of it?" instead;
		if player has termite emitter:
			all-say "Your work here is done." instead;
		otherwise:
			all-say "Why not search the mattress? Something's in there." instead;
	if player is in underside:
		if drainage is visible and elf row's flowers is visited:
			all-say "[one of]The drainage changes into something the faeries want. [plus][or]Any kind of flower? [plus]You could give the faeries a GARDENIA. [minus][cycling]" instead;
		if red camp is not visited or elf row's flowers is not visited:
			all-say "Look around a bit first, then come back. You haven't seen all the initial rooms, yet[if drainage is visible]. Though you can try to transform the drainage if you want[end if]." instead;
		if spis is not visited:
			if player has keycard:
				if barcode is part of the keycard:
					all-say "Nothing except badly hinted verbs should be stopping you from PUTting the keycard on the sensor." instead;
				if player has brocade:
					all-say "[one of]The keycard's a bit bare as is. But the brocade you have could liven it up. [plus][or]What's white with lots of black lines and a few numbers below? [plus][or]That'd be a BARCODE. [plus][or]You'll need to PUT BARCODE ON KEYCARD. [minus][cycling]" instead;
				otherwise:
					all-say "[one of]Something worthless to the faeries may help you. In some form. [minus][cycling]" instead;
			if player has dry cake:
				all-say "[one of]The dry cake isn't useful as-is, but it's part of something that will be. [plus][or]The cake becomes a KEYCARD. [minus][cycling]" instead;
			all-say "You will need to get by that door. You should find items lying around in the camp and the flower shop." instead;
		all-say "You seem to have done everything this location needs." instead;
		the rule succeeds;
	if player is in cramped red camp:
		if player does not have dry cake and player does not have keycard:
			all-say "The passage north gets you free dry cake[if player does not have emitter], if you can disrupt things[end if]." instead;
		if player does not have velcro and player does not have clover:
			all-say "The clover should be handy in another form." instead;
		if player does not have controls and player does not have lost corn:
			all-say "The lost corn will be useful as something else." instead;
		all-say "You've gotten everything you need here. There are no hidden puzzles." instead;
	if location of player is elf row's flowers:
		if player does not have brocade and barcode is off-stage:
			all-say "The brocade is free. Why not take it?" instead;
			the rule succeeds;
		if gardenia is off-stage:
			all-say "[one of]The faeries would like you to clean up the city a bit and give them a flower. [plus][or]What is the most obvious mess? [plus][or]The drainage in the Undesired Underside is an eyesore. [plus][or]It can become a GARDENIA, which you can give to the faeries. [minus][cycling]" instead;
			the rule succeeds;
		if player has gardenia:
			all-say "Try GIVEing the gardenia to the faeries. Either flower gives a solution." instead;
			the rule succeeds;
		if player has heaths or player has begonias:
			all-say "The alley is where you can transform your flower and use it." instead;
			the rule succeeds;
		if player has sheath or player has noise bag:
			all-say "You've transformed your flower[if words are not in aya], and you've gotten the words from the alley. Nice work. This puzzle's done[otherwise], but you need to use your new item in the alley[end if]." instead;
			the rule succeeds;
		all-say "Take a flower. Either flower." instead;
		the rule succeeds;
	if player is in aya:
		if motto is in aya:
			all-say "[one of]The, I mean, a motto probably belongs in the trash in the long term. But it's helpful to you. [plus][or]Who's it by? Too-Apt Pat Ott. How does the song go?[or]Potato, ...[or]TOMATO. [minus][cycling]" instead;
		if words are not in aya:
			all-say "You've done what you need here." instead;
			the rule succeeds;
		if player has heaths:
			all-say "[one of]You can change the heaths into something that can hold something you can change the words into[if player has tagged gadget]. The tagged gadget turns up all reds, but it can't detect letter shifts[end if]. [plus][or]Make a SHEATH. [minus][cycling]" instead;
			the rule succeeds;
		if player has sheath:
			all-say "[one of]What fits in a sheath? [plus][or]Make a SWORD. [minus][cycling]" instead;
			the rule succeeds;
		if player has begonias:
			all-say "[one of]Two words[if player has tagged gadget], says your gadget[end if]. Something that holds something. [plus][or]Also, one word is related to sound. [plus][or]Create a NOISE BAG. [minus][cycling]" instead;
			the rule succeeds;
		if player has noise bag:
			all-say "[one of]The words would fit nicely in the noise bag [plus][or]OPEN THE NOISE BAG [minus][cycling]." instead;
		if gardenia is in lalaland:
			all-say "Pick a reward from the fairies." instead;
		otherwise:
			all-say "Come back when you get a flower from the faeries." instead;
		the rule succeeds;
	if player is in spis:
		if player has lit-up tulip:
			all-say "Nothing more to do here." instead;
			the rule succeeds;
		try nerd-hinting instead;
	if player is in ett or player is on fuzzy looking wall:
		if velcro is off-stage:
			all-say "[one of]That wall is funny-looking, but you have nothing to climb it with. [plus][or][if red camp is visited]The clover from the camp is what you need[otherwise]Come back once you've visited the camp[end if]. [plus][or]The clover from the camp becomes VELCRO. [minus][cycling]" instead;
		if player is not wearing velcro:
			all-say "You can wear the velcro to climb the walls[if controls are off-stage], but you don't have the right item yet[end if]." instead;
		if controls are off-stage:
			all-say "[one of]You haven't done anything with the lost corn yet. [plus][or]The kernels look like buttons. [plus][or]The 'lost' adjective is important. [plus][or]The lost corn can become CONTROLS. [minus][cycling]" instead;
		if neon pig is in ett:
			all-say "[one of]That neon pig needs to go. For aesthetic and technical reasons. [plus][or]You can scan it, and that gives you some information. [plus][or]You'd rather have nothing there than the pig, too. [plus][or]It can become an OPENING. [minus][cycling]" instead;
		if power-shut is true:
			all-say "You've done what you can here[if player does not have tulip]. You'll need a light source to go east across the rail[end if][if spis is unvisited]. You haven't figured what's behind the door to the north, yet[end if]." instead;
		otherwise:
			all-say "[one of]The opening at the top of the climbable wall is interesting. [plus][or]It appears to have a bunch of circuit-ends, etc. [plus][or]The controls can override the power surge to the east. [plus][or]PUT CONTROLS IN OPENING. [minus][cycling]" instead; [?? once opening is surged: now tracks are spitting more than ever]
	if location of player is Bassy Abyss:
		if stickyhanded is false:
			all-say "[one of]The siren is irritating. [plus][or]You can disable it and make something you need. [plus][or]It becomes RESIN. [minus][cycling]" instead;
		if player has sword or player has sheath:
			if beats are visible:
				all-say "[one of]The sword can't beat anything non-physical. [plus][or]The beats can be changed into a BEAST. [minus][cycling]" instead;
			all-say "ATTACK BEAST. You win." instead;
		if player has noise bag:
			if words are in noise bag:
				all-say "[one of]The words can't shout over the beats. [plus][or]You need a more physical confrontation. [plus][or]Make a SWORD. [minus][cycling]" instead;
			all-say "[one of]The sword can't beat anything non-physical. [plus][or]The beats can be changed into a BEAST. [minus][cycling]" instead;
		all-say "[bug-report] I let you in with an item I should not have." instead;
	if player is in cordoned red condo:
		if player has dry cake:
			all-say "You've got the cake. You can just leave." instead;
		all-say "[one of]The dry cake is important. [plus][or]People won't let you take it unless they're gone. [plus][or]What could scare people? [plus][or]The tomato would be a one-off. You need more than that. [plus][or][if location of night thing is toh]There's something to uncover in the Ol['] Hotel[otherwise]A device may spook these people[end if]. [plus][or][if location of night thing is toh]Something you find in the hotel[otherwise]The termite emitter[end if] can be switched. People will flee. [minus][cycling]" instead;
	all-say "[bug-report] I ran out of hints and should not have.";
	the rule succeeds;

carry out resort-hinting:
	if player is in tiles:
		all-say "[one of]Hm, weird colored tiles. Blue all around. Brown. Sort of like store F and Forest. [plus][or]The TILES become an ISLET. [minus][cycling]" instead;
	if player is in islet:
		all-say "[one of]Two ways to get across the water. [plus][or]Two items lying around.[plus][or][if wings are visible]The wings you have will work. You don't even need to wear them[otherwise]The SWING can really make you fly[end if].[plus][or][if cork is visible]The cork will help you across the water as-is[otherwise]Alternatively, the ROCK can become much lighter--a CORK[end if].[plus][or]Either way, you only need to grab one and go east. [minus][cycling]" instead;
	if player is in olw:
		if poles are visible:
			all-say "[one of]The poles are just too vertical. What's something spelled like them that isn't? [plus][or]They're not ski poles, for going down a ski whatchamajigger... [plus][or]They can be a SLOPE. [minus][cycling]" instead;
		if ropes are visible and grips are visible:
			all-say "You can just go east now, up the slope." instead;
		if ropes are visible:
			all-say "[one of]The ropes provide safety but no traction. [plus][or]The sprig can change into something providing traction. [plus][or]It seems suctioned to the ground. [plus][or]It becomes GRIPS. [minus][cycling]" instead;
		if grips are visible:
			all-say "[one of]The grips provide traction but no safety. They're not 100% reliable. [plus][or]The spore can change into something providing safety. [plus][or]What do people use to climb up rock walls? [plus][or]ROPES. [minus][cycling]" instead;
		if toolshed is visible:
			all-say "[one of]You need a way up the slope. It's too smooth and steep. [plus][or]There are two ways through--flip the toolshed, or flip the ropes and spore. [plus][or]The toolshed isn't very useful, but maybe it can be. [plus][or]It's got some odd protrusions, which are a hint. [plus][or]They can HOLD the edge of your feet [plus][or]They're TOEHOLDS. [minus][cycling]" instead;
		all-say "You're done here." instead;
	if player is in Potshot Hotspot:
		if riot is visible:
			all-say "[one of]It's a riot! Way more than three people! [plus][or]How would you cut it down to three? [plus][or]Make them a TRIO. [minus][cycling]" instead;
		if protest is visible:
			all-say "[one of]Three people are still a protest. How can you cut them down? [plus][or]They have REAL jobs. They make something, they say! [plus][or]Apparently they use clay and say 'It's kiln me!' [plus][or]The people in the PROTEST are really POTTERS. [minus][cycling]" instead;
		if potters are visible and kilns are not visible:
			all-say "[one of]The potters need something to do with their hands. [plus][or]The chain links have been dropped. Perhaps you could use them for less violent purposes. [plus][or]They can become KILNS. [minus][cycling]" instead;
		if red bull burdell is visible:
			all-say "[one of]He does like yelling GET OUT. [plus][or]And changing stuff won't help you here. [plus][or]So maybe you need to change what he says? [plus][or]The gadget [if player has gadget]would be[otherwise]was[end if] marginal help with six reds. [plus][or]He has but one weakness, blah blah blah... [plus][or]TUG TOE. [minus][cycling]" instead;
		all-say "There is nothing more to do here except go east." instead;
	if player is in Roman Manor:
		all-say "[one of]Just one thing to do. Or not do. [plus][or]You want to do something that completely ignores the EXITS. [plus][or]Another verb. [plus][or]Your work's done. Time to EXIST. [minus][cycling]" instead;
	all-say "Unhandled case. I need to fix this, so a transcript would be an enormous help." instead;
	the rule succeeds;
	
ever-nerd-hint is a truth state that varies. ever-nerd-hint is usually false.

carry out nerd-hinting:
	if ever-nerd-hint is false:
		if player has begonias or player has noise bag:
			all-say "[one of]With the flower choice you made, there's an alternate way through. [plus][or][if player has noise bag]The nerds are protected from noise[otherwise]You can change the begonias into something that annoys the nerds[end if]. [plus][or][if player has begonias]NOISE BAG[otherwise]Well, you have the noise bag.[end if] [plus][or]Where are noises more concentrated? [plus][or]In A Yell Alley. [plus][or]Then, in Special Place, OPEN NOISE BAG. [plus][or]Alternatively you can just ASK NERDS[plus][or]ABOUT the DARKNESS. [minus][cycling]" instead;
		all-say "[one of][if east-tried is false]You haven't looked east of the train tracks[otherwise]East of the train tracks is something you can't work through[end if]. The absence of light is DARKNESS. DARKNESS - ASK = ?.[or]ASK the NERDS about the DARKNESS.[cycling]";
	otherwise:
		all-say "You need to ASK the nerds about something. They seem to know a lot.";
		now ever-nerd-hint is true;

ingred-check is a truth state that varies. ingred-check is usually false.

carry out sortie-hinting:
	if location of player is the nick:
		all-say "[one of]There's no gateman to help with the grate. [plus][or]You're feeling very hungry. You wish you could go somewhere to eat, or even make something to eat.  [plus][or]The tagged gadget gives a reading without you examining anything. [plus][or]This room becomes a KITCHEN. [minus][cycling]" instead;
	if location of player is kitchen:
		if taco is visible or coat is visible:
			all-say "Your work in the kitchen is done[if taco is visible], though the taco isn't quite what you want[end if]." instead.;
		if ingred-check is false:
			now ingred-check is true;
			all-say "It's a kitchen. You need to make something. But there are no ingredients around--just weird non-food items. Hmm." instead;
		if skate is visible:
			all-say "The skate can become steak." instead;
		if cult tee is visible:
			all-say "The cult tee can become lettuce." instead;
		if cathouse is visible:
			all-say "The CATHOUSE perfume can become hot sauce." instead;
		if spearman is visible:
			all-say "The spearman can become parmesan." instead;
		if tall trio is visible:
			all-say "The tall trio can become a tortilla." instead;
		if taco is not visible:
			all-say "You still have ingredients to put in the taco." instead;
		all-say "Your work here is done[if taco is visible], though you don't need to EAT the taco[end if]." instead;
	if warts are visible:
		if the nick is unvisited:
			all-say "This transformation shouldn't be too bad, but there's another hint in another room you should visit." instead;
		all-say "[one of]After a few turns, the warts will bother you. [plus][or] What did the guards say before laughing at your face? [plus][or]'No, too stupid a fad. I put soot on...' [plus][or]It's most of a famous palindrome. [plus][or]It begins with STRAW. [minus][cycling]" instead;
	if player is in ufc:
		if numset of dial is 16:
			all-say "You don't need to do anything else here." instead;
		all-say "[one of]You'll get hints for turning the dial wrong, but the EXITS at right angles are a clue. [plus][or]You need to figure what the right directions are. [plus][or]If you've tried going diagonally, that doesn't work, leaving four possibilities. [plus][or]You can also just think of a number with X in it, with seven letters. [plus][or]Binary search works, too, but the next clue shows how you can get things in one move. [plus][or]The right number is SIXTEEN. [minus][cycling]" instead;
		all-say "Nothing more to do here." instead;
	if roomroom is visited and sack is off-stage:
		all-say "[one of]The cask is versatile. Did you look at it? [plus][or]The tagged gadget gives it away. [plus][or]It is also a SACK, which can carry different things. [minus][cycling]" instead;
	if player is in roomroom:
		if moor is not visited:
			all-say "[one of]The room is just a plain room. You feel like you want to get out, though. [plus][or]It can become a MOOR. [minus][cycling]" instead;
		if shoes are not in lalaland:
			all-say "[one of]Your current footwear isn't cutting it. [plus][or]You need something to keep the wet out. Rubber would work well. [plus][or]The pair of hoses can become... [plus][or]...a pair of shoes. [minus][cycling]" instead;
		if player does not have coat:
			if kitchen is unvisited:
				all-say "If you're cold, look around a bit more. There's a room you haven't been to yet." instead;
			all-say "[one of]The moor is too cold without something warm. [plus][or]You've been to the kitchen[if taco is off-stage], but you need to make some food[otherwise], but you missed the last bit[end if]. [plus][or][if taco is off-stage]Make the food, then come back[otherwise]The TACO can become a COAT[end if]. [minus][cycling]" instead;
		all-say "Nothing more to do here." instead;
	if player is in ne-room:
		if scraped wall is not visible:
			all-say "Nothing more to do. You opened a secret passage." instead;
		if moor is unvisited:
			all-say "You haven't visited everywhere in this area. There's one place you can zap to, but not from this room." instead;
		if hay is off-stage:
			all-say "You can trade for wall stuffing in the moor." instead;
		otherwise:
			if hay is in lalaland:
				all-say "[one of]You stuffed the hay in the wall. What does the gadget offer? [plus][or]The hay wall can become a HALLWAY! [minus][cycling]" instead;
			if player has the straw or straw is in sack:
				all-say "Hm, the straw is too bulky to fit in the wall. Something like it?" instead;
			if player has the hay or hay is in sack:
				all-say "[one of]The hay might fit in well with the wall. [plus][or]PUT HAY IN WALL. [minus][cycling]" instead;
	if player is in sacred cedars:
		if oils are in cask:
			all-say "You can't pour the oils anywhere in this enclosed area. Try going back to the moor." instead;
		if caskfillings is 0:
			all-say "You can FILL CASK here, for starters." instead;
		if caskfillings is 1:
			all-say "You can fill the cask with more oils." instead;
		if caskfillings is 2:
			all-say "You have done all you need to here." instead;
	if player is in moor:
		if anapest is visible:
			all-say "[one of]The poem you hear is an ANAPEST. [plus][or]It's so snotty and elitist, you'd like to change it. [plus][or]Change it to a PEASANT. [minus][cycling]" instead;
		if peasant is visible:
			all-say "[one of]The peasant has a bunch of hay he doesn't want. Lord Ablemiser said it was 'almost the same thing.' What've you seen that's almost the same thing?[plus][or]The straw. [if sack is off-stage]The cask can be made into a sack[otherwise]It can fit in the sack[end if]. [plus][or]GIVE STRAW TO PEASANT. [minus][cycling]" instead;
		if soil is not visible or silo is not visible:
			if cedars is unvisited:
				all-say "You need to figure what's east of the trellis first." instead;
			if cask is in lalaland:
				all-say "You need to change the sack into a cask to carry the oils from Sacred Cedars." instead;
			if oils are not in cask:
				all-say "You need to fill the cask[if soil is in moor]. Yes, again[end if]." instead;
			if soil is not visible:
				all-say "[one of]What can the OILS become? [plus][or]First, a foundation. [plus][or]They'll make good SOIL. [minus][cycling]" instead;
			all-say "[one of]What else can the OILS be? [plus][or]Something to build. [plus][or]Let's continue the weapon theme[if missile is not off-stage], especially since you've got a missile[end if]. [plus][or]Make a SILO. [minus][cycling]" instead;
		if roadblock is visible:
			all-say "[one of]The roadblock seems misplaced, almost inviting you to enter. [plus][or]It's also cut 5/9 of the way down. [plus][or]What's an entryway? A DOOR. [plus][or]That leaves the word BLACK. [plus][or]It's a BLACK DOOR! [minus][cycling]" instead;
		if black door is not part of the silo:
			all-say "You don't need to carry the black door to PUT BLACK DOOR IN/ON SILO. The door will allow a way to put stuff in the silo." instead;
		if smilies are not in lalaland:
			all-say "[one of]The smilies seem like they don't belong on the page. [plus][or]What'd you hear about weapons that weren't harmful? Killing with love? [plus][or]They can become a MISSILE. [minus][cycling]" instead;
		if missile is not in silo:
			all-say "You can PUT MISSILE IN SILO." instead;
		if poem is visible:
			if poem is folded:
				all-say "[one of]The plane is useless. What can it become? [plus][or]It can become a PANEL. [minus][cycling]" instead;
			all-say "The paper has creases. You can fold it into something more useful." instead;
		if panel is not part of the silo:
			all-say "How to get the silo working? PUT PANEL ON SILO." instead;
		if shoot button is not visible:
			all-say "[one of]The hoots button isn't up to much. [plus][or]You don't hoot with a missile... [plus][or]...you SHOOT. [minus][cycling]";
		if steer button is not visible:
			all-say "[one of]The trees button isn't much help launching a missile. [plus][or]You don't want that missile going just anywhere. [plus][or]STEER BUTTON. [minus][cycling]";
		all-say "You should just need to push the launch button now." instead;
	all-say "Oops. I did not account for this hint. This is a [bug-report], and you'll need to see the walkthrough.";
	the rule succeeds;

check helping:
	if hintfull is true:
		say "Ignoring hint nag due to hint-every-move debug flag set. Only testers should see this. [bug-report] in final release." instead;

meaningful-hint is a truth state that varies.

carry out helping:
	if location of player is busiest subsite:
		say "You're not in the area where the magic happens, yet. The cardinal directions, including the boring lecture east, are out." instead;	
	if meaningful-hint is false:
		now meaningful-hint is true;
		say "You call out for the Magic Hint Fairy. 'Please! Please!'[paragraph break]All you hear in return is 'Asleep! Asleep!'[paragraph break]You pause. You've heard she can be a cruel cluer--not that she can spirit you to the cheaters['] hectares--but even a thin hint could probably make you enjoy your journey less if you rely on her too much.[paragraph break]Do you really want to poke her now?";
		if player consents:
			say "You see by the look on her face she's thinking 'Spiel or Spoiler?' But the look on your face shows you're ready to take the Perilous trip to Spoiler U with an idea aide. You won't spit on tips. [hintblah]";
			continue the action;
		otherwise:
			say "Her cheats sachet tempts you, but you remain chaste. [hintblah]" instead;

to say hintblah:
	say "(To resist the temptation of summoning the Hint Fairy later, you can use the HINTS OFF command to disable hints until you restart.)";

chapter helpoffing

helpoffing is an action applying to nothing.

understand the command "help/info/hint/hints off" as something new.

understand "help off" and "info off" and "hint off" and "hints off" as helpoffing.

carry out helpoffing:
	if hintsoff is true:
		say "Hints are already disabled." instead;
	say "This will shut off hints for the remainder of the play session. You will need to restart to see them again. Are you sure?";
	if the player consents:
		say "Done.";
		now hintsoff is true;
	otherwise:
		say "You can always shut hints off again if you change your mind.";

book silly verbs

before exiting:
	if red bull burdell is visible:
		say "You may believe in brawns over brain, but I don't. I believe in you! And your ability to exploit what Red Bull keeps repeating!" instead; 
	if player is in ufc and centrifuge-stopped is false:
		try going west instead;

chapter sing-wave-jump-sleep

rule for supplying a missing noun when tasting: now the noun is the player.

instead of tasting:
	if noun is saltine:
		try eating noun instead;
	say "I have nothing to state about taste."

instead of saying yes:
	say "[yn]";
	
instead of saying no:
	say "[yn]";
	
to say yn:
	say "I am not yes-no nosey...am I? NOT.[no line break][one of] (Yes and no aren't useful commands here--you'll be prompted when you need to make that choice.)[or][line break][stopping]"

rule for supplying a missing noun when swinging: now the noun is the swing.

does the player mean swinging the swing: it is very likely.

the block swinging rule is not listed in any rulebook.

instead of swinging:
	if noun is the swing:
		say "Whee! Sitting on the twine hurts a bit, but with a little shifting around, you imagine the swing could make you fly." instead;
	if noun is sword or noun is shotgun:
		say "Don't play with weapons." instead;
	say "That's probably just silly. Or impossible." instead;

understand "drink" as drinking.

rule for supplying a missing noun when drinking: now the noun is the player.

instead of drinking:
	say "I show no kind grin on drinking!"

instead of waking up:
	say "Weak."
	
instead of sleeping:
	if mattress is visible and night thing is not visible:
		say "On that mattress? Eww." instead;
	say "You might sleepwalk and slip on some peels."

instead of squeezing:
	say "That's either icky or impossible or both.";

instead of singing:
	say "Oddly, when you start, you see a big sign telling you not to.";

the block waving hands rule is not listed in any rulebook

instead of waving hands:
	if player is in busiest subsite:
		say "You don't see any friends nearby, sadly." instead;
	if night thing is visible:
		say "Bad idea. Don't know what might make it charge." instead;
	if deadbeat is visible:
		say "He nods in return." instead;
	if talkers are visible:
		say "They pointedly ignore you." instead;
	if beast is visible:
		say "It's not very friendly at all." instead;
	if goat is visible:
		say "Do you expect it to wave back? It'd probably fall over trying." instead;
	if gateman is visible:
		say "'That it? Waving? Why wave when you can ask?'" instead;
	if number of visible people > 1:
		say "There's someone else here, but just TALK TO them or ASK them about something instead." instead;
	say "An Inform 7 programmer waves at you from behind the fourth wall and points meaningfully to your keyboard or whatever you're typing on.";
	
instead of jumping:
	if player is on fuzzy wall:
		say "Bad idea. Just climb down instead." instead;
	say "What am I supposed to do with THAT four-letter mess?"

understand the command "eat [something]" as something new.
understand "eat [something]" as eating.

instead of eating:
	if location of player is the nick:
		say "Man, you'd sure like to be somewhere you could make some food." instead;;
	if noun is info-lit:
		say "If this were a spy game, that would be ideal. But it isn't." instead;
	if noun is cake pan:
		say "Err, it's kind of metal." instead;
	if noun is not edible:
		say "New odd metabolism is not you magic power." instead;
	say "[if location of player is the nick]Your hunger is a hint, not a necessity[otherwise]You won't strictly need any food for this game. Well, your character won't[end if]."

xray-vision is a truth state that varies. xray-vision is usually false.

asked-almond is a truth state that varies.

the asking yourself for something rule is not listed in any rulebook.

Check asking something for (this is the reflexive asking for item rule):
	if player has the second noun:
		say "You already have that." instead;

the nobody gives anything rule is listed before the translate asking for to giving rule in the check asking it for rulebook.

Check asking something for (this is the nobody gives anything rule):
	if noun is gateman:
		say "'Everything here is yours. No need to ask me.'" instead;
	if noun is nerds:
		say "'You can't just bully us, man! This isn't the playground any more!'" instead;
	if noun is faeries:
		say "[if fairy-worthy is true]You've already pleased them[otherwise]They remind you that you need to get a flower for them, first[end if]." instead;
	if noun is talkers:
		say "Welfare? Certainly not." instead;
	say "Asking directly for stuff doesn't work. If it were that easy, someone already would've. Maybe you want to ask ABOUT." instead;

check objasking gateman about saltine:
	now asked-almond is true;

instead of eating the saltine:
	if asked-almond is false:
		if gateman is visible:
			say "Old Man Almond coughs as you move to eat the saltine. He seems to have something to say. Listen?";
			if the player consents:
				try objasking gateman about saltine instead;
			otherwise:
				say "You throw back the saltine. Old Man Almond shrugs.";
		say "If you abscond with restaurant saltine packets in real life, that's not my business, but you're on an adventure now. Sure you want to eat it [if location of player is notices section]before you know what it's for[otherwise]even though you realize you could've asked the old man[end if]?";
		if the player consents:
			say "Gulp. Tastes decent enough.";
		otherwise:
			say "Good call[if location of player is notices section]. Maybe you can find or summon someone who'll tell you about it[otherwise]. Probably not poisonous, but still[end if]." instead;
	otherwise:
		say "You gulp it down. It's bland, but if Old Man Almond was right, that saltine will help you see through something that could be something else.";
	now saltine is in lalaland;
	now xray-vision is true;
	now undo-code is 1;
	prevent undo instead;

before examining (this is the saltine-check rule):
	if xray-vision is true:
		if noun is scraped wall:
			now xray-vision is false;
			now undo-code is 2;
			prevent undo;
			say "You see [if scraped wall is hayfilled]a hay wall, then [end if]a hallway." instead;
		if noun is dial:
			say "The number sixteen takes prominence on the dial for a moment. You shake your head, and it disappears. Weird.";
			now xray-vision is false;
			now undo-code is 2;
			prevent undo instead;
		if noun is flippable:
			now xray-vision is false;
			now undo-code is 2;
			prevent undo;
			try saltining the noun instead;

[understand the command "throw [something preferably held]" as something new.
understand "throw [something preferably held]" as throwing it at.]

check throwing tomato at nerds:
	say "They're not a real menace, just annoying." instead;

check throwing it at:
	say "No worth in that throw[if noun is tomato], but it could make a mess elsewhere[end if][if second noun is night thing], but the right item mite (sic) get to him[end if]."

rule for supplying a missing second noun when throwing:
	if noun is tomato:
		if night thing is visible:
			now second noun is night thing;
		if nerds are visible:
			now second noun is nerds;
	otherwise:
		say "No need to go pitching items. [line break]";
		continue the action;
	
the block throwing at rule is not listed in any rulebook.

check touching:
	if noun is missile:
		say "You hear a small FWEE." instead;
	if noun is velcro or noun is fuzzy wall:
		say "Kind of rough and gritty." instead;
	if noun is thorn:
		say "Pain, like a pin." instead;
	say "Touch's not [italic type]tons[roman type] couth." instead;

test toma with "gonear underside/n/w/tomato/e/s/w/throw tomato"

chapter throwing

understand the command "throw" as something new.

understand "throw [something preferably held] at [something]" as throwing it at.

chapter buying

rule for supplying a missing noun while buying: 
	if player is in elf row's flowers:
		if heaths are in elf row's flowers:
			now noun is heaths;
		if begonias are in elf row's flowers:
			now noun is begonias;

check buying:
	if location of player is elf row's flowers:
		if noun is brocade:
			say "The scratchings said the brocade was free." instead;
		if fairy-worthy is true:
			if player has heaths or player has sheath or player has begonias or player has noise bag:
				say "You've already traded them a favor. Plus, you have no money." instead;
			say "You can just take the flowers you want, since you did the faeries a favor.";
		otherwise:
			say "These magic faeries are above such unmagical things as currency.";
		do nothing instead;
	if noun is dry cake:
		say "It's free to invited guests. But the only way to be invited is to have money. Which you don't." instead;

chapter summoning

summoning is an action applying to nothing.

understand the command "summon" as something new.

understand "summon" as summoning.

carry out summoning:
	the rule succeeds;

chapter kicking-attacking

kicking is an action applying to one thing.

understand the command "kick [something]" as something new.

understand "kick [something]" as kicking.

carry out kicking:
	if noun is goat:
		say "That'd be cruel. He's a lazy sort of ornery, not violent." instead;
	if noun is beast:
		say "Maybe you know beast-fu in real life. But I'd advise using a weapon." instead;
	try attacking the noun instead;
	the rule succeeds;

instead of attacking:
	if noun is the player:
		if player has shotgun:
			try attacking the player;
		otherwise:
			say "I hope it's not the puzzles that made you try this." instead;
	if noun is not visible:
		say "You don't have any weapon with that sort of range." instead;
	if noun is riot:
		say "You don't want to see the mob act in combat." instead;
	if noun is trio:
		say "Even this threesome would leave you smote here." instead;
	if noun is phial:
		say "The lube would ooze out and be useless." instead;
	if noun is windows:
		say "The windows are too small to enter through. They're not worth breaking." instead;
	if noun is scraped wall:
		say "Hm, pounding on it doesn't help[if hay is part of scraped wall]. You've filled it, though, and maybe you can change it more easily now[otherwise]. The reading on the trellis seems to indicate you need to fill it first[end if]." instead;
	if noun is neon pig:
		say "No poking a pig even if your name's Penny." instead;
	if noun is nerds:
		say "They're just annoying and superior, not evil." instead;
	if noun is the metallic door:
		say "As you rush at the door, you notice one of the stick figures in the warning sign attacking a door as big and solid as this one! To save your self-esteem, you turn away and forget what you saw." instead;
	if noun is sandwich:
		try opening noun instead;
	if noun is mattress:
		if bump is part of the mattress:
			say "You might to better to look at that bump before disturbing it." instead;
		otherwise:
			say "Hey, it's not the mattress's fault it only yielded a lousy termite emitter." instead;
	if noun is deadbeat:
		say "[one of]You consider it, but out of mercy, you stay your hand. Mercy for the people who'd have to listen to his account of the fight[or]He doesn't deserve to be beat-up. Or upbeat, but that's another matter[at random]." instead;
	if noun is tomato:
		say "Why not get it all over someone, or something, else instead?" instead;
	if noun is red bull burdell:
		say "No chance of even a few slick licks. Maybe you have one last bit of verbal trickery. Or maybe you'll have to do something, too." instead;
	if noun is a disguise-piece:
		say "You are not the fashion police, alas." instead;
	if noun is woeful pat:
		say "Violence against the English language should not beget physical violence." instead;
	if noun is siren:
		say "It's surprisingly resilient, and your [if sword is visible and stickyhanded is true]sword[otherwise]fist[end if] bounces off it. Plus, it's probably hooked up to a hidden alarm that sounds even worse." instead;
	if noun is night thing:
		say "Just getting close to it fills you with terror. How could you attack from afar?" instead;
	if noun is gateman:
		say "But he's here to help you. And he's, uh, bigger than you, too." instead;
	if noun is bucket:
		say "You give the bucket a good kick, but fortunately this game is about anagrams and not figures of speech." instead;
	if noun is wolves:
		if player has shotgun:
			try shooting wolves instead;
	if noun is beats and player carries words:
		say "You have words, but they're drowned out by the beats." instead;
	if noun is beats and player carries sword:
		say "The beats seem to pulse harder, as your sword-swinging looks remarkably like some techno dance." instead;
	if noun is beast and player has sword:
		metro-victory instead;
	if noun is beast and player has sheath and sword is in sheath:
		try taking sword;
		if player has sword:
			metro-victory instead;
		otherwise:
			do nothing instead;
		say "[bug-report] Sorry it's so near the end of the game. All I can do is ask your help to weed it out." instead;
	if location of player is frost forts:
		try shooting the noun instead;
	if noun is gadget:
		say "Boy, if that's how you treat stuff that's there to HELP you..." instead;
	say "It'll be clear when you need violence. And how. Hopefully."

chapter stabing

stabing is an action applying to one thing.

understand the command "stab [something]" as something new.

understand "stab [something]" as stabing.

carry out stabing:
	if player has sliver:
		if noun is drapes or noun is spread:
			try cutting noun instead;
	if player has sword or sword is in sheath:
		try attacking the noun instead;
	say "Stab? With what you have? Bats." instead;
	the rule succeeds;

chapter kissing

the block kissing rule is not listed in any rulebook.

understand "kiss [something]" as kissing.

check kissing:
	if noun is faeries:
		say "You're not their type." instead;
	if noun is night thing or noun is beast:
		say "Oh no." instead;
	if noun is Woeful Pat:
		say "'Please! I am not finished!'" instead;
	if noun is peasant:
		say "He has a wife. Sorry." instead;
	say "Affection's so fine. Fact." instead;

check kissing the player:
	if notices section is unvisited:
		say "Hooray for loving yourself even after losing your job. I guess." instead;
	otherwise:
		say "You love yourself! You can do it! You can solve this game! You can...try something more serious next move!"

book area-save

chapter retrying

retrying is an action applying to nothing.

understand the command "retry/terry" as something new.

understand "retry" as retrying.
understand "terry" as retrying.

retried is a truth state that varies. retried is usually false.

carry out retrying:
	if location of player is trips strip:
		if retried is true:
			say "This is the place you go after retrying--as you probably know. So nothing happens." instead;
		say "Nothing happens. Perhaps this is where you'd be sent back to[if sf is visited or underside is visited or ufc is visited]after going somewhere like you've been[otherwise]once you're able to look around[end if]." instead;
	if trips strip is unvisited:
		say "You haven't been to the Trips Strip yet, whatever that is, and you suspect you can't just jump ahead. Besides, you don't want to risk retrying the busiest subsite." instead;
	if map region of location of player is resort:
		say "You don't really need to[if number of solved regions is 4]. In fact, you have nowhere else to go[otherwise], though there's one more region to solve[end if]. Are you sure?";
		if the player consents:
			do nothing;
		otherwise:
			say "Okay, back to the endgame." instead;
	item-warp;
	now retried is true;
	now player is in trips strip;
	cabinet-check instead;

a thing can be warpable. a thing is usually not warpable.

table of lastlocs
r	l	first-text	return-text
Forest	forest-sc	"You go to the forest."	"You return to the forest."
Sortie	sortie-x	"You return to the [if last-loc of sortie is moor]moor[otherwise]weird underground area[end if]."
Metros	metros-x	"You return to the hustle, bustle, etc."

a region has a room called last-loc.

a region has a list of things called item-list.

a region has a list of things called worn-list.

book onewaying

onewaying relates various things to various things.

the verb to oneway (it oneways, they oneway, it onewayed) implies the onewaying relation.

book anagramming

anagramming relates things to each other.

The verb to anagram (it anagrams, they anagram, it anagrammed) implies the anagramming relation.

book start and turns

part dope tan notepad

the dope tan notepad is a thing. understand "pad/note" and "note pad" as dope tan notepad.

the dope tan notepad is warpable.

instead of dropping the dope tan notepad:
	say "But it's compact and can be useful to take notes.";
	
vaguing is an action applying to nothing.

understand the commands "consult pad" and "pad" as something new.

understand "consult pad" and "pad" as vaguing.

carry out vaguing:
	say "[cur-has]";
	the rule succeeds.

description of dope tan notepad is "You just like it. The color, the feel. Other people have computers, but your notepad is yours. [cur-has]"

check examining dope tan notepad for the first time:
	say "It's your workbook. It holds bookwork. Not a thrilling color to most, and one of many you bought cheap a long time ago, and they haven't run out yet. It's small enough to fit in a pocket, and you have clipped a pen over it, too[one of]. It helped you through a lecture from this morning[or][stopping].[paragraph break](To use the notepad, you can type PAD (subject) or CONSULT PAD ABOUT (subject). If you forget what you've written about, you can type PAD or CONSULT PAD.)" instead;

to say cur-has:
	say "Currently you can consult it about:[line break]--";
	let A be 0;
	repeat through table of pad-stuff:
		if known entry is true:
			if remainder after dividing A by 4 is 0:
				say "[if A > 0].[line break]--";
			otherwise:
				say ", ";
			increment A;
			say "[short entry]";
	if A > 0:
		say "."

the pen is part of the dope tan notepad.

description of the pen is "It's clipped there so you don't misplace it."

instead of taking the pen:
	say "Bad idea. You tend to lose those things like nobody's business if they're not nailed down."

to say found-status of (a - a region):
	if a is solved:
		say " (solved)";
	otherwise:
		say " (opened)";

to say reso-maybe:
	if resort-known is true:
		say " (resort, not opened)[run paragraph on]";

table of pad-stuff
topic (topic)	known	blurb	short	verify
"notepad/pad/note" or "note pad"	true	"You keep the date you started using your notebooks on the inner front cover. This one's from three years ago."	"notepad"
"lecture"	true	"An hour-long pre-lunch lecture by some fellow named Curt Lee netted you three pages of doodles. You remember him saying how ONE WORD could open a NEW DOOR in this age of technological innovation if you picked the right one, and that's more magic than real magic, if real magic existed, which it doesn't.[paragraph break]But hey, at least lunch was nice."	"lecture"
"store/stores/malls/shop/shops/lots/mall"	false	"--[if forest-sc is visible]Forest[found-status of forest][otherwise]Store F[end if][line break]--[if sortie-x is visible]Sortie[found-status of sortie][otherwise]Store I[end if][line break]--[if metros-x is visible]Metros[found-status of metros][otherwise]Store M[end if][line break]--[if resort is visible]Resort[found-status of resort][otherwise]Store R[reso-maybe][end if]"	"stores"
"store 6/six/f" or "6/six/f"	false	"Store f is very leafy and woodsy."	"store f"
"store i/9/nine" or "i/9/nine"	false	"Store i appears to have a weird exit."	"store i"
"store m/13/thirteen" or "m/13/thirteen"	false	"Store m appears to open to a whole metropolis."	"store m"
"store r/18/eighteen" or "r/18/eighteen"	false	"Store r doesn't appear to be open yet."	"store r"
"macros"	false	"[tell-macros]"	"macros"
"c/cer/cert/certify"	false	"1 beep = 1 word, buzz = already changed, silence = nothing.[line break]red/green = right/wrong letters."	"certify"
"r/rec/rect/rectify"	false	"green dot = 1st letter. Red dot = last letter. Blue = the rest."	"rectify"
"secure"	false	"Pushing this before you enter anywhere new will make the area harder but pay off later."	"secure"
"recuse"	false	"If you secured, this will come in handy. You'll know."	"recuse"
"thor"	false	"THOR + N = THORN. Maybe. N + THOR = ?"	"Thor"
"talking"	false	"ASK for other people, or just say the word/s otherwise."	"talking"
"goat/toga"	false	"Certify = RGRR and Rectify = RBGB"	"goat"
"warts"	false	"[if warts are visible]'No, too stupid a fad. I put soot on warts.' But what's too stupid?[otherwise]This puzzle's crossed out. You feel good about that.[end if]"	"warts"
"red bull burdell"	false	"The guy I need to beat. Probably not with strength or anything like that."	"Red Bull Burdell"
"broad/board"	false	"[if location of player is notices section]It's got notes on needing two simple 7-letter portmanteaux[how-many-notices][otherwise]You're past the bit about the introductory puzzles.[end if]"	"board"
"terry" or "retry"	false	"--RETRY to get Terry to send you [if trips strip is visited]back to [end if]the Trips Strip[if trips strip is unvisited], wherever that is[end if]"	"retry"	false
"advice"	false	"--Nothing over eight words. Maybe one two-word deal.[line break]--Red Bull Burdell is much bigger and stronger than you. Traditional fighting won't work."	"advice"
"phial" or "lube"	false	"Worth looking into. Though you probably just want to ANGLE and GLEAN."	"lube"
"angle"	false	"Lets you see what can be changed."	"angle"	false
"glean"	false	"Lets you glimpse what you need to do."	"glean"	false
"todo"	false	"What's on tap right now: [if number of padded flippable not flipped-yet things is 0]nothing[otherwise][list of padded not inflexible not flipped-yet things][end if]"	"todo"	false

to say tell-macros:
	if marcos-trumped is true:
		say "CR/RC (something) = C (something) + R(something).";
	if h-short-told is true:
		say "ANGLE=AN to see changeable items, GLEAN=GL to see the story ahead."

to say whatever-they-are:
	if store f is in trips strip:
		if store i is in trips strip:
			if store m is in trips strip:
				say ", though you've found none."

to say how-many-notices:
	if gateman is in notices section:
		if getaway is in notices section:
			say ", both of which you got";
		otherwise:
			say ", of which you've got one--the nametag/gateman";
	otherwise:
		say ", which you need to figure out. The nametag seems to change to someone who can help you with whatever the gateway should become";

[how to include a topic IN WORDS]

chapter consulting

the block consulting rule is not listed in any rulebook.

consulting is an action applying to one thing and one topic.

check consulting it about:
	if noun is not notepad:
		say "You can only really consult your notepad. ASK something animate, if you want to." instead;
	repeat through table of pad-stuff:
		if known entry is true and topic understood includes topic entry:
			say "[blurb entry][line break]";
			the rule succeeds;
		otherwise:
			do nothing;
	say "You haven't written anything about that. [cur-has]";
	the rule succeeds;

chapter padding

understand the command "pad [text]" as something new.
understand "pad [text]" as padding.

padding is an action applying to one topic.

carry out padding:
	try consulting notepad about topic understood instead;

part begin-play

to assign-lgths:
	let UV be indexed text;
	let C be a number;
	repeat through table of odd-lgths:
		now C is which-word entry;
		now UV is "[odd-ana entry]";
		now lgth of odd-ana entry is number of characters in word number C in UV;
	repeat with UVX running through things:
		if UVX is flippable and lgth of UVX is 0:
			let UV be "[UVX]";
			replace the regular expression "\W" in UV with "";
			now lgth of UVX is number of characters in UV;
			[d "[UVX] =~ [UV] = [lgth of UVX] should = [legth of UVX].[line break]";] [this is a relic, slightly, & I should delete it]
	now lgth of motto is 6;

table of odd-lgths
odd-ana	which-word
magenta nametag	2
fuzzy clover	2
hoots button	1
trees button	1
pair of hoses	3
cheesy spearman	2
bottle of cathouse perfume	3
chain links	2
acne-bit cabinet	3

when play begins (this is the don't use any other when play begins rule):
	repeat with Q running through things:
		if Q is a sto:
			move Q to Trips Strip;
	now last-loc of sortie is ufc;
	now last-loc of resort is tiles;
	now last-loc of metros is underside;
	now last-loc of forest is sf;
	now the peasant has the hay;
	say "So you just got fired from the best company ever, and it's the best day of your life. New opportunities! New horizons! New ways to look at things! Like calling this stupid kiss-off job fair a 'convention.' As you are stuffed in a slow slow elevator up to the next lecture, you hope there's some way out...";
	move player to Busiest Subsite, without printing a room description;
	now player wears magenta nametag;
	now player has the dope tan notepad;
	now left hand status line is "[location of player] ([map region of location of player])";
	now right hand status line is "[cur-score of map region of location of player]/[max-score of map region of location of player]";
	repeat with Q running through things:
		if Q is a sto:
			now Q is in Trips Strip;
	repeat through table of anagrams:
		now the-from entry anagrams the-to entry;
		now the-from entry is flippable;
		if reversible entry is 1:
			now the-to entry is flippable;
		if there is no force-take entry:
			now force-take entry is false;
	set-other-default;
	if a random chance of 1 in 2 succeeds:
		now arena dig flyer is in underside;
	otherwise:
		now love letter is in underside;
	assign-lgths;

to set-other-default:
	now cask is not flipped-yet;

the magenta nametag is a thing. rgtext of magenta nametag is "[rc][gc][rc][gc][rc]*[rc]". lgth of nametag is 7. gpos of nametag is 7. rpos of nametag is 1.

does the player mean listening to the air: it is likely.

does the player mean doing something with the magenta nametag:
	if notices section is not visited:
		it is possible;
	if the nametag is not visible:
		it is very unlikely;
	if the instructional tag is examined:
		it is unlikely;
	if the tagged gadget is examined:
		it is unlikely;
	it is possible.

understand "tag" and "name tag" as nametag.

instead of taking off instructional tag:
	try taking instructional tag;

does the player mean examining the instructional tag when the tagged gadget is handled or tagged gadget is examined: it is likely.

description of magenta nametag is "[nametag-desc]"

check taking off magenta nametag:
	if player is in busiest subsite:
		say "Nobody really cares about nametags, but someone'd give you grief for it." instead;

to say nametag-desc:
	if player is in dy:
		say "[one of]You were the only person stuck with this color. Others got plain white or grey. Surely that can't mean anything...can it?[paragraph break]Because it seems totally useless now [if player wears nametag]on your shirt[otherwise]it's detached[end if][if goat is visible]. But after what you did to the goat, maybe it'll be useful somewhere[otherwise]. You'd like it to be useful for more than identification[end if][or]An ugly magenta[stopping].[no line break]";
		continue the action;
	if player is in notices section:
		say "It's still a bit scratched from your trip through the thorns--about 3/7 of the way from the right.[no line break]";
		continue the action;
	if player wears nametag:
		say "It's stuck to your shirt, all right. Somehow, you got a magenta one while most other people got plain white. It just took a tame nag to put it on, and you hate it but can't bring yourself to pull it off in public.[paragraph break]It does not make you feel like a powerful magnate, that's for sure.[no line break]";
	otherwise:
		say "It's cast off and harmless-looking. Be cool if it could become something useful.[no line break]"

check wearing the magenta nametag:
	if player wears the magenta nametag:
		say "You already are." instead;
	say "It's lost most of its sticky stuff on the back side. It'd just be a gnarled dangler if you reattached it." instead;

chapter Busiest Subsite

Busiest Subsite is a room. "This is the ritziest hotel you'll ever stay in. Terraces are west, a banner hangs over an auditorium entry to the east, and the elevator you came from is to the south. North's a function for later in the day.[paragraph break]There's a sign over an odd side passage the bustling, babbling masses seem to be ignoring."

the bustling masses are scenery in Busiest Subsite. understand "people" as bustling masses.

Include (-
	has transparent animate
-) when defining bustling masses.

the odd side passage is scenery in Busiest Subsite. "It's just under the NONE TRY sign. You could probably ENTER it, or go INSIDE. It's not really any direction you can tell, and you could have sworn it was near a different exit before you turned your back on it just a bit ago. It curves quickly so you can't see much of it."

instead of entering or following the side passage:
	try going inside;

description of masses is "Very diverse, yes, but equally drawn to whatever's to the east. You don't share their eagerness."

the auditorium entry is scenery in busiest subsite. "You'd love any excuse not to join everyone else going there."

terraces are plural-named scenery in subsite. "You can't see the caterers working on the terraces, and you probably don't want to disturb them."

The banner is scenery in Busiest Subsite. "'This way: BEING FIRED!'[one of] People walk by, discussing how the DEBRIEFING should make Mondays dynamos again.[or] You're run over by people saying 'This talk will be a FINE BRIDGE to a new job!'[or] Has nobody else noticed this sign? Or what it says?[stopping]"

instead of following banner:
	try going east instead;

the above-sign is privately-named scenery in Busiest Subsite. the printed name of above-sign is "sign above the passage". understand "sign" as above-sign when player is in subsite.

instead of taking above-sign:
	say "[grounds]";
	
instead of taking banner:
	say "[grounds]";
	
to say grounds:
	say "Appropriation of company property is potentially a fireable offense. Oh, wait."

instead of taking brazier:
	say "That'd probably be illegal.";	
	
description of above-sign is "It says NONE TRY, and it's just above a side passage people are ignoring."

instead of entering passage:
	try going inside instead;

check going north in Busiest Subsite:
	say "That way's a costlier cloister of rooms than recent firees deserve. The measure-a-resume event will be there, but not for another couple hours. You don't really want to get an agent, anyway.";
	been-check north instead;

check going east in Busiest Subsite:
	say "That way's an hour or two of listening to an under-action denunciator. You sure?";
	if the player consents:
		say "You look back at the [if above-sign is examined]NONE TRY passage[otherwise]passage you ignored--NONE TRY above it[end if]. Well, obviously it should say NO ENTRY. You hope whatever new job you find has a decent vision plan. But you realize you can't be fussy.[paragraph break]One more look back. Someone else walks through--he's wearing the same color nametag you are. During the talk--based on a horrendous bowdlerization of Robert Frost's [italic type]The Road Less Traveled[roman type]--you try to convince yourself you weren't really missing anything. It doesn't quite work. Maybe you could've done something else?";
		end the story;
	otherwise:
		say "You pause before entering. The lecture seems like it'd be fun to ditch.";
		been-check east instead;
		

check going south in Busiest Subsite:
	say "You already spent too much time in the overlate elevator.";
	been-check south instead;

check going outside in Busiest Subsite:
	if sign above the passage is examined or passage is examined:
		say "Do you mean trying the passage?";
		if the player consents:
			try entering the passage instead;
		otherwise:
			say "Well, it's still there, if you want to enter it." instead;
	say "There are many ways out. You may want to be more specific or have somewhere interesting or unusual in mind." instead;
	
the overlate elevator is scenery in Busiest Subsite.

description of elevator is "It's closed, now, on its way to being late for the next batch of people.";

check opening elevator:
	say "Foolhardy." instead;

check going west in Busiest Subsite:
	say "The Caterers['] Terraces are west. They're rendin['] dinner. You see someone sweating over a bizarre brazier, hear how this next lecture will 'Prep us for supper,' and promptly lose your will to do anything for a minute.";
	been-check west instead;

the brazier is scenery in subsite. "It's a brazier. [bug-report]"

instead of doing something with the brazier when brazier is visible:
	say "Next meal's too far off, and you've got no business on the terraces.";

check examining (this is the examine-dirs rule):
	if noun is a direction:
		say "Just try going that way instead." instead;

check going inside in Busiest Subsite:
	say "Someone tugs you by the arm. 'Can't you read? It says NO ENTRY! Quit dawdling and get to the DEBRIEFING already!'[paragraph break]You check and make sure. No, still NONE TRY. As you brush him off, you hear 'Nice. Smart. Miscreant!'[paragraph break]";
	wfak;
	say "The voice cuts off. The passage behind is gone. You shake off a momentary fear missing the lecture will go in your life file.";
	now player is in dy instead;
	
instead of thinking:
	say "Yes, I hope this game makes you do a little of this, but nothing painful."

check dropping magenta nametag:
	if player is in subsite:
		say "Enough people have discarded their nametags carelessly. No littering please." instead;
	if player is in dy:
		say "Littering in a nature-filled area like this is not a good idea." instead;

dirs-missed is a number that varies. dirs-missed is 0.

to been-check (subdir - a direction):
	choose row with mydir of subdir in table of subsitedirchecks;
	if explyet entry is true:
		continue the action;
	now explyet entry is true;
	increment dirs-missed;
	choose row dirs-missed in table of subsitedirchecks;
	if there is a blabola entry:
		say "[blabola entry]";

table of subsitedirchecks
mydir	explyet	blabola
north	false	--
south	false	"[line break]Your eye catches the side passage, which looks more interesting than the standard ways out. Maybe you could give it [if sign above the passage is examined or odd side passage is examined]another [otherwise]a [end if]look.[line break]"
east	false	"[line break]You feel sort of cornered by all the main passages. Sneaking off anywhere would be kind of fun.[line break]"
west	false	"[line break]The standard, usual directions don't seem to cut it. But there has to be somewhere else.[line break]"

check exiting in busiest subsite:
	try going outside instead;

chapter dy

dy is a privately-named room. printed name of dy is "A Dry Yard".

instead of exiting in dy:
	if thorn is visible:
		say "It looks like there's a passage beyond that thorn, but you'll need to get rid of it, first.";
	otherwise:
		say "You feel a sudden sense of vertigo as you walk towards the passage where the thorn was. You realize you have lost your sense of direction, and for whatever reason, you're preoccupied with finding it before you can continue. So, which way is the thorn?"

report looking in dy for the first time:
	say "[italic type]So what was that lecture supposed to be about? One word changing how we look at things, how things are? Rubbish. It has to be. You think.[line break][roman type]"

description of dy is "Nothing particularly beautiful grows here. All manner of prickly things and dried-up weeds lie around[if thorn is visible]. You see a big huge thorn blocking an exit. You're not sure what direction--you've gotten spun around--but the smaller thorns shooting off from it block any hope of slipping by[otherwise]. You see an exit in some direction or other where the thorn was[end if]. [if goat is not visible]The carefully pruned shrub and the brush laid next to it seem much more orderly[otherwise]The shrub and brush the goat wrecked are here, too[end if]."

the carefully pruned shrub is scenery in dy. "[if goat is not visible]It's been trimmed into the shape of a cat[otherwise]The goat ate enough of it, it doesn't look like anything any more[end if]."

understand "cat" as pruned shrub.

instead of taking the shrub:
	say "[worked]";
	
instead of taking the brush:
	say "[worked]";
	
to say worked:
	say "But someone worked hard on that. And besides, you just need[if toga is not visible]ed[end if] to take a message from it.";

the brush is scenery in dy. "The broken branches lying here in a pile [if goat is visible]no longer spell anything[otherwise]seem to spell GATO[end if]."

dry weeds are plural-named scenery in dy. understand "weed/things" as weeds when dry weeds are visible.

description of dry weeds is "In some areas, so dense they seem sewed together."

The gateman is a person. "Old Man Almond, the gateman, is grousing around here."

understand "old man" and "almond" as gateman when gateman is visible.

instead of taking weeds:
	say "Gardening will not help your current situation."
	
section gato

part begin-actual-play

part by-turnstile

every turn when player is in silent rut (this is the turnstile-guys-through rule):
	if coe is unvisited:
		say "[guy-by-description]";

to say guy-by-description:
	say "Dude, an undead! He walks by through the turnstile, which glows briefly. He's got [burns] and is wearing [glasstype] and [nosey]."

to say burns:
	say "[one of]muttonchop sideburns[or]a fu-manchu mustache[or]a glorious handlebar mustache[or]a carefully-trimmed Vandyke[or]a mustache your high school buddies derisively referred to as a 'molestache'[or]general patchy scuzz[or]one of those thin beards mostly along the jawline[or]a pencil-thin mustache[or]a soul patch[or]a goatee, no mustache[at random]"

to say glasstype:
	say "[one of]browline glasses[or]sixties horn-rimmed glasses[or]reading glasses[or]granny glasses[or]bifocals[or]clip-on sunglasses[or]a pretentious monocle[or]rimless glasses[or]specs with a hideous elastic strap at the back[or]racquetball goggles[at random]"
	
to say nosey:
	say "[one of]a fake warty witch's conker[or]a fake parrot beak[or]a red plastic ball under his glasses[or]a fake pig snout[or]a Pinocchio-style wooden rod sprouting from his face[or]a pickle-shaped and -colored schnoz[at random]"

book putverb

book items

section blank inventory

check taking inventory:
	if the first thing held by the player is nothing:
		say "Not very 'in.'" instead;
	say "Item time! [run paragraph on]";

after printing the name of the emitter while taking inventory:
	if emitter is angstgnatted:
		say " (full of angst gnats)";
	else if condo-evac is false:
		say " (full, you, uh, think)";
	otherwise:
		say " (nothing inside)";
	
after taking inventory:
	if xray-vision is true:
		say "Your possessions seem a bit--well, you don't know, more transparent. Perhaps it's that saltine you ate.";
	continue the action;

book rooms

part metros

chapter aya

aya is a privately-named room. "This dead-end alley is littered with [if words are visible]words, which almost drown out the intense beats heard elsewhere in the city, and [end if]garbage. Only way out is back east."

garbage is scenery in aya. "The garbage smell would make a bear gag--stink bug, gunk bits, etc."

the stink bug is scenery in aya. "[bug-report]"

understand "gunk bits" as stink bug.

instead of doing something with the stink bug:
	if the current action is eating:
		say "Um, no." instead;
	say "Not important[one of]. Just here for local flavor. Not very palatable flavor, but flavor nonetheless[or][stopping]."

understand "trash" as garbage.

instead of taking garbage: say "Don't know what you'd hope to take from it--a stink bug? Gunk bits? Plus you'd feel like a beggar[if player wears velcro]--and it might gunk up your velcro mittens[end if]."

instead of smelling garbage:
	try examining garbage;
	
check going nowhere in aya:
	say "Only way out's back east." instead;
	
some words are a thing in aya. rgtext of words is "[rc][rc][rc][rc][rc]". gpos of words is 5. rpos of words is 4. legth of words is 5.

description of words is "They're not visible[if words are in bag], but you can feel them pulsing against the bag[otherwise], but they certainly sound angry. Sharp. Pointed. Lethal.[end if]."

the sword is a thing. it is not fixed in place.

description of sword is "[if sword is in sheath]It's sheathed now. It's not really worth admiring. The handle looks slippery[otherwise]It's for great justice, you're sure. But instead of blood gutters it's shaped more like a microphone. Odd[end if]."

the hilt is part of the sword.

description of hilt is "It's bright and polished and slippery-looking, [if stickyhanded is true]but with your sticky hands, maybe you could grab it[otherwise]and you probably couldn't hold onto it for long right now[end if]."

instead of taking hilt:
	try taking sword;
	
check taking sword:
	if stickyhanded is false:
		say "The sword slips from your hands[if player wears velcro], and there's nothing the velcro mittens can hook onto[end if][if resin is not visible]. Hmm, you haven't found anything that'd help you grasp it[end if]." instead;
	say "The sword is yours, and you grab it firmly in your resined hands!";
	now player has sword instead;

check propfliptoing sword:
	if player is in aya and player does not have sheath:
		say "The words become nastier for a second, forming into a sword. But it is too large and terrifying to carry. The talk resumes as the sword disappears. You'll probably need something to put the sword in[if player has noise bag], and that noise bag won't work[end if]." instead;
	if sword is off-stage:
		continue the action;
	if player is not in Abyss and player does not have sheath:
		say "The words are sharp enough, but you have nothing to hold a sword in." instead;
	if player has sheath and sword is visible:
		say "With your sheath handy, yeah, it's time to get yourself a sword. And if those words are anything to go by, it's gonna be a NASTY sword.";
		continue the action;
	if player is not in Bassy Abyss:
		if player has noise bag:
			say "The time's not right, yet. There's no threat near you." instead;
		say "You can't go waving a sword about the city. Probably some laws or something." instead;

check opening noise bag:
	if player is in aya:
		if words are in aya and player has noise bag:
			now words are in noise bag;
			say "You open the bag, and you feel the waves and sounds of the words swirling into the bag. You close it." instead;
	if words are in bag:
		if location of player is spis:
			say "The nerds put their hands to their ears. 'You're holding us hostage!' they yell. 'Look, it's not our fault others can't insulate themselves from the noise, but if you want to let dumber people be as happy as we are, here. Stupid idealist. Here. Want the tulip? Trade the keycard. And get out.' You do.";
			now keycard is in lalaland;
			now player has lit-up tulip;
			reg-inc;
			choose row 2 in the table of tulip-acq;
			now chosen entry is true;
			move player to underside, without printing a room description instead;
		if location of player is Abyss:
			if beast is visible:
				say "The beast has no time for words! You need to be more direct.";
			otherwise:
				say "The beats will drown the words out.";
			the rule succeeds;
		say "The words in the bag would do no good here.";
		the rule succeeds;
	if player is not in Bassy:
		say "The beats aren't focused enough here to go pouring in the bag." instead;

for printing a locale paragraph about the words:
	set the locale priority of the words to 0.

printed name of aya is "A Yell Alley"

the words are scenery in aya.

the sword is a thing.

section motto-tomato

A motto is a thing in aya. "A motto is here. It's on a placard, which would be wholly unnoticeable and useless among all the garbage but for how [italic type]bad[roman type] a motto it is."

understand "placard" and "writing" as motto when motto is visible.

instead of putting on garbage:
	if noun is garbage:
		say "I'm sure the garbage would feel insulted." instead;
	say "You're a hero, not a garbage man!";

instead of inserting into garbage:
	if noun is garbage:
		say "I'm sure the garbage would feel insulted." instead;
	say "You're a hero, not a garbage man!";

rgtext of motto is "[rc][rc][rc][rc][gc][gc]". gpos of motto is 4. rpos of motto is 2. legth of motto is 6.

description of the motto is "[ex-placard]Somehow, someone named Too-Apt Pat Ott has managed to make a hash of grammar, logic, math, science and basic human decency in sixty letters. Sixty-four if it were spelled correctly.[paragraph break]I'd tell you what it said, but you-the-person would be the worse for it. It's like Monty Python's funniest joke in the world that way. Even considering it makes you feel worried you'll forget simple stupid stuff, like what's a fruit and what's a vegetable."

to say ex-placard:
	if the player's command matches the text "placard":
		say "[one of]Fourth wall note: it's not really, primarily, a placard. People seeing it probably think more, 'Hey! A motto!' than 'Hey! A placard!'[paragraph break][or][stopping]"
	
after taking the motto:
	say "You manage take the motto without taking it to heart.";
	
instead of eating motto:
	say "I hope you're not dumb enough to swallow that. Figuratively or literally.";

understand "rotten" and "rotten tomato" as tomato.

the description of the tomato is "It's as rotten as the motto of Pat Ott's you extracted it from. It is probably even more rotten on the inside, not that you have the bravery to check. Yet it also has that staying power. It doesn't have any mold on it, yet."

instead of eating tomato:
	say "That might give you a gas saga."


chapter Cramped Red Camp

Cramped Red Camp is east of aya. "Here in this cross between a platform and farm plot lie the shocking red tents of the terminally un- and under-employed. You are quickly excluded from this once-mum commune, since you seem actually busy. An alley lies to the west, a posh shop is to the east, and back south is where you started from[if dry cake is in red condo].[paragraph break]A driveway with a yard view leads north[otherwise].[paragraph break]The cordoned red condo you're not welcome back at is to the north[end if]. You hear painful music."

the music is scenery in red camp. "Wretchedly sharp harps, fake-smile upbeat protest stuff, but a change from the beats."

the driveway is scenery in cramped red camp. "It's immaculate compared to the rest of the city[if condo-evac is true], but you probably shouldn't follow it again[end if]."

the yard view is scenery in cramped red camp. "You can't see much of it until you're on the driveway, what with the buildings in the way, but once you're on the driveway, you'll need to move it or get nicked for loitering. Too bad. It's probably even nicer than the driveway."

the tents are scenery in red camp. understand "commune" as tents.

description of tents is "Dingy, ripped, and should not be described like this, because I had a blanket command."

instead of doing something with the tents:
	say "Best not look too far into the tents. People live there. People less helpful than the deadbeat. Who need their no-suds privacy from sounds."

understand "way" and "drive" as driveway when driveway is visible.

instead of following driveway:
	try going north;

the dead beat deadbeat is a male person in Cramped Red Camp. "A deadbeat is sitting on the ground here, looking dead beat[if condo-evac is false]. His eyes dart between you and the lost corn. Clearly, he resents the work your presence is forcing on him[otherwise]. Despite your heroism with the cake in the condo, he barely acknowledges you[end if]."

check scaning (this is the antlers and deadbeat silliness rule):
	if noun is antlers:
		say "[one of]Someone comes over to you and says 'Put that thing away. They're genuine. And new, too! The owner said he just bought them! Or shot a seventeen-point buck to get them! Well, I'm not sure, but I didn't see them here last time.'[or]The antlers seem too useless to allow scanning. Your gadget can't even be bothered to buzz.[stopping]" instead;
	if noun is deadbeat or noun is deadbeat-shirt or noun is goatee:
		say "'Whoah, don't tase me, bro. I have civil rights!'" instead;

check giving to deadbeat:
	if noun is a flower or noun is clover or noun is tulip:
		say "'Flowers, man? During this totally intense class warfare?'" instead;
	say "'One act of guilt-driven re-gifting is no plan of general redistribution and fairness, man!'" instead;
	
instead of giving cake to dead beat deadbeat:
	say "'I didn't, like, want the cake, man. I just, like, wanted a symbolic blow against... (insert class-warfare cliche here.) Plus, totally not enough icing.'"

to say corn-vis:
	say "[one of]. He points to the corn, says 'You've earned it, dude,' and shrugs. Wow, you got promoted from man to dude[or]. He looks over at the corn, motions to pick it up, but gets tired and curses litterbugs in general[stopping]";
	
instead of showing emitter to deadbeat:
	try objasking deadbeat about emitter instead;

instead of giving emitter to deadbeat:
	try objasking deadbeat about emitter instead;

understand "man" as deadbeat when deadbeat is visible.

to say are-were:
	if controls are in op:
		say " were";
	otherwise:
		say "[']re";

description of deadbeat is "He's wearing some of the hardest-trashed threads you've seen. You know he's a deadbeat because his t-shirt says DEAD BEAT DEADBEAT. Which fits in with this world, yes, but about as half-donkeyed as possible. Oh, and he has a ridiculous goatee, too."

the ridiculous goatee is part of the deadbeat.

description of the ridiculous goatee is "It seems half braided and wretchedly uneven, only partially by design. 'Babes like my chillin['] ill chin.' he says.".

instead of taking the goatee:
	say "It looks kind of icky. And you don't need to start a fight." instead;
	
the deadbeat-shirt is privately-named. the deadbeat wears the deadbeat-shirt. understand "threads" and "t-shirt" and "shirt" as deadbeat-shirt.

instead of taking deadbeat-shirt:
	say "Bad idea. He likes no suds with the sounds."

description of deadbeat-shirt is "The DEAD BEAT DEADBEAT text is as teleological as most of the shirt wearer's arguments, you suspect."

the posh shop is scenery in cramped red camp. "It's apparently called ELF ROW'S FLOWERS and says 'Go floral or go fall!'"

the alley is scenery in cramped red camp. "It looks a bit seedy but not that dangerous."

the barrier is scenery in cramped red camp. "[if dry cake is in red condo]It doesn't look electrical[otherwise]You could sneak back under, but you'd get chased[end if]."

check going north in Cramped Red Camp:
	if condo-evac is true:
		say "You'd probably be recognized and arrested." instead;

check going nowhere in cramped red camp:
	if noun is up or noun is down:
		continue the action;
	say "There're condos north[if dry cake is not off-stage], which you were booted out of[end if], an alley west, a flower shop east, and a seedier area south." instead;

the fuzzy clover is a thing in Cramped Red Camp. "A small clover is growing here.". rgtext of clover is "[rc][rc][rc][rc][rc][rc]". lgth of clover is 6. gpos of clover is 4. rpos of clover is 3.

description of clover is "You touch the leaves a bit, and they seem to hook into your skin. You count two layers of five leaves each."

instead of wearing clover:
	say "You can't wear that as-is, but it'd be helpful to stick to something if it became wearable.";

The leaves are part of the clover.

description of leaves is "Rather fuzzy--they seem almost to try to stick to you."

instead of taking the leaves:
	say "Picking the clover apart will just ruin it."

Report taking the clover:
	say "You pluck it.";
	the rule succeeds;

some velcro is a singular-named thing. it is not fixed in place. understand "mittens" and "gloves" as velcro when velcro is visible.

instead of tying velcro to fuzzy wall:
	say "You test to make sure it sticks. It does. You could climb up, if you wanted.";

instead of putting velcro on fuzzy wall:
	try tying velcro to fuzzy wall instead;

after doing something with controls:
	set the pronoun it to controls;
	continue the action;

after doing something with velcro:
	set the pronoun them to velcro;
	set the pronoun it to velcro;
	continue the action;

the indefinite article of velcro is "some".

description of velcro is "You're not smitten by these mittens--they're like what some kids had to wear when they were younger. Two glovey bits, with a long connecting string. They're made of a bunch of hooks and can probably catch on something if need be."

the string is part of the velcro. description of string is "Must've been made from the clover stem. It's not switchable, trust me--it's part of the ex-clover."

the hooks are part of the velcro. the hooks are plural-named. description of hooks is "You know, small little hooks that attach to loopy bits."

the velcro is wearable.

some lost corn is a singular-named thing. rgtext of lost corn is "[rc][gc][rc][gc][rc][gc][rc][rc]". lgth of lost corn is 8. gpos of lost corn is 5. rpos of lost corn is 3. the indefinite article of lost corn is "some".

initial appearance of the lost corn is "That lost corn is still in that corner. Sorry if that sounds corny."

the discolored buttons are part of the corn. description of discolored buttons is "They give you this weird urge to push them."

instead of taking mattress:
	try pushing mattress instead;

before doing something with the mattress:
	if emitter is off-stage and night thing is in lalaland:
		if current action is taking or current action is pushing or current action is pulling:
			say "As you move the mattress, but it seems that the bump moves with it. So you decide to search the mattress instead.";
			try examining mattress instead;
	
before pushing:
	if noun is buttons or noun is lost corn:
		say "Maybe if the corn were hooked up to something it could do something to. And it weren't, well, lost corn." instead;

the corner is scenery in cramped red camp. description of corner is "[if corn is in cramped red camp]It's not very interesting, but that's where the corn rolled[otherwise]You don't even remember which corner the corn rolled into, now[end if]."

check going to cramped red camp for the first time:
	say "As you walk north, you run into someone who is carrying a basket full of corn. You help him pick everything up. Or so you think. As he runs off, cursing you, you look in a corner--a corn ear!";
	now lost corn is in cramped red camp;

description of lost corn is "It's got a bunch of different kernels, and some look like buttons because they're not yellow[if player has corn]. The deadbeat might have something to say about it, if you could put up with him. His information would certainly be off the mark, but in Yorpwald, that's better than maybe being wrong[end if]."

check propfliptoing keycard:
	if condo-evac is false and dry cake is visible:
		say "The chatterers would have something to talk about at their next get-together. Like the weirdo who did something scary to that cake and got arrested. Ix-nay on the agic-may [']til they're gone." instead;

check propfliptoing controls:
	if condo-evac is false and lost corn is visible:
		say "The dead beat deadbeat might not beat you dead, but you'd get maligned ere you could complete that legerdemain." instead;

check taking lost corn:
	if condo-evac is false:
		say "The deadbeat almost gets up[one of]. 'Hey, Man, like, how do we know someone undisreputable doesn't, like, like the Man, man? Hey? Hey?' he says, with no dearth of hatred as he[or]but just[stopping] jerks his thumb north. He might not do anything, but his friends might pour out from their tents. You don't need a fight right now.[paragraph break]Maybe you need to strike a blow against some rich people to show you belong." instead;

the yellow kernels are part of the lost corn. understand "buttons" as kernels.

does the player mean examining the discolored buttons: it is likely.

does the player mean examining the air: it is very likely.

description of kernels is "Most are yellow but others are discolored--red and black and white."

understand "red/black/white buttons/kernels" and "red/black/white" as discolored buttons.

some controls are a thing.

check pushing controls:
	if controls are in op:
		say "No need to fiddle further.";
	otherwise:
		say "They're not hooked up to anything.";
	the rule succeeds;
	
description of controls is "This panel features all manner of black and yellow and red doohickeys and there's a small warning message saying REPLACEMENT CONTROLS IN CASE OF SURGE TO TRACKS. They've got instructions, which are surprisingly clear and even leave you wiser about u-boat wires."

the icontrols are part of the controls. the icontrols are privately-named. understand "instructions" as icontrols when icontrols are visible.

description of icontrols is "[if controls are in op]They served you well. No need to reread.[otherwise]They'll be useful once you find a place to put the controls.[end if]"

description of instructions is "Surprisingly readable, enough so you could just stuff these controls somewhere appropriate and things would get working. Some days you're just lucky, eh?"

chapter cordoned red condo

The Cordoned Red Condo is north of Cramped Red Camp.

check going inside in cordoned red condo:
	say "You're already inside." instead;
	
check going outside in cordoned red condo:
	try going south instead;
	
the talkers are plural-named scenery in cordoned red condo. understand "talker" as talkers.

after doing something with talkers:
	set the pronoun it to talkers;
	continue the action;

Include (-
	has transparent animate
-) when defining talkers.

description of talkers is "They're obviously part of the upper crust. They speak disdainfully of the red camp outside and how if they can't tolerate the beats, they can just buy a condo too."

does the player mean throwing the tomato at the talkers: it is likely.

instead of throwing tomato at talkers:
	say "That would get rid of one of them, but the rest would oust you. Perhaps a more pervasive threat would disperse them."

the antlers are plural-named scenery in cordoned red condo. rgtext of antlers is "[rc][rc][rc][rc][rc][rc][gc]". lgth of antlers is 7. gpos of antlers is 6. rpos of antlers is 7.

instead of taking the antlers:
	say "They are utterly useless, even for impressing people. Actually, they're all just here for the free food, too."

description of the antlers is "The antlers are just the sort of annoying thing that people only buy to show they own something more expensive than you do. They seem more chipped and worn than you'd expect for something so showy. As if they'd been moved around a lot."

the rentals are plural-named scenery.

description of the rentals is "The antlers are gone."

check going to condo for the first time:
	say "Since you shave, you can pass as one of the 'haves.' The beats drown out as you enter the condo. 'Eat On!' cries the hired help.[paragraph break]'Neato! ... No tea? ATONE!'[paragraph break]You are inspected and deemed less unworthy than that deadbeat who tried to enter--the one still sitting in the camp--but all the same, you're warned not to try anything funny."

A barnacled candelabra is scenery in Cordoned Red Condo.

description of candelabra is "It's tasteful and clever, as much as you hate to admit it."

instead of doing something with candelabra:
	if current action is not scaning:
		say "It's too high above to note if those are even real barnacles.";
	otherwise:
		say "You couldn't even jump to get close to it. But it's not important. Someone just put it there because they could.";

description of cordoned red condo is "[if condo-evac is true][cake-blab][otherwise]The beats aren't audible here, but conversation voices rant on. The talkers seem to take you for a stalker, but nobody wants to get too close[end if]. A barnacled candelabra hangs above[if antlers are visible], antlers are attached to a wall[end if], and carpets provide garish spectra on the floor[if condo-evac is false]. [paragraph break]All in all, this place probably has a high property value, and people would be horrified if something happened to make it drop[end if]."

the carpets are scenery in cordoned red condo. understand "spectra" and "carpet" as carpets.

after doing something with carpets:
	set the pronoun it to carpets;
	continue the action;

description of carpets is "Tacky but not worth taking."

instead of looking under carpets:
	say "This is a tribute to Nord and Bert, not Zork.";

instead of taking carpets:
	say "Too bulky, and what if they turned into something precast?"

to say cake-blab:
	say "You probably want to [if cake is in red condo]take the cake and run[otherwise]get going before everyone gets back[end if]"

before going nowhere in cordoned red condo:
	if condo-evac is true:
		say "Best not hang around the scene of the crime." instead;

check going north in cordoned red condo:
	say "It's dark that way. You step on a pliers and turn back before hitting any more perils." instead;

check going west in cordoned red condo:
	say "A man with a handkerchief over his nose welcomes you to the Allergy Gallery. You turn back." instead;

check going east in cordoned red condo:
	say "Ladies with ideals grab your garb. 'Step, pest!' they cry. 'You'll startle a starlet!'" instead;
	
check going down in cordoned red condo:
	say "An officious water informs you there are no wines for a swine like yourself. They are reserved for actors and a costar named Castro." instead;

check going up in cordoned red condo:
	say "You hear a warden being warned, and soon you're routed to a detour back where you started." instead;

check going in cordoned red condo:
	if noun is diagonal:
		say "You skulk in a corner for a bit. People not-notice you a bit more than before." instead;

Some dry cake is a singular-named thing in Cordoned Red Condo. "Some dry cake is here[if talkers are visible]. Nobody else is lower-class enough to eat it, but everyone's making sure nobody lower-class nabs it[otherwise]. And nobody is here to stop you from taking it any more[end if].". rgtext of dry cake is "[rc][rc][gc][gc][gc][rc][rc]". lgth of dry cake is 7. gpos of dry cake is 6. rpos of dry cake is 1. indefinite article of dry cake is "some".

description of dry cake is "The least stale of what's left, but weird and silvery and in no state to taste[if player does not have cake]. Someone sees you eyeing the cake, and you suddenly wish to steal it out of spite[end if]."

Instead of eating the dry cake:
	if cake is unexamined:
		say "It's in no state to taste[if cake is unexamined], as mentioned earlier[end if], and it won't be.";

check taking the dry cake:
	if condo-evac is false:
		say "[one of]You overhear tales of what happens to those that steal. [or]You feel a slap on your hand from someone who mentions Doc Bleak gave a creep a crepe once. A bold cake blockade! [or]Someone shakes a can of Best Cola and points it at you. Obstacle! [or]'Greedbag! Er, bad egg!' someone calls out. [or][cycling]Someone else takes a turn watching you." instead;

after taking dry cake:
	say "You take care not to let the cake tear as you pick it up.";

understand "steal [something]" as taking.

condo-barred is a truth state that varies. condo-barred is usually false.

after taking the lost corn:
	say "'All yours, man,' says the deadbeat. 'It's some pretty good corn, too. Ask me if you like want details.'";

check going south in Cordoned Red Condo:
	if condo-evac is true:
		if player does not have dry cake and player does not have keycard:
			say "A caveat before you vacate: some loot here may be a tool later. [run paragraph on]";
			try taking the dry cake;
		say "You migrate, ragtime music in your head. From the south, some camped folks decamp for har-hars and rah-rahs. 'Roaches scare! Ho!'[paragraph break]You're the camp's scamp now, so to speak. I bet you could even take that corn now!";
		now condo-barred is true;
			
report taking the dry cake:
	say "Taking it is a piece of cake with everyone gone, you note drily.";
	the rule succeeds;

check taking heaths:
	if the player's command matches the regular expression "\bheath\b":
		say "(all of them, why not?)";

check taking begonias:
	if the player's command matches the regular expression "begonia\b":
		say "(all of them, why not?)";

the keycard is a thing.

understand "key card" and "card" as keycard.

description of keycard is "It's plain and metallic[if barcode is part of the keycard], with a barcode that might help with optical scanners[otherwise], and the only thing that indicates it's a keycard is small text saying FOR AUTHORIZED ACCESS ONLY[end if]."

check putting barcode on keycard:
	say "You unpeel the barcode, slap it on the keycard, and it is a perfect fit! Also, you stuff the adhesive backing in your pocket, where you forget about it until you can find a proper garbage can. You haven't seen any in this city yet[if player is in alley], and you don't want to add to the garbage piles here[end if].";
	now adhesive backing is in lalaland;
	now barcode is part of the keycard instead;

check tying barcode to:
	try putting barcode on second noun instead;

check tying to barcode:
	try tying second noun to noun instead;

check taking barcode:
	if barcode is part of the keycard:
		say "The barcode is where it should be. It you unstuck it, it probably wouldn't re-stick." instead;
			
check putting barcode on:
	if barcode is part of the keycard:
		say "The barcode is where it should be. It you unstuck it, it probably wouldn't re-stick." instead;
	if second noun is controls:
		say "The barcode might be detected by some sort of electronic controls, but it doesn't belong on them." instead;
	if second noun is tomato or second noun is dry cake:
		say "Boy. When this hero gig ends, maybe don't become a grocery store manager, okay?" instead;
	say "You can't really see what the barcode would do there." instead;
	
check inserting barcode into keycard:
	try putting barcode on keycard instead;

chapter Elf Row's Flowers

check looking in elf row's flowers for the first time:
	say "'What low serf seeks our flowers?' you hear as you enter. 'He must do better than that primrose promiser! A rose leaves us sore! No succor in a crocus either! Players with parsley are rewarded sparely!'";
	
Elf Row's Flowers is east of Cramped Red Camp. "You're in a flower shop manned (elfed?) by faeries. You can go back west to the camp."

check going nowhere in Elf Row's Flowers:
	say "The only safe way is back west." instead;
	
check going east in Elf Row's Flowers:
	say "[one of]You suspect the fairies will smack the boor who explores the back rooms[or]You wouldn't want to disturb them while they're measuring geraniums or anything back there[or]That's where all the magic happens. It'll happen to you if you get nosy[at random]." instead;

[A loner loaner is a person in Elf Row's Flowers.]
Some freesia faeries are people in Elf Row's Flowers. understand "freesia/ fairies" as faeries.

instead of taking freesia faeries:
	say "They're too fast for you. And they are not souvenirs.";

check objasking faeries about gardenia:
	if player has gardenia:
		say "They see your gardenia and snatch it away before you can offer properly. 'Yes! Yes!' [run paragraph on]";
		try giving gardenia to faeries instead;

description of freesia faeries is "They are maybe a foot high, but each one of them carries a small wand you're sure you don't want pointed at you."

the wands are part of the faeries. understand "wand" as wands.

after doing something with wands:
	set the pronoun it to wands;
	continue the action;

description of wands is "[one of]You lean over to peer at the wands and realize the faeries are watching you stare. You feel embarrassed[or]You've got enough magic for one person. You don't need more[stopping]."

instead of doing something other than examining the wands:
	say "The wands are the faeries[']. Best not meddle.";

to say he-be:
	if heaths are in flowers or begonias are in flowers:
		if heaths are in lalaland or begonias are in lalaland:
			say "You have performed your magic admirably on our gift. Go and slay the beast![no line break]";
		otherwise:
			say "'Your magic will work on what you have. But you must use your ingenuity.'[no line break]";
	otherwise:
		say "'You may take one of the groups of flowers if you bring us some. They are the last of the batch from before the Night Thing came.'";

to say sure-you:
	say "[if heaths are not in flowers or begonias are not in flowers]! We are sure you can do something with your reward[end if]";

some heaths are a plural-named flower in Elf Row's Flowers. they are not fixed in place. rgtext of heaths is "[rc][rc][rc][rc][rc][rc]". lgth of heaths is 6. gpos of heaths is 6. rpos of heaths is 5.

understand "heath" as heaths.

description of heaths is "They're tied together so their stems seem to form a sort of narrow tube. But surely flowers alone aren't going to stop the thumping here."

stems are part of the heaths. understand "narrow tube" as stems when stems are visible.

description of stems is "The stems seem to be forming a narrow tube. It's weird--you can't see anything holding the flowers together."

the sheath is a wearable container.

after taking sheath:
	say "It's a bit awkward, but fortunately it can hook on to your pants or shorts or whatever you're wearing.";
	now player wears the sheath;
	
check taking off sheath:
	say "It'd be too awkward to carry." instead;

description of sheath is "It's a dull grey, with the words FOR GREAT QUIETNESS inscribed in some Gothic font."

some begonias are a plural-named flower in Elf Row's Flowers. they are not fixed in place. rgtext of begonias is "[rc][rc][rc][rc][rc][rc][gc][rc]". lgth of begonias is 8. gpos of begonias is 5. rpos of begonias is 3.

understand "begonia" as begonias.

the noise bag is an openable closed container. understand "noisebag" as noise bag.

instead of wearing bag:
	say "Oh, come on, you're not doing that badly. Figuring the begonias was pretty clever, really[if words are not in bag]--you just haven't found what to put in the bag, yet!";

description of the noise bag is "It's labeled OPEN TO CAPTURE NOISES."

check inserting into the noise bag:
	if noun is words and words are not held:
		say "You can't take them, though they do seem to be louder with the bag around." instead;
	if noun is beats:
		say "They will not get sucked in like the random noises did." instead;
	say "It seems like the bag can only contain non-substantial things. Noisy things." instead;

description of begonias is "Reddish, pinkish, white. Kind of fragile, but they seem to silence things a bit, like they could somehow contain the sounds if you knew how to mess with them."

fairy-worthy is a truth state that varies. fairy-worthy is usually false.

check giving to faeries:
	if noun is clover:
		if fairy-worthy is true:
			say "It's not as valuable as the gardenia, and you already have a flower, anyway." instead;
		say "[one of]'That is not a flower! It is just a silly weed!'[or]They probably want a real flower.[stopping]" instead;
	if noun is tulip:
		say "'That's not a real flower! That's technology!' they cry." instead;
	if noun is heaths or noun is begonias:
		say "[if player has noun]It's yours now[otherwise]It's not even yours[end if]." instead;
	say "They have no need of adventuresome things." instead;

check giving clover to faeries:
	if fairy-worthy is true:
		say "The faeries smile pityingly at you. Clovers grow everywhere[if fairy-worthy is true]. But they don't need anything more from you[otherwise]. They're more like weeds than flowers[end if]."

instead of showing to faeries:
	try giving noun to faeries;

check giving gardenia to faeries:
	say "They're visibly impressed. 'Perhaps you are the word smyth who will fulfill the sword myth! We offer a flower of your choice. But only one[if brocade is in elf row's flowers]. And take the brocade, too[end if]. Our flowers, or what you make of them, cannot provoke or participate in violence, but they may be able to contain the weapon you need.'";
	now the gardenia is in lalaland;
	now fairy-worthy is true instead;

report taking heaths:
	say "'Remember, adventurer! The flowers cannot be used for violence. They may be a suitable vessel.'";
	the rule succeeds;

report taking begonias:
	say "'Remember, adventurer! The flowers cannot be used for violence. They may be a suitable vessel.'";
	the rule succeeds;

check taking:
	if noun is spam and spam is part of sandwich:
		say "You peel the meat from the bread and take both.";
		now player has bread;
		now player has spam;
		now sandwich is in lalaland;
	if location of player is Elf Row's Flowers:
		if noun is brocade:
			if fairy-worthy is false:
				say "The faeries buzz around for a bit, peeved at your boldness. 'Well, maybe you will do something for us some day.'";
			otherwise:
				say "'It is the least we can do for you! Thank you for helping us!'";
			say "[line break]The FREE TO [if fairy-worthy is true]FREEDOM FIGHTERS[otherwise]FREELOADERS[end if] scratching by the brocade vanishes as you take it. Nice magic touch, that.";
			now scratchings are in lalaland;
			now player has brocade instead;
		if noun is begonias and player has heaths:
			say "[greedy-greedy]" instead;
		if noun is heaths and player has begonias:
			say "[greedy-greedy]" instead;
		if fairy-worthy is false:
			say "The faeries buzz around a bit. 'Bring us a gift first, human! [one of]New florae, loafer[or]No loaner for a loner[cycling]!' they cry." instead;
			
to say greedy-greedy:
	say "They cry 'We already gave you one clump of flowers for your one flower! So unfair, sour naif!'";

the brocade is a thing in Elf Row's Flowers. "A small piece of brocade is discarded away from the merchandise. You see scratchings above it.". rgtext of brocade is "[gc][rc][rc][gc][rc][gc][gc]". lgth of brocade is 7. gpos of brocade is 1. rpos of brocade is 7.

the merchandise is scenery in elf row's flowers. "All sorts of flower-related semi-magical stuff spans the store. You wouldn't know what to do with any of it, [if begonias are not in elf row's flowers and heaths are not in elf row's flowers]but you've got something for your troubles[else if fairy-worthy is true]but you can probably take the heaths or begonias[otherwise]but maybe if you help the faeries, they'll give you something[end if]."

does the player mean objasking the faeries about the merchandise: it is likely.

understand "flowers/seeds" as merchandise when merchandise is visible.

description of brocade is "It's rather plain and white with lots of black bars. It would take a rabider braider than most to weave such a design[if player does not have brocade]. Scratchings above it indicate it's not part of the normal merchandise[end if]."

scratchings are plural-named scenery in Elf Row's Flowers.

description of scratchings is "FREE TO [if fairy-worthy is true]FREEDOM FIGHTERS[otherwise]FREELOADERS[end if]."

the counter is scenery in Elf Row's Flowers.

description of counter is "It divides you from the fairies. It doesn't divide them from you, though, since they can fly[if brocade is in elf row's flowers]. There's a brocade here, under some scratchings[otherwise]."

the barcode is a thing. understand "code" and "bar" as barcode when barcode is visible.

description of barcode is "[if barcode is part of keycard]It's stuck to the keycard now, and it'd be unwise to peel it off[otherwise]It's got the adhesive backing so it looks like you could PUT it ON something. Barcodes just don't exist by themselves[end if]."

the adhesive backing is part of the barcode. adhesive backing is undesc.

instead of doing something with the adhesive backing:
	if the current action is examining:
		say "It's clear, and it'll keep the sticky bit on the barcode until you know where to put it." instead;
	say "You don't need to futz with the adhesive, explicitly. Just do what you want/need with the barcode.";

to say dont-peel:
	say "When you're ready to put the barcode on something, you don't need to peel it. Just PUT it ON whatever.";
	
section propflipping Elf Row

to say rude:
	say "You turn your back and focus so the faeries can't see you mangle their gift. Or so they aren't jealous, or they don't laugh at how dumb your magic is"

check propfliptoing sheath:
	if player is in Elf Row's Flowers:
		if fairy-worthy is false:
			say "The heaths are (still) merchandise. You don't want to know what faeries do to vandals." instead;
		if begonias are not in elf row's flowers:
			say "You already got the begonias--don't mess with the merchandise." instead;
		if player does not have heaths:
			try taking heaths;
		say "[rude].[line break]";

check propfliptoing barcode:
	if player does not have the brocade:
		say "You take the brocade first--you're a bit embarrassed your magic seems more, well, applied than theirs.";
		try taking brocade;
	if player is in Elf Row's Flowers:
		say "[rude].[line break]";

check propfliptoing noise bag:
	if the player's command does not include "noise":
		say "Yes, the begonias should become a bag, but you need to state what sort fully. Magic isn't THAT easy." instead;
	if player is in Elf Row's Flowers:
		if fairy-worthy is false:
			say "The begonias are (still) merchandise. You don't want to know what faeries do to vandals." instead;
		if heaths are not in elf row's flowers:
			say "You already got the heaths--don't mess with the merchandise." instead;
		if player does not have begonias:
			try taking begonias;
		say "[rude].[line break]";

to say in-sheath:
	now sword is contained in the sheath;
	do nothing;

the block buying rule is not listed in any rulebook.

instead of buying:
	if noun is player:
		say "Silly." instead;
	if player has noun:
		say "You already have that." instead;
	if location of player is subsite:
		say "This is not exactly a shopping mall." instead;
	if location of player is notices section:
		if noun is gateway:
			if gateman is visible:
				say "Bribery won't work, even if you had money, which you don't." instead;
			say "It's more real than a piece of the Brooklyn Bridge, but no." instead;
		if noun is gateman:
			say "Just ask him for what he knows. Knowledge is free." instead;
		say "Awfully generous, but your quest-aiding items here are all free." instead;
	if noun is tulip and nerds are visible:
		say "You have nothing to barter with. Either find what to ask, or outflank them somehow." instead;
	if location of player is trips strip:
		say "Your job is presumably to raise property values here, not take advantage of them." instead;
	if player is in elf row's flowers:
		if noun is brocade:
			say "It is free." instead;
		if noun is heaths or noun is begonias:
			say "You don't have any money[if fairy-worthy is true], and the faeries don't seem to want to barter any more[otherwise], but perhaps if you gave the faeries a flower, they would be willing to trade[end if]." instead;
	if peasant is visible and noun is hay:
		say "Maybe there's something you can give him, instead." instead;
	if peasant is visible and noun is peasant:
		say "Things aren't THAT feudal here." instead;
	say "Nothing seems to be on sale here.";

chapter Undesired Underside

check going inside in Undesired Underside:
	say "That's ambiguous with the hotel west and the metal door east." instead;
	
check going outside in undesired underside:
	say "You already are in the not-so-great outdoors." instead;

Undesired Underside is south of Cramped Red Camp. "This is some sort of seedy underground intersection with a train station south. You can also go north to what looks like a camp. You see the down escalator you came by. A building to the west advertises itself as The Ol['] Hotel[if bus-pass is false], and a bus pass seems like it was left on the ground to annoy you[end if]."

the down escalator is scenery in underside.

description of down escalator is "It's the down escalator you came here by. You can't see the end of it, way above."

instead of climbing down escalator: try going up instead.

drainage is in Undesired Underside. "That drainage you stepped in is [one of][or]still [stopping]around, taunting you as much as drainage can. You see [letter-or-flyer] in it. Cleaning it up would help take this city back."

drainage is fixed in place.

instead of taking drainage:
	say "It smells too powerful to even consider taking in its present form. Looks bad, too.";

check propfliptoing gardenia:
	if drainage is visible:
		now flyer is in lalaland;
		now can of beer is in lalaland;
		now soggy love letter is in lalaland;

description of drainage is "It smells and looks disgusting. Just the sort of thing that needs to be cleaned up or reprocessed to get this city looking nice again. A beer can and a [if flyer is visible]flyer[otherwise]soggy note[end if] float in it.". rgtext of drainage is "[rc][rc][rc][rc][rc][rc][rc][rc]". lgth of drainage is 8. gpos of drainage is 7. rpos of drainage is 6.

to say letter-or-flyer:
	say "a can of beer and a ";
	if arena dig flyer is visible:
		say "flyer";
	otherwise if soggy letter is visible:
		say "soggy note";
	otherwise:
		say "heap of gunk where an item should be ([bug-report])";

the can of beer is auxiliary scenery in Undesired Underside. description of can of beer is "It appears to be A[gc]ED [rc]AIN brand beer. Hooray, truth in advertising?". rgtext of can of beer is "[rc][rc][rc][gc][rc][rc][gc][rc]". lgth of can of beer is 8. gpos of can of beer is 2. rpos of can of beer is 1.

understand "aged rain" and "aged" and "rain" and "beer/ can/" as can of beer when can of beer is visible.

does the player mean doing something with can of beer:
	if can of beer is not visible:
		it is very unlikely;
	it is likely.

instead of doing something other than examining or reading or scaning the can of beer:
	if current action is drinking:
		say "Yuck. The beer's name may be even truer now than when some poor sap bought it." instead;
	say "You don't really want to get close to it." instead;

does the player mean doing something with the love letter when the love letter is visible: it is very likely.

the soggy love letter is auxiliary scenery. description of soggy love letter is "Someone has written [bold type]DEAR INGA:[roman type] in big letters. The waters have smudged the smaller writing, which is probably for the best.". lgth of soggy love letter is 8. gpos of soggy love letter is 7. rpos of soggy love letter is 3. rgtext of soggy letter is "[rc][rc][rc][rc][rc][gc][rc][gc]".

understand "dear inga" and "soggy note" as soggy love letter when soggy love letter is visible.

instead of doing something other than examining or reading or scaning the soggy letter:
	say "You don't really want to get close to it." instead;

the arena dig flyer is auxiliary scenery. lgth of dig flyer is 8. gpos of dig flyer is 8. rpos of dig flyer is 1. rgtext of dig flyer is "[rc][rc][rc][rc][rc][rc][gc][rc]". description of arena dig flyer is "The flyer seems like an invitation to an ARENA DIG--perhaps an arena that never got built, or a shindig in some arena."

the arena dig flyer is fixed in place.

instead of doing something other than examining or reading or scaning the arena dig flyer:
	say "You don't really want to get close to it." instead;

a gardenia is a flower.

description of gardenia is "It's white, and it's just one flower instead of the whole bush. But it's the only one you've seen in the city proper[if elf row's flowers are visited], and people or humanoids who go in for this sort of thing might value it[end if]."

check going east in undesired underside:
	if player has the keycard and barcode is part of the keycard:
		now metallic door is open;
		say "You [if spis is unvisited]decide the keycard might work, and what do you know, it Does. You[otherwise]swipe your keycard and[end if] go east as the door opens.";

bus-pass is a truth state that varies.

the bus pass is scenery in underside. "[bug-report]"

instead of doing something with the bus pass:
	now bus-pass is true;
	say "[one of]When you get near the bus pass, it ups the bass. Ouch.[or]You aren't doing that again. It will have to stay as a symbol of the over-urbanization of Mt. Rose until you fix things.[stopping]"

chapter The Ol' Hotel

does the player mean throwing the tomato at the night thing: it is likely

toh is a privately-named room. the printed name of toh is "The Ol['] Hotel".

toh is west of Undesired Underside. "What would a beaten-down city be without an ol['] hotel? Both have seen better days. Still hard not to loathe a hotel in this state. Everything's boarded up. The only way out is east."

The Night Thing is a person in toh. "A scary Night Thing is here, sitting on some sort of mattress."

after doing something with the night thing:
	set the pronoun it to night thing;
	continue the action;

description of Night Thing is "It's grown wrong, a gigantic hairy eyeless potato clearly not the right girth. And it's wired weird, with a cruel ulcer for a mouth. It seems more in the mood for bellowing than fighting, but it still probably doesn't belong in the hotel. It's too big to fight and too fearsome to get near. It appears to be sitting on some sort of mattress."

understand "potato" as night thing when night thing is visible.

instead of showing something to the night thing:
	try giving noun to night thing instead;

instead of giving something to the night thing:
	if noun is tomato:
		say "It stops roaring a bit to sniff the tomato, but it clearly doesn't trust you, doesn't have arms, and doesn't understand or care why you'd hate if it bit off your hand along with the tomato. So you decide to back up a bit and throw the tomato.";
		try throwing the tomato at the night thing instead;
	say "It's roaring too much. It seems upset about something, and you look unlikely to help.";

the cruel ulcer is part of the night thing. description of cruel ulcer is "It's nasty and yellowed and an indication maybe scaring people in the hotel is more stressful than you think. You wonder how food goes down there."

the ketchup bottle is a thing in toh. "A ketchup bottle lies here, torn apart--and you have a prime suspect in the Night Thing that [if night thing is visible]is roaring nearby[otherwise]was here[end if]."

the description of the ketchup bottle is "It is THE PUCK brand ketchup. They apparently put, heck, whatever they want in it! And you'll like it that way!"

instead of taking the ketchup bottle:
	if night thing is not visible:
		say "Trust me. You don't need it.";
	otherwise:
		say "The night thing roars. Though it's drained the bottle of its contents, it still feels an emotional attachment you would be wise not to break."

instead of eating the ketchup:
	say "You don't know where it's been. Oh, wait, even worse, you DO.";
	
check going nowhere in toh:
	say "Bad idea to explore further. This place should've probably been condemned long ago." instead;
	
check going outside in toh:
	try going east instead;
	
The termite emitter is a device. "A termite emitter lies here, forgotten. It still looks operational. Perhaps it's part of why the hotel is so beaten up."

the termite emitter can be angstgnatted. the termite emitter is not angstgnatted.

does the player mean switching on the termite emitter: it is very likely.

instead of opening termite emitter:
	say "You can't figure how. There's a reason for that[if condo-evac is true and flies-in-emitter is false], and it's empty anyway[otherwise]. Better to switch it, probably[end if]."

description of termite emitter is "It's a black box the side of maybe two CD cases stacked[if condo-evac is true]. It's empty now, you hope[otherwise]. It probably holds a lot of termites. You can switch it when you need to[end if]. Someone has scratched a circle with a line through it over a dollar sign[if condo-evac is false]. You can't hear or see the termites that should be in there, but hey, faith[else if flies-in-emitter is true]. The deadbeat put some angst gnats in it for you[otherwise]. It's probably empty after your hijinks[end if].";

check examining the termite emitter:
	ignore the examine devices rule.

understand "black box" and "machine" as termite emitter.

condo-evac is a truth state that varies. condo-evac is usually false.

flies-in-emitter is a truth state that varies. flies-in-emitter is false.

check switching on the termite emitter:
	if flies-in-emitter is true:
		if deadbeat is visible:
			say "It'd be rude to throw the gift back in the deadbeat's face." instead;
		if faeries are visible:
			say "The faeries are on your side. Try someone else, maybe." instead;
		if player is in Abyss:
			say "Those angst gnats have no chance. You need something much more powerful." instead;
		if location is not spis:
			say "There's nobody here worth bumming out with the angst gnats." instead;
		say "The angst gnats rise, fitful, like fruit-flies or as furies flit. The nerds gag and cough, but it isn't til they start complaining to each other or worrying about the poor repressed deadbeats they didn't do enough for that you snatch the lit-up tulip as they scalp-clasp and flail about.[paragraph break]'Why didn't you ASK about the DARKNESS?' one of them moans as he swipes or the tulip--and only knocks the keycard out of your hand.[paragraph break]They continue moaning, but now it is about economic equality. You hear one point out that all this sensitivity to social stuff will make them more sensitive for the ladies. Then another one argues that it's a Heisenberg Uncertainty Principle sort of thing that if people know you might be angling for that emo stuff, it won't work.[paragraph break]You're almost sucked in, until you realize you have a city to save, and you chuck your keycard at them in frustration over how dumb smart people can be before running out.";
		now nerds-unwelcome is true;
		now player has the lit-up tulip;
		choose row 3 in the table of tulip-acq;
		now chosen entry is true;
		reg-inc;
		now emitter is not angstgnatted;
		move player to underside instead;
	if condo-evac is true:
		say "It's empty now." instead;
	now termite emitter is switched off;
	if location is toh:
		say "This place has seen enough abuse. Maybe find one that hasn't?" instead;
	if location is elf row's flowers:
		say "Trust me. You don't want to know what the faeries would do back to you." instead;
	if location is spis:
		say "You're being watched too closely." instead;
	if location of player is cordoned red condo:
		if condo-evac is true:
			say "It's already crawling with termites." instead;
		now condo-evac is true;
		now talkers are in lalaland;
		reg-inc;
		say "You activate. They vacate it. It's like a petard--those who prated, depart and ratchet down the chatter. The emitter dies down after the last termite escapes." instead;
	if location is ett:
		say "The city's public transport has taken enough of a beating. It doesn't need that." instead;
	if location is Bassy:
		say "The termites would put up a valiant but futile effort." instead;
	say "The termites don't have enough to eat here." instead;

the smartest mattress is a container in toh. the mattress is fixed in place.

instead of doing something with the mattress when mattress is visible:
	if night thing is visible:
		say "Can't do or see much with the night thing on it.";
	otherwise :
		continue the action;

rule for printing a locale paragraph about the smartest mattress:
	if night thing is not visible:
		say "The mattress still conforms to the shape of the night thing that sat on it[if emitter is off-stage]--but it also has a bump in one corner[end if].";
	now mattress is mentioned;

description of smartest mattress is "The Smartest Mattress doesn't have a one-page proof of Fermat's Last Theorem in it or anything. It just 'remembers' the shape of whatever sat on it for a long, long time[if emitter is off-stage][get-the-bump][otherwise]. It doesn't seem to have held anything other than the termite emitter[end if]."

to say get-the-bump:
	say ". The bump in the mattress is obvious enough that you have to take another look. You find a small machine labeled, apparently, a termite emitter, and you take it";
	now bump is in lalaland;
	now player has the emitter;

the bump is part of the mattress. description of the bump is "Why not examine the whole mattress instead?"

instead of doing something with bump:
	say "On getting close to the bump you realize it's due something wedged inside the mattress. You search around, find a small machine that is apparently a termite emitter, and take it.";
	now bump is in lalaland;
	now player has the emitter;

instead of examining the bump:
	try examining the mattress;

understand "something" as bump when bump is part of the mattress and mattress is visible.

instead of taking cruel ulcer:
	say "That's neither useful nor lucre."

check throwing tomato at Night Thing:
	say "A direct hit! The tomato goes into the cruel ulcer, and the Night Thing smiles at first at the ketchupiness of the tomato. Then the inner rottenness and backwards logic from the (oops, a) motto kicks in. The beast wonders what sort of villain would DO that to ketchup. It looks at you in fear, sure you have more where that came from, and retreats into an unseen hole, scratching and clawing.[paragraph break]The mattress the Night Thing was on is slightly but noticeably bumpy. It's also ripped open, and it's concealing something rather badly.";
	now tomato is in lalaland;
	now night thing is in lalaland;
	reg-inc;
	the rule succeeds;
	
does the player mean throwing the tomato at the night thing: it is very likely.

rule for supplying a missing second noun when throwing:
	if night thing is visible:
		now the second noun is the night thing;

rule for supplying a missing noun when throwing:
	if night thing is visible:
		now the noun is the tomato;


chapter Special Place is Special

spis is a privately-named room. the printed name of spis is "Special Place is Special".

[get dealing, delegate]
[unlit until]
[a gofer forage]

every turn while player is in spis:
	if nerds are visible:
		say "[one of]'Since I'm not omniscient...' Ish, note his tone. Sarcastic, arctic ass.[or]'...How I do it, idiot.'[paragraph break]'Who?'[or]'...butler I rebuilt!'[or]Someone shows how [one of]mediums are for dummies[or]he cut down a stupider disputer[stopping]. Others agree, eager.[or]They debate whether dumb people deserve clamor or calm.[or]Someone blames the dim-led middle for This Whole Mess.[or]A conceders['] crescendo follows a claim nerds'd LIKE to help everyday stupid people.[or]You hear a MENSA-y yesman encourage someone to repeat himself.[or]You hear them recite ironies noisier than necessary.[or]The nerds give amens that MENSA is the greatest social group ever.[at random]"

after going to spis for the first time:
	reg-inc;
	continue the action;

description of spis is "It's quiet here, mostly due to the soundproofed walls. It'd make a nice mob combine, but it's a small bunker for nerds who just got sick of the thumping[one of]. You suspect the actual entertainment center's in back, and you'd have to learn a secret handshake or something to get there. But you've got a city to save[or][stopping]."

soundproofed walls are plural-named scenery in spis. "They've got those perpendicular ridges in sets of three that kill sound. The nerds won't shut up, though."

check going nowhere in spis:
	say "There's nowhere to go but back west. Well, okay, the nerds have a few cool secret passages you won't get to see, but those don't count." instead;
	
The metallic door is a door. it is east of underside and west of spis.

the initial appearance of the metallic door is "A large metallic door is to the [if player is in spis]west. It's got no sensor on this side, so you can probably just walk back through it[otherwise]east. It has a warning graphic repelling any old chump from entering. There's also a sensor off to the side[prob-reads][end if]."

to say prob-reads:
	say "[if spis is unvisited]. Probably reads something-or-other to let you in[end if]"

description of metallic door is "It's very solid[if metallic door is open] when it's closed[end if][if player is in underside], and the warning graphic gives it some semblance of character[end if]."

the warning graphic is part of the metallic door. understand "sod" and "figure" and "stick figure" and "stick-figure" as warning graphic.

instead of taking the warning graphic:
	say "Since it's part of the door, you take the warning graphic in the metaphorical sense and vow not to act like the poor schelps depicted on it.";

description of warning graphic is "It's a red circle with a line through it. The poor struck-through sods in the center appear to be stick-figure parodies of stupid people. You note one [one of]stuffing a metal fork in a toaster[or]moving a rook pawn to start a chess game[or]thought-bubbling that 8+8=17[or]picking his nose[or]with his arm around Clippy[or]reading a tabloid magazine[or]hitting a computer that is not plugged in[or]about to walk over an open manhole[or]about to misplace a square manhole[or]watching a sporting event on TV[or]with a barbell crushing his neck[or]carrying a boombox[or]driving an SUV[or]getting an easy Sudoku all wrong[or]wearing a baseball hat backwards[or]wearing a sports jersey[or]drinking and smoking[or]feeding a shark[or]getting a tattoo[in random order]. Then you lose where it was in all the dizzying variety."

The sensor is scenery in undesired underside. description of the sensor is "It has a small red optical beam."

to say want-wanted:
	if player has tulip:
		say "got the tulip from";
	otherwise if spis is visited:
		say "guard the tulip";
	otherwise:
		say "want to see"
		
understand "senser" as sensor.

the priv-camp is a backdrop. the priv-camp is in underside and cramped red camp. understand "camp" as priv-camp.

description of priv-camp is "It's pretty dirty and meager[if red camp is visited], as you already saw[otherwise], but maybe you'll get help there[end if]."

the red optical beam is part of the sensor. description of red optical beam is "I wouldn't advise staring into it, but you could put something on it."

instead of inserting into the beam:
	try putting noun on beam;

keycard-put is a truth state that varies. keycard-put is usually false.

check putting barcode on sensor:
	if barcode is part of the keycard:
		try putting keycard on sensor instead;
	say "You try holding the barcode up to the sensor, which pulses a bit. But the door doesn't seem to open. Perhaps the barcode needs to be a part of something." instead;
	
check putting keycard on sensor:
	if barcode is not part of keycard:
		say "Hm. The blank keycard doesn't seem to work. Maybe you could jazz it up a bit." instead;
	if keycard-put is false:
		say "What do you know? It works! The door slides open!";
	otherwise:
		say "The door slides open again.";
	reg-inc;
	now metallic door is open instead;

nerds-unwelcome is a truth state that varies.

check opening the metallic door:
	if location of player is spis:
		say "The door slides open as you step near it.";
		now metallic door is open;
		the rule succeeds;
	if player has tulip:
		if nerds-unwelcome is true:
			say "You aren't welcome back[if keycard is in lalaland], and besides, no keycard[end if]." instead;
		say "It'd be awkward to revisit. 'What?! We gave you the tulip! We can't solve all you dumb people's problems. We have our own!' Then they'd go on to bemoan how only dumb people seem to enter politics." instead;
	if keycard-put is true:
		say "You use the keycard to open the door again.";
		try putting keycard on sensor instead;
	if location of player is undesired and keycard is off-stage:
		say "You hear braying laughter behind the door. 'Hey! Some unintellectual's trying to get in. Like we'd make it a piece of cake for them to.' Then someone else admonishes the speaker for ending a sentence with a preposition." instead;
	if noun is open:
		say "But it already is. For now." instead;
	say "The metallic door's got no handle. It's probably operated by the sensor[if player has keycard], which might open if you put something like your keycard on it[otherwise], but you don't seem to have anything to activate it[end if]." instead;

check putting on the sensor:
	ignore the can't put onto what's not a supporter rule;
	if noun is gadget:
		say "The gadget is more for gauging than doing, but yeah, you probably need something high-tech. Oh, the gadget remains silent, too." instead;
	if noun is the dry cake:
		say "Opening an electronic door is not such a piece of cake. Something more metallic." instead;
	say "The sensor remains deactivated. You'll probably want to go with something more high-tech." instead;

check putting on the red optical beam:
	ignore the can't put onto what's not a supporter rule;
	try putting noun on sensor instead;

check putting it on:
	if second noun is the red optical beam:
		try putting noun on sensor instead;

every turn when metallic door was open:
	now metallic door is closed;
	if metallic door is visible:
		say "The metallic door slides shut.";

some nerds are people in spis.

check giving something to nerds:
	if player has lit-up tulip:
		say "You have nothing they want, and they gave you what you need." instead;
	if noun is gadget:
		say "They have their own techie toys to futz with. With which to futz." instead;
	say "They aren't big on material things. They probably prefer to show how smart they are." instead;

initial appearance of nerds is "Some nerds are here, nattering about technology and such."

description of nerds is "They're sitting around nattering about nerd things. They don't look particularly nerdy, but most of their conversation seems to revolve around how smart other people aren't. They're babblier than most but don't seem bribable."

the lit-up tulip is a thing in spis.

instead of taking lit-up tulip:
	if nerds are visible and player does not have tulip:
		say "Someone grabs your wrist. 'No light for the unenlightened! You have to ASK NERDS the right thing before you can get it. What do you want it for? Why? And ASK us, don't TELL us.'[paragraph break]That was kind of abrasive[one of]. You think of the places you haven't been and the obstacles to getting there[or]. You wonder if there's a way to disrupt their little gathering if you can't figure what to ask[or]. These intellectual elites are almost as annoying as the economic elites[cycling].";

check dropping the lit-up tulip:
	say "No. It's useful."

description of lit-up tulip is "No fickler flicker or slight lights--its twinkle lets it wink for a long while, with no way to turn it off or on."

instead of switching off tulip:
	if player does not have the tulip:
		say "That's the nerds[']." instead;
	say "Don't worry. It's very, very energy efficient. Designed that way.";

instead of switching on tulip:
	say "It's already lit."

to say nerd-dark:
	if player does not have tulip:
		say "'What, you want us to do everything for you? We're exhausted from building this soundproofing. But whatever. Here. Have this lit-up tulip.'";
		now player has the tulip;
		reg-inc;
		choose row 1 in the table of tulip-acq;
		now chosen entry is true;
	otherwise:
		say "'What?! We gave you the tulip! We can't solve all you dumb people's problems. We have our own!' They go on to bemoan how only dumb people seem to enter politics.[run paragraph on]";

chapter Er, Train Terrain

to say if-piggy:
	say "[if neon pig is visible]neon pig embedded in the wall seems even sillier[else if controls are in op]controls are working nicely in the opening you made[otherwise]empty opening you made needs something to fit[end if][run paragraph on]"

ett is a privately-named room. the printed name of ett is "Er, Train Terrain". ett is south of Undesired Underside. "You're in a disused train station[if power-shut is false]. It's a bit dark, but not as bad as to the east, where noise seems to be echoing[otherwise]. The tracks are east and, it seems, so is the noise[end if]--it's louder here than anywhere else. You reckon you should strain for the darkness to the east, with more rats in than trains.[paragraph break]You could go back north, [if player is on fuzzy looking wall]though you may need to get off the wall first[otherwise]and there's a fuzzy looking wall to the west, too. You notice pointed art and a faded ad on it, but the [if-piggy][end if]. A rail is to the east[if power-shut is false]--it seems to be sparking quite a bit, so best not to step on it, yet[otherwise]--it's no longer sparking[end if]."

check scaning in ett (this is the can't scan some things til you climbed rule):
	if noun is neon pig or noun is op:
		if player is not on fuzzy looking wall:
			say "You are too far away at the moment." instead;

check going west in ett:
	say "The big fuzzy wall is that way.";

check going nowhere in ett:
	say "The tracks are east, and the city is back north." instead;

the faded ad is scenery in ett. "It references some dead fad while proclaiming TRANSIT FOR ARTISAN AND TSARINA."

the pointed art is scenery in ett. "It's a typical train depot PSA saying 'DON'T PIRATE.' You hope the creator wasn't paid. Rotten stuff."

the fuzzy looking wall is scenery in ett. "The wall looks and feels like steel wool, from top to bottom."

the fuzzy looking wall is a supporter.

instead of climbing the down escalator:
	try going up instead;

check going up in ett:
	try climbing the fuzzy looking wall instead;
	
check going up in underside:
	say "Everything starts okay, but just as you get a glimpse of the top, the escalator starts speeding down more quickly. You try jumping a step, but it doesn't matter. Eventually, you're thrown to the bottom of the escalator, a bit more tired than before." instead;

power-shut is a truth state that varies. power-shut is usually false.

east-tried is a truth state that varies. east-tried is usually true.

check going east in ett:
	now east-tried is true;
	if power-shut is false:
		say "There may be an evil lair that way, but it's behind a live rail in the darkness. You'd touch it before you got there[if shoes are in lalaland]. Even those rubber shoes won't insulate you[end if]." instead;
	if player does not have tulip:
		say "You go east but reach a thin ledge. It needs to be lightened before you go further. You have no item to help with that." instead;
	if player does not have sheath and player does not have noise bag:
		say "You hear an even worse roaring than usual. Nothing you're carrying seems up to defeating it. You double back." instead;
	if player has noise bag and noise bag does not contain words:
		say "You hear an even worse noise as you walk across the tracks. You run back across where you can reflect how it would be nice if you had something in that noise bag to combat...whatever it is there." instead;
	if player has sheath and sword is not in sheath:
		say "You hear an even worse noise as you walk across the tracks. You run back across where you can reflect how it would be nice if you had a weapon in that sheath." instead;
	say "Your lit-up tulip gives a welcoming anemic glow. You're ready to tackle whatever's lying craven in the cavern. Unfortunately, the glow, while it helps you avoid tripping over rails, doesn't save you from a passage weakened by the beats. You tumble down.";

does the player mean climbing the fuzzy looking wall: it is very likely.

check scaning op (this is the too far from opening rule):
	if player is not on fuzzy looking wall:
		say "You're too far away." instead;
		
check climbing:
	if player is in ett:
		if noun is not fuzzy looking wall:
			say "Hm. Perhaps you meant to climb the fuzzy wall instead.";

check climbing fuzzy looking wall:
	if player is on fuzzy looking wall:
		say "You can't see anything worth getting to above the recess. And you're up dangerously high as is." instead;
	if player has velcro and player is not wearing the velcro:
		say "You can't climb up there with bare hands, so you decide to wear the velcro. [run paragraph on]";
		try silently wearing the velcro;
	if player is not wearing the velcro:
		say "The wall's vertical. You'd need something that sticks to the bits of steel wool." instead;
	otherwise:
		say "The velcro strip is just the thing to climb the wall and reach the small recess.";
		now player is on the fuzzy looking wall instead;

the neon pig is scenery in ett.

gpos of neon pig is 3. rpos of neon pig is 7. rgtext of neon pig is "RRRGRRG".

the op is privately-named scenery. understand "opening" as op. the printed name of op is "the opening"

description of op is "[if controls are in op]It's not an opening any more, what with the controls covering it. But since you put the controls where they belong, I won't get pedantic[else if player is on fuzzy looking wall]You can see that the opening isn't just a blank area, though it doesn't lead anywhere much. Once you got rid of that neon pig, there are still all kinds of receptacles and such that could be attached to something electrical[otherwise]You look back up at the opening you made. Seems something belongs in there[end if]."

the tracks are scenery in ett. understand "rail" and "rails" and "track" as tracks.

instead of taking tracks:
	if power-shut is false:
		say "They're charged, and they're too heavy, anyway.";
	otherwise:
		say "And what would you do with the tracks?[if sword is visible] You have a sword as a weapon[otherwise] There's got to be a better weapon if you need it[end if]."

instead of putting something on tracks:
	say "Good way to get electrocuted. You'll need to find some place to switch them off before you get near them.";
	
understand "use [text]" as a mistake ("Use is a bit too vague for this game's parser to understand. You can SWITCH something, or PUT something IN/ON something, or even ATTACH something to something. You can PUSH a button or OPEN something, as well. Specific items should clue you what to do if you EXAMINE them.")

after examining tracks:
	set pronoun it to noun;
	set pronoun them to noun;

description of tracks is "[if power-shut is false]Sparking dangerously. If you want to get by, you'll probably have to figure where to shut the tracks off[otherwise]Much calmer now that you switched the power off[end if]."

check climbing the op:
	say "(I'm assuming you meant the wall leading to it.)";
	
description of neon pig is "[if player is not on fuzzy looking wall]It's obviously awful and tacky and worse than nothing, even from this distance. It's eating a bunch of glowing flowers and is probably hogging (sorry) electricity and causing the tracks to sputter[otherwise]It's even tackier in person. Thankfully, the neon isn't painfully bright, but it would be nice to just dispose of it somehow[end if]."

after examining the neon pig:
	if player is on fuzzy looking wall:
		say "The artist seems to have signed his name--do you want a look?";
		if the player consents:
			say "The swine who conceived this is, according to the red glowing cursive script, named [first custom style]INPENGO[roman type].";
		otherwise:
			say "Don't blame you.";
	continue the action;

the red glowing cursive script is part of the pig.

description of cursive script is "It spells [first custom style]INPENGO[roman type]."

[certifying ?? new items]

instead of doing something with the script:
	if current action is examining:
		continue the action;
	if current action is scaning or certifying or cring or rectifying:
		say "[if player is on fuzzy wall]The script doesn't play nice with your gadget[otherwise]Too far away[end if]." instead;
	otherwise:
		say "It's part of the pig--not much else to do but examine it.";

the glowing flowers are part of the neon pig. glowing flowers are undesc.

instead of doing something with the glowing flowers:
	say "Their symbolism is obvious and probably meant to be. You probably want to deal with the whole hog (sorry.)"

instead of taking neon pig:
	say "You'd let go of the wall and fall to your doom.";

check inserting into op:
	if noun is not controls:
		say "Won't fit, wouldn't stay anyway. You need something squarish and electrical. Or electronic." instead;
	if power-shut is true:
		say "You've already crossed things up." instead;
	if player is not on fuzzy looking wall:
		say "You're too far away to do that." instead;
	otherwise:
		say  "They fit well. One of the red buttons starts blinking and whirring, and you flip the switch by it and, BAM! You hear a rush of power. The tracks below seem to stop sparking.";
		reg-inc;
		now controls are in op;
		now power-shut is true instead;

understand "controls" as op when controls are in op.

chapter Bassy Abyss

Bassy Abyss is east of ett. "Well, this is it[if beats are visible]. You feel like a movie star, with the beats pulsing in the background to lead you on to defeating--oh, wait. The beats ARE what you're trying to defeat[rieuw][else if beast is visible]. The beast is growling, probably to frighten or distract you into doing nothing constructive. I guess it worked for so long when it was incorporeal and it got lazy[rieuw][otherwise]. You should've been kicked back to the trips strip, since you won. This is a BUG[end if]."

every turn when player is in Abyss:
	if siren was visible:
		if siren is visible:
			say "[bold type]RRIIEEUUWW.[roman type] That siren just won't shut up. Man, you just want to pound it into a powder.";
	if noise bag is visible and words are in noise bag:
		say "The words in your noise bag [one of]rattle it enough that you feel a sharp poke in your leg[or]continue to rattle[stopping]."

to say rieuw:
	say "[if siren is visible][one of].[paragraph break]A siren off to the side kicks in, to make things even more head-achingly techno up in here. Too loud to get near[or]. And still, that stupid siren near you.[stopping][end if]";

check going nowhere in Bassy Abyss:
	say "There's no easy way out. But you sense you have what you need[if siren is visible]. Or you can get that last bit[end if]." instead;

section beats-beast

the beats are a backdrop. rgtext of beats is "[gc][gc][gc][rc][rc]". legth of beats is 5. gpos of beats is 1. rpos of beats is 4. understand "thumping" as beats.

description of beats is "You're sure some idiot would call them funky, but you don't have time for aesthetics."

instead of taking the beats:
	say "Hmph. Your ears are already sort of taking a beating, but that's not really the same thing."

check examining beats:
	if night thing is visible or player is in Bassy or location of player is spis:
		say "You can't hear the beats right now, but then, there's [if nerds are visible]chatter[otherwise]chatter[end if] to contend with." instead;

The beast is a thing.

description of beast is "It's blob-shaped with one tooth and one eye and inside-out headphones that still transmit music."

Include (-
	has transparent animate
-) when defining beast.

section siren-resin

the siren is scenery in Bassy Abyss. rgtext of siren is "[rc][rc][rc][rc][gc]". rpos of siren is 5. gpos of siren is 3. legth of siren is 5.

instead of doing something other than examining or attacking or scaning or propfliptoing the siren:
	if siren is not visible:
		continue the action;
	say "It's too darn loud, even a few feet away. Getting close to it would incapacitate you.";

description of siren is "Blue and red and flashing and very loud. You'd love to pound it into a powder."

the resin is a thing. printed name of resin is "some resin[if resin is held] (all over your hands)[end if]"

instead of dropping resin:
	say "You'll have to wash it off, and there's no running water nearby." instead;

description of resin is "It feels sticky on your hands."

stickyhanded is a truth state that varies. stickyhanded is usually false.

part Sortie

chapter Ug, Frenetic Centrifuge

ufc is a privately-named room. "[if centrifuge-stopped is true]Exits lie north and east[maybe-shift]![otherwise]Two exits are spinning counterclockwise. They're a right angle apart.[end if]"

to say maybe-shift:
	say "[if kitchen is unvisited and roomroom is unvisited], and you're pretty sure they won't fake you out at the last minute[otherwise], and they've been behaving since you fixed that dial[end if]";

centrifuge is scenery in ufc.

description of centrifuge is "You're dizzy enough looking around without trying to focus on anything. The dial in the middle of the room seems to control it."

description of exits is "[if numset of dial is 16][otherwise]They're at right angles. You're not sure what direction they're in, or they're supposed to be in, because they're spinning, too. That's part of the puzzle[end if]."

the printed name of ufc is "Ug, Frenetic Centrifuge".

The dial is in ufc. The dial has a number called numset. The numset of the dial is 0. the dial is fixed in place.

the ufcex are privately-named scenery in ufc. understand "exits" and "angle" as ufcex when ufcex is visible. "[ufcex-descrip]."

to say ufcex-descrip:
	if numset of dial is 16:
		say "They're behaving nicely now. That is, not at all. They won't jump if you go through them";
		continue the action;
	say "The exits are spinning but at a right angle to each other[if dial is unexamined]. You glance at the dial, which seems to refer to them and be flashing various letters as well[otherwise], and maybe between the dial and the exit positions you can work out which number you need, here[end if]"

check turning the dial:
	if numset of dial is 16:
		try dialsetting dial to 0 instead;
	say "You need to specify a number: 0-99." instead;

initial appearance of the dial is "There's a dial here in the middle of the room. It is at [numset of the dial] and [if centrifuge-stopped is true]should probably be kept it that way[otherwise]is not spinning with the rest of the room, so you can probably turn it[end if]."

description of dial is "You see EXITS [if numset of dial is 16]N E [otherwise]? ?--you can see two letters, but they're scrolling through the four cardinal directions--[end if]written in the center of its circle. It's currently set to [numset of dial], and you [if numset of dial is 16]want to keep it that way, thank you very much[otherwise]can set it anywhere from 0 to 99[end if]."

dialsetting it to is an action applying to one thing and one number.

understand the command "set [something] to [number]" as something new.

understand "set [something] to [number]" as dialsetting it to.

understand the command "switch [something] to [number]" as something new.

understand "switch [something] to [number]" as dialsetting it to.

understand the command "turn [something] to [number]" as something new.

understand "turn [something] to [number]" as dialsetting it to.

numming is an action applying to one number.

understand the command "[number]" as something new.

understand "[number]" as numming when player is in ufc.

does the player mean examining a sto when player is in ufc: it is unlikely.

check numming:
	try dialsetting dial to number understood instead;

before going in ufc:
	if centrifuge-stopped is false:
		say "You can't get your bearings long enough to lurch at or through an exit. You're getting hit with a lot of G's." instead;


numset-ever-greater is a truth state that varies. numset-ever-greater is false.

check dialsetting it to:
	if noun is not dial:
		say "That doesn't set to a number." instead;
	if numset of dial is 16:
		say "Oh no. You don't need to start that again[if number understood is 16]. Or even think about it[end if]." instead;
	if number understood is numset of dial:
		say "That's where the dial's already set." instead;
	if number understood > 99:
		say "If you tried to spin it that high, you'd break the dial." instead;
	if number understood < 0:
		say "This negative attitude isn't helping you solve the puzzle." instead;
	increment total-tries;
	if number understood is 16:
		now numset of dial is 16;
		say "The room warps a bit. You wonder if you made things worse, but then things start to slow down...yes! You did it![paragraph break][if dial is unexamined]You glance at the dial, see EXITS N E and wonder if it was always there[otherwise]The flipping bits in the dial lock in to say EXITS N E[end if]. And you see, yes, exits are north and east now. My hat, that was mathy.";
		reg-inc;
		now centrifuge-stopped is true instead;
	if numset of dial > 17:
		now numset-ever-greater is true;
	if numset of dial < 16 and number understood < numset of dial:
		binary-eval;
	otherwise if numset of dial > 16 and number understood > numset of dial:
		binary-eval;
	otherwise:
		increment consecutive-binary-tries;
		say "[if number understood and numset of dial skip]The room seems to slow down for a bit before picking up speed and spinning as fast as before[otherwise]The room seems to spin a bit slower. You think[end if]. [run paragraph on]";
		if consecutive-binary-tries is 5:
			say "You feel you're getting closer. Perhaps you'll get it soon.";
		otherwise if consecutive-binary-tries is 8:
			say "You feel and hope you're getting there, but maybe dividing and conquering the possible numbers left will work, if you don't understand the dial clues[if dial is unexamined] it'd be worth examining[end if].";
		else if consecutive-binary-tries is 10:
			say "Inching along with the dial will get there, but maybe you should be--or should have been--more clinical. Each guess can cut half the numbers, though maybe there's an even quicker way.";
		else if consecutive-binary-tries is 12:
			say "Well, you know you'll get there eventually. P.S. I hope you are using the up-arrow and not typing the numbers out.";
		otherwise:
			say "[line break]";
	[d "Currently [total-tries] tries, [consecutive-binary-tries] consecutive binary tries.";]
	if total-tries is 5:
		say "[line break][if dial is unexamined]You wonder if the dial has any sort of hint on it--an odd word or number or letter--anything.[otherwise]Maybe the EXITS on the dial provides a hint--you're starting to figure this world out, and since you need a number, which would contain all these letters?[end if]";
	now numset of dial is number understood instead;

consecutive-binary-tries is a number that varies. consecutive-binary-tries is usually 0.

to binary-eval:
	now consecutive-binary-tries is 0;
	say "[one of]The room seems to spin a bit faster, and you worry you've made things worse[if dial is examined]--the dial contains a hint[otherwise]--maybe examining the dial can help[end if].[or]You feel like you are just flailing with the numbers, now, as the room picks up speed[if numset-ever-greater is true]. Maybe you should try guessing some higher numbers--you might pass over the right one[otherwise]. The room slowed down a bit when you moved up several numbers. Maybe if you keep going past the right number, you'll eventually hit it[end if].[or]You think that a binary search may not be a bad way to do things, even if you-the-player don't know what a binary search is.[or]Getting colder.[stopping]";

successful-binary-tries is a number that varies. successful-binary-tries is usually 0.
total-tries is a number that varies. total-tries is usually 0.

to say slowdown:
	say ", but the room seemed to slow down for a moment as you moved the dial [if numset of dial < number understood]up[otherwise]down"

to decide whether (a - a number) and (b - a number) skip:
	if a < 16 and b < 16:
		decide no;
	if a > 16 and b > 16:
		decide no;
	decide yes.
		
some warts are a thing. warts are plural-named.

instead of taking the warts:
	say "They're kind of stuck on, and you've heard cures like burning make them worse."

description of warts is "You can only feel them--no mirrors around--but they're plenty bumpy.". rgtext of warts is "[rc][rc][gc][rc][rc]". lgth of warts is 5. gpos of warts is 5. rpos of warts is 1.

some straw is a singular-named thing. "Some straw, too much to lug around in your hands, is here."

the indefinite article of straw is "some"

does the player mean giving the straw to the peasant: it is very likely.

description of straw is "Yellowy, rough edges, semi-musty. Just straw."

to say dial-hints:
	if odd-dial-hint is false:
		now odd-dial-hint is true;
		say ".";
		the rule succeeds;
	now odd-dial-hint is false;
	say "[one of]. Maybe [if dial is examined]you should look at the dial again for a partial hint about the number[otherwise]there is a partial hint to the number on the dial[end if].[or]. You glance over at the dial and briefly wonder what numbers have X's in them.[or]. The dial seems to indicate there are seven letters--one is X, and two are a direction. But you haven't had to go diagonally a lot. That leaves four possibilities.[or]. Where, abbreviated, should the exits point?[or]. The room NEXT IS Easier.[stopping]"

odd-dial-hint is a truth state that varies. odd-dial-hint is usually false.

check setting the dial to:
	say "Try a number instead. Or, if you typed out a number more than twenty, use the digits." instead;

centrifuge-stopped is a truth state that varies. centrifuge-stopped is usually false.

check looking in ufc:
	if player was not in ufc:
		if centrifuge-stopped is false:
			say "You climb down a larded ladder, slip, and hear people chattering about Lord Ablemiser, finances fan since... once a money yeoman, the frugalest till the fear gluts, but who'd nag nary a granny for late fees. Generous? Gone sour. Got antsy, then nasty... more garnish, less sharing... stern rents, a splinter should rent slip... a testier treatise from his Taxman, Tan Max... claiming to feel raw and that the welfare era flew! 'No grace in ignorance, caring one!' A cold clod, now. He may even have aborted a debtor...";
			wfak;
			say "You hear cases for destroying his castle and for showing love. You see and avoid tripwires, and just as you feel your wit's riper, a snare nears. You swat at a pest and miss a step, then tumble madly wide of a middle way...";
			wfak;
			say "You roll down a gradient, tirade, nag to yourself...";
			wfak;
			say "Thump. ('Drat, poor trapdoor!' you hear.)";
			wfak;
			say "Oh, ";
			wfak;
			say "Man, ";
			wfak;
			say "My ";
			wfak;
			say "head's ";
			wfak;
			say "spinning...[paragraph break]";
			wfak;
			say "Oh dear it's not me it's the room...";

chapter ROOM

roomroom is a privately-named room. the printed name of roomroom is "Room". "[if moor is unvisited]This room is a little too undescribed. It's nice to have a break from all this puzzling, but it's almost too easy a break[tagit][otherwise]Zapping yourself to the moor hasn't made this room any more exciting[end if].[paragraph break]A passage leads west back to the centrifuge, and another leads north."

to say tagit:
	if kitchen is visited:
		say ". It feels like the Nick, but you can't explain why";
	if player has gadget:
		say "[one of]. Your gadget seems to vibrate a bit[or][cycling]";

check scaning air (this is the air scan rule):
	if player does not have gadget:
		continue the action;
	if player is in the nick or player is in spis or player is in moor or player is in roomroom or player is in manor:
		if player has gadget:
			try examining gadget instead;

instead of going nowhere in roomroom:
	say "You can only go west back to the centrifuge or north[if ne-room is visited] to the trellis[end if].";

roomroom is east of ufc.

the sack is in lalaland. "A sack is here. It's made of some rough material. Call it burlap. That's a safe one."

the sack is flipped-yet.

the sack is a container. rgtext of sack is "[rc][gc][rc][gc]". lgth of sack is 4. gpos of sack is 3. rpos of sack is 4.

description of the sack is "Sturdy burlap[if number of things in sack is 0]. It's empty[otherwise]. In it you see [list of things in sack][end if]."

the cask is in roomroom. "You see a plain cask here[if cask is closed]--it seems unbroached[otherwise]--someone has cut a small hole in the top[end if]."

the cask is a transparent container. the cask is closed. rgtext of cask is "[rc][gc][rc][gc]". lgth of cask is 4. gpos of cask is 3. rpos of cask is 4.

description of the cask is "It's plain and cedar[if cask is open], with a small hole on top[otherwise], and it seems airtight[end if][in-cask]."

to say in-cask:
	if cask is open:
		say "[if number of things in cask is 0]. It's empty[otherwise]. In it you see [list of things in cask][end if]";

check taking cask:
	if cask is closed:
		say "It is too heavy right now. There seems to be a liquid in there." instead;

check opening cask:
	if cask is closed:
		say "There's no obvious way to open it." instead;
	otherwise:
		say "It already is." instead;

the description of the hole is "Wide enough to FILL the cask, but not to put anything in."

check propfliptoing the cask:
	if sack is visible:
		if number of things in sack > 0:
			say "As the sack changes, [the list of things in sack] falls out.";
			repeat with Q running through things in sack:
				move Q to location of player;

check propfliptoing the sack:
	if cask is visible:
		if cask is closed:
			say "The cask seems to bulge, and you hear a sloshing, but it doesn't do anything. Maybe you need to walk around and find someone who can open the cask." instead;
		if oils are in cask:
			say "That's inadvisable. The oils would leak out." instead;

check taking the straw:
	if player has sack:
		if straw is in sack:
			say "Better to keep it in the sack." instead;
		say "You lump the straw in the sack.";
		now straw is in the sack instead;
	otherwise:
		say "Too unwieldy to carry anywhere." instead;
		
instead of eating the straw:
	say "Eww.";
	
instead of eating the hay:
	say "Eww.";
	
[rule for implicitly taking the straw:
	say "(gathering it up)";
	
rule for implicitly taking the hay:
	say "(gathering it up)";]

the straw-hay drop rule is listed before the can't drop what's not held rule in the check dropping rules.

check dropping (this is the straw-hay drop rule):
	if noun is straw or noun is hay:
		if noun is in sack:
			say "No use making a mess. It's easier to lug around that way.";

the straw-hay insert rule is listed before the can't insert what's not held rule in the check inserting it into rules.

the straw-hay put-on rule is listed before the can't put what's not held rule in the check putting it on rules.

the silo-put rule is listed before the can't put onto what's not a supporter rule in the check putting it on rules.

the silo-put rule is listed before the can't put what's not held rule in the check putting it on rules.

before tying to:
	if noun is second noun:
		say "[tau]" instead;

before inserting into:
	if second noun contains noun:
		say "Things are already like that." instead;
	if noun is second noun:
		say "[tau]" instead;
		
before putting on:
	if noun is second noun:
		say "[tau]" instead;
		
to say tau:
	say "Tautology?! Got ya, lout!";
	
check tying it to (this is the silo tying rule):
	if noun is second noun:
		say "[tau]" instead;
	if noun is silo:
		try tying second noun to silo instead;
	if noun is panel:
		if second noun is silo or second noun is rectangle:
			try putting noun on silo instead;
	if noun is missile:
		if second noun is silo and black door is part of the silo:
			say "(You want to PUT the missile IN, I think.)";
			try inserting noun into silo instead;
	if noun is black door or noun is missile:
		try putting noun on second noun instead;

check putting it on (this is the straw-hay put-on rule):
	if noun is straw or noun is hay:
		try inserting noun into second noun instead;

check putting it on (this is the silo-put rule):
	if second noun is sack or second noun is cask:
		try inserting noun into second noun instead;
	if noun is the missile:
		ignore the can't put what's not held rule;
	if noun is the black door:
		ignore the can't put what's not held rule;
		if second noun is boundary:
			try putting black door on silo instead;
		if second noun is not silo:
			say "The black door probably needs to fit on, or in, a big structure." instead;
	if second noun is rectangle:
		if noun is panel:
			try putting noun on silo instead;
		say "Not really the right size[if noun is not door] or shape[end if]." instead;
	if second noun is silo:
		ignore the can't put onto what's not a supporter rule;
		if noun is black door:
			ignore the can't put what's not held rule;
			say "The door fits the dotted line in the silo perfectly, swinging slightly before sliding in place with a click.";
			now dashed boundary is in lalaland;
			now black door is part of the silo instead;
		if noun is panel:
			now panel is part of the silo;
			now dotted rectangle is in lalaland;
			say "The panel fits into that rectangle handily and even starts to glow. Two buttons appear on it--one says HOOTS, the other TREES.";
			now the hoots button is part of the panel;
			now the trees button is part of the panel instead;
		if noun is missile:
			say "I assume you mean IN the silo...";
			try inserting missile into silo instead;
		say "That doesn't seem to fit anywhere on the silo." instead;
	if second noun is the straw or second noun is the hay:
		say "That'd crush it." instead;
	if noun is the missile:
		say "You roll the missile around, but you can't find anywhere to put it." instead;
		

before doing something with the missile when the missile is visible:
	if missile is in lalaland:
		say "It's where it needs to be. No need to worry about it." instead;

check inserting missile into silo:
	if black door is part of silo:
		say "You schlep the missile through the door. You hear some mechanics and whizzing, then 'SMILIES MISSILE PROPERLY ALIGNED FOR DEALING A RESULT WITH LUSTER.'";
		now missile is in silo instead;
	otherwise:
		say "No way to put the missile in or on the silo--and nothing on the side to attach it to. If there were any door to the silo, you could put the missile in there." instead;

check pushing:
	[if player has the noun:
		say "It's not pushable." instead;]
	if noun is panel and panel is part of the silo:
		say "Brute force is no way to figure what the panel does! Well, you could try mental brute force..." instead;
	if noun is black door:
		if silo is visible:
			try putting black door on silo instead;
		otherwise:
			say "Unfortunately, you don't have anything to push the door onto. Yet." instead;
	if noun is missile:
		if silo is visible:
			try inserting missile into silo instead;
		otherwise:
			say "Unfortunately, you don't have anything to push the missile into. Yet." instead;
	if noun is panel:
		say "You give a few hups but fail to push[if panel is part of the silo]. Probably better to put it on something[otherwise]. Maybe push the buttons individually[end if]." instead;
	
instead of examining the panel:
	if panel is not part of the silo:
		say "It looks kind of electrical. It doesn't seem to have wires, so hopefully you can just slip it where it needs to go." instead;
	say "There are two buttons, on the left [if trees button is visible]('TREES')[otherwise]('STEER')[end if] and right [if shoot button is visible]('SHOOT.')[otherwise]('HOOTS.')[end if] Both are an important shade of red, and you'll probably have to push them to figure what they do." instead;

the hoots button is a thing. description is "It's on the right, and it's labeled HOOTS.".

rgtext of hoots button is "[rc][rc][gc][rc][rc]". rpos of hoots button is 4. gpos of hoots button is 5. lgth of hoots button is 5.

the shoot button is a thing. description is "It's on the right, and it's labeled SHOOT.".

the trees button is a thing. description is "It's on the left, and it's labeled TREES."

rgtext of trees is "[rc][rc][gc][gc][rc]". rpos of trees button is 2. gpos of trees button is 5. lgth of trees button is 5.

the steer button is a thing. description is "It's on the left, and it's labeled STEER."

understand "right button/" as hoots button when hoots button is visible.
understand "right button/" as shoot button when shoot button is visible.
understand "left button/" as trees button when trees button is visible.
understand "left button/" as steer button when steer button is visible.

check pushing hoots button:
	say "You hear loud hoots. You're not sure if people are laughing, or alerting you to the fact that this button doesn't do what it's supposed to, or if maybe it's some high-tech/magic war dance or even an air raid siren. Whatever it is, it's clear noise won't get the missile launched." instead;

missile-steered is a truth state that varies. missile-steered is usually false.

check pushing shoot button:
	if missile-steered is false:
		say "[one of]You hear a loud siren! Unfortunately, the missile doesn't seem to be pointing anywhere besides straight up. Guess where it'll land?[or]The missile doesn't have a specific target. You can't just go chucking it anywhere.[stopping]" instead;
	say "Boom! The force of the missile launch knocks you back, and you know you'll see no finer inferno than the rainbows that burst before you pass out. You are knocked unconscious, unable to feel the wave of simple, heartfelt, plainly-worded concerns that explode across the land.[paragraph break]You wake up in a castle. Lord Ablemiser is beaming, no longer big, mean.[paragraph break]'That launch deserves A LUNCH!' he booms. 'To be had now!'";
	wfak;
	say "And how! You learn he is, of course, Noble Ol' Ben. But he feels he hasn't re-earned the title, yet. (Yeh, ok, hokey.)[paragraph break]You try every tater treat and enjoy the flukiest lutefisk ever. You even meet the guards who imprisoned you[guards-say-hi]. After a fast, live festival with an orchestra, a carthorse returns you to the Trips Strip.";
	reg-inc;
	solve-region sortie instead;
	
to say guards-say-hi:
	if grist is in obligatory and cake pan is in obligatory:
		say ". They smile and apologize awkwardly, and you say no problem.";
	else if grist is in lalaland and cake pan is in lalaland:
		say ". They thank you for leaving them so much food even after they--well. They were too lazy when they woke up";
	otherwise:
		say ". They thank you for leaving enough food to tide them over but not too much. They needed room for the feast";

trees-is-pushed is a truth state that varies. trees-is-pushed is false.

check pushing trees button:
	now trees-is-pushed is true;
	say "Trees begin growing out of the side of the silo. The Trees button switches to a reset button which starts blinking violently. You push it before the trees crash down on you. They vanish, and it's a trees button again. Whew." instead;
	
check pushing steer button:
	if missile-steered is true:
		say "That'd be overkill. You imagine it's [if shoot button is visible]the other[otherwise]another[end if] one you need to push to launch.";
	otherwise:
		now missile-steered is true;
		say "The silo makes some ominous VVTT-KLK-BRRW noises and tilts slightly in the direction of what you assume and hope is Lord Ablemiser's palace.";
	the rule succeeds;

instead of tying to (this is the check for big quest item attachment rule):
	say "[one of]Note--while this game generally maps ATTACH/TIE X TO Y to PUT X ON/IN Y, [bold type]PUT ON/IN[roman type] is a bit more specific and is thus recommended.[or][run paragraph on][stopping]";
	if noun is missile or noun is hay or noun is straw or noun is panel or noun is black door:
		try inserting noun into second noun instead;
	if second noun is missile or second noun is hay or second noun is straw or second noun is panel or second noun is black door:
		try inserting second noun into noun instead;
	continue the action;

instead of taking the silo:
	say "Counterproductive and, uh, very unlikely.";

instead of taking the soil:
	if soil is in moor:
		say "It belongs here[if silo is visible], as a foundation for the silo[end if]." instead;
	otherwise:
		say "It'd be a mess to carry.";

check inserting it into (this is the straw-hay insert rule):
	if second noun is cask:
		ignore the can't insert what's not held rule;
		if cask is closed:
			say "The cask is probably for holding liquids, and it's not open. Well, not yet." instead;
		if noun is straw or noun is hay:
			say "Too messy--it'd stick out and fall out. [run paragraph on]";
		say "The cask is really made to hold liquids." instead;
	if second noun is sack:
		ignore the can't insert what's not held rule;
		if noun is in sack:
			say "It already is." instead;
		if noun is missile or noun is panel or noun is door:
			say "That wouldn't fit in the sack[if straw is in sack or hay is in sack], even if the sack were empty[end if]." instead;
		if noun is not straw and noun is not hay:
			say "[if straw is in sack or hay is in sack]You've already got something in there[otherwise]The sack's probably better for bulkier stuff[end if]." instead;
	if noun is black door or noun is panel:
		ignore the can't insert what's not held rule;
		try putting noun on second noun instead;
	if noun is missile:
		ignore the can't insert what's not held rule;
		if second noun is not silo:
			if noun is black door and black door is part of the silo:
				try inserting missile into silo instead;
			say "The missile needs to fit in or on a big structure." instead;
	if noun is hay or noun is straw:
		ignore the can't insert what's not held rule;
		if second noun is silo:
			say "Farming isn't what you need to do, here[if door is not part of silo], and anyway, you've got no door to schlep things through." instead;
		if noun is straw:
			if second noun is scraped wall:
				say "That would be perfectly sensible, to make the wall impermeable. The scraped wall seems to indicate the need for a hallway, though." instead;
			otherwise if second noun is not sack:
				say "The straw is probably more useful thatching a cottage or something. You haven't seen any, yet." instead;
	if second noun is silo:
		if noun is not missile and noun is not panel and noun is not door:
			say "That doesn't belong in or on the silo." instead;

check putting on scraped wall:
	try inserting noun into scraped wall instead;

check inserting it into (this is the sack-into-wall rule):
	if noun is sack:
		if number of things in sack is 1:
			say "I'm assuming you mean what's in the sack.";
			try putting a random thing in the sack on the noun instead;
		otherwise if number of things in sack is 0:
			say "The sack wouldn't look right, there.";
		otherwise:
			say "[bug-report] Should not have 2 things in the sack." instead;

check inserting hay into:
	if second noun is cask:
		say "The cask should probably store liquids instead." instead;
	if second noun is not scraped wall:
		say "You're not sure how the hay could augment anything, there." instead;
	now hay is part of the scraped wall;
	now scraped wall is hayfilled;
	reg-inc;
	say "Hey, well, the wall is made of hay now. Anyhow, I hope you know what you're doing." instead;

before taking hay:
	if hay is part of the scraped wall:
		say "Don't undo what you did. The haywall is close to what you want." instead;

section hoses-shoes

a pair of hoses are a plural-named thing. "[one of]A familiar pair of hoses lies by the cask. The guards must have forgotten them while, err, celebrating[or]The pair of hoses is still lying around[stopping].". rgtext of hoses is "[rc][rc][rc][gc][gc]". gpos of hoses is 3. rpos of hoses is 2. lgth of hoses is 5.

after doing something with hoses:
	set the pronoun it to hoses;
	continue the action;

the shoes are a plural-named undesc.

understand "hose" as hoses.

description of hoses is "Thick, probably better for keeping water out than storing it."

instead of tying hoses to something:
	say "They are too thick to bend and twist and tie to anything."

instead of tying something to hoses:
	say "They are too thick to bend and twist and tie to anything."

section pouring

pouring is an action applying to one visible thing.

understand the command "pour [any thing]" as something new.

understand "pour [any thing]" as pouring.

does the player mean filling the cask: it is very likely.

carry out pouring:
	if player has sack and cask is in lalaland:
		say "You'll probably want that cask back." instead;
	if player does not have cask:
		say "Nothing to pour anything out of." instead;
	if noun is cask:
		if oils are not in cask:
			say "Nothing in the cask to pour." instead;
	if noun is soil:
		if oils are in cask and noun is soil and soil is not visible:
			try propfliptoing soil instead;
	if noun is silo:
		if oils are in cask and noun is silo and soil is not in moor:
			try propfliptoing silo instead;
	if noun is not visible:
		say "You can't see any such thing." instead;
	if noun is oils and location of player is sacred cedars:
		if oils are in cask:
			say "You've already got some oils in the cask." instead;
		otherwise:
			try filling cask instead;		
	if noun is oils or noun is cask:
		say "You need to decide what you want the oils to be before pouring. Don't worry, there are only twenty-odd possibilities. And some don't make sense." instead;
	if noun is spout:
		say "(I assume you mean the oils into the cask.)";
		try filling the cask instead;
	say "You can't pour that!";
	the rule succeeds;

chapter Stiller Trellis / Crashing Archings

ne-room is east of Kitchen and north of roomroom. "[if cedars are not visited]This room feels close to something important. [end if][the-trellis]. You can also go west or south.". ne-room is privately-named. printed name of ne-room is "Stiller Trellis".

trel-priv is privately-named scenery in ne-room. printed name of trel-priv is "the trellis". understand "trellis" as trel-priv.

understand "decoration" as trel-priv.

description of trel-priv is "[if cedars is visited]The trellis feels appropriate to guard or indicate the room to the east, now you've been there[otherwise if scraped wall is visible]It arches over the scraped bit of wall, as if the wall wasn't supposed to be there. You could picture a passage through it, somehow[otherwise]You could just wallk under it to the east, now[end if]."

to say the-trellis:
	if cedars is visited:
		say "The trellis feels appropriate to guard or indicate the room to the east, now you've been there";
	otherwise:
		say "The trellis to the east is the only decoration you've seen in this area[if scraped wall is not hayfilled], and the wall below it seems scraped and beaten up a bit[end if]"

the scraped wall is scenery in ne-room. the scraped wall can be hayfilled. the scraped wall is not hayfilled.

instead of climbing scraped wall:
	say "No good handholds, and it goes up too high anyway."

instead of climbing trel-priv:
	say "No good handholds, and it goes up too high anyway."

the crashing archings are plural-named scenery. "The archings blocked the passage to the east once they crashed down. You burble like a lubber when you think of what you cannot see again and that there is no way back."

understand "rubble" as crashing archings.

instead of climbing archings:
	say "You can't get through. The rubble has hardened."
	
instead of taking archings:
	say "The rubble seems to have dried and stuck like lava."
	
description of scraped wall is "It's scraped away, but not enough to bust through[if scraped is hayfilled]. In fact, you stuffed hay where the wall was sort of peeled away[otherwise]. You see HALLWAY UNDER (UN-)UN-CONSTRUCTION scrawled on it[end if].";

instead of reading wall:
	say "[if scraped is hayfilled]You can no longer see where it says [end if]HALLWAY UNDER (UN-)UN-CONSTRUCTION.";

check going east in ne-room:
	if scraped wall is visible:
		say "You're going to have to do something about that wall, first[if scraped wall is hayfilled]. You already have, but just one more thing[end if]." instead;
	if caskfillings is 2:
		say "The rubble from the archings is too high and too solid." instead;
		
carry out scaning scraped wall:
	if player does not have gadget:
		say "You have nothing to scan the wall with." instead;
	if scraped wall is hayfilled:
		if player has gadget:
			say "Your gadget beeps and flashes [gc][gc][rc][rc][rc][rc][rc]." instead;
		otherwise:
			say "The gadget lights up with [gc][bc][rc][bc][bc][bc][bc]." instead;
	say "The lights on the gadget pulse a bit, but nothing really stays. Maybe the wall's not ready, yet." instead;

check going nowhere in ne-room:
	say "You can only go west or south[if the room east of ne-room is sacred cedars] or, since you opened the hallway, east[end if]." instead;

check inserting into scraped wall:
	if noun is not straw and noun is not hay:
		say "That doesn't seem to fit. You need something stuffing-like." instead;

instead of throwing cask at scraped wall:
	say "The cask bounces off harmlessly, no worse for the wear. You take it back." instead;

chapter hallwaying

hallwaying is an action applying to nothing.

understand the command "hallway" as something new.

understand "hallway" as hallwaying when player is in ne-room.

carry out hallwaying:
	if scraped wall is hayfilled and sacred cedars is unvisited:
		say "Of course! The hay wall you made collapses into a hallway leading east. One you won't even have to duck through!";
		reg-inc;
		now scraped wall is in lalaland;
		the rule succeeds;
	if scraped wall is hayfilled:
		say "Yes, you made a hallway east." instead;
	say "You can't make the hallway until you add something to the wall.";

chapter Sacred Cedars

Sacred Cedars is east of ne-room. "You are in a room with thick entwined cedars as walls. On one, a spout pouts, waiting for a stoup. You can go back west to the generic room. Oddly, you feel no reason to be scared. You see drawings and holy verse on the walls."

check going west in Sacred Cedars:
	if caskfillings is 2:
		say "You hear a creaking above as you exit. You see the trellis's archings crashing on you and quickly run from under them. The archings smash against each other, and their rubble blocks the hallway. You hear a voice. It must be Lois. 'WIN,' she whines.";
		now printed name of ne-room is "Crashing Archings";
		now crashing archings are in ne-room;
		now trel-priv is in lalaland;

lois is scenery in sacred cedars. "Lois is beyond trivial standard I7 descriptions, so if you see this, it is a [bug-report] Report to the author with a transcript."

instead of doing something with lois:
	if current action is attacking:
		say "No. Lois is here to help.";
	say "Looking for evidence of Lois proves lack of faith."
	
drawings are scenery in sacred cedars.

description of drawings is "You see a vermian minerva, a cliche chalice, a lanced candle, and a clean lance. They're probably not for anything, but they're solid idols and add atmosphere nicely."

check going to sacred cedars for the first time:
	say "You follow the haywrought throughway..."

the holy verse is scenery in sacred cedars.

description of holy verse is "It's all calligraphic, and it reads[paragraph break][italic type]From here, where lumbers slumber, is, lo! Purest oil snout erupts: solution! Ground unsod becomes sound. Lovely volley. A stone atones![paragraph break]  --LOIS[roman type]"

check going nowhere in sacred cedars:
	say "There is no other secret door. You can only go back west." instead;
	
the sc are privately-named plural-named scenery in sacred cedars. printed name of sc is "cedar wall/walls". understand "cedars" and "cedar/ wall/walls" as sc.

description of sc is "They look nice and even smell nice. A holy verse is carved on one."

the spout is scenery in sacred cedars. description of spout is "It can be used to FILL the appropriate receptacle here.";

check switching on spout:
	if player has cask:
		try filling the cask instead;
	if player has sack:
		try filling the sack instead;
	say "It'd be cannier to have a container." instead;

chapter Kitchen

Kitchen is north of ufc.

does the player mean putting an ingredient on the tortilla: it is very likely.

description of Kitchen is "This is a kitchen, though it's a bit bare, except for the obligatory fridge[if tall trio is visible] and some really corny tune. There's also a tall trio of chef statues[end if]. South is the centrifuge room and east is [if ne-room is unvisited]somewhere less hectic[otherwise]the room with the trellis[end if]."

the tune is scenery in the kitchen.

instead of doing something other than listening to the tune:
	say "You can't do anything other than listen to it, so you do for a bit.";
	try listening instead;

description of tune is "You should be listening instead.";

check going nowhere in Kitchen:
	say "You can only go south to the centrifuge or east[if ne-room is visited] to the trellis[end if]." instead;

[abb is a privately-named room. the printed name of abb is "Ah! Blocked, Blockhead."]

the tall trio is scenery in Kitchen. rgtext of tall trio is "[gc][rc][rc][rc][rc][rc][rc][rc]". lgth of tall trio is 8. gpos of tall trio is 1. rpos of tall trio is 2.

understand "chefs/chef/statues/statue" as tall trio.

description of tall trio is "Well, they are actually only six inches tall, the middle chef patting the other's shoulders. But they're skinny enough that you sort of assume they're tall. Well, I'm assuming. There's a loose label attached to the middle one, who is half-hugging both his pals."

the middle chef is part of the tall trio. description of middle chef is "[mid-chef]"

the left chef is part of the tall trio. left chef is undesc. instead of doing something with left chef: say "He's not nearly as interesting as the middle one."

the right chef is part of the tall trio. right chef is undesc. instead of doing something with right chef: say "He's not nearly as interesting as the middle one."

to say mid-chef:
	say "He's got a big smile and is semi-hugging his friend. The loose label sticks out awkwardly. Maybe it has some actual useful information on it.";
		
the loose label is part of the tall trio.

instead of taking the loose label:
	say "You can read it okay without taking it."

does the player mean doing something with the loose label when the tall trio is not visible: it is very unlikely.

description of loose label is "[loo-la]"

to say loo-la:
	say "'We can help be something to base a whole spicy dish on! A truly MALE meal, not lame! It may help you make a run through a border!' Red writing is on the other side of the label--flip it over to see what it is?";
	if the player consents:
		say "--crafted with care by [first custom style]AL TILTRO[roman type].";
	otherwise:
		say "Ok.";

an ingredient is a kind of thing.

the tortilla is an ingredient. understand "taco" as tortilla.

to say tort-desc:
	say "tortilla";
	if tortilla is visible:
		if ingredients-in-tort is 0:
			say " (boring and plain)";
		if ingredients-in-tort is 1:
			say " (not totally bland)";
		if ingredients-in-tort is 2:
			say " (edible looking, but could be better)";
		if ingredients-in-tort is 3:
			say " (almost perfect)";
		
does the player mean doing something with the tortilla: it is likely.

check inserting an ingredient into an ingredient (this is the tortilla-into rule):
	try putting noun on second noun instead;

check inserting something into ingredient:
	if noun is not an ingredient:
		say "That wouldn't taste very good." instead;

the tortilla-into rule is listed before the can't insert into what's not a container rule in the check inserting it into rules.

check putting something on the tortilla (this is the tortilla-onto rule):
	if noun is an ingredient:
		tort-add noun;
		the rule succeeds;

ingredients-in-tort is a number that varies. ingredients-in-tort is 0.

instead of opening hotsauce:
	if tortilla is visible:
		try inserting the hotsauce into the tortilla instead;
	otherwise:
		say "You don't seem to have anything to put the hot sauce in. Or on." instead;

understand "covering" as hotsauce when hotsauce is part of tortilla;

the printed name of the tortilla is "[tort-desc]";

to tort-add (x - a thing):
	if x is part of the tortilla:
		say "You've already put [x] in the tortilla.";
		the rule succeeds;
	now x is part of the tortilla;
	increment ingredients-in-tort;
	if x is hotsauce:
		say "You open the packet and squirt the sauce in the tortilla. You place the empty packet in a pocket, where you forget about it. [run paragraph on]";
		now printed name of hotsauce is "covering of hot sauce";
	say "[one of]Hm, a good start. The tortilla's not plain any more.[or]The tortilla's coming together.[or]The tortilla's almost there![or]You got it[now-taco]! A full-blown taco! You can't help but pick it up and admire it![stopping]";
	if ingredients-in-tort is 4:
		now player has the taco.

instead of eating taco:
	say "That might give you a gas saga. Plus, this game has no fully implemented bathrooms. Or any at all."

to say now-taco:
	now tortilla is in lalaland;
	now taco is in Kitchen;
	reg-inc;

description of tortilla is "Flat circular dough[if number of things that are part of the tortilla > 0]. It's got [a list of things that are part of the tortilla] on it[otherwise]. Plain[end if].";

understand "tortilla" as taco.

instead of eating the tortilla:
	if ingredients-in-tort is:
	-- 0: say "Too plain by itself. Put something in it?";
	-- 1: say "It's still too plain to eat.";
	-- 2: say "It's about halfway to being something.";
	-- 3: say "It looks pretty good but needs one more ingredient.";
	-- otherwise:
		say "[bug-report]";

check putting something on an ingredient:
	if noun is part of the tortilla and second noun is not part of the tortilla:
		say "(Switching the two)";
		try putting second noun on noun instead;
	if noun is not an ingredient:
		say "That wouldn't taste very good." instead;
	if noun is tortilla and second noun is ingredient:
		say "(I'm switching the [noun] and [second noun])";
		try putting second noun on noun instead;
	if second noun is part of the tortilla:
		try putting noun on tortilla instead;
	if second noun is not tortilla:
		say "That's not the best ingredient to make a base for a dish--maybe something more bready." instead;

check eating an ingredient:
	if ingredient is part of the tortilla:
		try eating tortilla instead;

check taking an ingredient:
	if noun is part of the tortilla:
		say "Don't futz with your culinary creation." instead;
	
the cult tee is a thing in Kitchen. rgtext of cult tee is "[rc][rc][rc][gc][rc][rc][gc]". legth of cult tee is 7. gpos of cult tee is 3. rpos of cult tee is 7.

instead of wearing cult tee:
	say "You appreciate the drawing, but [if moored is true]it's not warm enough to get you to the moor, and [end if]you never were one for bold t-shirts anyway."

instead of wearing skate:
	say "You can't skate, and there's only one of it."

understand "shirt" as cult tee.

description of cult tee is "You remember seeing someone wearing it. It's got a picture of a carrot, a cucumber, an olive and several different colored onions all in a bowl. They're drumming their fingers, waiting for someone or something. You remember this t-shirt being briefly in fashion back home, wherever that is."

the head of lettuce is an ingredient.

description of the lettuce is "It's green and crispy and healthy and without flavor on its own."

instead of eating the lettuce:
	say "Too plain by itself.";

the skate is a thing in Kitchen. rgtext of skate is "[gc][rc][rc][rc][rc]". legth of skate is 5. gpos of skate is 1. rpos of skate is 2.

description of the skate is "Ice, not roller. [one of]Apparently, it's Kate's. The blade is sharp enough to cut meat, but unfortunately, there's not any around here. Yet. Plus, the blade's probably not clean enough. You don't see any sink around. Never mind[or]It's a meaty question, what an ice skate's doing somewhere this warm[stopping]."

the slab of steak is an ingredient.

description of steak is "Pre-cooked, and you can still smell the sizzle if you reach down to smell it. You think."

instead of eating the steak:
	say "It's cooked, but really, it'd go better with something. And you're not hungry."

after doing something with the spearman:
	set the pronoun him to the spearman;
	continue the action;

the cheesy spearman is a male thing in Kitchen. "A cheesy spearman is here. Well, he's [italic type]posed[roman type] all cheesy.". rgtext of spearman is "[rc][rc][rc][rc][rc][rc][gc][gc]". lgth of spearman is 8. gpos of spearman is 2. rpos of spearman is 8. understand "man" as spearman when spearman is visible. the spearman can be female or male. the spearman is male.

description of spearman is "[spearman-descrip]".

to say spearman-descrip:
	say "He's got a stupid smile and is giving a thumbs-up with his free hand. It's cheesy, not corny. You know the difference. But it is not a REAL cheese. Yet. You notice some writing on his armor--do you wish to read it?";
	if the player consents:
		try examining armor;
		continue the action;
	otherwise:
		say "OK.[no line break]";
	
the armor is part of the spearman.

description of armor is "Don't put me [first custom style]NEAR SPAM[roman type]. The red all-caps seem more overdone than his thumbs-up[if player does not have spearman]. The spearman wobbles a bit as you look closely at the writing--you can probably pick him up."

report taking the spearman:
	say "The base of the spearman has something on it. Maybe a signature.";
	
the signature is part of the spearman. understand "base" as signature.

check scaning (this is the scan the spearman and not his parts rule):
	if noun is signature or noun is armor:
		say "That's part of the spearman, really, even with the writing on it. At least, the eight lights from your gadget seem to indicate that. [run paragraph on]";
		try scaning spearman instead;

description of signature is "[if player does not have spearman](looking under the spearman the signature is beneath, which is a slight cheat, but I won't judge)[end if]You see [first custom style]A. SMERNPA[roman type] on the bottom, in bright red. That must be the sculptor, or whomever."

instead of doing something other than examining the armor:
	say "The armor is tied together with the spearman."

understand "spear man" as spearman.

the spear is part of the spearman. description of spear is "It's about as tall as he is."

the pile of grated parmesan cheese is an ingredient. understand "cheese" as parmesan when parmesan is visible.

description of parmesan is "Already grated, for which you are grateful. (Sorry if the description grates.)"

instead of eating the parmesan:
	say "Too strong by itself. You could've when you were three. Not now."

the taco is a thing. rgtext of taco is "[rc][rc][rc][rc]". lgth of taco is 4. gpos of taco is 3. rpos of taco is 1.

description of taco is "It's got meat, lettuce, cheese, sauce and a soft tortilla shell--but you know that, since you made it yourself!"

the coat is a wearable thing.

does the player mean wearing the coat: it is likely.

report wearing the coat:
	say "It's nice and warm. Not enough to make you sweat. But handy if you [if moored is true and moor is unvisited]try to poke around the moor[otherwise]find a way outside of here[end if].";
	the rule succeeds;

description of coat is "It's warm and fuzzy and shapeless and as multi-colored as the taco you made it from."

check taking off the coat:
	if player is in moor:
		say "Too cold." instead;

the bottle of CATHOUSE perfume is a thing. "A bottle of CATHOUSE perfume is here.". rgtext of CATHOUSE is "[rc][rc][gc][rc][rc][gc][rc][gc]". lgth of cathouse is 8. gpos of cathouse is 4. rpos of cathouse is 8.

understand "cat house perfume" as CATHOUSE.

CATHOUSE is in Kitchen.

description of the CATHOUSE is "It's yellow-gold, like most. The bottle seems to advertise a spicy scent too hot for most people's tastes.[paragraph break]By the way, you were never sure if it was one word or two, and what's more, the company who made it probably didn't care, either[one of]. There's something about how it'll free-m-up to love you, which you try to forget[or][stopping]."

instead of drinking the CATHOUSE:
	say "Deadly."

the large packet of HOTSAUCE is an ingredient.

description of hotsauce is "It's some disturbing mix of reddish shades of orange-red. The ungrammatical HOTSAUCE on the packet blocks out any list of ingredients, which is probably for the best."

understand "packet" and "sauce" as hotsauce when hotsauce is visible.

understand "hot sauce" as hotsauce.

instead of eating hotsauce:
	say "If you really want to do this, you are obviously too wild and crazy for text adventures.";

check propfliptoing hotsauce:
	if the player's command does not include "hot" and the player's command does not include "hotsauce":
		say "Close, but you need to say the full name of the sauce." instead;

section fridge

the obligatory fridge is scenery in the kitchen. understand "refrigerator" as fridge.

the obligatory fridge is a transparent openable container.

check examining fridge:
	ignore examine containers rule;

rule for printing room description details of fridge:
	omit contents in listing;

the manila animal is undesc. it is part of the obligatory fridge. understand "lamina" as manila animal.

instead of doing something other than examining the manila animal:
	say "You've half-destroyed a store, but you draw the line at aesthetically altering a fridge."
	
instead of examining the manila animal:
	say "Written on it: '? No, too stupid a fad. I put soot on warts.' But you've seen no soot, and you're not sure what should be behind the question mark."

description of obligatory fridge is "A manila animal forms a lamina over it. It doesn't appear locked or anything."

check opening obligatory fridge:
	say "[one of][or]You reopen and peer on. [stopping]You see [a list of things in obligatory fridge] inside." instead;

some grist is in the fridge.

description of grist is "It's a heapin['] helpin['] of the stuff, haphazardly lumped at the bottom of the fridge."

understand "heapin/ helpin/" as grist

understand "heapin/ helpin/" as grits

the grits are a thing. lgth of grist is 5. gpos of grist is 1. rpos of grist is 4. rgtext of grist is "[gc][gc][gc][rc][rc]".

description of grits is "They look more edible than the grits, but you're not THAT hungry right now."

the cake pan is in the fridge. lgth of cake pan is 7. gpos of cake pan is 5. rpos of cake pan is 4. rgtext of cake pan is "[rc][gc][rc][rc][rc][rc][rc]".

understand "cakepan" as cake pan.

description of cake pan is "It's empty, and you wonder what it's doing in the fridge."

the pancake is a thing.

description of pancake is "It's too plain to eat. And probably too cold."

instead of eating pancake:
	say "Too cold and bland. Plus, it's someone else's[if taco is off-stage]. You might want to fiddle with what's on the counter[otherwise]. The stuff on the counter was what was really important, anyway[end if]."

instead of eating grits:
	say "You have an attack of conscience and decide to leave them for someone else."

chapter The Nick

check going to kitchen:
	if kitchen is unvisited:
		say "Two nuts stun you with rubber hoses. 'A blockhead! Ha, blockade![paragraph break]'Passed our tamest net, but that's no statement. Outrage if you get past our gate.' After this rare accident, you're incarcerated.[paragraph break]As they walk away, you hear '...made mead. Much, chum.'[paragraph break]Then drunken reveling. Boy, those pinheads made your head spin.";
		now cask is open;
		now the hole is part of the cask;
		now the hoses are in roomroom;
		if warts are visible:
			pad-rec "warts";
		move player to the nick instead;

before going in the nick:
	say "You're stuck here. Well, physically, anyway." instead;

The Nick is a room. "You're locked in this arty suite of austerity by a great grate and also bars. It's a more forbidding version of the gateway in the Notices Section. You doubt even Old Man Almond could magic it open. There appears to be no standard way out. It has no accommodations, not even unsoft futons. This is a saner snare than the centrifuge, but it doesn't look like you'll drug a guard or reveal a lever to escape[if player has gadget][beepity-nick][end if]."

to say beepity-nick:
	say ".[paragraph break]Your gadget seems to shake a bit";

after looking in kitchen for the first time:
	say "You notice some warts on your face. Stress causes them, you know, and between the centrifuge and the nick, you've had a bit lately.";
	now warts are part of the player;
	now startmod5 is the remainder after dividing (turn count + 4) by 5;

printed name of The Nick is "The Nick".

The great grate is scenery in The Nick.

the branding is part of the great grate. description of branding is "The grate seems to be made by [first custom style]HECK TIN[roman type]. It's not steel, but it's still too thick. The redness of HECK TIN surprises you a bit.". understand "brand" as branding.

description of great grate is "Huge. Eugh. Old Man Almond couldn't magic it, and not even Flexi-Felix could slip through the holes! You notice a branding on it that you can probably read."

check opening great grate:
	say "You utterly fail to move it. If it could speak, it might sarcastically say [one of]Boring? Brig? No![or]Give up! And mean it, inmate![or]Denied, Indeed![or]Weak Try, Raw Tyke![at random]" instead;

understand "gate" as great grate when grate is visible.

the bars are scenery in The Nick. description of bars is "[the-bars]"

to say the-bars:
	say "Too narrow and strong to allow escape. In red, below the bars, you see a something scrawled. It might be a hint. Check it out?";
	if the player consents:
		say "In red crayon: [first custom style]N. THICKE[roman type] was here.[no line break]";
	otherwise:
		say "Ok.[no line break]";

instead of taking the bars:
	say "Not likely."

nick-turns is a number that varies. nick-turns is 0.

startmod5 is a number that varies.

every turn:
	if turn count > nick-turns:
		if player is in The Nick:
			say "[one of]Boy. Incarceration's making you hungry[or]You think back to your home and all its comfy rooms[or]Mmm, Foood. Even a lame meal. You'd even make it yourself[or]You feel the heat of Lord Ablemiser's hate even from here. You can't take it, so you want to get out of the The Nick. The Nick[or]You counter your urge to sink in despair. Or maybe you just table it[or]You doze off and dream you're watching a cooking show featuring ten hicks. They work in multiple...oops, you wake up and forget[or]You'd do housework to avoid this workhouse[or]You doze off--some guy named Chen has a kit for getting out of here--you wake up. Dang[or]You feel the plot thicken, maybe even your brain. You're hungry for a new location and whatever new puzzles it has cooked up[stopping].";
		if the remainder after dividing turn count by 5 is startmod5:
			if warts are visible:
				if player is in the nick:
					say "[line break]";
				say "You almost pick at your new warts[if player has gadget], and your gadget activates slightly[end if]. Man! If only you could say something to make the warts grow backwards, or just make them disappear. Or both.";
		now nick-turns is turn count;
	if player is in trips strip:
		if two-of-three is true:
			if number of solved regions is 3:
				if given-pass-yet is false:
					say "So. Apparently, you have one store area left to clear. But you don't need to. Store R suddenly looks invitingly idyllic, but if you're in the mood for adventure, you may want to look somewhere else first--[last-store], maybe.";
					now given-pass-yet is true;

given-pass-yet is a truth state that varies. given-pass-yet is false.

to say last-store:
	if store f is visible:
		say "store f";
	if store i is visible:
		say "store i";
	if store m is visible:
		say "store m";
	if metros are visible:
		say "the metros";
	if forest is visible:
		say "the forest";
	if sortie is visible:
		say "the sortie";

chapter kitchening

kitchening is an action applying to nothing.

understand the command "kitchen" as something new.

understand "kitchen" as kitchening when player is in the nick.

carry out kitchening:
	reg-inc;
	say "That does it! The heck with that silly old grate. Your prison dissolves, and it becomes the place you meant to go all along[if straw is in the nick]--the straw remains intact, too[end if].";
	if straw is in the nick:
		now straw is in kitchen;
	now player is in kitchen;
	the rule succeeds;

to say gad:
	say "Your gadget's not near anything, but it's registering ";
	
check examining the tagged gadget:
	if player is in the nick:
		say "[gad][if gadget is cert]seven reds on its screen[otherwise][bc][bc][bc][rc][bc][bc][gc][end if]." instead;
	if player is in spis:
		if player has tulip:
			say "The gadget shows nothing now." instead;
		say "You scan the area, wondering what you could ask nerds that'd get that tulip. You see [asknerds]." instead;
	if player is in tiles:
		say "Your gadget's not near anything, but it's registering [if gadget is cert][rc][rc][gc][gc][rc][otherwise][rc][gc][bc][bc][bc][end if]." instead;
	if player is in roomroom:
		say "Your gadget's not near anything, but it's registering [if gadget is cert][rc][gc][gc][rc][otherwise][rc][bc][bc][gc][end if]." instead;
	if player is in moor:
		if anapest is visible:
			try scaning anapest instead;
		say "Your gadget's not near anything, but it's registering [if gadget is cert][rc][gc][gc][rc][otherwise][rc][bc][bc][gc][end if]." instead;
	if player is in roman manor:
		say "Your gadget's not near anything, but it's registering [if gadget is cert][gc][gc][gc][rc][rc][otherwise][gc][bc][bc][rc][bc][end if]. That can't be for the manor, though." instead;

part resort

chapter Roman Manor

Roman Manor is east of Potshot Hotspot. "Your new home. Um, yo, here. Now. I could ramble about the marble, praise a spire, note the toaster rotates, or sanction what it contains, but really--you sense one last hurdle hurled in your way to happiness.[paragraph break]The darned [italic type]exits[roman type] are all around, but you just want to do less than nothing[if player has gadget]. Your gadget rattles for hopefully the last time[end if]."

the final-exits are privately-named plural-named scenery in roman manor. "They are tempting you to adventure, but you'd rather be than do."

understand "exits" as final-exits when final-exits are visible.

the marble is scenery in roman manor. "It makes you calm and relaxed enough to sit back and do nothing with a purpose."

the spire is scenery in roman manor. "It makes you calm and relaxed enough to sit back and do nothing with a purpose."

the toaster is scenery in roman manor. description of toaster is "It's just a stupid harmless luxury, here for a last bit of cheap wordplay."

instead of doing something other than examining the toaster:
	say "You'll be able to fiddle with it later. Right now, just sit back and soak it all in."

before going in roman manor (this is the clue final verb rule):
	say "[one of]You don't feel like getting up. Suddenly you're in the middle of an exitstential crisis.[or]You've had it with doing and going[or]Sit, ex-adventurer.[or]Where would you go? Texis?[or]Seeing exits just makes you want to...[or]Aww, c'mon, this one's just switching TWO WHOLE LETTERS. You had other tougher ones to MAKE it here! Maybe you're overthinking[stopping]." instead;

chapter tiles

Tiles is a room. "Tiles, tiles, tiles. As far as you can see. Different colors, but this doesn't look like a comfortable place for a nice long vacation[if player has gadget][beepity-tiles][end if]."

to say beepity-tiles:
	say ". Your gadget jostles a bit";

instead of going in Tiles:
	say "It's all tiles as far as you can see. You've barely had to walk great distances all game, and you're not starting now."
	
ts are plural-named privately-named scenery in tiles. understand "tiles" and "tile" as ts. printed name of ts is "tiles".

understand "colors" as tiles.

the ts are flippable. the rpos of ts is 2. the gpos of ts is 1. the legth of ts is 5. the rgtext of ts is "[rc][rc][gc][gc][rc]".

description of ts is "They're all sorts of weird shapes, but the colors are what you find curious. Light brown where you are, in a twenty foot radius, with blue around them. There's a lot of brown beyond that[one of]. You see no further details in da tiles. Yeah, sorry[or][stopping]."

instead of doing something other than examining or scaning the ts:
	say "It looks like you probably just have to work your magic on them again. There can't be that many possibilities."

check going in tiles:
	say "One way looks as good as the others. You'd probably collapse from exhaustion before you got anywhere.";

section isleting

isleting is an action applying to nothing.

understand the command "islet" as something new.

understand "islet" as isleting.

carry out isleting:
	if player is in tiles:
		reg-inc;
		say "The tiles crumble and shift around.";
		move player to Islet;
	the rule succeeds;

chapter islet

Islet is a room. "Well, this is a little more idyllic. But it's all sand and so small, it can't be the one you were promised. You can see land way to the east."

section scenery

the sand is scenery in islet.

description of the sand is "It's not worth digging through[if cork is visible or wings are visible]. You've got something useful, anyway[otherwise]. Maybe the stuff lying around is worthwhile[end if]."

instead of doing something other than examining the sand:
	say "No use searching for anything like a handset in the sand. Islets have sand, so I sort of had to put it here."

the tree is scenery in islet.

instead of climbing tree: say "That'll only get you off the sand on the island, not the island."

description of tree is "It has ALOFT scratched into it[one of]. An arrow points down to the swing, as if to say, the tree can't help you fly[or][stopping]."

section swing - wings

the swing is a thing in islet. "You see a rudimentary swing hanging from a tree here."

instead of taking the swing:
	say "It's tied, and you can't climb to untie it."
	
description of swing is "The swing is made of twine and hangs from a tree branch high up. You won't get to untie it, but you can see something scratched on the tree.". rgtext of swing is "[rc][rc][rc][rc][rc]". lgth of swing is 5. gpos of swing is 2. rpos of swing is 1.

understand "twine" as swing

the wings are a plural-named wearable thing.

description of wings is "They're pretty standard magical wings you can flap to make fly. You probably don't even need any fancy verbs or anything--just go the direction you want.";

section cork - rock

the rock is a thing in islet. "You see a large-ish rock here[if player has wings], but it seems less potentially useful now you have the wings[end if]."

description of rock is "It's holey, like a sponge, but not any less dense for that--it's scratched to read FLOAT.". rgtext of rock is "[rc][gc][rc][gc]". lgth of rock is 4. gpos of rock is 3. rpos of rock is 4.

instead of taking the rock:
	say "It's too heavy. You can't see anything to do with it.";

description of cork is "It's porous and really very light."

the cork is a thing. "A chunk of cork is here."

section going

check going in islet (this is the islet escape rule):
	if player has wings and noun is up:
		try going east instead;
	if player does not have cork and player does not have wings:
		say "It's too far to swim, and you've got no way to go over the lake or whatever it is." instead;
	if noun is not east:
		say "Maybe something's that way, but you see land far off to the east." instead;
	if player has wings and wings are not worn:
		say "You put on the wings for your flight. [run paragraph on]";
		now player wears the wings;
	if wings are worn:
		say "The wings last just long enough for you to get to the landmass you see before shredding into the wind[if player has cork]. You also, unsurprisingly, drop the cork as you operate your wings[end if]. But it probably doesn't matter now--you're across to some sort of mainland.";
	otherwise:
		say "With the cork under your chest, you dog-paddle to the land you saw from the islet. It's soggy and gross from your swim, and the seaweed or whatever has already started to eat at it. You scrabble past a few silent inlets...";
	now cork is in lalaland;
	now wings are in lalaland;
	now player is in olw instead;

section flying

flying is an action applying to nothing.

understand the command "fly/flap" as something new.

understand "fly" and "flap" as flying.

carry out flying:
	if player has wings and player is in islet:
		try going east instead;
	say "You have nothing that can help sustained flying." instead;
	the rule succeeds;

section swiming

swiming is an action applying to nothing.

understand the command "swim" as something new.

understand "swim" as swiming.

carry out swiming:
	if player has cork and player is in islet:
		try going east instead;
	if player is in islet:
		say "No land anywhere close. You'd need help to swim." instead;
	the rule succeeds;

chapter olw

olw is a privately-named room. "You've reached an odd low woodland. There's a gigantic [if poles are visible]set of poles[otherwise]slope you could climb[end if] off to the east[if toolshed is visible]. Also, a toolshed that's in bad shape[otherwise]. You see toeholds on the [p-s][end if]."

understand "odd low woodland" as olw.

printed name of olw is "An Odd Low Woodland".

check going nowhere in olw:
	say "You'd probably get lost that way. Besides, the vague commotion to the east, past the rising [p-s] seems worth seeing." instead;

section poles-slope

the poles are plural-named scenery in olw. understand "pole" as poles.

description of poles is "They are too vertical [if toeholds are not visible]and smooth [end if]to climb. You can't push them--they look quite sturdy and are at least twenty feet tall--but maybe, if they were at an angle..[if toolshed is visible]. Or you could get a grip somewhere[end if].". lgth of poles is 5. gpos of poles is 5. rpos of poles is 4. rgtext of poles is "[rc][rc][rc][rc][rc]".

the slope is scenery.

description of slope is "It's forty-five degrees. Too steep to climb without any traction[if toeholds are visible], so thank goodness you dismantled that toolshed[otherwise][check-tools][end if]."

check climbing poles:
	try going east instead;

check climbing slope:
	try going east instead;

section toolshed - toeholds

the toolshed is scenery in olw. rgtext of toolshed is "[gc][gc][rc][rc][rc][rc][rc][rc]". description of toolshed is "It's metal, with no visible entry. A forbidding u-lock (a padlock could become an old pack, after all,) is welded to it, just to make sure, and the windows are so dark you can't see inside. You notice outcroppings scattered around the toolshed walls.". lgth of toolshed is 8. gpos of toolshed is 1. rpos of toolshed is 5.

the windows are part of the toolshed. the windows are plural-named.

carry out scaning windows:
	say "The windows that held soot turn up [if gadget is cert][rc][rc][rc][rc][rc][rc][rc][rc][otherwise][bc][bc][bc][bc][rc][bc][bc][gc][end if].";

description of windows is "They have held soot enough to turn your eyes all red just thinking of what's inside."

the outcroppings are part of the toolshed. the outcroppings are plural-named.

carry out scaning outcroppings:
	say "They don't seem to give a different scan than the rest of the toolshed. But you might be able to climb on them, if you were smaller. They show [if gadget is cert][gc][gc][rc][rc][rc][rc][rc][rc][otherwise][gc][bc][bc][bc][rc][bc][bc][bc][end if]. Maybe they are a clue for what the toolshed can become." instead;

description of outcroppings is "They are something you could put your feet--no, the edge of them--on. They look sturdy enough to hold you, but the problem is, they don't lead anywhere."

the u-lock is undesc. the u-lock is part of the toolshed. understand "lock" as u-lock.

instead of doing something with the u-lock:
	if current action is scaning or current action is cring or current action is certifying or current action is rectifying:
		say "The gadget is silent. It's probably the toolshed.";
	say "The u-lock is too strong to break. You have nothing sharp. Maybe you can get a clue from its brand name. Take a look?";
	if the player consents:
		say "It's from the [first custom style]OLD ETHOS[roman type] company. Locks should be functional and not beautiful.";
	otherwise:
		say "You don't bother with it."

instead of entering toolshed:
	say "You couldn't break into any stores, and you're not breaking into the toolshed."

description of toeholds is "They [if poles are visible]could almost help you up the poles, if the poles weren't so vertical[otherwise]should make it easy to climb the slope to the east."

section woodland text shortcuts

to say check-tools:
	if ropes are visible or grips are visible:
		say ", but you might be able to just walk up with the tools you have";
	otherwise if ropes are visible:
		say ", and with just a little, the ropes would provide enough safety";
	otherwise if grips are visible:
		say ", which the grips give--but it's too risky. You need some safety";
	otherwise:
		say ", so maybe a couple tools--or a way to change the smoothness of the slope--will make it climbable";

to say p-s:
	if poles are visible:	
		say "poles";
	otherwise:
		say "slope";

section woodland escape

check going east in olw:
	if poles are visible:
		say "The poles are too vertical to climb[if toeholds are visible]. Even with the toeholds[else if ropes are visible and grips are visible]. Even with your tools[else if ropes are visible or grips are visible], and your one climbing tool wouldn't be quite enough anyway[end if]." instead;
	if toeholds are visible:
		say "You make it up the [p-s] with some effort and surprisingly little risk. The toeholds are more like footholds, really[if player has grips or player has ropes]. You didn't even need any climbing gadgets--and in fact you drop them in surprise when you see what awaits[end if].";
		now grips are in lalaland;
		now ropes are in lalaland;
		now player is in Potshot Hotspot instead;
	if player has ropes and player has grips:
		say "Tying the rope around your waist, throwing it til it catches on a rock above, and using the grips, you make it up the slope. What you see makes you drop them in surprise!";
		now grips are in lalaland;
		now ropes are in lalaland;
		now player is in potshot hotspot instead;
	otherwise if player has ropes:
		say "You aren't skilled enough to make it up there with the ropes. You might need one more gadget that lets you hold your place on the slope." instead;
	otherwise if player has grips:
		say "The grips seem stable, but you probably want some safety precaution--something to hold you--just in case." instead;
	say "You try climbing the slope, but it's still a bit too steep for you to hold and get traction. If only you had something that magically appeared on the slope--or failing that, a couple of trusty tools to work your way up." instead;

section spore - ropes

the spore is in olw. "A spore, about a foot long, is here. Imagine what it could grow into."

description of spore is "It looks too delicate as is to support any weight."

instead of taking the spore:
	say "You might crush it in its present form."
	
rgtext of spore is "[rc][rc][rc][rc][rc]". rpos of spore is 1. gpos of spore is 4. lgth of spore is 5.

section sprig - grips

the sprig is in olw. "A sprig of some tree or weed you can't identify is here.". understand "weed" as sprig when sprig is visible.

instead of taking the sprig:
	say "Uprooting it would probably kill it."

rgtext of sprig is "[rc][rc][rc][rc][rc]". rpos of sprig is 1. gpos of sprig is 5. lgth of sprig is 5.

description of sprig is "Very fragile in its present form."

some grips are a plural-named thing. some ropes are a plural-named thing.

check climbing ropes:
	if poles are visible:
		try climbing poles instead;
	if slope is visible:
		try climbing slope instead;
		
description of grips is "They appear to be suction-based, but you wouldn't climb anything TOO steep with them."

description of ropes is "They look sturdy enough, knotted at the ends and so forth. You could probably use them to climb up a slope, but you might need some help."

chapter Potshot Hotspot

Potshot Hotspot is east of olw.

description of Potshot Hotspot is "An arid, trod dirt road. East is siesta.[paragraph break][if riot is visible][one of]Oh dear. A horde. Uprisers--surprise--protesting YOU! A full-blown riot[or]There's a riot going on here! A bunch of people seem to be protesting...you, accusing you of things you'd never be brave enough to do[cycling][else if protest is visible]Three can still be a protest as well as a crowd, apparently. They're blocking your way east[else if potters are visible]The potters aren't just pottering around[pottiness][else if red bull burdell is visible]Red Bull Burdell is here, looking down at you[otherwise]It's nice and peaceful here, but it's not quite home[end if].";

to say ri-tri:
	if riot is visible:
		say "rioters";
	otherwise:
		say "remaining protesters";

every turn while player is in potshot hotspot:
	if potters are visible and kilns are visible:
		continue the action;
	if riot is visible or protest is visible:
		say "[one of]The [ri-tri] accuse you of animal cruelty: making monotremes into metronomes[or]'GO, CURSE-SCOURGE!' the [ri-tri] chant[or]The [ri-tri] claim none will CAGE OUR COURAGE[or]The [ri-tri] frame themselves as lackers, you as a slacker[or]The [ri-tri] seem particularly upset you destroyed store I[if store i is in trips strip] (okay, they're wrong)[end if] & R[or]The visceral cavilers continue their masculine calumnies[or]The tough-on-sin [ri-tri] continue on shouting[or]The mob rages at teargas you'll never fire[or]The [ri-tri] discuss what to hate next. A 'BINGO! AMEN!' restarts their bemoaning[or]No skeptic pickets the picketers.[or]Someone in the crowd yells that they will not be your assistant satanists. Everyone rebukes him for a moment--that's a bit far--then continues protesting.[or]The [ri-tri] characterize you as a leper to repeal or repel[or]Some bitter lug says you're the litterbug throwing darn glass in his grassland[or]Some total fruit loop's sign says ANARCHIST? NAH, RACIST[or]One sign says 'Can't BEAR US, ABUSER?' Actually, it's a rebus, drawn with a corny crayon. Clever, really.[or]One sign calls you COP-SHY PSYCHO.[or]One person, in a total existential crisis, waves an 'ARREST SARTRE' sign.[or]The [ri-tri] vent their hates with haste.[or]'HILLSIDE SHILL, DIE!' they chant[or]One carries an AVENGE GENEVA sign. You've never been there[or]The [ri-tri] continue their bashin['], hoping to banish you[or]The [ri-tri] lambaste your work in the forest with a meat slab[or]The [ri-tri] decry The Villain In The Villa. The one they're blocking you from. Well, mobs aren't big on logic[or]You note someone casting his own blames at you, keeping lame tabs on your reaction[or]The [ri-tri] have little wish to turn shouting into hugs[at random].";
	if potters are visible and kilns are not visible:
		say "The potters eye you with a mixture of contempt, boredom and expectation.";

to say pottiness:
	if kilns are visible:
		say "--they're working happily at their kilns";
	otherwise:
		say "--they have nothing better to do with their hands right now than wave signs and pump their fists and point at you"

before going in potshot hotspot:
	if red bull burdell is visible:
		if noun is east:
			say "Not with Red Bull Burdell around you aren't." instead;
		say "I don't think Red Bull Burdell wants you to take him literally. Villains just talk like that, you know?" instead;

check going west in Potshot Hotspot:
	say "No going back now." instead;
	
check going east in potshot hotspot:
	if riot is visible:
		say "That rabble seems mad at you. You'd better disperse them first." instead;
	if red bull burdell is visible:
		say "Not with Red Bull Burdell around you aren't." instead;
	if protest is visible:
		say "They still outnumber you. And they're still angrier, too. Gotta calm them down." instead;
	if potters are visible and kilns are not visible:
		say "They seem a civil trio, buy as you inch east, they get vitriolic. 'Give us something to do!'" instead;

check going nowhere in potshot hotspot:
	say "The only way you want to try to go is east." instead;

section riot

the riot is plural-named scenery in Potshot Hotspot. lgth of riot is 4. gpos of riot is 4. rpos of riot is 3. rgtext of riot is "[rc][rc][rc][rc]".

Include (-
	has transparent animate
-) when defining riot.

before asking protest about:
	say "You're not going to reason with them. Maybe reason about them, though." instead;

description of the riot is "Well, you have to admit, there's a wide variety of people protesting you, even if they're not explaining why. The riot will probably be blocking your way to the east, but they don't seem to be attacking you, which is nice.[paragraph break]You probably can't take the whole riot at once, but it'd be nice to have a way to weed out most of them."

the protest is privately-named scenery. printed name of protest is "trio".

legth of protest is 7. gpos of protest is 1. rpos of protest is 6. rgtext of protest is "[gc][rc][rc][gc][gc][rc][rc]".

understand "trio" as protest when riot is not visible and tall trio is not visible.

understand "three" and "protesters" and "crowd" and "uprisers" and "horde" and "rioters" and "rabble" and "mob" as protest when protest is visible.

understand "protest" and "protesters" and "crowd" and "uprisers" and "horde" and "rioters" and "rabble" and "mob" as riot when riot is visible.

Include (-
	has transparent animate
-) when defining protest.

description of protest is "They're all smeared with--huh? Looks like clay?"

understand "lacy" as a mistake ("No, the clay is a clue to what the trio really wants to do.") when protest is visible.

the potters are plural-named scenery. understand "trio" as potters when potters are visible.

Include (-
	has transparent animate
-) when defining bustling masses.

before asking potters about:
	say "They don't want talk. They want action. Maybe even bribery." instead;
	
description of potters is "[if kilns are visible]They're happily working away at their kilns[otherwise]They're looking around antsily. But they don't look saintly. They're a bit upset you've given them nothing to do with their hands[end if]."

rule for printing a locale paragraph about kilns:
	if potters are visible:
		now kilns are mentioned;

after propfliptoing potters:
	kiln-potter-check;

after propfliptoing kilns:
	kiln-potter-check;
	
to kiln-potter-check:
	if potters are not visible or kilns are not visible:
		continue the action;
	wfak;
	now Red Bull Burdell is in Potshot Hotspot;
	now potters are in lalaland;
	now kilns are in lalaland;
	say "You hear a shout from the south and see a man too ham-nosed to be handsome. He washes down some walnuts of wan lust with liquid from one purple metal can labeled Sado-Soda, then a bottle of Renegade Green-Ade.[paragraph break]'I am RED BULL BURDELL!' he shouts, in a voice neither earthy nor hearty. 'From my ROOTS to my TORSO!'[paragraph break]He's the size of a large hut, bellowing real thug laughter. 'Armed with a dream in a canoe on the ocean, I forged my RESPECT SCEPTER from iron noir and went from Da Prominent to Predominant! Once, you rodents snorted.'[paragraph break]Moving from angered to enraged, he puts his battle tablet in a back pocket and pulls out his cruelty cutlery forged of [italic type]iron noir[roman type].[paragraph break]'I will not waste sweat. GET OUT!!!! My ideal time? IMMEDIATELY! Nuance is nuisance!'[paragraph break]The potters flee with their kilns, which had handy wheels you didn't notice, to a mowed meadow. Bad time to maunder unarmed. Hope you have one last reverse in reserve!";
	now cutlery is in hotspot;
	wfak;

cutlery of cruelty is scenery. "It's built for cruel looks, not practicality. But it looks impressive. Not that you want to look too closely."

instead of doing something other than examining cutlery:
	try examining instead.

every turn when Red Bull Burdell is visible:
	say "'GET OUT!!!!' yells Red Bull Burdell. [bull-bellow][line break]His boot swings within an inch of your chin[one of][or] again[or] yet again[or], prompting one of your finest feints[or]. But his smiting timings are just off[or]. Be neat if he got beaten[or]. He's stout-armed, but he can be outsmarted[or]. He's getting set-up for an upset[or]. He's on a roll, but you're rolling around ok[cycling].";
	
to say bull-bellow:
	say "[one of]'See how A Red Bull draws REAL BLUD!' Stunned by his terrible spelling, you almost do.[or]'One of those pesty types, eh?'[or]'I'll win flat-out, fat lout!'[or]'GOTCHA! ACT, HOG!' he yells, prematurely.[or]He notes the lack of plumage on your pale mug.[or]'I kneecap, peacenik!'[or]'You should BLUSH, SHLUB!'[or]'I'm too sure for any ruse!'[or]'You're SO THIN, you'll get NO HITS on me!'[or]'Sit, bum! Submit!'[or]'I'm A TREE EATER and you like YOGURT, RUG TOY!'[or]'Lurch, churl![or]'Turn, runt!'[or]'Soon I cavort, a victor!'[or]'Loser role's yours, pasty patsy!'[or]'No DETOUR from getting ROUTED!'[or]Oh, to rebut the brute![or]You need actions as tonic but think 'I so can't.'[at random]";

Red Bull Burdell is a person.

rule for printing a locale paragraph about Red Bull Burdell:
	now Red Bull Burdell is mentioned;

description of Red Bull Burdell is "His face is covered with crass scars. He has a toe that sticks out--a partial bare foot inside a fear boot. You think back to Achilles for some reason."

the crass scars are part of red bull burdell.

description of scars is "Every villain has a few. Red Bull Burdell maxed out his useful attributes, but no amount of leveling up could improve his looks."

instead of doing something with crass scars:
	say "He'd want to beat you up for looking at them even if he didn't already want to beat you up.";

The toe is part of Red Bull Burdell.

understand "his toe" as toe.

description of toe is "A toe just juts out from Burdell's boot, looking infected."

understand "fear/ boot/" and "bare/ foot/" as toe.

before scaning when red bull burdell is visible:
	if gadget is in hotspot:
		say "You can't reach the gadget, but you remember six reds, then [bc][rc][gc][bc][bc][bc]." instead;
	say "[kicks]" instead;
	
to say kicks:
	say "He kicks the gadget out of the way! You were only able to read [if gadget is cert]six reds, then [bc][rc][gc][bc][bc][bc][otherwise][bc][rc][gc][bc][bc][bc], then six reds[end if] as he apparently kicked the switch, too[if gadget-secured is true], which is impressive given how you locked it[end if]. He's not going to let you near it.";
	if gadget is cert:
		now gadget is rect;
	otherwise:
		now gadget is cert;
	if player has gadget:
		now tagged gadget is in Potshot Hotspot;

the chain links are a plural-named thing. rgtext of chain links is "[rc][gc][rc][rc][gc]". legth of links is 5. gpos of links is 4. rpos of links is 5. description of chain links is "A bunch of ovular links hooked together. Out of the hands of rioters, it'll cause less achin[']. You notice they are Made In China. But they're still more violent things than you'd like, this near your manor."

understand "chains" as chain links when chain links are visible.

instead of taking chain links:
	say "Bad idea. You don't want to take up arms against them. They're just shouting slogans. Plus, they outnumber you."
	
some kilns are a plural-named thing.

instead of taking the kilns:
	say "[one of]Way too heavy. (fourth wall time: try this again for a horrible spoiler pun.)[or]How would you potter around with such heavy things?[stopping]"
	
description of kilns is "You certainly managed to summon the deluxe version. I guess all that practice earlier in the game, you learned how to give your anagrams OOMPH. You gaze at your work proudly, [if potters are visible]as do the potters[otherwise]but it's not work you can work with further[end if]."

section tugtoeing

tuging is an action applying to one thing.

understand the command "tug [something]" as something new.

understand "tug [something]" as tuging.

carry out tuging:
	if red bull burdell is not visible or noun is not toe:
		try pulling the noun instead;
	if red bull burdell is visible:
		if the player's command includes "boot":
			say "The boot is too big to grab. Think smaller." instead;
		if the player's command matches the regular expression "tug *toe":
			reg-inc;
			say "'No go, goon!' you yell, grabbing his toe.[paragraph break]'Gritty try, git,' he says, with a sporting pig snort.[paragraph break]You hold on as he yells 'HA! Inept!' ...[run paragraph on]";
			wfak;
			say "[paragraph break]'THE PAIN!' Burdell's toe yellows, and he yells 'OW!' first irately, then tearily, before facing reality--he is shrinking! 'My levels!' he cries. 'My statistics!'[paragraph break]'No terms, monster!' You ignore the predator's teardrops through a rain of 'No fair!'[paragraph break]Then he moans 'No Mas,' fleeing with feeling. 'I predate someone adepter!'[paragraph break]'Who? How?' you reply, instinctively.[paragraph break]Ya won! No way!";
			now red bull burdell is off-stage instead;
		otherwise:
			say "Six letters. You're close." instead;
	otherwise:
		say "[reject]";
	the rule succeeds;

part Trips Strip

Trips Strip is a room. "You see what was once a bunch of small malls. Most lots appear vacant or dilapidated. You're currently standing on some disused pavement[if sidewalk is unexamined]. There's a cracked sidewalk here, too, with graffiti, if you need a distraction[end if]."

pavement is scenery in trips strip. "It's not just a pavement but an event map. [event-map]."

to say event-map:
	if forest is unsolved:
		say "You see yourself walking past innumerable trees, wearing a disguise through a turnstile, and finally creating some sort of explosion somewhere very cold";
	else if sortie is unsolved:
		say "you see yourself making food, entering a sacred area behind a secret passage, and creating some sort of fireworks that seem to land on a castle without harming it";
	else if metros is unsolved:
		say "You see yourself picking up several hours and taking a light source into a very deep place underground. You seem to have a hand over one ear in each picture.";
	otherwise:
		say "You see yourself hopping from one island to another, climbing a plateau, and dispersing a crowd."

understand "graffiti" as sidewalk.

check going down in trips strip:
	if metros is solved and sortie is solved:
		say "You've solved both the metros and sorties." instead;
	if metros-x is visible and sortie-x is visible:
		if metros is solved:
			try entering sortie-x instead;
		if sortie is solved:
			try entering metros-x instead;
		say "You've opened two ways down, so you'll specifically need to ENTER SORTIE or ENTER METROS instead." instead;
	if metros-x is not visible and sortie-x is not visible:
		say "You can't find any place to go down[if sortie is solved or metros is solved] that you need to any more[end if][if sortie is not solved or metros is not solved], but there's at least one to find[end if]." instead;
	if metros-x is visible:
		try entering metros-x instead;
	if sortie-x is visible:
		try entering sortie-x instead;
	say "Unfortunately, there are no cool hidden passages beneath the stores. That you can see. Yet." instead;
	
check going inside in trips strip:
	if number of visible available portals is 1:
		try entering a random visible available portal instead;
	if number of visible available portals is 0:
		say "You'll need to figure a store out to go inside[if number of not unsolved regions > 1], on top of what you solved[otherwise], first[end if]." instead;
	say "That's ambiguous--you can currently enter [the list of visible available portals] to explore areas you haven't solved yet." instead;
		
check going nowhere in trips strip:
	say "You'd probably get lost. There's a long expanse of nothing. Best to browse through the stores. Examine them a bit." instead;

the cracked sidewalk is scenery in Trips Strip.

description of sidewalk is "[one of][gen-graf][or][random-name][cycling]"

to say gen-graf:
	say "[one of]FORK SPOONERISMS![paragraph break]There's a lot more, and you could waste a lot of time here.[or]Palindromes are for half-wits![or]Dale Wisk was liked. [italic type]Liked?![roman type][line break][or]Ash Ewer was here![or]COOL LOCO has left his signature.[cycling][run paragraph on]"
	
to say random-name:
	choose random row in the table of full names;
	say "[bold type][anname entry] was here!!![roman type]"

check smelling in Trips Strip:
	say "There's a mixture of smells coming from the shops. You could probably go up to one and examine it." instead;

the wooden sign is scenery in the Trips Strip. description of wooden sign is "'A store is not a shop.'"

instead of taking the wooden sign:
	say "It's too heavy. And besides, what would you do with it?"
	
instead of singing in the Trips Strip:
	say "The sign seems to vibrate. Not enough to pick it up. It's too heavy[if score is 0]. But still, curious[end if]."

chapter dope tan notepad

chapter stores

section stos

a sto is a kind of thing. a sto is usually undescribed. a sto is usually fixed in place.

understand "unlock [something]" as unlocking it with.

rule for supplying a missing second noun while unlocking:
	now second noun is notepad;

before opening:
	if noun is a sto:
		say "It's locked, and you won't find the key." instead;
		
before closing:
	if noun is a sto:
		say "It's already closed. And locked." instead;
		
instead of unlocking:
	if noun is a sto:
		say "Forget it. It's good and shut.";
	else if noun is metallic door:
		say "It has no visible keyholes. The only possible interface seems to be the sensor.";
	else if noun is rut-door:
		say "No keyholes. Maybe you can break it down some other way?";
	otherwise:
		say "You don't need to.";

instead of searching:
	if noun is not a store:
		say "[one of](Nothing in this game has anything deeply hidden, so searching works like examining.)[or][run paragraph on][stopping]";
		try examining the noun;
	otherwise:
		say "It's closed and locked, so you take a look at it.";
		try examining the noun;

to say vac-lot:
	say "Just a vacant lot.[no line break]";
	if all-seen is false:
		say "[paragraph break]It might take some time to inspect all 26 stores in order. Would you like a general overview?";
		if the player consents:
			now all-seen is true;
			try examining storeall;
			the rule succeeds;
		otherwise:
			say "Ok. You can X STORES again if you change your mind.[run paragraph on]"

all-seen is a truth state that varies. all-seen is usually false.

section stores

to say store-summary:
	if number of visible flippable stos is 0:
		say "You don't see any more stores. You should probably get going to that resort";
	otherwise:
		say "Of the various stores you see, only [list of visible flippable stos] look[if number of visible flippable stos is 1]s[end if] like you can do anything with [if number of visible flippable stos is 1]it[otherwise]them[end if]. Store r [if you-can-advance]finally looks like you can visit[otherwise]has a COMING SOON sign by it[end if]"

to show-em-all:
	say "[one of]They're labeled with letters, not numbers, though you can refer to them by number if you want. [or][stopping][store-summary]. A few others seem greyed-out, like in a video game or something. There can't be anything in any of them. [no line break]";
	pad-rec "stores";

to pad-rec (q - text):
	repeat through table of pad-stuff:
		if there is no short entry:
			say "[bug-report] need short entry for [q].";
		else if short entry is q:
			if known entry is false:
				now known entry is true;
				if there is no verify entry or verify entry is true:
					say "(You record the information in your [one of]dope tan [or][stopping]notepad.)[no line break]";
			the rule succeeds;

to pad-rec-q (q - text):
	repeat through table of pad-stuff:
		if there is no short entry:
			say "[bug-report] need short entry for [q].";
		else if short entry is q:
			if known entry is false:
				now known entry is true;
			the rule succeeds;

instead of taking storeall:
	say "[ment-not-phys]." instead;

to say ment-not-phys:
	say "Your extreme powers are mental, not physical";

storeall is plural-named privately-named scenery in trips strip. understand "all/stores/lots/mall/malls/strip" and "small malls" as storeall. printed name of storeall is "all the stores".

instead of entering storeall:
	say "They're all closed, and even if they weren't you'd have to pick a specific one.";

description of storeall is "[stos-desc]".

first-warned is a truth state that varies. first-warned is false.

to say stos-desc:
	if first-warned is true:
		say "They'd probably take too long to look at one-by-one. So you decide for a general overview.";
		now first-warned is true;
	show-em-all;

section store

disamb-store is privately-named scenery in trips strip. printed name of disamb-store is "the entire store bank". understand "store" and "entire/ store bank/" as disamb-store when player is in trips strip.

instead of entering disamb-store:
	say "Hmm, you can't enter all the stores at once. You'll need to pick a store--you can also refer to it by its abbreviation, e.g. Q for Store Q.";
	
does the player mean examining disamb-store:it is likely.
does the player mean entering disamb-store: it is likely.
[does the player mean doing something to disamb-store: it is likely.]

[last-sto-done is a sto that varies. last-sto-done is usually store-a.

understand "store" as a store when player is in trips strip.
understand "store" as a store when the item described is last-sto-done.

does the player mean doing something ]

description of disamb-store is "[sto-desc]".

instead of burning:
	if noun is a sto:
		say "The stores have suffered enough.";
	otherwise if noun is warts:
		say "That's a false cure that makes them worse.";
	otherwise:
		say "You've got no source of fire[if player has lit-up tulip]. The tulip doesn't count--it's child safe and stuff[end if].";

instead of pushing a sto:
	say "Physical activity won't do it, here.";
	
instead of entering a sto:
	say "It's closed and locked, and you won't find a key[if noun is flippable]. Maybe you can do something else to it[otherwise]. It doesn't seem like you need to enter, anyway[end if]."

check taking a sto:
	say "[ment-not-phys]." instead;

[understand "window" as the pronoun it when it is set to a sto.?? does the player mean doing something with last-sto: it is likely.]
	
to say store-overview:
	if number of visible flippable stos is 0:
		say "You've gotten rid of all four stores you could have.";
		the rule succeeds;
	say "Most seem greyed out. [run paragraph on]";
	if number of visible flippable stos is 1:
		say "Only [list of visible flippable stos] seems worth a closer look.";
		the rule succeeds;
	say "You might be able to do something with [list of visible flippable stos].";
	the rule succeeds;

to say sto-desc:
	say "There are 26 of them. Would you like a general overview?";
	if the player consents:
		say "[store-overview]";
		the rule succeeds;
	otherwise:
		if store f is visible:
			say "Okay, the first one you see is store f...";
			try examining store f;
			the rule succeeds;
		if store i is visible:
			say "Okay, the first one you see is store i...";
			try examining store f;
			the rule succeeds;
		if store m is visible:
			say "Okay, the first one you see is store m...";
			try examining store f;
			the rule succeeds;
		if store r is visible:
			say "Okay, the first one you see is store r...";
			try examining store f;
			the rule succeeds;
		the rule succeeds;

section sto by sto

the plural of sto is stos.

to say gd:
	say "It's oddly greyed out, like an area in a video game you can't get to. [no line break]"

to say na:
	say "There isn't even a store here any more--just a plot. [no line break]"

store-a is a privately-named  sto. understand "store/ 1/one/a" as store-a when player is in trips strip. the printed name of store-a is "store a".

description of store-a is "[na]"

store b is a sto. understand "store/ 2/two" as store b when player is in trips strip.

description of store b is "It's oddly greyed out, like an area in a video game you can't get to. A flyer inside the front door says SORBET SERVED HERE. Though someone wrote NO LONGER after SORBET."

store c is a sto. understand "store/ 3/three" as store c when player is in trips strip.

description of store c is "[one of]It's bolted up but seems to be advertising a corset sale. Useful for people who work in the escort sector. This game is word-weird, not sex-weird, though[or]You briefly wonder if you'll find sapient panties in there that might give a hint, then you feel kinda guilty. Whether you're male or female[or]'This is a LEGIT AD to help you (censored)!!!'[or]You see striptease tapestries and are sorry you did[or]'No shower here. Don't ask.'[or]Apparently employees of L'Harem Harlem pay only 69% of retail here[or]All Lipsopolis items are 20% off, apparently. Whatever those were.[or]It's adults only, but no Lady Lust will visit any time soon[or]Oh dear. Not a legit ad. 'Get L...'? Whoah.[or]You think you hear 'You perv, voyeur :p' but the emoticon makes it moot. Nice.[or]'Helpin' you deuces seduce!'[or]That's all the innuendo I got. There are a lot better sites on the Internet to find information on THAT sort of thing than wherever you downloaded this game from. Go on. Harp. Pry[stopping]."

store d is a sto. understand "store/ 4/four" as store d when player is in trips strip.

description of store d is "[na]"

store e is a sto. understand "store/ 5/five" as store e when player is in trips strip.

description of store e is "It's bolted up but seems to be advertising a stereo sale."

store F is a sto. understand "store/ 6/six" as store f when player is in trips strip. rgtext of store f is "[rc][rc][rc][rc][rc][rc]". lgth of store f is 6. gpos of store f is 6. rpos of store f is 2.

description of store f is "This is odd. It's kind of green on top, brown on the bottom. You smell pine needles and dirt and leaves you can't quite discern[if forest is unvisited]. You think back to the gateway and how you got here. There might be a way to do something to this store[end if]."

the roadsign is undesc. the roadsign is part of store f. understand "road sign" as roadsign.

instead of doing anything other than examining roadsign:
	say "It's too high to do anything except examine. Seeing as how it's green with red lettering, you can read it, as high up as its writing is.";

understand "writing" as roadsign.

check examining roadsign:
	say "It's a strain to read, and you wonder if maybe you can look for more natural clues as to what to do with store F. Try and pick out the writing anyway?";
	if the player consents:
		say "The sign appears to say, in all red, [rc][one of]E FORTS/OSTFER[or]TRESFO/S EFTRO[cycling][roman type] before a gust of wind picks it up and flips it around.";
	
store g is a sto. understand "store/ 7/seven" as store g when player is in trips strip.

description of store g is "A flyer stapled to the front of the door says [if you-can-advance]REST?! GO![otherwise]GO REST![end if]. There's nothing else interesting about the store, though."

the go rest flyer is scenery in trips strip. go rest flyer is undesc.

instead of examining the go rest flyer:
	say "It's just a subtle hint to get on with things[if you-can-advance]--in this case, to your reward[end if]."

store h is a sto. understand "store/ 8/eight" as store h when player is in trips strip. [description of store h is "Spray painted on this locked store is: others are in the throes of being reshot."]

description of store h is "Inside store H you see such a miscellany that you doubt it was successful even in better times."

store I is a sto. understand "store/ 9/nine" as store i when player is in trips strip. lgth of store i is 6. gpos of store i is 1. rpos of store i is 5. rgtext of store i is "[gc][rc][rc][rc][rc][rc]".

the Tories are plural-named scenery in trips strip. lgth of tories is 6. gpos of tories is 6. rpos of tories is 5. rgtext of tories is "[rc][gc][gc][rc][rc][rc]". The tories are auxiliary.

description of tories is "A closer look indicates that they are glued to the inside of the store window, instead of hanging from a wire on a nail or something."

understand "pictures" and "picture" and "prime minister" as tories when tories are visible.

[note we need to activate TORIES]

description of store i is "It has pictures of famous political figures in the window--Winston Churchill, Benjamin Disraeli and that great charmer, Margaret Thatcher[one of] (*)[or][stopping]. You think you remember what party they belonged to. The pictures are big enough to obscure--almost--the outline of a trap door behind them[one of].[paragraph break](*) semantically, too good to pass up. Feel free and welcome to picture it being said sarcastically[or][stopping]."

Margaret Thatcher is part of the Tories. description of Margaret is "Her looks are not important. Okay, the Iron Lady, etc[tory-clue].";

Winston Churchill is part of the Tories. description of Winston Churchill is "His looks are not important. Okay, bulldogish face, cigar, etc[tory-clue].";

Benjamin Disraeli is part of the Tories. description of Benjamin Disraeli is "He's wearing one of those long coats with vest and pocket-watch you guess were popular in Victorian times. You know you should know who he is more than you do, but his face is familiar[tory-clue].";

to say tory-clue:
	if player has gadget:
		say "[one of][or]. Scanning the pictures for clues physically is all well and good, but maybe your gadget would help[or]. You bring your gadget too close to the pictures, and it begins behaving weirdly[or][cycling]";

store j is a sto. understand "store/ 10/ten" as store j when player is in trips strip.

description of store j is "[na]"

store k is a sto. understand "store/ 11/eleven" as store k when player is in trips strip. rgtext of store k is "[rc][rc][rc][rc][rc][rc]"

description of store k is "A peculiar smell of smoke and incense seeps from store k, but that's probably not the BIG reason this place went downhill. No use prying."

store l is a sto. understand "store/ 12/twelve" as store l when player is in trips strip. 

description of store l is "You take a peek inside but feel loster and loster as you do."

store M is a sto. understand "store/ 13/thirteen" as store m when player is in trips strip. rgtext of store m is "[rc][rc][rc][gc][rc][rc]". lgth of store m is 6. gpos of store m is 6. rpos of store m is 1.

store n is a sto. understand "store/ 14/fourteen" as store n when player is in trips strip. 

description of store n is "It's grayed out, but you see a music book inside, songs for tenors, apparently."

store o is a sto. understand "store/ 15/fifteen" as store o when player is in trips strip. 

description of store o is "[na]"

store p is a sto. understand "store/ 16/sixteen" as store p when player is in trips strip. 

description of store p is "[gd]A poster covers the entire window. It's turned away from you, as if to say, don't worry about solving it this game, pal."

the poster is part of store p. description of poster is "Wait! There is some writing, if you look closely.[paragraph break]'This poster intentionally left blank.'"

store q is a sto. understand "store/ 17/seventeen" as store q when player is in trips strip. 

description of store q is "[na] There's no secret passage to store u, either."

store R is a sto. understand "store/ 18/eighteen" as store r when player is in trips strip. rgtext of store r is "[rc][rc][rc][rc][rc][rc]". lgth of store r is 6. gpos of store r is 4. rpos of store r is 2.

description of store r is "[how-far-along]. You also see a roster in the front door."

to say how-far-along:
	if number of solved regions is 1:
		say "It's almost all grey, except for a touch of red at the corners";
	otherwise if number of solved regions is 2:
		say "There's a thin band of red paint around the bottom";
	otherwise if number of solved regions is 3 and two-of-three is false:
		say "Store R is now about halfway red";
	otherwise:
		say "Store R is now completely red";
	say ". You [if you-can-advance]should be able to do something to it like with the other stores[otherwise]probably need to see what's behind at least one other stores first[end if]"
		
the roster is part of store r. the roster is auxiliary. rgtext of roster is "[gc][rc][gc][rc][rc][rc]". gpos of roster is 1. rpos of roster is 4. lgth of roster is 6.

instead of taking the roster:
	say "It's taped inside the front door, and besides, it's just there for information."

to decide whether you-can-advance:
	if number of solved regions < 3:
		decide no;
	if number of solved regions is 3 and two-of-three is false:
		decide no;
	decide yes.

description of roster is "The roster on the front of store R reads: 'People who have earned a vacation: you[if you-can-advance] (well, go on!)[otherwise] (well, soon!)[end if]'"
	
store s is a sto. understand "store/ 19/nineteen" as store s when player is in trips strip. 

description of store s is "As you get near, you hear GO AWAY YOU TOSSER. The accent doesn't sound British, so you suspect this person just needed to use a new swear."

store t is a sto. understand "store/ 20/twenty" as store t when player is in trips strip.

description of store t is "The window of store T is really lovely stained-glass--it's a depiction of a family of otters."

store u is a sto. understand "store/ 21/twentyone" as store u when player is in trips strip. 

description of store u is "[gd]The floor looks like some sort of cross-country map."

store v is a sto. understand "store/ 22/twentytwo" as store v when player is in trips strip.

description of store v is "[gd]You see people voting glumly and you think you hear them whisper it doesn't matter. Their votes go in the stove anyway. But politics wasn't in your job description. There are, apparently, more basic needs than that right now."

store w is a sto. understand "store/ 23/twentythree" as store w when player is in trips strip. 

description of store w is "You see replicas of two tall, narrow buildings."

store x is a sto. understand "store/ 24/twentyfour" as store x when player is in trips strip. 

description of store x is "[na]"

store y is a sto. understand "store/ 25/twentyfive" as store y when player is in trips strip. rgtext of store y is "[rc][rc][rc][rc][gc][rc]".

description of store y is "[gd]However, the store seems filled with water. There's a slimy bit and a small glistening sphere."

store z is a sto. understand "store/ 26/twentysix" as store z when player is in trips strip.

description of store z is "[na]"

section basic descriptions

section basic actions

test j with "j/x 1/x 2/x 3/x 4/x 5/x 6/x 7/x 8/x 9/x 10/x 11/x 12/x 13/x 14/x 15/x 16/x 17/x 18/x 19/x 20/x 21/x 22/x 23/x 24/x 25/x 26/x a/x b/x c/x d/x e/x f/x g/x h/x i/x j/x k/x l/x m/x n/x o/x p/x q/x r/x s/x t/x u/x v/x w/x x/x y/x z/x one/x two/x three/x four/x five/x six/x seven/x eight/x nine/x ten/x eleven/x twelve/x thirteen/x fourteen/x fifteen/x sixteen/x seventeen/x eighteen/x nineteen/x twenty/x twentyone/x twentytwo/x twentythree/x twentyfour/x twentyfive/x twentysix"

does the player mean examining store u: it is likely.
does the player mean examining store d: it is likely.
does the player mean examining store n: it is likely.
does the player mean examining store s: it is likely.
does the player mean examining store e: it is likely.
does the player mean examining store w: it is likely.

check examining a direction in trips strip:
	if player's command matches the regular expression "\b<a-z>$", case insensitively:
		try examining the store instead;

to say see-tories:
	say "[one of](NOTE: this is not a political statement on their party.)[or][stopping]";
	now td is in trips strip;
		
td is privately-named scenery in trips strip. printed name of td is "trap door". understand "trap/ door" and "trapdoor" and "place" and "outline" as td when player is in trips strip.

description of td is "There's a message on it. You have to move around to see all the letters behind the pictures, but when you do, it says:[paragraph break][fixed letter spacing]PROPERTY OF[line break]  STORE I[paragraph break]TO ATTACKING[line break]   MISSION[variable letter spacing]. [paragraph break]Whoah, heavy stuff."

instead of opening td:
	say "You can't get in the store[if cur-score of stores is 0]. It might be easier to try your luck/skill in Store F first.[otherwise]. How did you find a passage through [other-store], again?[end if]";

to say other-store:
	say "[if forest-sc is visible]Store F[otherwise]that other store[end if]";

[description of store b is "You smell different types of fruits as you get closer. You also feel slightly cold."

]

[description of store k is "Smoke permeates the store. You can't see through."

check entering the oddly sweet-smelling smoke cloud:
	adj instead;]
	
the metros-x are a plural-named and privately-named portal. "You see a down escalator here[if metros is solved], but it's not worth revisiting the metros[end if]."

description of metros-x is "The escalator has TO THE METROS scrawled on it."

the printed name of metros-x is "metros".

understand "metros" as metros-x.

understand "escalator" as metros-x when metros-x is visible.

check climbing metros-x:
	try entering metros-x instead;
	
to check-2-of-3:
	if number of solved regions is 3:
		say "You think you hear oohing and ahing in the background at the adventurer who has decided to do more than the bare minimum. (Fourth wall note: you can UNDO and recuse if you want to.)"

check entering metros-x:
	if metros is not unsolved:
		say "You've been there, done that[if metros is bypassed], or had it done, at any rate[end if]. Big city life is not for you." instead;
	check-2-of-3;
	say "The elevator's descent is rapidly scented worse. After you pass [one of]a[or]that[stopping] 'Tasers? You bet! Asters? Out, bye!' billboard, an automated voice announces that Mt. Rose has made its freedom more def, so ordinary citizens are motivated to level up just like Red Bull Burdell, and a funky thumping beat helps them to action.[paragraph break]You jog down the elevator to escape it, but of course it's coming from the city[if drainage is in underside]. And you step into some drainage at the bottom. Eww[end if][if underside is visited]. Which you should've remembered from last time[end if].";
	wfak;
	move the player to last-loc of metros;
	recover-items instead;
	
to recover-items:
	let JJ be the map region of the location of the player;
	repeat with JJJ running through item-list of JJ:
		now player has JJJ;
	repeat with JJJJ running through worn-list of JJ:
		now player wears JJJJ;
	
description of store m is "There's no window in store m--just a detailed map of a subway."

subway map is part of store m. description of subway map is "It's for a city called Mt. Rose, apparently."

[description of store p is "SAY THE MAGIC WORD TO ENTER is on the door here. There seems to be a sale of gimmicky joke items here."]

[description of store t is "Lovable baby otters peek back out at you through the unbreakable window. You have no knowledge of how to care for or adopt them, though, sadly, and you won't get any in this game."]

[description of store u is "You see a dizzying array of branching possibilities inside. Roads looping across each other and themselves."]

[description of store v is "'Piles of stuff are lying around. You think they're valuable, but you're not sure."]

[description of store w is "It should be much taller than it currently is, you sense."]

[description of store y is "A glimpse through the windows shows a slimy interior. The door seems like you could pry it if you were very strong or skilled. Which you aren't."]

chapter taping

taping is an action applying to one thing.

understand the command "tap [something]" as something new.

understand "tap" as taping.

carry out taping:
	if noun is pat:
		say "Not apt." instead;
	if noun is spout:
		if cask is visible:
			try filling the cask instead;
		otherwise:
			say "The oils would go to waste." instead;
	say "You may want to PUSH or ATTACK instead." instead;

does the player mean taping the spout: it is very likely;
	
chapter moor

Moor is a room. description of moor is "You're on a moor. Woods all around don't look inviting, but they also don't look like they can stop you returning to the room[if anapest is in moor][one of][or]. You hear bad poetry[stopping][end if]."

understand "reset" and "reset button" as a mistake ("[reset-goof]") when player is in moor and panel is part of silo.

to say reset-goof:
	if trees button is visible:
		say "[if trees-is-pushed is true]The reset button only appears in emergencies. Maybe there's one more thing you can change the trees button to[otherwise]A reset button is not practical once a missile is launched, but you are on the right track[end if].[run paragraph on][line break]";
	otherwise:
		say "You don't need to reset the steer button [if trees-is-pushed is true]like you did the trees button[end if]. It's what it's supposed to be.[run paragraph on][line break]"

the woods are scenery in moor. "They're forbidding--both for travel and for changing them into something else."

Woeful Pat is a person in the moor. "Woeful Pat, the awful poet, [one of]sees he has an audience and administers a dose of his odes[or]is well into a ballad. It's all bad[stopping]."

understand "awful poet" and "awful" and "poet" as woeful pat.

to say pat-blather:
	say "He holds up his hand for silence. He is too busy reciting his poem. You probably can't ask him about anything other than the poem[if anapest-clued is false]. Or maybe its meter[end if].[run paragraph on][line break]";

description of Woeful Pat is "His clothes are dark and strategically scuffed. Good thing chain wallets and other ugly stuff haven't made it here yet. But looks aren't really important. It's that poetry you find awful."

Woeful Pat wears the dark clothes.

description of dark clothes is "All dark clothes start to look the same after a while."

instead of taking dark clothes:
	say "A bad idea on so many levels."

every turn while player is in moor:
	if woeful pat is visible:
		say "The poem [one of]establishes its singsong rhythm early[or][drones][stopping].";

anapest-clued is a truth state that varies. anapest-clued is false.

to say drones:
	say "drones on[one of], poor and laboring--good qualities in a person but not a poem[or], and Pat gestures with a hand to the ear and nod of his head to make sure you're listening, not breaking his meter[or][cycling]. Da da DA, da da DA, da da DA[one of][or][if anapest-clued is true]. You'd recognize the rhythm if you studied harder in English class[otherwise]. That anapest is getting annoying, and you'd like to get rid of it, somehow[end if][or][ifnv][cycling]";

to say ifnv:
	if player does not have expo flyer:
		say ". Pat pauses for a minute, and involuntarily nod your head in gratitude. You don't notice when he shoves a flyer into your hand. 'I will be performing minor works soon. A poetry appreciator like you is welcome to attend.' Thankfully, there's no sample on the flyer you have";
		now the player has the expo flyer;

the expo flyer is a thing. description of expo flyer is "Stick Figure Connoiseur Expo and Open Poetry Session at [first custom style]ST. PAEAN[roman type] the Consummate Counterexample this Saturday! You probably didn't need to see St. Paean all in such bright red. It seems all wrong for the name of a church, or, well, anything, really."

check scaning the expo flyer:
	say "[if anapest is in moor]Your gadget makes no noise, but you can't help feeling the name St. Paean is totally wrong[otherwise]No sound[end if]." instead;

poetry-listening is a truth state that varies. poetry-listening is usually false.

to say anapest-clue:
	say "[one of][if anapest-clued is false]'I noticed you nodding your head to the beat. You probably don't even know it's called ANAPEST. But the proper meter of my poem is but the surface of its multifaceted splendor!'[otherwise]'Ah, an appreciator of fine ANAPEST. I will waste little time discussing its details, since you already seem to know them.'[end if][line break]Before you realize he'd given you a break from his poetry, he starts up again.[or]The explanation about anapest was bad enough the first time around. Memorably bad, at least.[stopping]";
	now anapest-clued is true;

check listening to poetry:
	if anapest-clued is false:
		say "[anapest-clue]" instead;
	say "[one of]He at least has the meter--what was it again, anapest?--right. 'End of canto,' he says.[or]'Are you reflecting on the inner meaning? Or are you just amazed I have gotten the anapest right in every single line? Even--pff--mathy types who can't count meter right.'[stopping]" instead;

instead of examining poetry:
	try listening to poetry instead;

the poetry is scenery in the moor. poetry is undesc.

understand "poem" as poetry when poetry is visible.

the roadblock is a thing in the moor. "You see a roadblock that isn't actually blocking any roads.". rgtext of roadblock is "[rc][rc][gc][rc][rc][rc][gc][rc][rc]". lgth of roadblock is 9. gpos of roadblock is 5. rpos of roadblock is 1.

understand "block" as roadblock when roadblock is visible.

the road is scenery in moor. "[if roadblock is visible]You'd expect to see a road, what with the roadblock, but you don't[otherwise]Removing the roadblock did not make a road appear. But hey, free black door[end if]."

instead of climbing roadblock:
	say "There's no path behind it. Some roadblock, eh? It might be better reincarnated as something else."
	
the roadblock is fixed in place.

instead of taking roadblock:
	say "Even dented, it won't let you move it."

a black door is a thing. description of black door is "It's a swinging door. No knob or bolt or anything."

black door is pushable between rooms.

check pushing black door to:
	if second noun is inside and silo is visible:
		try pushing black door instead;
	otherwise:
		say "You don't need to get rid of the black door or move it anywhere else." instead;
	
instead of opening black door:
	if black door is part of the silo:
		say "A quick peek inside, and you see enough machinery to know this is not a grain silo.";
	otherwise:
		say "Other doors in this game may lead somewhere for no reason, but this doesn't.";
	
check taking the black door:
	if black door is part of the silo:
		say "It's where it needs to be." instead;
	if player does not have black door:
		say "It's too heavy to lug around for any amount of time, but maybe it can fit in somewhere if you push it, or put it on something.";
		stop the action;
	
check taking the panel:
	if panel is part of the silo:
		say "It's where it needs to be." instead;

instead of opening silo:
	if black door is part of the silo:
		try opening black door instead;

instead of entering black door:
	say "Nope--it won't lead to new worlds. You're already sort of in one.";

description of roadblock is "It's about eight feet long and really dark (you pause and hope you don't need to learn COBOL) and dented about halfway through--well, four-ninths of the way from the right. You are utterly unsure what it could possibly be guarding, and you also notice bright red graffiti[one of].[paragraph break]You think back to how your nametag was creased, and how the roadblock probably doesn't change into anything too easy like a broad lock. The moor [if sack is not off-stage]and the sack [end if]already felt a bit easy[or][stopping]."

instead of doing something with the bright red graffiti:
	if current action is examining:
		continue the action;
	say "You can really only examine it.";
	
description of bright red graffiti is "DOC OLBARK has chosen to deface this roadblock, in bright red paint, no less."

anapest is scenery in moor. rgtext of anapest is "[rc][rc][gc][rc][rc][rc][gc]". lgth of anapest is 7. gpos of anapest is 4. rpos of anapest is 7. anapest is undesc.

understand "anapests" as anapest.

instead of scaning poetry:
	say "Yes, this sort of poetry doesn't require deep reading, just scanning";
	if player has gadget:
		say ". But seriously, your gadget seems to blink with the beat. [run paragraph on]";
		try scaning anapest instead;
	otherwise:
		say ". Too bad you don't have a device to scan it more practically." instead;
	
instead of doing something other than scaning anapest:
	if the current action is objasking generically:
		continue the action;
	say "The beat is drilled in your head: da da DA da da DA da da DA (repeated. I'll spare you the words, but...)"

instead of going nowhere in moor:
	say "[one of]You'd probably get lost in the forest around. There're no paths[or]You get the feeling walking won't do much good, here[stopping]. But nothing's stopping you from leaving the way you came."

the peasant is a man. "A peasant is here[if peasant has hay], carrying some hay over his shoulder[end if]."

understand "peasants" as peasant when woeful pat is visible.

before doing something with the hay:
	if the peasant has the hay:
		say "The hay is the peasant's, and he is bigger than you. But maybe you can give him something for it, if you want it[one of]. You notice him occasionally looking ruefully at it[or][cycling]." instead;

some hay is a singular-named thing.

indefinite article of some hay is "some".

description of hay is "It's hay, hey? Well...there's enough to stuff something pretty substantial."

instead of taking the hay:
	if peasant has the hay:
		say "That's the peasant's. Someone already ripped him off, giving it to him." instead;
	if hay is in sack:
		say "Keep it in the sack [']til you need it." instead;
	if player has sack:
		say "You can't take it all at once, but it can fit in the sack. So you put it there.";
		now hay is in sack;
	say "You couldn't see around it if you carried it in your hands, and you've got nothing that'd hold it."

description of peasant is "He can't help glancing at the hay he is carrying, frowning a bit."

check giving sack to peasant:
	say "No, wait, you might still need that, after. But you could maybe give him something inside[if number of things in sack is 0], once you have something there[end if]." instead;

understand "trade" and "trade [text]" as a mistake ("Instead of trading, use GIVE X to Y.");

check giving cask to peasant:
	say "No, wait, you might need that. But you could maybe give him something inside." instead;
	
before showing to peasant:
	ignore the can't show what you haven't got rule;
	try giving the noun to the peasant instead;
	
before giving to peasant:
	if noun is sack and straw is in sack:
		say "(I'm assuming you meant the straw in the sack)";
		try giving the straw to the peasant instead;
	say "The peasant probably doesn't want any adventuring goodies. Maybe something more practical for him?" instead;

before giving straw to peasant (this is the straw-peasant rule):
	ignore the can't give what you haven't got rule;
	say "'Oh, thank you! Lord Ablemiser stuck me with this hay when I needed much stronger material to re-patch my house. He pretended not to know the difference. But he did once!'[paragraph break]'Here's a little something I wrote. [italic type]He[roman type] said even an economist wouldn't put emoticons in a poem.'[paragraph break]He hands you a paper and heads off. You're sick of poetry, but with the hay weighing you down, you'll never catch the peasant.";
	now player has the poem;
	reg-inc;
	now peasant is in lalaland;
	now straw is in lalaland;
	now hay is in sack instead;

a poem is a thing. lgth of poem is 5. gpos of poem is 1. rpos of poem is 2. rgtext of poem is "[gc][rc][rc][rc][rc]".

does the player mean folding the poem: it is likely.
does the player mean doing something with the tinfoil info-lit when the poem is visible: it is unlikely.

understand "limerick" as poem when poem is not folded.

understand "paper" as poem when poem is visible.

smilies are part of poem. smilies are plural-named.

instead of taking smilies:
	say "You can't just pull them off the page."
	
description of poem is "'I'm just a guy who loves his work[smy][line break]Not like that poet, he's a jerk[smy][line break]He says he rhymes neater[smy][line break]And rips on my meter[smy][line break]I'd like to punch him in his smirk[smy]'[paragraph break][if missile is off-stage]Whoah, lots of smilies, there. They'd drive a grammar pedant ballistic[otherwise]The limerick doesn't make you want to lick the mire like Woeful Pat's stuff[end if]. The paper seems creased on a few diagonals."

check examining poem:
	if poem is folded:
		say "It's a paper plane now, devoid of evidence anyone ever scrawled poetry on it." instead;

to say smy:
	if missile is off-stage:
		say " :) :)";

chapter Cruel Ones' Enclosure

coe is a privately-named room. coe is north of Silent Rut. "A creepy acre, yep[if liches are in coe]. Liches block your way north[one of]. Drat! You were hoping for a ghost, whom you could zap to goths, and then if they had any spunk, you'd show them up as punks[or][stopping][end if][if drapes are in coe]. Along the north wall you see drapes fluttering. They're too thick to work through, but they seem to have some sort of message on them[else if spread is in coe][if-spread][end if][if banshee is visible]. A banshee also wails about his former life, and it'd be nice to get rid of him, but not necessary[otherwise].[end if]."

to say d-s:
	say "[if drapes are visible]drapes[otherwise]spread[end if]";

understand "spared" as a mistake ("That is what the drapes say.") when drapes are visible.

understand "spared" as a mistake ("Actually, the spread must NOT be spared if you want to get beyond here.") when spread is visible.

check going nowhere in coe:
	say "There seems to be nothing, or worse than nothing, that way." instead;

cavern is scenery in coe.

description of cavern is "[if liches are visible]You try to crane your neck around the liches, but they move to block you from even LOOKING. Tough crowd[otherwise]You don't know where it leads[end if]."

instead of entering cavern:
	try going north.
	
printed name of coe is "Cruel Ones['] Enclosure"

to say if-spread:
	if spread is cut:
		say ". That spread you cut in half is sagging to either side, revealing a doorway north";
	otherwise:
		say ". A spread blocking the way north flutters in some breeze you cannot feel";

Include (-
	has transparent animate
-) when defining banshee.

the banshee is scenery in coe. rgtext of banshee is "[rc][rc][rc][rc][rc][gc][rc]". gpos of banshee is 5. rpos of banshee is 3.

description of banshee is "You can't see him, but you can hear him."

the hasbeen is scenery. the hasbeen is undesc. understand "has been" as hasbeen.

instead of doing something with hasbeen:
	if hasbeen is visible:
		say "That's just what he wants. People thinking about him, remembering his misery, even when he is gone. You have more important things to do!";
	otherwise:
		continue the action.

turnstile-score is a truth state that varies. turnstile-score is usually false.

check going north in Silent Rut:
	if player is wearing the beard:
		say "You are 'recognized' and let through. A voice whispers 'Many vowels, one flow.'";
		if turnstile-score is false:
			reg-inc;
			now turnstile-score is true;
		continue the action;
	otherwise:
		say "The turnstile jams. There's no way to jump over it. Maybe it doesn't like your face[if number of things that are part of beard is 2]. Perhaps that disguise could fool it[end if]." instead;

to check-score (a - indexed text):
	say "We need to zap the Check-Score for [a] for the new version.";
	the rule succeeds;

chapter Ghouls' Slough

Ghouls' Slough is north of coe. "This is as ethereal as its inhabitants. You're not sure where the marsh starts and the lake ends. You think you hear a dog bark in this dark bog."

the dog is scenery in slough. dog is an undesc. instead of doing something with the dog when dog is visible:
	say "Where was it? Where is it? Eegh. You don't REALLY want to think about it."

understand "god" as a mistake ("You want to tempt the supernatural in somewhere this hellish? Phew.") when dog is visible.

the bog is scenery in slough. bog is an undesc.

understand "gob" as a mistake("The bog expands into a big ol['] gob and sucks you down. Congratulations, there aren't many ways to die in this game, but you found one. Fortunately, you can undo this.[end-it]") when bog is visible.

to say end-it:
	end the story;

instead of doing something with the bog when bog is visible:
	say "The bog feels like a big old gob that could suck you in. You just want to find a way away from it."

the marshlake are privately-named plural-named scenery in Ghouls' Slough. understand "marsh/lake" as marshlake. the printed name of marshlake is "The marsh and lake".

description of marshlake is "On almost all sides. You worry you may get sucked down into it."

nowhere is south of Ghouls' Slough.

check going in Ghouls' Slough:
	say "Useless without any sort of guide to look at[if player has maps]. Hey, those maps might be useful[end if]." instead;

check examining maps in Ghouls' Slough:
	say "You read the maps. So much to avoid in this direst stride: ah, the heath. Scare Acres, Harm's Marsh, Hell's Shell, Dirge Ridge, Nowt Town, the Icy Vat Cavity and even a grocery that'd make an ogre cry. And you note the warning to hide from the Direr Rider and Purple Pulper. Then the chilling cries of 'BRAAINS, SABRINA!' distract you, until you look up in horror at one of the iciest cities...";
	move player to Frost Forts instead;

chapter Frost Forts

Frost Forts is a room. "It'd take a chimera to do the ice harm here, there's so much of it. The forts all around seem to frown at you, and six-foot-high iced dice are placed all around. An icecap is near you, smelling of ipecac.[paragraph break]All exits seem to lead somewhere even darker[if wolves are in Frost Forts], though I doubt those werewolves will let you get there[end if][one of]. You steel yourself against the sleet[or][stopping]."

the ipecac icecap is scenery in frost forts. "Just there for local flavor."

instead of doing something with icecap:
	say "It's just there for ambience."

the sleet is scenery in frost forts. "It's painful when it hits you, but since it's rain mixed with snow, it means you won't freeze...right away."

understand "steel" as a mistake ("You'd be crushed if the sleet turned to steel.") when sleet is visible

the iced dice are scenery in frost forts. "Each side seems to have a big skull-and-crossbones on it. Their fare is fear, with no warm cafe hidden in any face."

instead of taking dice:
	if wolves are visible:
		say "If you could pick the dice up, you wouldn't need another weapon to trash the wolves." instead;
	otherwise:
		say "The dice could crush the vowels easily. But you're not twenty feet tall, so they just sort of add to the atmosphere."

understand "ice" as iced dice.

the fos are privately-named scenery in frost forts. the frowns are part of the fos. understand "forts" and "frost forts" as fos.

description of fos is "Frowny, as if they want you to fry now. Also, impassable. They have no doors but you're worried something will jump at you from them."

description of frowns is "Probably either sharp or decaying teeth behind them, but the wolves won't get near you."

Frost Forts is north of Ghouls' Slough. nowhere is south of Frost Forts.

vowels are in Frost Forts. vowels are plural-named and privately-named. "Six large aplastic capitals sway when there is no breeze and stand still when there is."

rgtext of vowels is "[rc][gc][rc][rc][rc][gc]". lgth of vowels is 6. gpos of vowels is 3. rpos of vowels is 6. [describe canister in room better]

understand "a/e/i/o/u" as vowels when player is in frost forts.

instead of taking vowels:
	say "They're just way too big. Like that LOVE statue in Philadelphia. But they fill you with fear, unlike the LOVE statue."

understand "vowels/letters/five/six/tall/aplastic/capital/capitals/pyramid" as vowels when vowels are visible.

description of vowels is "[vowel-desc]"

to say vowel-desc:
	if the player's command includes "vowels":
		say "Hm...yes...they are that, but what do they need to be?[no line break]";
	otherwise:
		say "A, E, I, O, U and Y, piled up in a pyramid. Hm, they are not just ANY letters. Yup, they're pretty much shouting out (figuratively) 'We're vowels!' Yup, even the Y. The vowels also have bumps where they end.[no line break]"
		
the bumps are part of the vowels. description of bumps is "They look almost like fangs or very long nails."

check going in Frost Forts:
	say "[if vowels are in frost forts]The rest of the area seems even more forbidding. You sense the vowels would roll after you and crush you if you tried. You have no idea how to get back to warmer climates[otherwise]The werewolves will catch you easily[end if]." instead;

chapter Softer Forest

sf is a privately-named room. the printed name of sf is "Softer Forest".

rdir is a direction that varies.

smelled-sf is a truth state that varies. smelled-sf is usually false.

the lapsing sapling is scenery in sf. "It seems to be bending [opposite of rdir], away from the nice-smelling breeze."

check examining lapsing sapling:
	now smelled-sf is true;
	
check going rdir in sf:
	if smelled-sf is false:
		try going opposite of rdir instead;
	say "You walk for a while, following your nose, and...";
	now odor is in rf;
	now player is in rf instead;

check going in sf:
	say "Hm. You don't have any reason to believe anything's that direction. Always the risk of [one of]having a snake sneak[or]a fauns snafu[or]a stray satyr[or]a stag or gators[or]a stoat making toast of you[at random] or something." instead;
	the rule succeeds;
	
description of sf is "There's no clear sign of clearings, but the leaves are soft here. There's a smell, too. Unusual but pleasant. There's also a lapsing sapling, half bent over from a breeze."

check smelling in sf:
	now smelled-sf is true;
	say "It's a weird smell, but you like it. You feel a breeze from the [rdir], so that's the direction the smell must be from." instead;
	
the odor is scenery. the odor is undesc. it is in sf. rgtext of odor is "[rc][rc][gc][gc]". lgth of odor is 4. gpos of odor is 2. rpos of odor is 4.

check propfliptoing odor:
	if player is in sf:
		say "You sense the odor swirling, but it's not thick or intense enough to become something physical." instead;

understand "smell" and "breeze" as odor.

instead of taking odor:
	say "Impossible."

instead of examining odor:
	try smelling instead;

chapter Rest of Forest

rf is a privately-named room. the printed name of rf is "Rest of Forest".

the forest-leaves are a privately-named backdrop. "They've already undergone nature's transformation, so they're beautiful rather than useful.". printed name of forest-leaves is "leaves".

the forest-leaves are in rf and sf.


description of rf is "This is the place where you tracked the odor to[if rut-door is visible], though it's gone now that you magicked it into a door[end if]."

rut-door is a privately-named fixed in place thing. the printed name of rut-door is "the improbable door[if bulge is part of the rut-door] with the bulge in it[end if]". understand "door" and "improbable" as rut-door when odor is visible or rut-door is visible.

rut-opened is a truth state that varies. rut-opened is false.

the musical note is part of the rut-door. description of musical note is "It's black, while the door is white. Perhaps it's meant to clue what the bulge is, or how to open the door.". understand "music" as musical note.

does the player mean doing something with the musical note: it is likely;

the bulge is part of the rut-door. rgtext of bulge is "[gc][gc][rc][rc][gc]". lgth of bulge is 5. gpos of bulge is 1. rpos of bulge is 5.

description of the bulge is "It's shaped like a racetrack, except for an upper corner that fans out. There's no way to pry it from the door. [run paragraph on][bugle-clue]."

to say bugle-clue:
	say "[one of]You give it a few taps, but it's wedged in[or]Maybe it could become some sort of instrument to get rid of the door[or]It's stuck to the door proper, like a tattoo[or]You guess it's your charge to open the door with it[or]You've got no reveilleation, err, revelation, what it should be, yet[or]You pay closer attention--[if player has the gadget]your gadget leaves it solved[otherwise]it can't be that hard to shift around. Maybe you should've taken one of the devices in the cabinet. It's only going to get tougher[end if][cycling]"

instead of taking the bulge:
	say "[bugle-clue]."

the bugle is a thing. description of bugle is "It's brass, you guess, and it probably can't be too hard to play."

understand "horn" as bugle when bugle is visible.

understand "doorway" as rut-door when rut-opened is true and rut-door is visible.

description of rut-door is "[rut-descri]."

to say rut-descri:
	if rut-opened is true:
		say "It's really just the outline of a doorway now. You can just enter it";
		the rule succeeds;
	say "It's got a big musical note painted on the front above [if bulge is visible]a bulge which doesn't seem to belong on the door[otherwise]where the bulge was[end if][if rut-door is open]. Maybe the novelty of seeing a door lead somewhere it shouldn't has worn off for you, but the world beyond the door seems less vibrant than this forest. Still, you don't know where else you could go[otherwise]. It probably leads somewhere interesting, if you can get through it[end if]"

check opening rut-door:
	if rut-opened is false:
		say "There's no real handle to grab[if bulge is part of the rut-door]. Not even that bulge, though you probably shouldn't snub nubs like that completely[end if]." instead;
	say "It's already open. Would you like to walk through?";
	if the player consents:
		try entering rut-door instead;
	otherwise:
		say "There's nowhere else to go, though." instead;

check entering rut-door:
	if rut-opened is false:
		say "You try walking through the closed door, but physical force doesn't seem to work[if player has bugle]. But that bugle might do something[end if]." instead;
	say "You walk away from dew odors and redwoods to a leveraged everglade. You slip into a fissure and feel fussier, then make a vainer attempt to crawl back up the ravine you slid down. Looks like you're stuck here. Way to burgle, bugler!";
	now player is in silent rut instead;

check propfliptoing door:
	if player is in rf:
		say "The odor's faint here, so it would just make a small door. Nothing you could fit through." instead;

check smelling in rf:
	if rut-door is not in rf:
		say "It's an odor of new places and adventure and puzzles and how to get there. Which is all very good, but you'd like something more physical to help you." instead;
	otherwise:
		say "The door is of some odd wood you don't recognize." instead;

check going inside in rf:
	try entering rut-door instead;
	
check going nowhere in rf:
	if odor is visible:
		say "That smell seems to be drawing you to new places. If you left it, you might not find it again.";
	otherwise:
		say "And maybe lose where this door you made is? Bad idea.";
	the rule succeeds;

chapter Silent Rut

Silent Rut is a room. "You feel a breeze from the west, and there's a plainer passage east. Weird solitary figures pass through a turnstile to the north at regular intervals."

check going inside in silent rut:
	try going north instead;
	
check going nowhere in silent rut:
	say "No going back. Storing's west, sorting's east and that turnstile's north." instead;

Include (-
	has transparent animate
-) when defining gy.

gy is privately-named scenery in Silent Rut. understand "man" as gy when player is in silent rut. gy is undesc.

printed name of gy is "bearded man".

understand "bearded/ man/guy/figure" as gy.

instead of doing something with gy:
	say "Too late. He's gone. But here's another.";

Include (-
	has transparent animate
-) when defining gy2.

gy2 are plural-named and privately-named scenery in Silent Rut. printed name of gy2 is "bearded men". gy2 is undesc.

understand "bearded/ men/guys/figures" as gy2.

instead of doing something other than scaning gy2: 
	say "They're just--there. Their faces are different, but they seem the same, and they take no notice of you."

turnstile is scenery in Silent Rut. "It's one of those full-height deals, so you won't be able to jump it. It doesn't look quite vicious enough to grind you backwards with its teeth, but still, it's shiny and imposing. The top reads TO THE LUPINE LINEUP."

instead of entering turnstile:
	try going north;

beard-unmade is a truth state that varies. beard-unmade is usually false.

chapter Storing Area

Storing area is west of Silent Rut.

check going nowhere in storing area:
	say "You can only go back east." instead;

section canister / red ring

the scantier canister is a fixed in place container. "[one of]A scantier canister, which you can probably call a can without hurting its feeling,[or]The scantier canister[if livers are off-stage], gaping red ring at the top,[end if] is welded to the shelf.[stopping]"

the canister is on the flesh shelf. understand "can" as canister.

instead of taking the canister:
	say "It's welded to the shelf.";

the red ring is part of the canister. description of red ring is "It's on top of the canister."

the grinder is part of the canister. description of grinder is "It's a pair of sabled blades that rotate opposite to each other when activated."

instead of inserting into grinder:
	try inserting noun into canister;

instead of inserting into red ring:
	try inserting noun into canister;

instead of putting on grinder:
	try putting noun on canister;

instead of putting on red ring:
	try putting noun on canister;

instead of putting on canister:
	say "(I'm going to assume you meant in the canister.)";
	try inserting noun into canister instead;

understand "sabled/ blades/blade" as grinder. understand "blender" as grinder.

instead of doing something other than examining the red ring:
	say "It's not that important, but it had to be SOME color. You probably want to do stuff with the canister and not the grinder.";

instead of doing something other than examining the grinder:
	say "The grinder's too sharp to touch.";

description of canister is "It's small, nowhere near the size of a cistern. It's got a red ring you can put stuff in, and you can see a grinder of sabled blades inside. The grinder reads [italic type]team meat, tame [']em at...[roman type] and appears to have no switch or anything."

instead of entering canister:
	say "It's too small."

the blades are part of the canister. description of blades is "Sharp and warm to the touch."

the canister can be broken. the canister is not broken.

check examining canister:
	ignore the examine containers rule;

does the player mean inserting chicken liver into canister when chicken liver is in canister: it is very unlikely.

does the player mean inserting cow liver into canister when cow liver is in canister: it is very unlikely.

check inserting into the scantier canister:
	if noun is sandwich:
		try taking the spam;
	if canister is broken:
		say "It's broken now." instead;
	if noun is spam:
		say "The canister whirrs threateningly before you drop the spam in. It's probably already been mixed, and the canister possibly even has some bad mechanical memory of being the guilty party. Technology!" instead;
	if noun is in canister:
		say "It's already in there.";
	if noun is a glopmeat:
		if number of glopmeats in canister is 1:
			say "Bam! The canister chokes, sputters, and then wheezes--you see a mist arise from it and vanish. It coughs out a combined glop that can only be described as LIVERS. Then it wheezes and seems to die.";
			reg-inc;
			now chicken liver is in lalaland;
			now cow liver is in lalaland;
			now canister is broken;
			now player has the livers;
			now description of canister is "It's completely broken." instead;
		otherwise:
			say "The canister seems to whir. The meat you put in is shortly unrecognizable.";
			now noun is in canister instead;
	if noun is bread:
		say "You don't need to futz with the bread that way." instead;
	say "The canister's for processing food, and that doesn't count." instead;
	
check switching on the canister:
	say "There appear to be no buttons to switch it on or off."
	
check switching off the canister:
	say "There appear to be no buttons to switch it on or off."
	
check touching the blades:
	say "They begin to whir at your finger's presence." instead;

section flesh shelf

the flesh shelf is a fixed in place supporter in storing area.

initial appearance of flesh shelf is "[if number of things on flesh shelf is 0]You've taken everything from the shelf[otherwise]You see [a list of things on flesh shelf] on the shelf[end if]."

description of flesh shelf is "NO SHAM HAMS ON FLESH SHELF is traced into its top[if number of things on flesh shelf is 0]. It's clear of items[otherwise]. You can see [a list of things on flesh shelf] on it[end if]."

description of Storing Area is "This is a storing area with various unpalatable-looking meats[if number of things on flesh shelf is 0], and you've taken the stuff that isn't totally nasty[otherwise] on a flesh shelf[end if]."

check examining the flesh shelf:
	ignore the examine supporters rule.

A glopmeat is a kind of thing. A glopmeat can be stuck. A glopmeat is usually stuck.

a sandwich is on the flesh shelf. the sandwich is not fixed in place.

instead of pulling sandwich:
	try opening sandwich;

instead of opening sandwich:
	say "You [if player does not have the sandwich]take the sandwich and [end if]peel the bread from the meat.";
	now player has bread;
	now sandwich is in lalaland;
	now player has the spam;
	
instead of eating sandwich:
	say "It doesn't seem appetizing, and you don't need the energy it might give.";

description of sandwich is "Just bread and[one of]--oh dear--[or] [stopping]spam."

check taking the bread:
	if bread is part of the sandwich:
		now player has the bread;
		now sandwich is off-stage;
		now spam is not fixed in place;
		now player has spam;
		say "It was somewhat lumped to the spam inside it, but you were able to peel them apart. The spam probably doesn't need to go on the floor, so you take that, too." instead;

instead of peelxing bread:
	if bread is part of the sandwich:
		try taking bread instead;
	otherwise:
		say "Peeling crust is a bad habit.";

some bread is part of the sandwich. rgtext of bread is "[gc][rc][rc][rc][gc]". lgth of bread is 5. gpos of bread is 1. rpos of bread is 5. the bread is singular-named.

description of bread is "It appears gritty, like a five o'clock shadow. Not mold, more like it was toasted some time ago[if bread is part of sandwich]. It's kind of stuck to the spam it's covering[end if]."

some spam is part of the sandwich. rgtext of spam is "[rc][rc][rc][rc]". lgth of spam is 4. gpos of spam is 4. rpos of spam is 1. spam is singular-named. the indefinite article of spam is "some".

understand "meat" as spam when spam is visible. does the player mean doing something with spam: it is likely.

instead of peelxing spam:
	if spam is part of the sandwich:
		try taking spam instead;
	otherwise:
		say "Don't play with your food. Even if it's not very foody.";

check taking bread:
	if bread is part of the sandwich:
		move bread to location of player;
	if spam is part of the sandwich:
		move spam to location of player;
	now sandwich is off-stage;

check taking spam:
	if bread is part of the sandwich:
		say "You unpeel the bread before taking the meat. What the heck, you take the bread, too. [run paragraph on]";
		now player has bread;
	now player has spam;
	say "It's sort of gross, but hey, it's not moldy.";
	now sandwich is in lalaland instead;

[check taking spam:
	now bread is not part of sandwich;
	now spam is not part of sandwich;]

description of spam is "It's nastily marbled with mostly pink and some white."

section meats in freezer

check putting it on:
	if noun is a glopmeat and second noun is a glopmeat:
		say "Don't be so literal. Well, not that way. What's a better way to join them?" instead;

the cur-liv is a glopmeat that varies. the cur-liv is the chicken liver.

understand "meat" as chicken liver when cur-liv is the chicken liver.
understand "meat" as cow liver when cur-liv is the cow liver.

liver-disambig-yet is a truth state that varies.

before doing something with the cow liver when cow liver is visible:
	now liver-disambig-yet is true;

before doing something with the chicken liver when chicken liver is visible:
	if liver-disambig-yet is false:
		now liver-disambig-yet is true;
		if the player's command does not match "chicken":
			say "(You aren't sure which liver to handle, but the cow liver looks viler. So, yeah, the chicken.)";

after doing something with a glopmeat:
	if noun is in canister:
		now cur-liv is a random glopmeat that is not the noun;
	now cur-liv is the noun;

does the player mean putting the cur-liv on: it is likely.
does the player mean inserting the cur-liv into: it is likely.

The chicken liver is a glopmeat on flesh shelf.

The cow liver is a glopmeat on flesh shelf.

description of a glopmeat is usually "[if item described is not in canister]It's ugly and frozen and stamped 100% amalgamatable[otherwise]It's all chopped up in the canister right now[end if]."

does the player mean taking cur-liv when cur-liv is held: it is unlikely.

does the player mean taking chicken liver when chicken liver is held and cow liver is held: it is very likely.

check taking a glopmeat:
	if player has noun:
		say "You've already got it." instead;
	if glopmeat is in canister:
		say "You shouldn't put your fingers near the blades." instead;
	if glopmeat is stuck:
		say "It's stuck, well, frozen to the wall[if player does not have chisel]. And you can't change that. Well, not with what you've got[end if][if player has chisel][chiz-chiz][end if].";
		now noun is not stuck;
		if player has chisel:
			now player has noun;
		do nothing instead;

check putting bread on spam:
	ignore the can't put what's not held rule;
	if bread is part of sandwich:
		say "Already is." instead;
	say "No sense remaking the sandwich." instead;

check putting spam on bread:
	ignore the can't put what's not held rule;
	if bread is part of sandwich:
		say "Already is." instead;
	say "No sense remaking the sandwich." instead;
	
to say chiz-chiz:
	say ". But you look through your inventory, and that chisel [if cow liver is stuck and chicken liver is stuck]looks made for scraping stuff. It easily peels the liver, which you take[otherwise]looks up to the task again. It works"
	
instead of eating a glopmeat:
	say "Uggh. No way[if noun is examined]. Even though someone stamped it as eatable[end if]."

instead of eating livers:
	say "Wow. No. It might be safe to, but it wouldn't be fun to."

instead of eating spam:
	say "The thought of eating spams gives you a spasm. Eww."

livers are things. description of livers is "Amalgamated well enough that you can't recognize which parts were cow and chicken. You suspect they're an upgrade from the individual livers but need to find out how.". lgth of livers is 6. gpos of livers is 6. rpos of livers is 5. rgtext of livers is "[rc][rc][rc][rc][rc][rc]".

understand "glop" and "glob" as livers.

chapter Sorting Area

Sorting Area is east of Silent Rut. "This is where, presumably, stuff gets sorted. A notes stone over a stack of tacks relates some sort of silly myth that's probably relevant to you. [container-desc]."

check going nowhere in sorting area:
	say "You can only go back west." instead;

the stick figures are scenery in Sorting Area. understand "stick/ figure" as stick figures.

description of stick figures is "Each one's not particularly detailed, but it probably provides a clue as to the stuff in each bucket."

the notes stone is scenery in Sorting Area. understand "myth" as notes stone.

the stack of tacks is scenery in Sorting Area. "[bug-report]"

instead of doing something with the stack of tacks:
	say "The tacks are glued to themselves. And, oddly, to the notes stone."
	
Instead of taking the notes stone:
	say "it seems to be glued to the stack start. Anyway, you probably only need it for its information."

description of notes stone is "'[italic type]Attention, word-boy/girl/woman/man! You may think letters are the basic building blocks of everything, but the computer age has shown it's 1[']s and 0[']s! We don't stow twos here! You won't find simpler piles, Mr.! Or a simpler pile, Mrs.![roman type]'"

to say container-desc:
	if number of containers in sorting area is 0:
		say "You've pretty much ransacked this place. There's nothing much left";
		the rule succeeds;
	say "You see [number of containers in sorting area in words] container[if number of containers in sorting area > 1]s[end if] here: [a list of the containers in sorting area]";

rule for choosing notable locale objects for sorting area:
	repeat with item running through containers in sorting area:
		set the locale priority of the item to 0;

the bucket is an opaque container in the Sorting Area. printed name of bucket is "bucket[if ones are in bucket] full of ones[end if]".

the shell is a container in the Sorting Area. printed name of shell is "shell[if dashes are in shell] full of dashes[end if]".

the bubble is a container in the Sorting Area. the printed name of bubble is "bubble[if noughts are in bubble] full of noughts[end if]".

description of bucket is "It's got HERE BE ONES scrawled on it, and a stick-figure with a question mark between a mouth and a pair of eyes[if ones are in bucket]. A bunch of narrow rectangular tiles are piled in[otherwise]. It's empty[end if]."

description of shell is "It's upturned, so it can hold stuff, with HERE BE DASHES scrawled on it[if dashes are in shell]. A bunch of narrow rectangular tiles are piled in, a bit more efficiently than the other containers. A stick figure on it has X-eyes from looking into the sun[otherwise]. It's empty[end if]."

description of bubble is "It's [if noughts are in bubble]labeled NOUGHTS. You see a stick figure falling over, clutching at more spilling innards than a stick figure has a right to have[otherwise]empty now that you flipped the noughts[end if]. You also spy some sort of elegy written."

the elegy is part of the bubble.

the description of the elegy is "It's in a bright red that seems entirely wrong: [italic type][first custom style]Uthgon's Hut-Song:[paragraph break]Hognuts! Un-ghost no thugs![line break]Og hunts gnus, tho['] gush not tons, ugh.[roman type]"

instead of taking bucket: say "Buckets be stuck here, unsurprisingly."

instead of taking bubble: say "It's welded to the wall."

instead of taking shell: say "It's welded to the wall."

check inserting into:
	if second noun is trel-priv:
		say "Maybe you mean to put something in the wall instead?" instead;
	if second noun is silo:
		if noun is panel or noun is black door:
			try putting noun on second noun instead;
	if noun is notepad:
		say "Goodness, no, you might leave it there." instead;
	if second noun is bubble or second noun is bucket or second noun is shell:
		say "But it's already been sorted. No need to muck it up." instead;

check examining (this is the ignore-container-examine rule):
	if noun is sack or noun is cask:
		ignore the examine containers rule;
	if noun is a container in sorting area:
		ignore the examine containers rule.

Rule for printing room description details:
	if location of player is sorting area:
		omit contents in listing;
		stop.

part lll

chapter lalaland

[this is for items that have been consumed, but we don't want to recycle them. It's easier to check if they're off-stage or in Lalaland than to define a boolean. Well, for me, anyway.]

Lalaland is a room. "[bug-report] I have no idea how you got here. Definitely, let me know, so it doesn't happen to anyone else. This room should be inaccessible. Anything that appears here should be pretty much dealt with. Perhaps I could've used a boolean called dealt-with, but I didn't."

instead of doing something in lalaland:
	say "You probably need to undo things." instead;

book error checking

rewarn-val is a number that varies. rewarn-val is usually 0.
rewarn-next is a number that varies. rewarn-val is usually 2.

to say warn-inc:
	increment rewarn-val;
	if rewarn-val is rewarn-next:
		if rewarn-next < 7:
			increment rewarn-next;
		now rewarn-val is 0;
		say ". You may want to use or examine the tagged gadget or maybe ANGLE to see what you can tinker with";

part understanding mistakes

section intro

understand "magnet" and "a magnet" as a mistake ("[blurby][line break]") when gateman is visible or nametag is visible.

understand "believe" as a mistake ("Yes, but what do you believe you can do?") when toga is visible.

understand "gato" as a mistake ("Nice try, but wrong language[if toga is visible]. Another animal would be more likely to eat through the thorns[otherwise]. Plus, the goat's been through enough[end if].") when player is in dy.

to say blurby:
	if nametag is visible and location of player is notices section:
		say "A magnet is two words--but 'a' doesn't fully count. It wouldn't be strong enough to pull the gate--you need someone to help you, not something, as the broad board says.[run paragraph on]";
		continue the action;
	if nametag is visible and location of player is dy:
		say "A magnet [if goat is visible]wouldn't have helped you[otherwise]won't help you[end if] get past the thorns[if goat is not visible]. Right idea, though. Wrong item[end if].[run paragraph on]";
		continue the action;
	say "[reject][run paragraph on]";

understand "no entry" as a mistake ("[if player is in subsite]Hm, interesting. You wonder if anything is REALLY stopping you from entering.[otherwise]You solved that puzzle already. Onward![end if]")

understand "get a man" as a mistake ("[get-a-man][run paragraph on]")

section forest

understand "steam" as a mistake ("Very good try. But you need to change the meats into something else entirely. Something you can handle and use. Plus, too cold.") when player is in storing area.

understand "padres" as a mistake ("Bringing holy men down here would probably kill them.") when drapes are visible or spread is visible.

understand "rasped" as a mistake ("The [d-s] makes a raspy noise, but you probably need to think up something more concrete. Not, like, cement concrete.") when drapes are visible or spread is visible.

understand "spader" as a mistake ("James Spader is pretty cool, but he can't help with supernatural stuff. Okay, maybe he could, but I couldn't afford the appearance fee.") when drapes are visible or spread is visible.

understand "parsed" as a mistake ("You can just read the [if drapes are visible]drapes[otherwise]spread[end if] instead. The messages here aren't cryptic enough to be parsed.") when drapes are visible or spread is visible.

understand "amps" as a mistake ("You are trying to save a magic world, not start a garage band.") when spam is visible or maps are visible.

section metros

understand "sap" as a mistake ("The PSA is sappy, but it's not literally sappy. And sticky stuff won't get you up the wall.") when art is visible.

understand "asp" as a mistake ("That's just what you need, a snake chasing you with everything else going on.") when art is visible.

understand "spa" as a mistake ("You haven't earned a break yet.") when art is visible.

understand "smitten" as a mistake ("No matter how hard you try to convince yourself they[are-were] useful, you're not smitten with them. They're just ugly. But they could stick really good on the right surface.") when player has velcro.

understand "stalker" as a mistake ("Hm. If the talkers became a stalker, the only person to stalk would be you. You can't change who they are inside, but maybe you can give them a fright.") when talkers are visible.

understand "snores" as a mistake ("Any noise would be better than the beats, but without the sensor, you're probably not getting by the door.") when sensor is visible.

understand "senors" as a mistake ("The senors you [want-wanted] are behind the door.") when sensor is visible.

understand "baste" as a mistake ("I'm sorry, Mario (or Moira,) but the kitchen is in another castle. I mean, store.") when beats are visible or beast is visible.

understand "tar" as a mistake ("That'd be neat, make the wall sticky so you could climb and deal with that [if neon pig is visible]stupid pig[otherwise]opening you made[end if]...but you might get stuck to it, too. Need something clingy but less gooey.") when player is in ett.

understand "rat" as a mistake ("No, you don't need a rat chasing you around.") when player is in ett.

understand "lair" and "lairs" as a mistake ("That might work, if the rail did not lead to a lair already.") when tracks are visible.

understand "liar" and "liars" as a mistake ("The number of liars the rails would create would be shocking.") when tracks are visible.

section resort

understand "prigs" as a mistake ("Grumbling about prigs, or showing those prigs, won't help now. You should probably [if grips are visible]use the grips to climb[otherwise]make the sprig into something useful[end if] instead.") when grips are visible or sprig is visible.

understand "prose" as a mistake ("There's not much of that left in this game, and talking won't help you up the [p-s][if ropes are visible]. The ropes will[otherwise]. The spore can be something sturdier[end if].") when player is in olw.

understand "pores" as a mistake("[if ropes are visible]No, you had it[otherwise]Almost[end if].") when player is in olw.

section sortie

understand "pray" as a mistake ("[if player is in sacred cedars]It's been so long, you only mouth a 'Yarp!'[otherwise]Things aren't THAT bad.[end if]")

understand "stake" as a mistake ("A stake would have no use here. So that might be a miskate. Err...") when steak is visible or skate is visible.

understand "takes" as a mistake ("You do a take, then a double take, at how odd this all is.") when steak is visible or skate is visible.

understand "teaks" as a mistake ("Hm. An indoor forest. [if cedars is unvisited]That would be neat to see. But this is a kitchen[otherwise]But you've already seen one[end if].") when steak is visible or skate is visible.

understand "pears" as a mistake ("I think you have the wrong food group. The spearman is cheesy. Plus, he wouldn't be a spearman without his spear.") when spear is visible.

understand "limes" as a mistake ("I think you have the wrong food group. Besides, it wouldn't be nice to wipe the spearman's smile off.") when spear is visible.

understand "thicken" as a mistake ("You might rather thin-nen to slip through the bars, but you can't.") when player is in nick.

understand "black rood" and "rood" and "blackrood" as a mistake ("Hm, no, keep religion out of it[if black door is visible]--you have what you need[end if].") when roadblock is visible or black door is visible.

understand "rood" as a mistake ("[if rut-door is visible]You already changed the odor[otherwise]This game is religion-neutral[end if].") when player is in rf.

understand "odor" and "rood" as a mistake ("That door's reinforced. [if spis is visited]The nerds[otherwise]Whoever's behind there[end if] made sure it was protected against all sorts of crazy things.") when player is in underside.

understand "odor" and "rood" as a mistake ("It would stink to be you if you did that. The nerds outnumber you and would crucify you.") when player is in spis.

understand "lacy" as a mistake ("No, the clay is a clue to what the trio really wants to do.") when trio is visible.

understand "tap" and "tap [text]" as a mistake ("'Ah! You are TAPping your feet to my anapest beat!'[paragraph break](You may want to PUSH or ATTACK something instead.)") when woeful pat is visible.

understand "similes" as a mistake ("[if smilies are visible]Adding similes would mean the limerick wouldn't scan. They'd be more forced than the smilies. Though forced smilies can be a powerful weapon. Hmm.[else if missile is visible]You can't create something abstract from something concrete. Or plutonium. Or whatever that missile's made of.[otherwise][reject]")

understand "sever" as a mistake ("Violence or vandalism in such a holy place? For shame.") when player is in cedars.

understand "serve" as a mistake ("You must serve by going forth on your mission of warlikeness and love.") when player is in cedars.

understand "mope" as a mistake ("[if woeful pat is visible]Woeful Pat can't decide if you're mocking him or trying to outdo him. But he is sure you're not listening carefully enough to his poem, so he coughs briefly to get your attention.[else if smilies are visible]You try, but the smilies on the poem prevent you.[else if poem is visible]You mope, but it doesn't change the poem. Or even give you an idea for one.[else]You mope, but you imagine no poem.[end if]");

section other stuff

understand "try" as a mistake ("That's more sensible than the five alternatives, but it's not concrete enough.") when player is not in subsite

understand "try" as a mistake ("You're sure the lecture to the east will be about trying versus doing. Ugh[if above-sign is examined]. Maybe the sign means something different[end if].") when player is in subsite.

part parser errors

Rule for printing a parser error when the latest parser error is the not a verb I recognise error:
	say "[reject]";

Rule for printing a parser error when the latest parser error is the nothing to do error:
	let A be indexed text;
	now A is "[the player's command]";
	replace the regular expression " .*" in A with "";
	say "There's nothing useful or small enough for you to [A] here.";
	the rule succeeds;

long-word-warned is a truth state that varies. long-word-warned is false.

Rule for printing a parser error when the latest parser error is didn't understand error:
	[let hop be hash of the player's command;
	d "[hop] = [the player's command].";]
	if deadbeat is visible:
		if the player's command includes "hipster":
			say "He trips (tough, when you're sitting down) at being called a hipster. Whether that makes him more likely to be one, I don't know.";
	say "[reject]";
	reject the player's command;

[Rule for printing a parser error when the latest parser error is the noun did not make sense in that context error:
	if the player's command matches the regular expression "\bpassage\b", case insensitively:
		if location of player is subsite:
			say "(Going down the passage)";
			try entering passage instead;
	if the player's command matches the regular expression "\bgato\b", case insensitively:
		if location of player is dy:
			say "The shrubbery cat is not moved. Perhaps you need to meddle with something else.";
			try entering passage instead;
	if the player's command matches the regular expression "\bbugle\b", case insensitively and bugle is visible:
		if player has bugle:
			say "Looks like you may be trying to play the bugle, but you have the wrong verb. Would you like to PLAY it?";
			if the player consents:
				try playing the bugle instead;
			otherwise:
				say "Ok.";
				reject the player's command;
	if long-word-warned is false:
		if the player's command matches the regular expression "^<a-z>{9}", case insensitively:
			now long-word-warned is true;
			say "You won't need to use a word that long in this game. They should all be eight letters or less, except for [if black door is off-stage]one compound word[otherwise]a compound word you already found[end if].";
			reject the player's command;
	say "That's not something you can say, do or see here[if player has the tagged gadget or tagged gadget is visible][warn-inc][end if].";]

to say youlose:
	end the story;

check going inside in notices section:
	if getaway is visible:
		try entering getaway instead;
	otherwise:
		try entering gateway instead;

check going nowhere in Notices Section:
	if getaway is visible:
		say "That'd be a cop-out, running away. You'd break Old Man Almond's heart, too. You probably want to go in and enter the gateway.";
	otherwise:
		say "You can't even remember the way you came through the thorns. And the gateway's right in front of you. You sense it's probably where you want to go.";
	the rule succeeds;

check looking in dy for the first time:
	change north exit of dy to nowhere;
	
check going nowhere in dy:
	say "[one of]You hit a snag, and the [if toga is visible]toga[else if thorn is visible]thorn[otherwise]passage[end if] nags you. Or seems to[or]You see a snipe among some pines and lose your spine[or]You're feeling negative to vegetation, so you can't see a way through[or]A stick crawling with ticks gives you pause[or]You'd run into snarled darnels[or]I won't let snag-tangles get at you that way[or]That way's some of the thickest thickets ever[cycling]." instead;

check going nowhere:
	say "This is a generic message to say you can't go that way. It should be changed to add variety and flavor.";
	the rule succeeds;
	
book parsing

Rule for printing a parser error when the latest parser error is the I beg your pardon error:
	say "[one of]Be daring, you pro![or]Broaden your grip.[or]Go yon, bud! Repair![or]Go, do pure brainy![or]Peg your brain. Do![or]Probing, you read...[or]'No prayer, bud,' I go.[or]No drab gripe, you![or]You're poring. Bad.[or]Go, bud. Reap irony![or]Be young or rapid![or]Yip on, drab rogue![or]Go yon, rapid rube![or]Yep, I guard no orb.[or]Yup, I err. Dang. Boo.[or]Broody gape? Ruin![at random]" instead;

curfloor is a number that varies.
curtower is a number that varies.

to decide which indexed text is alf of (A - indexed text):
	let B be indexed text;
	let C be indexed text;
	let L be a list of indexed text;
	let Q be the number of characters in A;
	repeat with X running from 1 to Q:
		let B be character number X in A;
		add B to L;
		sort L;
	repeat with X running from 1 to Q:
		now C is "[C][entry X in L]";
	decide on "[C]";

part numtxt

say-check is a truth state that varies. say-check is false.

the block answering rule is not listed in any rulebook.

does the player mean answering the player that:
	it is very likely;
	
instead of answering:
	say "You'll want to ASK about a SUBJECT, instead.";

the block asking rule is not listed in any rulebook.

period-warned is a truth state that varies. period-warned is usually false.

ignore-transcript-nag is a truth state that varies.

after reading a command:
	if the player's command matches the regular expression "^\p" or the player's command matches the regular expression "^\*":
		if currently transcripting:
			say "Noted.";
		otherwise:
			if ignore-transcript-nag is false:
				say "You've made a comment-style command, but Transcript is off. Type TRANSCRIPT to turn it on, if you wish to make notes. Or if you want to eliminate this nag, for instance if you have an interpreter that does so independently, say yes now.";
				if the player consents:
					now ignore-transcript-nag is true;
		if helpdebugflag is true and hintfull is false:
			now just-print is false;
			all-say "[the player's command]";
			now just-print is true;
		reject the player's command;
[	if the player's command matches the regular expression "\..*<\w\d>":
		if period-warned is false:
			say "I noticed you used a period to separate commands. Unfortunately, the parser in Inform seems to have an extremely obscure bug (#1011 at inform7.com) that interacts with the solution to this game if you separate commands with periods. It was noticed near the release deadline, and it manifests itself when you refer to an in-game item or person or thing as one in a series of commands, usually three or more.[paragraph break]I do this not to call out the people who've worked so hard on making Inform 7 robust and accessible but to warn you that there are certain commands you may wish to type by themselves, because this game taxes the parser[if two-of-three is true]. I accept if you choose to downgrade this game as a judge because of this, or because I may be wrong that the bug is on Inform's end[end if].[paragraph break]This warning won't appear again. Hit any key to continue.";
			now period-warned is true;
			wait for any key;]
	remove stray punctuation;
	let XX be indexed text;
	let XX be the player's command in lower case;
	change the text of the player's command to XX;
	if the cow liver is visible or the chicken liver is visible or livers are visible or spam is visible:
		if the player's command matches the regular expression "\bmeats\b", case insensitively:
			say "What kind of meats they are is important. As are their physical characteristics. So you'll need to be specific.";
			reject the player's command;
[	if trips strip is unvisited:
		if the player's command matches "\b?i<mfr>\b":
			say "[reject]";
			reject the player's command;]
	if the player's command matches the regular expression "^say ", case insensitively:
		if the player's command includes " to ":
			do nothing;
		otherwise:
			say "(I[one of]n the future, i[or][stopping]f you want to SAY something, you can just say the single word. To talk to someone, ASK someone ABOUT something.)";
			replace the regular expression "^say " in XX with "";
			change the text of the player's command to XX;
	if toga is visible:
		if the player's command matches the regular expression "\bgoat\b", case insensitively:
			if the player's command matches the regular expression "^goat", case insensitively:
				do nothing;
			otherwise:
				say "(In the future, you can just say what you need to summon. No verbs or frills.)";
				try propfliptoing goat;
				follow the notify score changes rule instead;
	if panel is visible and panel is part of silo:
		if the player's command matches the regular expression "buttons":
			say "One button at a time, please." instead;
	if goat is visible and toga is not visible:
		if the player's command matches the regular expression "\bgoat\b":
			say "[one of]It doesn't seem to realize it did anything for you. It's probably still confused. Probably irritable from eating all those big gigantic thorns. Don't fluster the restful[or]Don't fluster the restful[stopping].";
			reject the player's command;
	if player is in elf row's flowers:
		if the player's command matches the text "fairies":
			say "[one of]Something you did or thought causes the freesia faeries to buzz...I guess they're magical enough to detect your misspellings. You feel a tingling, but it passes[or]They can see in your eyes you're spelling things wrong, but they'll have to deal.[or][run paragraph on][stopping]";
		if player's command matches the regular expression "\bshoot\b", case insensitively and hoots button is visible:	[this is sadly necessary for my current coding skills, since SHOOT is a proper verb, too]
			try propfliptoing shoot button;
			follow the notify score changes rule instead;
	repeat through table of anagrams:	[this code may slow things down]
		if the-from entry is visible:
			if there is an exact-text entry and the player's command matches exact-text entry:
				try propfliptoing the-to entry;
				follow the notify score changes rule instead;
			if there is a text-back entry and the player's command matches text-back entry:
				try propfliptoing the-from entry instead;
	repeat through table of anagrams:
		if the-to entry is visible:
			if there is a text-back entry and the player's command matches text-back entry:
				say "You can't flip things back." instead;
			if there is an exact-text entry and the player's command matches exact-text entry:
				try propfliptoing the-to entry instead;
	continue the action;

definition: a thing (called x) is takeable:
	if x is not visible, decide no;
	if x is part of something, decide no;
	if x is the blue lube, decide no;
	if x is carried, decide no;
	if x is worn, decide no;
	if x is the player, decide no;
	if x is fixed in place, decide no;
	decide yes;

chapter Notices Section
	
Notices Section is a room. "[one of]You're not sure what to do here, and there's no welcoming committee. But there's a broad board with various scrawlings[or]The broad board has some scrawlings you [if broad board is unexamined]haven't read yet[otherwise]figure will be useful[end if][stopping]."

after looking in notices section for the first time:
	choose row with short of "Thor" in table of pad-stuff;
	if known entry is true:
		now known entry is false;
		say "(You get rid of the Thor entry in your notepad. You figured what he was there for.)"

instead of taking gateman:
	say "'Gateman, get a man, very clever. But thing is, you'll be working with things, not ideas or actions."

to say get-a-man:
	if notices section is unvisited:
		say "[reject]";
		continue the action;
	if trips strip is visited:
		say "Old Man Almond cannot help you now.";
		continue the action;
	if gateman is visible:
		say "You already did.";
		continue the action;
	say "Very good. Very close. But here's a meta-nag: what kind of man? What would he work with?";
	continue the action;

instead of going outside in notices section:
	try going north instead;
	
section broad board

the scrawlings are part of the broad board. the scrawlings are undesc.

instead of examining the scrawlings:
	try examining the broad board;

the broad board is scenery in notices section."[one of]Welcome to YORPWALD![paragraph break]WANTED:[paragraph break]selfless fellow (sorry, ladies--dictionary's fault, not ours) willing to turn self into humiliating identification to aid adventurers in rescuing our fair land. Must have strength or magic touch to raise big metal gate. Benefits include unlimited complaining to rescuer/prospective hero. Obligations include explaining situation to rescuer/prospective hero. Risks include hero getting stuck on requisite four- and three-letter word combination to release you.[paragraph break]--fellow named Terry to allow quick way for temporarily stuck adventurers to RETRY and return to Trips Strip.[paragraph break]You hear a big VOOP. The writing on the board is replaced by even more writing, but not before you write some basic stuff down in your notepad.[board-note][or]--quester request: text adventurer with savoir-savior to save shuffled Yorpwald from being fflushed. High chance of real fun, flaneur, low chance of funeral. No crudities like diuretics needed![line break]  Canny Nancy or Brian Brain preferred to Manly Lyman or Army Mary. Skill kills![line break]  Scrabble, MasterMind, and/or Boggle expertise a plus. Simple on-the-game training provi... oh! Hi there! You got to the gates stage. Two combos of a 4- and 3-letter word here. Futz with the dollhouse if you want. That's a six-letterer[if nametag is visible]. Nametag: it grants starting. Gateway helps you --- ---- (Not [italic type]that[roman type] bad. Less to switch than the goat.)[end if][if two-of-three is true].[paragraph break]   [bold type]!!!!LIMITED TIME TEXT ADVENTURING SEASON OFFER FOR THE BUSY TEXT ADVENTURER: ONLY SOLVE 2 OF 3 AFFLICTED AREAS TO EARN ACCESS TO YOUR ROMAN MANOR!!!![roman type][otherwise].[end if][stopping]"

to say board-note:
	pad-rec-q "board";
	pad-rec "retry"

understand "billboard" as broad board.

section dollhouse

a dollhouse is a thing in notices section. "You see [if static is visible]an incomplete-looking dollhouse with a crackling noise coming from a few loud holes[otherwise]that dollhouse you put those attics on[end if]. It is labeled FOR INSTRUCTIONAL PURPOSES ONLY."

instead of entering dollhouse:
	say "It's way too small. As the house itself says, it's for instructional purposes[if static is flipped-yet], and you've sort of figured things already[otherwise], but you don't have to tinker with the static[end if]."

understand "house" as dollhouse.

instead of taking or attacking the dollhouse:
	say "Some hero you'd be, [if the current action is taking]stealing[otherwise]vandalizing[end if] a dollhouse."

description of the dollhouse is "[if attics are visible]It's much nicer and quieter now that you tacked the attics on[otherwise]You hear static hissing through some loud holes. Plus, the dollhouse isn't nearly as tall as it could be. It could use an extra floor or two[end if]."

the static is scenery in notices section. gpos of static is 3. rpos of static is 1. legth of static is 6. rgtext of static is "[rc][gc][rc][rc][rc][rc]".

description of static is "It's not especially painful to listen to, but you probably don't have to."

the attics are plural-named scenery. the attics are flippable and flipped-yet. gpos of attics is 6. rpos of attics is 5. legth of attics is 6. rgtext of attics is "[rc][gc][rc][rc][rc][rc]".

understand "attic" as attics when attics are visible or static is visible.

check taking the attics:
	say "You could flip them back to the static if you want, but it'd be mean to steal what you just built. And useless." instead;
	
description of the attics is "They fit perfectly on what was once the top of the house. Maybe there's no way a doll could get there, but the house looks nicer now."

section acne-bit cabinet

the acne-bit cabinet is an open openable transparent container in notices section. It is fixed in place. "[one of]It can't be... can it be...? A cabinet floating in mid-air. It looks acne-bit. It's open, too[or]The acne-bit cabinet is still floating here. It contains [the list of things in cabinet][stopping]."

rpos of cabinet is 7. gpos of cabinet is 5.

the nice bat is a thing. "A nice bat is here, standing on its rear claws."

description of nice bat is "The items it took from you are somewhere, you're sure. It's a big bat, six feet tall, with wings that probably open like a streetside watch seller's trenchcoat."

the bw are privately-named plural-named things. printed name of bw is "bat wings". understand "bat wings" and "wings" as bw.

description of bw is "They're not extended right now."

instead of doing something with the nice bat:
	if the current action is attacking:
		say "That'd be a spectacular way to fail." instead;
	if the current action is propfliptoing:
		if number of solved regions < 2:
			say "The cabinet hasn't earned it, yet." instead;
		continue the action;
	say "The bat does some bizarre cheerleading movement as you turn to look at it. You probably don't want to get too close.";

instead of opening acne-bit cabinet:
	say "It already is.";

instead of taking acne-bit cabinet:
	say "Whatever keeps it floating in place also keeps you from moving it."
	
description of acne-bit cabinet is "It's open. The cratered bits, once retraced, spell CAN BITE. You see [a list of things in cabinet] inside.";

check examining cabinet:
	if location of player is trips strip:
		say "You've no idea where the cabinet puts all its items. Maybe you didn't read the right fantasy books. But--they're there, helping keep your inventory free. A small favor." instead;

description of cratered bits is "They read CAN BITE[if doorslammed is true], as you found out when trying to take both those devices[end if]."

check examining cabinet:
	ignore the examine containers rule;

instead of closing the acne-bit cabinet:
	say "You don't need to hide anything in there from anyone.";

section salient saltine

the salient saltine is a warpable thing in the acne-bit cabinet.

report taking salient saltine for the first time:
	say "Got it. You'd have liked a beefier freebie, but with calm opining, you see no use complaining.";
	the rule succeeds;

description of salient saltine is "It says SLAINTE on the package."

Undo flag is a number variable. Undo flag is 0.

rule for deciding whether to allow undo:
	if helpdebugflag is true:
		append "UNDO: (no hint)[line break]" to the file of shufhints;
	if undo is prevented:
		if undo-code is 1:
			say "Un-eating the saltine entails a huge intent--to win it NOW.[paragraph break]";
			allow undo;
		else if undo-code is 2:
			say "You won't be arrested for such retreads, but I will provide a warning against your reusing sin urge.";
			allow undo;
		else if undo-code is 3:
			say "Very sensible indeed.";
			allow undo;
		otherwise:
			say "[bug-report] I've disabled undo on this command. for some reason I can't figure.";
			deny undo;
			
section tinfoil info-lit

the tinfoil info-lit is a thing in cabinet. understand "tin/info/lit/paper" as tinfoil info-lit. the tinfoil info-lit is warpable.

description of tinfoil info-lit is "[one of]It's tinfoil only in writing content--it's actually just paper. It's excessively all-caps and underlined, but I'll spare you that.[paragraph break][or][stopping][line break]--If you can change stuff, why can't the people who made the gadget? How do they know you can? Or why couldn't they find someone off the street? Perhaps they're the real enemy and you're the pawn.[line break]--If they really wanted you to clean things up, why can't you take both the cabinet items?[line break]--That NEW DOOR is not really ONE WORD. Same as the guy who can lift it! They are two short words, easier than one big scary seven-letter word.[line break]--Anyone smart enough to untangle this world won't get any fulfillment from doing so. So there, stupider disputer![paragraph break]--Sane Sean, Farce Facer, Saltier Realist[paragraph break]It'd all be distressingly persuasive gotcha-logic if you hadn't been able to flip that toga. But this is a fantasy world, [if getaway is visible or gateman is visible]and you've already shown cracks in his logic[otherwise]so there's a chance he's wrong[end if]."

section tagged gadget

a tagged gadget is in acne-bit cabinet. the tagged gadget is warpable. the tagged gadget can be broken. the tagged gadget is not broken.

rule for printing a locale paragraph about the tagged gadget:
	if location of player is hotspot:
		say "[if red bull burdell is visible]Your gadget lies here, but no way you get to it[otherwise]Your gadget, kicked from your hands by Red Bull Burdell, is here for the taking[end if].";
		now tagged gadget is mentioned;
	
does the player mean switching on the gadget: it is likely.

the pit is part of the tagged gadget. pit is undesc.

instead of doing something other than examining the pit:
	say "That's fiddling a bit."
	
instead of examining pit:
	try examining the tip;

the tip is part of the tagged gadget.

understand "side" as tip.

description of tip is "The tip points to [c-r][if gadget-secured is true] and is stuck there[end if].";

the handle is part of the tagged gadget. description of handle is "Just wide enough so the tag won't fall off, and just long enough you can grab it without blocking the screen."

instead of doing something other than examining the handle:
	if the current action is attacking:
		say "You couldn't hold the gadget, then." instead;
	say "The handle is about what it seems to be. Don't worry about it too much."

sg-warn is a truth state that varies.

instead of switching on gadget:
	if sg-warn is false:
		say "Fourth wall note: SG may save you keystrokes in the future. You can also C/CER/CERTIFY (object) or R/REC/RECTIFY (object) to skip this altogether.";
		now sg-warn is true;
	try sging instead;

understand "device" as tagged gadget.

the tagged gadget can be cert or rect. the tagged gadget is rect.

the cratered bits are part of the cabinet. understand "acne" and "bit" as cratered bits.

scanned-g is a truth state that varies.

instead of dropping the gadget:
	say "It seems far too useful[if scanned-g is false], if indeed it works[otherwise], especially since it actually works[end if]."

check examining tagged gadget (this is the did I break gadget rule):
	if gadget is broken:
		say "It's lifeless without the tag. Nice work!" instead;

description of tagged gadget is "It looks like a wall stud sensor, six inches by twelve, with a screen broken into nine mini-panels. An instructional tag [if instructional tag is examined]you read[otherwise]you can read[end if] is tied to the handle, which narrows where it meets the main part. The gadget crackles with greeny energy. A tip on one of its edges points to [c-r][if gadget-secured is true] and is stuck there[end if]."

the instructional tag is part of the tagged gadget.

does the player mean taking off the magenta nametag when the player is not wearing the magenta nametag: it is very unlikely.

the gadget-screen is undesc. it is part of the tagged gadget. the gadget-screen is privately-named. the printed name of the gadget-screen is "gadget's screen". Understand "gadget screen" and "screen" as the gadget-screen.

the s-r is part of the tagged gadget. the s-r is privately-named.

printed name of s-r is "SECURE button". understand "secure button" and "secure" and "button" as s-r.

description of s-r is "[if gadget-secured is true]It's been pushed down with no way to get it back up[else if gadget-unsecurable is true]It's sticking out slightly, but you can't push it in[otherwise]It's sticking out slightly from the bottom of the gadget[end if]."

gadget-secured is a truth state that varies.

[!!! take all: make sure gadget is last]

check pushing s-r:
	if gadget-secured is true:
		say "You already secured the gadget." instead;
	if gadget-unsecurable is true:
		say "You can't any more. Entering a store seems to have broken it." instead;
	if trips strip is unvisited:
		if gateman is visible:
			say "Old Man Almond coughs. 'Wait [']til you get through the gate to do that. Just a little kibitzing.'" instead;
		otherwise:
			say "You've no clue what it's for, and it seems important and possibly irreversible, unlike the CERTIFY/RECTIFY bit. You decide to wait until you find someone who can decipher it." instead;
	[choose row in table of pad- : why don't 2 tables work ?!]
	say "You lock the gadget into [c-r].";
	now gadget-secured is true instead;

to say c-r:
	say "[if gadget is cert]CERTIFY[otherwise]RECTIFY[end if]";

the blurb is part of the tagged gadget. description of blurb is "Ask a qualified person if the [s-r] button is right for you."

instead of examining gadget-screen:
	try examining gadget;

procedural rule while eating something: ignore the carrying requirements rule.

procedural rule while taking off instructional tag: ignore the carrying requirements rule.

the knot is part of the tagged gadget.

description of the knot is "It's simple but tight, with no way to start to loosen it."

instead of taking the knot:
	try untieing the knot instead;

instead of untieing the knot:
	say "There's no way to do much with the knot. It's simple and tight."
	
instead of untieing tag:
	try untieing knot instead;

instead of untieing gadget:
	try untieing knot instead;

check taking instructional tag:
	ignore the can't take component parts rule;
	
description of instructional tag is "'This gadget is certified untransmogrifiable. It can be used to SCAN objects to determine their transmogrification states. Removing tag invalidates warranty. One beep indicates changeabilty. Two mean a compound word.'[paragraph break]Above a small print disclaimer you see two helpful examples.[paragraph break]CERTIFY: [lemon-melon].[paragraph break]RECTIFY: [paste-tapes].[prg]"

to say lemon-melon:
	say "[first custom style]L[roman type][second custom style]E[roman type][first custom style]M[roman type][second custom style]O[roman type][second custom style]N[roman type] <-> [first custom style]M[roman type][second custom style]E[roman type][first custom style]L[roman type][second custom style]O[roman type][second custom style]N[roman type][one of]. A little thought and you realize only two reds on the gadget makes things trivial[or][stopping]"

to say paste-tapes:
	say "[bluetext]PA[roman type][first custom style]S[roman type][second custom style]T[roman type][bluetext]E[roman type] <-> [bluetext]TA[roman type][second custom style]P[roman type][first custom style]E[roman type][bluetext]S[roman type][one of]. The red S/green T in PASTE and green P/red E in TAPES are, you realize, the first and last letters in the other word[or][stopping]";

to say prg:
	pad-rec-q "certify";
	pad-rec "rectify";

the example is part of the instructional tag. understand "text" as example.

description of example is "LEMON <- RGRGG -> MELON. Which seems to indicate -E-ON is the way to go, then there's only one other possibility."

the small print disclaimer is part of the instructional tag.

description of small print disclaimer is "'Adventurers using the gadget frequently are no less certifiably heroic and no more certifiably practical than those who don't. Manufacturer not responsible for guilt and lack of intellectual self-satisfaction over using gadget too frequently or frustration from using gadget too rarely.'"

check taking off the instructional tag:
	try silently taking the instructional tag;

instead of taking the instructional tag:
	say "If it were a mattress tag, you'd laugh and pull it off, but everything's gotten a bit weird. You don't know about the warranty on it, but the tag seems pretty important. It's made of some weird plastic you can't just rip[one of].[paragraph break]Plus, embarrassingly, you've no clue how to undo the knot tying the tag to the handle--and you're not strong enough to pull the tag off[or][stopping].";
	if player does not have the gadget:
		say "[line break]By the way, it seems you don't even have the gadget. Did you mean to take the tagged gadget instead?";
		if the player consents:
			say "Ok, taking the gadget.";
			try taking the gadget instead;
		otherwise:
			say "Ok.";

check pulling instructional tag:
	try taking instructional tag instead;
	
does the player mean examining the gadget-screen when gadget is held: it is likely.

does the player mean examining the gadget: it is very likely.

[description of colored writing is "'Provably 79.12% hintier than a tagged gadget. [bold type]Green print[roman type] gives you the start, [italic type]red print[roman type] the end. The middle bits fall into place.'[paragraph break]The example reads PA[italic type]S[bold type]T[roman type]E <-> TA[bold type]P[roman type][italic type]E[roman type]S. (red S, green T, green P, red E. Everything else blue.)"]

section blue lube and phail phial

[a hack here is to put this below the gadget so the "Ow" appears last]

the PHAIL phial is a thing in acne-bit cabinet. the phial is warpable.

description of PHAIL phial is "It feels tingly and looks glinty. It contains some blue lube that sloshes about a bit[if phial is not in cabinet]. It also reads, in small print, PROACTIVELY HELPING ADVENTURERS IN LIMBO. Geez, acronyms can be so contrived[end if]."

the blue lube is part of the phail phial.

description of blue lube is "[if phial is in cabinet]You have to blink a couple times after looking at it, even from outside the cabinet. It has some weird powers, but you'd probably need to ask how to harness them[else if lube-asked is false]You sense it's powerful, but you need to look at it right[otherwise]You remember what Old Man Almond said. You can either ANGLE or GLEAN as you look at it, which may help your quest[end if]."

[Rule for deciding whether all includes blue lube while taking (this is the zap lube rule): rule fails.]

cabinet-bit-me is a truth state that varies.

check taking phial:
	if cabinet is visible:
		if lube-asked is false:
			now cabinet-bit-me is true;
			say "Ow! The cabinet bites you as you try to take the phial. It must really value that phial." instead;

instead of opening the phail phial:
	say "The blue lube would leak out."
	
instead of taking the blue lube:
	say "[if lube-asked is false]It probably belongs in the PHAIL phial, until you determine otherwise[otherwise]You wouldn't be able to GLEAN or ANGLE then[end if]."

instead of doing something other than examining the blue lube:
	if the current action is objasking generically:
		continue the action;
	if the current action is objasking about:
		continue the action;
	say "The lube's really only good for staring at."

last-lube-turn is a number that varies.
	
check examining blue lube:
	if lube is not in cabinet:
		if turn count < 5 + last-lube-turn:
			say "It's too soon since the last time you looked into the lube. Wait [5 + last-lube-turn - turn count] turn[if 5 + last-lube-turn - turn count > 1]s[end if], then try again.";
			decrement the turn count;
		now last-lube-turn is turn count;

report taking phail phial:
	if player has phail phial:
		say "The cabinet's doors seem to waver a bit[if cabinet-bit-me is true], but this time it doesn't nick you[otherwise], but they don't slam on you[end if]." instead;
	continue the action;

section secureing

secureing is an action applying to nothing.

understand the command "secure" as something new.

understand "secure" as secureing.

carry out secureing:
	if player has gadget:
		try pushing s-r instead;
	otherwise:
		say "[reject]" instead;

chapter recuseing

recuseing is an action applying to nothing.

understand the command "recuse" as something new.

understand "recuse" as recuseing.

carry out recuseing:
	if player does not have gadget:
		say "[reject]" instead;
	if player is not in trips strip:
		say "You can't recuse unless you are in the trips strip." instead;
	if gadget-secured is false and gadget-unsecurable is true:
		say "Not securing in your first trip means you can't use the gadget to recuse." instead;
	if number of solved regions is 1:
		say "You haven't even gotten beyond the strip yet." instead;
	if number of solved regions < 3:
		say "You can't recuse yet." instead;
	if forest is unsolved:
		say "Suddenly, you know what to do. You point the gadget at [if store f is visible]store F[otherwise]the forest you didn't get through[end if]. As if in a blur, you see werewolves in ice being shot, and you hear distant applause.";
		now forest is bypassed instead;
	if sortie is unsolved:
		say "Suddenly, you know what to do. You point the gadget at [if store f is visible]store I[otherwise]the sortie you didn't get through[end if]. As if in a blur, you see a missile launched, and it bathes a grey castle in bubbles and rainbows and other obnoxiously smiley stuff.";
		now sortie is bypassed instead;
	if metros is unsolved:
		say "Suddenly, you know what to do. You point the gadget at [if store f is visible]store I[otherwise]the sortie you didn't get through[end if]. As if in a blur, you see a missile launched, and it bathes a grey castle in bubbles and rainbows and other obnoxiously smiley stuff.";
		now metros is bypassed instead;
	say "You don't seem to have any regions to recuse yourself from. From which to recuse yourself. And you don't get to skip [if store r is visible]store R[otherwise]the resort[end if]." instead;
	the rule succeeds;

chapter intro-marcos

to intro-marcos:
	say "A shady figure sidles up to you. 'Hey! Pal! The name's Marcos. Sellin['] macros. Ways to lump actions together. Streamline your thinking. Actually, not selling them at all, just hoping you remember me once you fix things here and are an acclaimed hero with clout.'[paragraph break]He takes the gadget away from you, fiddles with it, and gives it back. You are impressed with his dexterity, and yet, you feel it would not be too tough to imitate him.[paragraph break]'Just, if you think of one scan as R or C--you can make both of them with RC. Or CR. I won't judge.'[paragraph break]As you note this in your notepad, he pops out of view.";
	pad-rec "macros";
	now marcos-trumped is true;

chapter gleaning

gleaning is an action out of world.

understand the command "nk/kr/hrk/glean" as something new.

understand "glean" as gleaning.
understand "hrk" as gleaning.
understand "nk" as gleaning.
understand "kr" as gleaning.

carry out gleaning:
	if phial is in cabinet and lube-asked is true:
		try taking phail phial;
		if player does not have phail phial:
			all-say "[bug-report] Please let me know how this happened." instead;
	if blue lube is not visible:
		all-say "[reject]" instead;
	all-say "You stare into the blue lube for a bit. You see a sort of story unfold.";
	if location of player is notices section:
		if nametag is visible or gateway is visible:
			all-say "You see yourself looking down at your nametag, mouthing a word and pointing to the gate as an old man appears. Then you see yourself tinkering with the tagged gadget on both settings, snapping your fingers and groaning, and walking through the gate." instead;
		all-say "You see Old Man Almond sneaking up from behind you and smacking you on the head, saying to get going already...[paragraph break]And a few seconds later, he does[one of]. Well, the clairvoyance bit seems to work[or]. Again. Hooray for double-checking, I think[stopping]." instead;
	if location of player is trips strip:
		if store f is visible:
			all-say "You see yourself walking into woods instead of store F. Well, not quite woods. You think you hear lumber rumble." instead;
		if store i is visible:
			all-say "You see yourself examining the pictures on store I, then store I, then watching it collapse. You see yourself walking down a narrow exit. You even hear something like a rocket launch." instead;
		if store m is visible:
			all-say "You see yourself examining the map on store M. You hear a subway--or maybe just a busy city street." instead;
		if number of solved regions < 3:
			all-say "You see yourself entering [random visible portal]." instead;
		if store r is visible:
			all-say "You see yourself walking into store R, which temporarily looks like a luxury hotel, with palm trees around." instead;
		all-say "You see yourself relaxing in a large manor far from here." instead;
	if map region of location of player is forest:
		if rf is unvisited:
			all-say "You see yourself walking forward, your hair blowing back." instead;
		if odor is visible:
			all-say "You see a door swirling from nothing." instead;
		if bulge is visible:
			all-say "You see the bulge in the door falling and becoming an instrument." instead;
		if silent rut is unvisited:
			all-say "You see yourself [if bugle is visible]playing the bugle and [end if]entering the door." instead;
		if coe is unvisited:
			all-say "You see yourself [if number of things that are part of beard < 2]putting something on your face and then [end if]walking through the turnstile with ease." instead;
		if livers are off-stage:
			all-say "You see the liches working their way under the meats, which fall into the grinder." instead;
		if sliver is off-stage:
			all-say "You see the livers narrowing to a sharp, semi-metallic point." instead;
		if drapes are not in lalaland:
			all-say "You see yourself pointing at the drapes, which billow out." instead;
		if shotgun is off-stage:
			all-say "You see yourself reading the Notes Stone poem and carrying something stick-like." instead;
		if silver is not in shotgun:
			all-say "You see yourself [if silver is off-stage]molding the sliver into something more compact and glowing and then [end if]loading your shotgun." instead;
		if spread is cut:
			all-say "You see yourself consulting some sort of paper for where to go next." instead;
		if location of player is frost forts:
			all-say "You are about to look in the phial, but it's survival time, here." instead;
		all-say "You see yourself tracing something on the [if spam is in lalaland]maps[otherwise]spam[end if] and then walking with a purpose." instead;
	if map region of location of player is sortie:
		if centrifuge-stopped is false:
			all-say "You see yourself flipping the dial back and forth about six or seven times[sortie-end]." instead;
		if the nick is unvisited:
			all-say "You see yourself walking north from the centrifuge and being captured!" instead;
		if hoses are in roomroom:
			all-say "You see yourself smiling at your feet with pride!" instead;
		if taco is off-stage and coat is off-stage:
			all-say "You see yourself making a meal, then dressing up to go out. You don't see yourself EATING the meal." instead;
		if moor is unvisited:
			all-say "You see yourself going to the southeast room and disappearing!" instead;
		if anapest is in moor:
			all-say "You see a rough but friendly fellow scaring Pat away." instead;
		if hay is not in lalaland:
			all-say "You see yourself filling up a wall then making it disappear." instead;
		if silo is not in moor:
			all-say "You see yourself building a huge structure without saying anything." instead;
		if black door is off-stage or panel is off-stage or missile is off-stage:
			all-say "You see yourself programming the silo, seeing it moving. It has something in it, a way to get in, and it's programmable." instead;
		if hoots button is visible or trees button is visible:
			all-say "You see yourself pointing at one of the buttons before it changes. Then you push it and hear a roar." instead;
		all-say "The blue lube shows you nothing. You probably just need to push a button or something." instead;
	if map region of location of player is metros:
		if night thing is in toh:
			all-say "You see yourself giving something reddish to [if toh is visited]the night thing in the hotel[otherwise]some weird monster[end if] before it runs off." instead;
		if cake is in condo:
			all-say "You see bugs crawling over rich people, while you sit in the center, smiling." instead;
		if player has heaths:
			all-say "You see a weapon at your side, as if tied to a belt." instead;
		if player has begonias:
			all-say "You are holding a bag with one hand and plugging an ear with another." instead;
		if neon pig is in ett:
			all-say "You see the neon pig exploding, the empty space being filled with--an ear of corn?!" instead;
		if spis is unvisited:
			all-say "You see yourself putting the [if brocade is in lalaland]barcode[otherwise]brocade[end if] and the [if cake is in lalaland]keycard[otherwise]dry cake[end if] on the sensor in the Underside. Nothing works until you put them together." instead;
		if player does not have tulip:
			if player has noise bag or player has begonias:
				all-say "You see yourself grabbing the tulip as the nerds close their ears!" instead;
			all-say "You see yourself opening the termite emitter so it swarms over the nerds[if emitter is not angstgnatted]. You wonder who would be dirty enough to give you a quick refill[end if]." instead;
		if player is not in Bassy:
			all-say "You see yourself (finally) walking across the tracks." instead;
		if siren is visible:
			all-say "You see yourself pounding the siren and then doing various gymnastic tricks--wait, no, that'd be cray. Too many verbs to guess." instead;
		if beats are visible:
			all-say "You see a crooked musical note replaced with a monster worse than the night thing. Then you take something from your [if player has noise bag]noise bag[otherwise]sheath[end if] and it collapses before you touch it." instead;
		all-say "[bug-report] I need a hint here." instead;
	if map region of location of player is resort:
		if player is in ts:
			all-say "You seem to see yourself double--swimming or flying across the water." instead;
		if player is in olw:
			all-say "You see two visions: one, the toolshed being destroyed, and another, the wall to the east tilting. In each, you climb up." instead;
		if player is in hotspot:
			if red bull burdell is visible:
				all-say "You don't have time for that." instead;
			if red bull burdell is in lalaland:
				all-say "You see yourself walking east." instead;
			all-say "You see the (former) rioters discussing art and wearing dirty smocks." instead;
		if player is in roman manor:
			all-say "You see yourself just being. No, abiding. No, it's hard to describe what you're doing, since it's so passive." instead;
	all-say "[bug-report] This part needs a story.";
	the rule succeeds;

to say sortie-end:
	say ". It all seems to end with some sort of rocket hitting a huge building. But there is no explosion, just rainbows and bubbles. Weird";

chapter angleing

angleing is an action out of world.

understand the command "rk/kn/hnk/glean" as something new.

understand "angle" as angleing.
understand "hrn" as angleing.
understand "rk" as angleing.
understand "kn" as angleing.

carry out angleing:
	if phial is in cabinet and lube-asked is true:
		try taking phail phial;
		if player does not have phail phial:
			all-say "[bug-report] Please let me know how this happened." instead;
	if player does not have phail phial:
		all-say "[reject]" instead;
	if number of visible flippable not flipped-yet things is 0:
		all-say "You stare into the blue lube for a bit, then look around. Nothing looks different here. Looks like you have very little magic to do." instead;
	all-say "You stare into the blue lube for a bit, then look around. You notice weird hazes around [the list of visible flippable not flipped-yet things] when you do.";
	now all visible flippable not flipped-yet things are padded;
	h-check;
	choose row with short of "todo" in table of pad-stuff;
	if verify entry is false:
		say " (You decide to keep a to-do entry in your notebook of stuff you haven't tackled, yet.)";
	pad-rec "todo";
	the rule succeeds;

to h-check:
	if h-short-told is true:
		continue the action;
	if the player's command matches "angle":
		increment h-short;
	if the player's command matches "glean":
		increment h-short;
	if h-short > 4 and location of player is not notices section:
		all-say "After this vision, you shake your head a bit on seeing yourself taking six foot-high letters labeled ANGLE and crumpling them into AN. Then you do the same for GLEAN, which folds to GL. (Fourth wall time, to be clear--you can use four abbreviations, now.)";
		now h-short-told is true;
		pad-rec "macros";

h-short is a number that varies. h-short-told is a truth state that varies.

chapter certifying

annoying-switches is a number that varies.

certifying is an action applying to one thing.

understand the command "c/cer/cert/certify [something]" as something new.

understand "certify [something]" as certifying.
understand "cert [something]" as certifying.
understand "cer [something]" as certifying.
understand "c [something]" as certifying.

carry out certifying:
	if gadget is rect:
		if gadget-secured is true:
			say "Your gadget has been locked in. You can only rectify." instead;
		say "(switching to certify first)";
		increment annoying-switches;
		now gadget is cert;
	try scaning noun;
	if annoying-switches > 3 and marcos-trumped is false and location of player is not notices section:
		intro-marcos;
	the rule succeeds;

chapter rectifying

rectifying is an action applying to one thing.

understand the command "r/rec/rect/rectify [something]" as something new.

understand "rectify [something]" as rectifying.
understand "rect [something]" as rectifying.
understand "rec [something]" as rectifying.
understand "r [something]" as rectifying.

carry out rectifying:
	if gadget is cert:
		if gadget-secured is true:
			say "Your gadget has been locked in. You can only certify." instead;
		say "(switching to rectify first)";
		now gadget is rect;
	if annoying-switches > 3 and marcos-trumped is false and location of player is not notices section:
		intro-marcos;
	try scaning noun instead;

chapter cring

cring is an action applying to one thing.

understand the command "rc/cr [something]" as something new.

understand "rc [something]" as cring.

understand "cr [something]" as cring.

[?? if something is not flippable, this rejects it. Perhaps what I really want to do is check for a scan, and if nothing happens, don't worry about it or don't try the next bit. Have a per-turn checklist for this.]

[?? 2 disambig in a row are a bug to consider]

marcos-trumped is a truth state that varies.

carry out cring:
	if gadget-secured is true:
		say "You can't use this shortcut. Your gadget is locked." instead;
	if noun is inflexible:
		if number of things that anagram the noun > 0:
			say "The gadget buzzes[buz-help]" instead;
		say "The gadget makes no noise." instead;
	now marcos-trumped is true;
	if noun is pig and player is not on fuzzy looking wall: [?? this is not perfect code. It is duplicated elsewhere.]
		say "You are too far away at the moment." instead;
	if gadget is cert:
		say "You get to scanning, twiddling from certify to rectify and back.";
		try scaning the noun;
		now gadget is rect;
		try scaning the noun;
		now gadget is cert;
	otherwise:
		say "You get to scanning, twiddling from rectify to certify and back.";
		try scaning the noun;
		now gadget is cert;
		try scaning the noun;
		now gadget is rect;

chapter sging

sging is an action applying to nothing.

understand the command "sg" as something new.

understand "sg" as sging.

carry out sging:
	if notices section is unvisited:
		say "[reject]" instead;
	if player does not have tagged gadget:
		try silently taking tagged gadget;
	if player does not have tagged gadget:
		say "You need the gadget to do that." instead;
	if gadget is cert:
		say "You switch the gadget to rectify.";
		now gadget is rect;
	otherwise:
		say "You switch the gadget to certify.";
		now gadget is cert;
	the rule succeeds;

chapter gating

gating is an action applying to nothing.

understand the command "gat" as something new.

understand "gat" as gating.

carry out gating:
	if instructional tag is visible:
		say "Hey! Guns are dangerous! It [if shotgun is not off-stage]wasn't[otherwise]won't be[end if] that easy to get one. Besides, changing the tag will break the tagged gadget.";
	the rule succeeds;

chapter scaning

rule for supplying a missing noun when scaning or certifying or rectifying or cring: now the noun is the air.

scaning is an action applying to one thing.

understand the command "scan [something]" as something new.

understand "scan [something]" as scaning.

understand "scan" as a mistake ("[reject]") when notices section is unvisited.
understand "scan [text]" as a mistake ("[reject]") when notices section is unvisited.
understand "c" as a mistake ("[reject]") when notices section is unvisited.
understand "c [text]" as a mistake ("[reject]") when notices section is unvisited.
understand "r" as a mistake ("[reject]") when notices section is unvisited.
understand "r [text]" as a mistake ("[reject]") when notices section is unvisited.
understand "cr" as a mistake ("[reject]") when notices section is unvisited.
understand "cr [text]" as a mistake ("[reject]") when notices section is unvisited.
understand "rc" as a mistake ("[reject]") when notices section is unvisited.
understand "rc [text]" as a mistake ("[reject]") when notices section is unvisited.

does the player mean scaning the player:
	if notices section is visited:
		it is unlikely;
	it is likely;

to say cant-change:
	if gateman is visible:
		say "[one of]'That means what you have can't be changed.'[or].[stopping]";
	otherwise:
		say "Hm, maybe you can't do anything with that.";

gadget-scanned is a truth state that varies. gadget-scanned is usually false.

before scaning (this is the take gadget if you can rule):
	if player does not have the tagged gadget:
		if player is in notices section:
			say "Take the gadget from the cabinet so you can scan things." instead;
		if gadget is visible:
			say "(Taking the gadget first)";
			try taking the gadget;
	if player has tagged gadget and gadget is broken:
		say "You broke it when you cut off the tag." instead;

to say asknerds:
	say "[if gadget is cert][rc][rc][rc][rc][rc][rc][rc][gc][otherwise][bc][rc][bc][bc][bc][bc][gc][bc][end if]";
	
to say nerdsask:
	say "[if gadget is cert][rc][rc][gc][rc][rc][rc][gc][rc][otherwise][bc][bc][bc][gc][rc][bc][bc][bc][end if]";

carry out scaning:
	if notices section is not visited and map region of location of player is not intro:
		say "[bug-report] You should not get to this code before the notices section. You probably jumped away from the intro." instead;
	if noun is storeall:
		say "Not all at once." instead;
	if noun is disamb-store:
		say "Pick a specific store, instead." instead;
	if player has gadget:
		now gadget-scanned is true;
	if player does not have gadget:
		if notices section is unvisited:
			say "[reject]" instead;
		say "You'll need the gadget to scan anything." instead;
	now scanned-g is true;
	if noun is banshee:
		if liches are visible:
			say "The banshee's annoying but not as obstructive as the liches. In fact, you can't tell where he is." instead;
	if noun is tulip and nerds are visible:
		say "[one of]One of them demands to see your gadget. They poke around at it, convince you they know more about it than you do, and hand it back[or]'Insanity is doing the same thing over and over again and expecting to get different results!' one of them says. Stiff high-fives follow all around.[or]You hear exasperated sighs.[stopping]" instead;
	if noun is faeries:
		say "You sense that would be a breach of some magical etiquette you don't understand." instead;
	if noun is bars or noun is grate:
		say "[if gadget is cert]Seven reds. Hm, you can't see seven letters, so maybe that applies to the message on there instead[otherwise]The gadget blinks oddly. Maybe rectify isn't the right setting here[end if]." instead;
	if noun is night thing:
		say "No way you're getting close enough." instead;
	if noun is nerds:
		say "[one of]'Whoah! Hey! What's that? You didn't make that, did you? Don't know the specifications? Thought not.'[or]'Still messing with superior technology, eh?'[stopping] As the nerds ask other questions, the gadget lights up, and you notice [nerdsask], until the nerds ask no more. Then it goes back to [asknerds] as you think how, why and what to ask nerds." instead;
	if location of player is Elf Row's Flowers:
		say "[one of]You feel a bit self-conscious waving your gadget around such intrinsically magical beings. But they do not seem to mind.[paragraph break][or][stopping]";
	if noun is poem and poem is not folded:
		if smilies are visible:
			say "Your gadget doesn't do much til it's over the smilies. So you look more closely at those.";
			try scaning smilies instead;
	if noun is tinfoil info-lit:
		say "Nothing happens. Which just proves the paper's premise that this is all too ridiculous, amirite?" instead;
	if noun is barcode:
		say "You receive a rush of insight unlike anything anywhere else in the game![paragraph break]";
		say "[one of]Bread, $1 per loaf[or]Condensed cream of mushroom soup, $.39[or]Cabbage, $.19/lb[or]Ice Cream, $2.50/gallon[or]Jack's Pizza, 5/$10, limit 5[or]Milk, $1.99/gallon[or]Root Beer, $.68 per liter bottle[at random] at [one of]Winn-Dixie[or]SuperFresh[or]Marsh[or]Kroger[or]Piggly Wiggly[or]Jewel Finer Foods[or]Aldi[or]Safeway[or]Moo and Oink[at random]!" instead;
	if noun is pat:
		say "'Fie!' yells Pat. 'My poem transcends technology! I am sure such a fancy gadget could not even tell me its meter!' Then he gets back to recitation." instead;
	if noun is td:
		say "Not likely--it's inside the store." instead;
	if noun is oils and gadget is rect:
		say "You stick the gadget down the cask's hole so it's almost touching the oils[if silo is in moor]. It's stuck on [rc][bc][bc][gc][otherwise]. It goes to [bc][bc][rc][gc] -- then [rc][bc][bc][gc] -- and back[end if]." instead;
	if noun is oils and gadget is cert:
		say "You stick the gadget halfway into the cask, and it reads [if silo is in moor][rc][gc][gc][rc][otherwise][rc]**[rc], the middle two dots flipping from both red to both green and back[end if]." instead;
	if noun is the player:
		if warts are visible:
			say "Hm, the warts are registering. [no line break]";
			try scaning warts instead;
		say "The gadget remains silent as you scan yourself. You're either too awesome for any funny changes, or too boring and inflexible. Whichever." instead;
	if noun is gateman:
		say "The gadget buzzes.[paragraph break]'What?! Were you hoping to change me back?' harrumphs Old Man Almond." instead;
	if noun is the broad board:
		say "The gadget makes no noise[if gateman is visible][cant-change][otherwise]." instead;
	if noun is tiles:
		say "[if gadget is cert][rc][rc][gc][gc][rc][otherwise][rc][gc][bc][bc][bc][end if], all up and down the tiles." instead;
	if noun is the rut-door:
		say "You see no activity until you wave your gadget over the bulge. So you leave it there, and you pick something up.[line break]";
		try scaning bulge instead;
	if noun is sandwich:
		say "Your gadget refuses to remain stable. As if it's trying to read two things at once. Perhaps if you pulled the components apart?" instead;
	if noun is oils:
		say "The feedback from the cask seems to interfere, and you don't want to drop your toy IN the oils--but really, there are only so many possibilities. You're pretty sure you can figure them.";
	if noun is candelabra:
		say "Way too high up." instead;
	if noun is dial:
		if numset of dial is 16:
			say "Your gadget is silent. You've figured what to do with the dial." instead;
		say "The gadget, when over the EXITS part, reads [if player has gadget][rc][rc][rc][rc][rc]??--flipping between [rc][rc] and [gc][gc][otherwise][bc][bc][bc][bc][gc]??--flipping reds and blues[end if]. The dial's solution probably has the letters EXITS in it, somehow." instead;
	if noun is dollhouse:
		if attics are visible:
			try scaning attics instead;
		otherwise if static is visible:
			say "Before your gadget gets close, you begin to pick something up. Perhaps it isn't the dollhouse you need to do something with. Something ambient--like the static. So you do that.";
			try scaning static instead;
		otherwise:
			say "[bug-report] One of attics and static sould be visible." instead;
	if noun is part of the gadget:
		say "If that part of the gadget were detectable, you'd have gone crazy from the beeping by now." instead;
	if noun is gadget:
		say "If it were detectable, you'd have gone crazy from the beeping by now." instead;
	if noun is the gateman:
		say "The gadget buzzes[one of]. 'It'd be easier to ask me questions than scan me.'[or]. 'What've you got against me? If you're ever a nametag, you'll feel sorry you even THOUGHT it.'[stopping]" instead;
	if noun is cabinet:
		say "Two beeps. It's [rc][rc][rc][rc][rc][rc][gc] over most of the cabinet but [if number of solved regions < 2]the cabinet seems sensitive about its acne[otherwise][rc][rc][rc][gc][gc][rc][gc] over the bits[end if]." instead;
	if noun is subway map:
		say "The reading's different over the map of Mt. Rose than the store proper. [if gadget is cert][gc][rc][rc][rc][rc][rc][else if gadget is rect][gc][bc][bc][bc][rc][bc][end if]." instead;
	if noun is poem and poem is not folded:
		if smilies are part of the poem:
			say "The gadget is more active on the right edge where the smilies are, so you move it over there.";
			try examining the smilies instead;
		say "The gadget makes an odd noise. Perhaps it's in some weird state between forms. If poems can mean different things, they can probably become different things." instead;
	if gadget is cert:
		if noun is the motto:
			say "Your gadget gives [rc][rc][rc][rc][gc][gc]. Hmm, motto is only five letters. But then you remember it's A MOTTO." instead;
		if noun is the magenta nametag:
			say "You see lights form on the screen--[rc][gc][rc][gc][rc]*[rc]--the sixth light starts red, then flashes to green and back, as if there are two aspects to the magenta nametag worth considering." instead;
		if noun is the odor:
			say "You hear occasional beeping--[rc][rc][gc][gc] flashes on and off, faintly." instead;
		if noun is not inflexible:
			say "[if noun is begonias or noun is roadblock or noun is acne-bit cabinet]You notice the gadget beeps twice. Hmm[otherwise]The gadget beeps once[end if]. A series of lights comes across: [rgtext of noun][one of] (R=red, G=green)[or][stopping][if noun is drainage and player has gadget]. Uh, oh. Not really helpful at all. Maybe you'll find a hint elsewhere, or in the stuff floating in the drainage[else if noun is tall trio]. You scan each of the tall trio to make sure nothing changes. It doesn't[else if noun is Tories]. Each portrait looks the same, and you note six letters--probably Tories[end if]." instead;
	if gadget is rect:
		if noun is the motto:
			say "Your gadget gives [bc][bc][bc][bc][gc][rc]. Since Motto is only five letters, you wonder what's up, but it's labeled A MOTTO." instead;
		if noun is the magenta nametag:
			say "The gadget's a bit of a mess. It starts at [bc][bc][rc][bc][gc][bc][bc] and goes to [gc][bc][bc][bc][bc][bc][rc] and back. Like there are two aspects to the magenta nametag worth considering. Perhaps the gadget would be better for this[if doorslammed is true]. You can always switch back later[end if]." instead;
		if noun is the odor:
			say "You hear occasional beeping--[bc][gc][bc][rc] appears on and off." instead;
		if noun is not inflexible:
			say "Most of the screen goes blue. Then a green dot and red dot bounce left and right across the gadget screen until they stabilize:[line break]";
			repeat with Q running from 1 to lgth of noun:
				say "[if gpos of noun is Q][gc][else if rpos of noun is Q][rc][otherwise][bc]";
			say "[line break]";
			if noun is tories:
				say "Hm, six letters--probably mainly important they're Tories.";
			if noun is tall trio:
				say "You scanned each of the tall trio, and the readout didn't change.";
			do nothing instead;
	if number of things that anagram the noun > 0:
		say "The gadget buzzes[buz-help]" instead;
	say "The gadget makes no noise." instead;
	the rule succeeds;
	[rehash so if Old Man Almond is visible...]

instead of putting something on cabinet:
	say "(I'll assume you meant in, not on.)";
	try inserting noun into cabinet instead;
	
doorslammed is a truth state that varies. doorslammed is usually false.

check taking gadget:
	if gadget is in hotspot and red bull is in hotspot:
		say "Red Bull Burdell kicks it out of the way." instead;

	[say "Old Man Almond coughs. 'Be a bit unfair if you had both. Y'wanna leave that other doohickey? I mean, I believe you're gonna fix things, but just in case, the next guy'll need a little...plus, I'm kinda proud of it...'"]

after taking the tagged gadget:
	say "[one of]It's a weird looking thing, all right. The instructional tag dangling from it seems relatively free of legalese. You notice a tip in a pit on the side of the gadget[or]You pick up the gadget, again[stopping].";
		the rule succeeds;
	continue the action;

to say buz-help:
	if gateman is visible:
		say "[one of]. 'That means you can't change the item back,' says the gateman.[or].[stopping]";
	otherwise:
		say ".";

saltining is an action applying to one thing.

carry out saltining:
	if number of things that anagram noun is 0:
		say "Nothing anagrams this. [bug-report] in a released version." instead;
	if noun is red bull burdell:
		say "You squint at Red Bull Burdell, trying to notice what he could become, but all you notice is his infected toe." instead;
	if noun is livers:
		say "Your vision blurs a bit, and you see a silver sliver." instead;
	if noun is not flippable:
		say "[bug-report] in a released version. This cannot be flipped back to anything." instead;
	say "Your vision blurs a bit, and instead you see [a random thing that anagrams noun].";

section gateway

a portal is a kind of thing. a portal can be enterable. a portal is usually enterable. a portal is fixed in place.

instead of taking a portal:
	say "It's probably easier to enter than to take.";

the gateway is a portal in Notices Section. "A gateway is here. You can see through its sturdy wire fencing, but it doesn't seem to lead anywhere[if carving is unexamined]. A carving on some of the fencing seems interesting, though[otherwise]. You can read the carving again, if you want[end if].".

the carving is part of the gateway. description of carving is "[if gateman is not visible]'Toga had more to rearrange than gateway.' Hmm. But there were only three letters there. You find yourself craving a solution.[otherwise]'Enter.'"

the sturdy wire fencing is undesc. it is part of the gateway.

instead of doing something with sturdy wire fencing: say "It's strong enough. You aren't getting through the gateway by force."

rgtext of gateway is "[gc][rc][gc][rc][gc][gc][gc]". rpos of gateway is 7. gpos of gateway is 1. lgth of gateway is 7.

description of gateway is "It's a fence gate, so you can see beyond and through it, though it seems too heavy to lift in any case[if gateman is not visible], but who knows, maybe if there was a guy who tended to gates, he'd help you[otherwise][end if]. It'd be neat if..."

the getaway is a portal. "Your getaway to a new land is here."

description of getaway is "The new land behind the getaway gateway looks exciting.  Maybe you'll spend long enough there, your severance check'll make it to your mailbox before you get back."

the new land is scenery in notices section.

description of new land is "You'll have to go there to see it all."

instead of entering new land:
	if gateway is visible:
		try entering the gateway instead;
	if getaway is visible:
		try entering the getaway instead;
	say "[bug-report] gateway and getaway both gone.";

instead of doing something other than examining or entering the new land:
	say "Go through the gate to learn more about it."
	
check opening the gateway:
	if gateman is not visible:
		say "Hmm. You're not strong enough. Just as you weren't hungry enough to eat those thorns." instead;
	say "[one of]The gateman coughs. 'I'm not allowed to open it til you figure what it really is. But I think you're up to it.'[or][gateway-hint][stopping]" instead;
	
to say gateway-hint:
	say "[one of]'No getting away from it. The gateway needs to GO somewhere.'[or]'There're 2519 possibilities, but the one you want is actually pretty close. As close as it can be.'[or]'Can't give any more hints, sorry.'[stopping]"

check entering the gateway:
	if gateman is visible:
		say "[one of]You hold your hand out in front of you to see if the gateway vanishes as you walk through. Nope. The gateman speaks.[paragraph break]'It's not that easy. But it's not that hard. It doesn't get anywhere yet. Til it does, you can't go far away.'[or][gateway-hint][stopping]" instead;
	say "It doesn't lead anywhere. Yet. Plus, you'd run into the, uh, gate part. Maybe if you had someone who could give information. And not stupid information like a silly old nametag." instead;
	
understand "gate" as gateway when gateway is visible.
	
understand "gate" as getaway when getaway is visible.
	
understand "gateway" as getaway when getaway is visible.
[This seems to hang inform if I SHOWME GATEWAY]

check propfliptoing the getaway:
	if gateway is not visible:
		continue the action;
	if gateman is not visible:
		say "The gateway rumbles. You see a shiny new terrain...and you try to lift the gate, but you can't. The terrain disappears, replaced with the regular city view.[paragraph break]That seemed right. Perhaps you can summon someone who can do better." instead;
	say "The gateway rumbles. 'Bout time you figured it!' says Old Man Almond. 'Let me use my trifle lifter here.' He tugs down on the gate with his index finger, and it flips up. 'What?! I can have magic powers, too! Couldn't a survived being a nametag otherwise!'"

gadget-warned is a truth state that varies. gadget-warned is false.

[I tried til Tired!]

check entering the getaway:
	if player does not have tagged gadget and gadget-warned is false:
		say "'Dang adventurers these days. It's not demo mode out there. If I'd had doohickeys nice as that gadget in the cabinet when I saved that ONE world...I'd of took it! And saved two or three worlds!'";
		now gadget-warned is true instead;
	say "Your path is clear. [if player has gadget]Old Man Almond gets his hanky out. 'Okay, hunt.'[paragraph break]'Thank you,' you say.[paragraph break]'Plod wary through warpy old Yorpwald!' he calls.[paragraph break][otherwise]'A fool! Aloof!' cries Old Man Almond as you pointedly ignore the cabinet. As you walk through, you hear 'Oh, nope! No hope!'[no line break][end if]";
	say "The gate disappears behind you[if static is visible] as you hear Old Man Almond grumble about being stuck with the static[end if], and you walk down a pathway, then yap 'What?!' when you see...";
	wfak;
	say "...a few strip malls. Ugh. There's nothing adventurous about THAT.[line break]";
	solve-region intro instead;

understand "fellow" as gateman.

section gateman subjects

asking generically is an action applying to one topic. Understand "ask about [text]" or "talk about [text]" or "a [text]" as asking generically.

objasking generically is an action applying to one visible thing. Understand "ask about [any thing]" or "talk about [any thing]" or "a [any thing]" as objasking generically.

check objasking generically (This is the check for only one sensible object converser rule):
	if the number of persons enclosed by the location is one:
		repeat with X running through visible scenery:
			repeat through table of default-gen-blather:
				if X is the default-talker entry:
					say "[gen-blah entry][line break]" instead;
		say "You inquire into your own thoughts. You gain no illumination." instead;
	if the number of persons enclosed by the location is two:
		repeat with candidate running through visible people:
			if the candidate is not the player:
				try objasking candidate about the noun instead;
	say "Ambiguous--more than two people here." instead;

does the player mean asking scenery about: it is unlikely.

does the player mean asking the player about: it is very unlikely.

objasking it about is an action applying to one thing and one visible thing.

understand the command "ask [thing] about [any thing]" as something new.
understand "ask [thing] about [any thing]" as objasking it about.

understand the command "ask [thing] about [text]" as something new.
understand "ask [thing] about [text]" as asking it about.

Check asking generically (This is the check for only one sensible converser rule):
	if the number of persons enclosed by the location is one:
		repeat with X running through visible scenery:
			repeat through table of default-gen-blather:
				if X is the default-talker entry:
					say "[gen-blah entry][line break]" instead;
		say "You inquire into your own thoughts. You gain no illumination." instead;
	if the number of persons enclosed by the location is two:
		repeat with candidate running through visible people:
			if the candidate is not the player:
				try asking the candidate about the topic understood instead;
	say "Ambiguous--more than two people here." instead;

to say lazy-adv:
	say "'Scan something. Like that dollhouse. Or even me. Then come back and ask for details. You're an adventurer, not a freeloader.'";

check objasking it about (This is the check for object information rule):
	if noun is gateman:
		if second noun is gadget or second noun is part of the gadget:
			if scanned-g is false:
				say "[lazy-adv]" instead;
	repeat through table of subject-blather:
		if second noun is the person-subj entry:
			if noun is the him-who entry:
				say "[him-say entry][line break]";
				the rule succeeds;
	repeat through table of default-sub-blather:
		if noun is gateman:
			if second noun is not goat:
				if second noun is in dy:
					say "'In the past now. Nothing more to learn from that.'" instead;
				if second noun is in subsite:
					say "You'd know more about that than he would." instead;
			if second noun is visible:
				say "Examining would probably work better." instead;
		if noun is him-who entry:
			say "[him-say entry][line break]" instead;
	repeat through table of default-gen-blather:
		if noun is default-talker entry:
			say "[gen-blah entry][line break]" instead;
	if noun is notepad:
		try consulting notepad about "[second noun]" instead;
	if noun is not a person:
		say "You can only ask people about things." instead;
	say "[bug-report]!!! [noun] does not have a default entry." instead;

stupid-temp is text that varies.

check asking about (This is the check for specific topics rule):
	if the topic understood is a topic listed in the table of general-blather:
		if him-who entry is the noun:
			if scanned-g is false:
				if topic understood matches "certify" or topic understood matches "rectify":
					say "[lazy-adv]" instead;
			say "[him-say entry][line break]" instead;
			[would like to pad-rec "[what-pad-gen entry]" instead; but inform is confused with 2 tables]
			do nothing instead;
	repeat through table of default-gen-blather:
		if the noun is default-talker entry:
			say "[gen-blah entry][line break]" instead;
	if noun is notepad:
		try consulting notepad about "[second noun]" instead;
	if noun is not a person:
		say "You can only ask people about things." instead;
	say "[bug-report] no default blather for [noun].";


to say goat-toga:
	say "[one of]'Yes, yes, you got the first puzzle. Good adventurer, good adventurer. If you're interested, the gadget would've said Red-Green-Red-Red in Certify mode and Red-Blue-Green-Blue in Rectify.' [run paragraph on][or]'Gosh durn it, I wasn't as forgetful as YOU when I was young!' (Good thing you wrote things in your notepad.)[no line break][stopping]";
	pad-rec "goat";
	
check asking gateman about "world peace":
	say "That goal's a slog, but I've got confidence." instead;

understand "man" as gateman when gateman is visible.

understand "gate man" as gateman.

description of gateman is "'Stop staring!' he booms. [if getaway is visible]'You're ready to go through that gateway.'[otherwise]'You summoned me. The next bit is easier. What to do about that gateway?'[end if]"

to say your-job:
	say "[one of]'We need someone to free us from the marauding tyranny of Red Bull Burdell! He has made a plowyard out of Yorpwald!'[or]'You. Red Bull Burdell. Win.'[stopping]";

to say oma-gate:
	if gateway is visible:
		say "'Just want to make sure you're up to it. Really, I shouldn't say this, but it's easier than the goat, sort of.'";
	otherwise:
		if player does not have gadget:
			say "'I'd get one of them devices in the cabinet before entering.'";
		otherwise:
			say "'Get going through there!'[run paragraph on][line break]";

check asking the player about:
	say "If only learning were so easy." instead;

does the player mean objasking about red bull burdell: it is very likely.

does the player mean objasking generically red bull burdell: it is very likely.

does the player mean doing something with the merchandise when the merchandise is visible: it is very likely;

to say faery-flower:
	say "[if fairy-worthy is true]'We have faith in you! You will make the city safe to grow flowers again!'[otherwise]'We trade in flowers! Flowers we have for flowers we want! What do you have?'[end if][run paragraph on]"

lube-asked is a truth state that varies.

to say about-lube:
	if player has the phial:
		say "'Good, good, you got the phial. Remember to ANGLE to see into things or GLEAN to see around as need be.";
	otherwise:
		say "'[if cabinet-bit-me is true]Cabinet not letting you have it, eh? [end if][one of]The phial with the lube can grease your way to a solution. He coughs. 'Sorry.'[or][stopping]If you ANGLE the lube just right so it reflects a light source, you can see what you can change. If you GLEAN, you see what you should be doing. How your story unfolds[if lube-asked is false].' He walks over to the cabinet, pats it, and says a few words. 'You can take the phial now.'[otherwise].'[end if]";
	pad-rec "lube";
	pad-rec "angle";
	pad-rec "glean";
	now lube-asked is true;

table of subject-blather
him-who	person-subj	him-say
deadbeat	lost corn	"[if condo-evac is false]Hey, Man, like, how do we know you don't, like, like the Man, man? Hey? Hey?[else if lost corn is visible]'Oh, yeah, it's yours, man, I guess. It's real corn, not that genetically modified stuff from,' he sees red for a sentence, '[first custom style]SLORNTCO[roman type] or that nasty dried stuff from [first custom style]ST ROLCNO[roman type].'[otherwise]'Whoah. You must've like did something with that corn. Biosustainibility, man.'"
deadbeat	termite emitter	"[if player does not have emitter]'What? Do you know something?' he looks furtively around, more convinced than ever you're with The Man, man.[else if condo-evac is false]'Whoah. No fingerprints, no proof.'[otherwise][try-flies][end if]"
faeries	gardenia	"[if drainage is in underside]You don't have a gardenia, yet. But you can find or make one.[otherwise][faery-flower][end if]"
faeries	heaths	"[faery-flower]"
faeries	begonias	"[faery-flower]"
faeries	merchandise	"[faery-flower]"
faeries	clover	"'That is just a weed! We have no use for it. But maybe it is useful to you.'"
faeries	tomato	"'A putrid vegetable! [if player has tomato]We should kick you out right now[otherwise]It must taste terrible[end if]!'"
faeries	tulip	"'Some flower that is!'"
gateman	board	"'Good idea to take notes on it. But I can tell you about other stuff. Like [if player has gadget]your gadget[otherwise]the gadget in the cabinet[end if]. Or even how it works.'"
gateman	dollhouse	"[if attics are off-stage]'A nice warm-up puzzle, but if you're in a hurry, I'll respect that too. Not the dollhouse that needs changing.'[otherwise]'You don't need to take it with or anything. Nice job with it, though.'[end if]"
gateman	static	"'[if attics are off-stage]If it's kind of annoying, you can get rid of it. Or you should be able to. But you don't have to.[else if attics are visible]Good job getting rid of it.[otherwise]Err...I know you're in a hurry, but can if you could change it back to the attics before you get going, my ears would thank you.[end if]'"
gateman	attics	"'[if attics are off-stage]Yes. Do--whatever it is--to make them appear. I hope you haven't lost your powers.[else if attics are visible]They're more appropriate to the dollhouse, definitely.[otherwise]I kind of prefer [']em to the static. You wouldn't mind switching them back?[end if]'"
gateman	blue lube	"[about-lube]"
gateman	phail phial	"[about-lube]"
gateman	Red Bull Burdell	"[one of]'Like he was thought up by a demonic comedian. Made a desireable age disagreeable. Seemed a poseur when he uprose from abuser to saboteur. He'd make a poignant nag point, acting like a rowdy pal. Heard his cleats in the castle as he did something called 'Leveling up.' Maximizing stuff called hit points and magic points. Things went awry before we were wary.' You sense there's more, if you can bear it.[or]'Rum toil became turmoil. We let the ploy draw us in...then Prowl Day made the world pay. Against Raw Rori, a warrior, the devil lived. We ignored eroding. He humbled Bud Helm. We have lost LOTS because of him. He beat us all. Rooftops of troops on Fort Oops. Profs too. Auctioned Education. I cautioned...' His eyes grow mistened. 'Ten dimes say it's end times. We've waited for an upturn to turn up...'[or]'The time for talk is over.[stopping][rbb][line break]"
gateman	tinfoil info-lit	"'Sane Sean. Hmph. Either he's dumb enough to believe Red Bull Burdell's a good guy, or he's a deep undercover double agent.'"
gateman	Bud Helm	"'Yeah, we were surprised the bum held Bud Helm.  Bud Helm bled, hum? He's a broken man, now.'"
gateman	Raw Rori	"'Pure strength. That won't get the job done here.'"
gateman	pen	"[one of]'How thoughtful! Nobody's asked me for my autograph for a long time! Wait, we don't have time for that nonsense.'[or]You probably know more about the pen than he does.[stopping]"
gateman	dope tan notepad	"'Good choice, taking notes.'"
gateman	new land	"'I can't tell you everything about it. But you'll see for yourself.'"
gateman	nametag	"'Oh, did you have problems with it? Sorry. I'd have helped, but I was--I mean, I wasn't [italic type]at all[roman type].' He rolls his eyes."
gateman	yourself	"[your-job][run paragraph on]"
gateman	gateman	"'Once I solved one of those hero mills with more hills...' he blinks. 'I'm not important. You are. And what you can do.'"
gateman	goat	"[goat-toga]"
gateman	toga	"[goat-toga]"
gateman	shrub	"[goat-toga]"
gateman	brush	"[goat-toga]"
gateman	thorn	"'Sorry if it scratched you. And by you I mean your nametag. And by your nametag I mean me.'"
gateman	gateway	"[oma-gate]"
gateman	getaway	"[if gateway is visible]'That sounds right. But don't tell me. Do -- whatever you did with the goat.'[otherwise][oma-gate][end if]"
gateman	gadget	"'Powerful. Not as a weapon. Won't be many fights. But it tells what you can make of an object. Ask about a specific button if you're interested.'"
gateman	s-r	"'It'll make it tougher and easier for you. Forces you into only certifying or rectifying if you push it. But you can RECUSE yourself later from one of the three areas to conquer. Once you get through the gateway, might want to scan everything, then decide.'[pad-sec]"
gateman	noise bag	"'Cities gotta have noise but there's one just constantly has it. Burdell's behind it, too, claiming he couldn't have leveled up without that techno-piffle and people get used to it if they try. Actually calls it music. He's got a lieutenant to enforce it.'" [!! HACK ... NOISE BAG comes before NOISE]
gateman	saltine	"'Miniscule, but more than mini-clues. The next item that baffles you, eat that and it'll make sense. Oh, apologies in advance: good flavors and good magic don't mix.'"
gateman	cabinet	"'[if lube-asked is false][lubeit]It's kind of picky and defensive. Might not give you everything til I tell it.' He goes over to the cabinet, pats it on its top, and returns. 'Now you should be able to get everything from there.[otherwise]Everything in there's yours.[end if]'"
peasant	hay	"'Not very useful for building something unless you want it to be destroyed quickly. Can't think of any reason I'd need that.'"
peasant	poetry	"'I don't expect to make a profit off it.'"
peasant	straw	"'Good, sturdy stuff. I could use some to repair my hut. Can't imagine anyone would want what I have, though.'"
peasant	Woeful Pat	"'He's better than I am at poetry, I'm sure. I'm not educated enough to know why. Frustrates me sometimes.'"
nerds	Tulip	"[if player has tulip]'It's not going to burn out for a long time.'[otherwise]'We've got plenty of light here. You can have it, if you know what to ask us about. And if the tulip would help.'[end if]"
talkers	dry cake	"Someone puts his hand to his lips. Someone else shakes his head. You don't DESERVE that cake, apparently."
woeful pat	peasant	"'There is one who tries his best. But he is too...obvious.'"
woeful pat	poetry	"[anapest-clue][run paragraph on]"
woeful pat	anapest	"[one of]'A mere creative writing teacher could educate you as to the pedantic details of writing doggerel that mimics my epic's flow!'[or]You didn't REALLY want the anapest explanation again, did you?[ana-true][stopping]"

to say pad-sec:
	pad-rec "secure";
	pad-rec "recuse";

to say rbb:
	pad-rec "Red Bull Burdell";

to say lubeit:
	now lube-asked is true;

to say ana-true:
	now anapest-clued is true;

the utterly unreadable book is undesc.

to say book-in-nerds:
	now utterly unreadable book is in spis.

check examining the utterly unreadable book:
	say "Even the author names are pretentious and important sounding. Think they're worth remembering?";
	if the player consents:
		say "[rc]Srs. Ek and Rensard-Knessard";
	otherwise:
		say "Hm, yes. The nerds may give a less cryptic hint."
	

to say rose-sore:
	if player does not have gardenia:
		say "[one of]. It will probably be tougher than getting a sore and changing it to a rose[or][stopping]"

table of default-gen-blather
default-talker	gen-blah
banshee	"It is too busy moaning in general."
beast	"It cares not for conversation, only noise."
beats	"Dude! They're clearly busy bringing sexy back. You text adventurers don't know how to act."
deadbeat	"He mumbles something about [if condo-evac is false]chillin['] with his ill chin and how you're probably down with the man's anthems and aligned with those yuppies to the north, glancing furtively at the lost corn[else if corn is visible]how you haven't put that lost corn to good use yet[otherwise]whatever you did with that corn better work[end if], and he also complains about [if spis is visited]those nerds you saw[otherwise]nerds down southish, for some reason[end if]."
faeries	"[if fairy-worthy is false]'You must bring a powerful flower up! For our magic garden-and-a-third! There must be one in the garbage in this city! All we can make with the beats pounding are freesias, and we are getting sick of those!' they exclaim[rose-sore].[otherwise]'You are our hero! Thank you so much[sure-you]! Now go and save the city.'[paragraph break]Hm, they didn't really seem to be listening to you, but fair enough, that's something."
goat	"It got here by magic, but it isn't. Don't fluster the restful."
gateman	"'Eh?! There's all sorts of things to ask me about! That goat back there! Your quest! Your purpose! The [if getaway is visible]getaway[otherwise]gateway[end if]! General advice! No time to scold clods asking about--whatever you just mumbled about.'"
gy	"The men passing through seem in a hurry."
gy2	"You could never get everyone's attention at once."
liches	"They are too busy moaning to each other."
masses	"You can't think of any small talk. Not that you're trying hard to."
nerds	"[one of]'He's not smart enough for us, is he?'[paragraph break]'Well, if he found his way in, he might be!'[paragraph break]'Well, he'll have to find the right thing to ask nerds like us about if he wants us to stop talking about him like he's not here.'[paragraph break]'Maybe he's a lateral thinker. Who knows several ways to get through things.'[paragraph break]'GROAN! Nice Ignorance.'[paragraph break]They don't seem to want to tell you what to ask them, out of principle, but if you keep nagging them, they may let some cryptic knowledge slip--out of principle, too.[paragraph break]'If he doesn't know what to ask nerds about, he deserves Ed's snark!' Ed leaps, pleased with what elapsed.[or]More annoying banter. Perhaps you could disperse them with the right item. They'd deserve it.[or]They form a grin ring, but you close your eyes a bit. As you do, you are hit with a book[book-in-nerds]![or]'You might as well ask dorks like Karkdoss about something that inconsequential!' An argument on the virtues of code versus frisbee golf follows.[or]Apparently you haven't seen the light for what to ask nerds about. A few smart-people in-jokes later, you feel too aliterate to retaliate. 'We can wait here a thousand years!' one laughs. They debate on what's worse, dumb people in power or out.[stopping]"
night thing	"Easier to argue a grue away."
peasant	"'I don't know much about that, that's sure. But I know from hay, being an abler baler.'"
red bull burdell	"[bull-blather]"
riot	"[riot-trio-blather]"
trio	"[riot-trio-blather]"
talkers	"You get a few pinched, strained smiles, which is more than you expected, but no actual words."
woeful pat	"[pat-blather]"
wolves	"Now is not the time for a heroic lecture. Now is a time for violence."
yourself	"'Self, what should I do now?' 'Self, don't start talking in the third person.' 'Self will make a point of it, self.'"

table of default-sub-blather
him-who	him-say
deadbeat	"He mumbles 'Mt. Rose Rots Me Moster' and giggles stupidly. You probably won't learn much from him except about instruments of subversion."
faeries	"They possess wisdom humans never will, but they can't help you with the physical realm, beyond flowers."
gateman	"'Can't say anything about that. You'll have to find out for yourself later.'"
nerds	"The nerds look at each other, wondering why you would ask about that, really. They are sure you have something important to ask them. Maybe somewhere you can't get to or through. They could figure it out if in your shoes.[paragraph break]Hmm. Apparently, when you ASK NERDS, you need to come correct[one of]. All the same, you're a bit annoyed, and you wonder if you can maybe play a mean trick on them to get that tulip, instead[or][cycling]."
yourself	"You're an interesting subject and all, but [if location of player is subsite]nobody else wants to bother[otherwise]there's stuff to be done, and stuff[end if]."

before objasking nerds about:
	if player has tulip:
		say "They don't want to be bugged any more. And, really, you don't want to bug them. You have what you wanted.";

before asking nerds about:
	if player has tulip:
		say "They don't want to be bugged any more. You have what you wanted.";

to say bull-blather:
	say "[one of]'Do nice? No dice!'[or]'Eat granite, ingrate!' You're not fooled. There's none around.[or]'You are the ANGRIEST of the INGRATES so far!'[or]'They will feel my anger in Germany!'[or]'I don't like when a STINKER TINKERS in my business!'[or]'You're not courageous! You just have a raucous ego![or]'EVIL? VILE? I just LIVE! And you will not OVERTAKE my TAKEOVER!'[or]'You are but a flea on a leaf!'[or]'The MAN CITES SEMANTICS!'[or]'The NERVE! NEVER!' he yells, with arching chagrin.[at random][line break]His bluster won't get subtler. Probably best to take action instead."

to say riot-trio-blather:
	say "You're not going to reason with them. Maybe reason about them, though.[no line break]"

Bud Helm is a person. description of Bud Helm is "He has been condemned to obscurity."

Raw Rori is a person. description of Raw Rori is "Condemned to obscurity."

does the player mean objasking generically the deadbeat: it is unlikely.
does the player mean objasking about the deadbeat: it is unlikely.

to say roo:
	pad-rec "advice";

table of general-blather
him-who	topic	him-say
deadbeat	"beats"	"They're nice and all for being an excuse not to work but they're lame and all because it's hard to concentrate on anything. I'm tired of them, [if condo-evac is true]dude[otherwise]you square, yo[end if]."
deadbeat	"slorntco"	"'Organized capital is so repressive, dude.'"
faeries	"brocade"	"[if player has brocade]'Do with it what you need.'[else if brocade is in lalaland]'You have disposed of it! We hope it was useful.'[otherwise]'It is free. But useless to us.'[end if]"
faeries	"gardenia"	"[if player has gardenia]'Yes! We like that! Give it!'[else if gardenia is in lalaland]'Yes! Thank you!'[otherwise]'We need something for our garden-and-a-third!'[end if]"
faeries	"drainage/flower/garden/garbage"	"[if gardenia is in lalaland]No need to remind them of your tasks.[otherwise]'Perhaps you can turn something in this city into a flower. We are out of ideas. How can drainage become a flower?'"
faeries	"beats/beast"	"'It is terrible! Our hearing is even more delicate than a clumsy human's. We can do no decent long-term magic with it around.'"
faeries	"heaths/begonias/heath/begonia"	"[he-be]"
gateman	"next guy"	"'Optimistic, aren't we?'"
gateman	"thor"	"'Poor guy. Forgotten well before Ragnarok.'"
gateman	"gato"	"[goat-toga]"
gateman	"general advice" or "advice" or "general"	"'Well, you figured a few things out. Probably want to focus more on changing stuff than weird actions. One thing, though. The things you need to change? Nothing too complex. Nothing over eight letters. Okay, maybe one two-word thing with nine. Red Bull Burdell often babbled about getting eighteen intelligence and wisdom from all that leveling up, but he never got a decent vocabulary. The philistine. Also, there will be clues--clues that seem all wrong, deliberately wrong, but their wrongness will stand out and help you.'[roo]"
gateman	"scold/clods"	"Like I said, no time for that. Time to fix things!"
gateman	"xyzzy"	"'Fool! That magic word has no place alongside your magic abilities.'"
gateman	"world"	"'Well, it wasn't a very SPECIAL world. In fact it was one of those hero mills. I remember more hills than anything else. Figure it had to break down for the next guy to save it.'"
gateman	"yorpwald"	"It's been shuffled, nearly f-flushed! You can rescue it from physical, economic and moral malaise. The stuff [if number of things in cabinet > 0]in[otherwise]from[end if] the cabinet should help. Ask me about it. I'm a hero emeritus, and stuff."
gateman	"lupine" or "lineup" or "lupine lineup"	"'Distant rumblings from somewhere beyond a very quiet place. Sadly, violence may be necessary.'"
gateman	"noble" or "ol' ben" or "noble ol' ben" or "ben"	"'He used to be fair and just. But he saw Red Bull Burdell [']leveling up,['] whatever that is. Red Bull bragged about how he'd multiplied his gold and achieved eighteen charisma, or something, before changing Noble Ol['] Ben.'"
gateman	"quest" or "purpose" or "my purpose"	"'Three major problems. Noise is one. A lupine lineup. Noble Ol['] Ben: gone. Lorn. No longer.'[paragraph break]'And I think I know who's behind it: RED BULL BURDELL!'"
gateman	"dollhouse/attic/attics/static/house"	"'Just something to get your feet wet, fiddling with the devices and such. Won't help with a quest[if attics are off-stage]. Though it's not the dollhouse but what's coming from it[otherwise]. Good job with it, yes[end if].'"
gateman	"certify" or "certifying"	"'Tells you what letters are right for what something should be. Red is wrong, green is right. Probably not as powerful as rectifying, but gets you some stuff right away. Also--there's some clues out there that're just plain red. They're all wrong.'[prcer]"
gateman	"rectify" or "rectifying"	"'Gives you the first and last letters of what something should be.'[prrec]"
nerds	"darkness"	"[nerd-dark]"
nerds	"dorks"	"They snicker knowingly. They assure you Karkdoss is the worst, if you want to ask dorks for any help."
nerds	"karkdoss"	"They snicker knowingly. You wonder if there is a Karkdoss, and if so, if they really think you should ask dorks."
nerds	"tulip" or "lit-up tulip" or "flower"	"[if player has tulip]'It's yours now! Don't trip and break it.'[otherwise]'You can't just ASK NERDS to give you something! This isn't the locker room after gym class! What do you need it for?'[end if]"
peasant	"lord/ ablemiser"	"'Used to be much nicer, I'd say!'"
peasant	"poetry/poem"	"'I try my best at it.'"
peasant	"peasant/himself"	"'They tell me I'm an abler baler.'"
peasant	"straw"	"'I could use some[if straw is in sack]. And it looks like you've got some. I'll give you the hay for the straw[otherwise].'"
peasant	"hay"	"'I can't use it. But I could use some straw.'"
woeful pat	"beat/rhythm/meter"	"[anapest-clue][run paragraph on]"
woeful pat	"st paean" or "paean"	"'It will be my best reading yet. Even better than this[if player does not have flyer]. Here you go, just to make sure[give-flyer].'"

to say prcer:
	pad-rec "certify";

to say prrec:
	pad-rec "rectify";

to say give-flyer:
	now player has flyer;

to say try-flies:
	if flies-in-emitter is false:
		say "'Dude! So you got it back from the night thing!' He takes the emitter, stuffs it in one of the more grungy-looking tents, and gives it back. 'Sorry there aren't any termites. But now these angst gnats inside may be handy later. We probably like infected them with social discontent when they bit us.'[no line break]";
		now flies-in-emitter is true;
		now termite emitter is angstgnatted;
	otherwise:
		say "'Man, I've done enough work today. I don't want to catch any more angst gnats. Even for you.'";

talktoing is an action applying to one thing.

understand the command "talk to [someone]" as something new.
understand "talk to [someone]" as talktoing.

check talktoing:
	if noun is riot or noun is protest:
		try asking noun about "mumblety-peg" instead;
	if noun is the player:
		try asking player about "mumbletypeg" instead;
	if noun is not a person:
		say "This place doesn't have that sort of magic." instead;
	say "You engage in small talk, not sure what to ASK about.";
	try asking noun about "mumbletypeg" instead;

check telling about:
	if noun is the player:
		say "You give yourself a pep talk." instead;
	say "You're an adventurer, not a ruler--probably better to ASK about stuff." instead;

part wolves-vowels

the wolves are in Lalaland. "Slavering werewolves aren't quite advancing on you. They're waiting for the first sucker to step forward and get shot. You think."

the wolves are plural-named.

Include (-
	has transparent animate
-) when defining wolves.
 
instead of taking wolves:
	say "Well, you can take them, but not in that sense."

understand "werewolves" as wolves.

description of wolves is "Drooling, hate in their eyes, etc. And a bit of fear. Of your gun."

part store i-sortie

the sortie-x is a portal. "Some stairs lead down from the wreckage of store I, further than you can see[if sortie is solved]. But no sense looking--you know what's there, and you've taken care of it[end if]."

check climbing sortie-x:
	try entering sortie-x instead;

check exiting in trips strip:
	if trips strip is visible:
		try entering sortie-x instead;

the printed name of sortie-x is "sortie". understand "sortie" and "stairs" and "stairway" as sortie-x. sortie-x is fixed in place.

description of sortie-x is "It's a stairway. For making a sortie. Exiting[sortie-have]"

to say sortie-have:
	if sortie is solved:
		say ". But you already did. It's not worth revisiting.";
	otherwise:
		say ". Though you probably want to ENTER it to exit the strips[if metros-x is not visible]. Or go DOWN[end if].[no line break]"

check entering sortie-x:
	if sortie is unsolved:
		check-2-of-3;
		move player to last-loc of sortie;
		recover-items instead;
	otherwise:
		say "You did what you needed in the moor. No sense retracing your steps." instead;

part forest-store f

Rule for supplying a missing noun while entering (this is the other stuff to enter rule):
	if passage is visible:
		now the noun is the passage;
		continue the activity;
	if location is trips strip:
		if number of visible available portals > 1:
			say "That's ambiguous--you can currently enter [the list of visible available portals] to explore areas you haven't solved yet." instead;
		otherwise:
			now the noun is a random visible available portal;
			continue the activity;
	otherwise:
		continue the activity.

does the player mean entering the forest-sc: it is likely.

the forest-sc is a privately-named thing. it is fixed in place. the printed name of forest-sc is "forest". "There's a forest here where store F was[if forest is solved], but you probably don't need to go back there[end if].". forest-sc is a portal. forest-sc is fixed in place.

understand "forest" as forest-sc.

description of forest-sc is "[if forest is solved]It's real verdant, yeah. I could tell you about what's inside, but you've been through it[otherwise]It's bigger on the inside than the outside, you're sure, or something like that[end if]."

gadget-unsecurable is a truth state that varies.

before entering a portal in trips strip (this is the deactivate the SECURE button rule):
	if gadget-secured is false and gadget-unsecurable is false and player has gadget:
		now gadget-unsecurable is true;
		say "You hear a click from your gadget as you pass through...you poke at it and notice the SECURE button is no longer pushable. (UNDO if you want to change this.)[line break]";
		continue the action;
	
check entering the forest-sc:
	if Forest is solved:
		say "You already solved the forest. Maybe you should look elsewhere?" instead;
	check-2-of-3;
	say "As you walk in the forest, you hear rumors of monsters ravaging the surrounding areas. Not zombies, or vampires, or skeletons. Trying to figure what it is, you daydream maybe YOU could be that hero--so successfully that you get lost!";
	wfak;
	say "Until you stumble on a small clearing. You think you smell something.";
	now player is in last-loc of forest;
	recover-items instead;

[the forest oneways store f.]

part resort-store r

the r-p is a privately-named thing. "You see a big resort is here. It seems like it's all for you! All you have to do is enter.";

the r-p is a portal.

the printed name of r-p is "resort". understand "resort" as r-p.

instead of taking r-p:
	say "Enter it instead.";

description of r-p is "Man, it's beautiful, spacious and sunny. Even sunnier than the Trips Strip, which is much nicer since you started cleaning up, but still..."

resort-known is a truth state that varies. resort-known is usually false.

check propfliptoing the r-p:
	if you-can-advance:
		continue the action;
	if store r is visible:
		now resort-known is true;
		say "That seems like it should work, but you sense you have not done enough yet[if islet is unvisited]. In fact, you probably haven't even visited some place you need to work through[end if]. But maybe, soon, you will." instead;

check entering the r-p:
	say "You stride into the resort thinking 'Gee. Damn. Endgame.' But it is a mirage! You're gamier than to let that bother you, though, even though you hardly seem to be in paradise.";
	now the player is in tiles instead;

part drapes-spread

the drapes are plural-named scenery. the drapes are flippable. lgth of drapes is 6. gpos of drapes is 6. rpos of drapes is 1. rgtext of drapes is "[rc][rc][rc][rc][rc][rc]".

understand "cloth" as drapes when drapes are visible.

instead of opening the drapes:
	say "This isn't some living room. (Sorry.) Those drapes are thick and oppressive."
	
description of drapes is "Too thick to walk or even cut through. They're made of heavy cloth packed together. Very dark, even in this semi-darkness. But if you look long enough, the various folds seem to make letters."

the folds are part of the drapes. the folds are plural-named. understand "letters" as folds.

instead of cutting the folds: try cutting the drapes instead.

understand "message" as folds when folds are visible.

description of folds is "The message in the folds is: NONE WILL | BE SP---D. Three of the letters seem smudged. Darn it."

the spread is scenery.

understand "cloth" as spread when spread is visible.

check taking the spread:
	say "It is too large and too heavy." instead;

description of spread is "[if spread is cut]It's been cut down the center.[otherwise]You can see traces of the old message on the drapes--[spread-d-u].";

to say spread-d-u:
	say "[if drapes are unexamined]which probably wasn't important[otherwise]you realize it was NONE WILL BE SPARED now[end if]"

the drapes oneway the spread.

the spread can be cut. the spread is not cut.

part sliver-livers-silver

the sliver is a thing. the sliver is flippable. rgtext of sliver is "[gc][rc][rc][gc][gc][gc]". lgth of sliver is 6. gpos of sliver is 1. rpos of sliver is 6.

description of sliver is "Surprisingly sharp and of an odd metal. It could cut something as long as you didn't have to saw much."

the livers oneway the sliver. the livers are flippable.

the silver is a thing. the printed name of silver is "silver cylinder". understand "silver cylinder" and "cylinder" and "silver bullet" and "bullet" as silver when silver is visible.

understand "shell" as silver when silver is visible.

does the player mean doing something with the shell when silver is visible: it is unlikely.

description of silver is "[if silver is in shotgun]It's locked and loaded, Junior[otherwise]It's shaped like a small rocket. It'd make a nifty projectile[end if]."

the sliver oneways the silver.

[the silver oneways the livers.]

part maps-spam

some maps are plural-named things.

understand "map" as some maps.

description of maps is "[if Ghouls' Slough is visited]They've worked fine. No need to use them again.[else if coe is visited]You notice the Cruel Ones['] Enclosure at the bottom.[otherwise]They're labeled 'ALMOST LOST, MA' and claim to lead to the slayer layers. Maybe if you get past the turnstile they'll make more sense.[end if]"

the spam is flippable.

part noughts-shotgun

the noughts are a plural-named thing in bubble. understand "naughts" as noughts. rgtext of noughts is "[rc][rc][rc][rc][rc][rc][rc]". lgth of noughts is 7. gpos of noughts is 7. rpos of noughts is 1.

understand "0s" and "zeros" and "zeroes" as noughts.

description of the noughts is "One look at them makes you feel like solving silly word puzzles, shooting stuff and chewing bubble gum.[paragraph break]There is no bubble gum in this game."

to say nau-nou:
	if the player's command matches "naughts":
		say ". You also remember some nonsense about there being two ways to spell naughts/noughts";

the shotgun is a container. understand "shot/ gun" as shotgun when shotgun is visible.

the shotgun can be loaded. the shotgun is not loaded.

description of shotgun is "It's pretty dingy but still intimidating. For whatever reason, it's six-barreled at the muzzle end[if shotgun is not loaded]. Not loaded, though[otherwise]. It's loaded[end if]."

the muzzle is part of the shotgun. description of muzzle is "Six-barreled. Weird."

chapter fireing

fireing is an action applying to one thing.

understand the command "fire [something]" as something new.

understand "fire [something] at [something]" as a mistake ("You can either FIRE the gun, or you can SHOOT the whatever you want.") when player has shotgun

understand "shoot [something] with [something]" as a mistake ("You can either FIRE the gun, or you can SHOOT the whatever you want. No need for a second noun.") when player has shotgun

understand "fire [something]" as fireing.

carry out fireing:
	if noun is missile:
		if missile is visible:
			if panel is visible:
				if panel is not part of silo:
					say "Maybe if you plugged that panel in." instead;
				otherwise:
					say "That panel probably holds the key to firing the missile." instead;
			say "You probably need some circuitry or machinery for that." instead;
	if noun is shotgun:
		try shooting noun instead;
	say "No, you were the one who got fired[if player has shotgun]. You may want to SHOOT something, though[end if]. And not just because you got fired. I hope not.";
	the rule succeeds;

part shades-dashes

the dashes are a plural-named thing in the shell. the dashes are flippable. rgtext of dashes is "[rc][rc][rc][rc][gc][gc]". lgth of dashes is 6. gpos of dashes is 3. rpos of dashes is 6.

the pattern is part of the dashes.

description of pattern is "Two upside-down semicircles next to each other, engraved in each of the dashes. Nothing too deep, just an imprint."

understand "semicircles" and "imprint" as pattern.

check taking the dashes:
	say "Too many of them. They'd spill all over." instead;

description of dashes is "Each seems to have a pattern carved into it."

the shades are a plural-named disguise-piece. the elevation of the shades is 2.

understand "shade" and "glasses" as shades when shades are visible.

description of shades is "They're not dark enough to prevent you seeing anything. You suspect they'd fit your face just fine. They also have a small hook between the lens bits[if shades are part of beard]. A fake beard hangs snugly over the stems[end if]."

to say nose-shades:
	if shades are part of nose:
		say ", which link the shades with a nose";
	if shades are part of beard and nose is part of beard:
		say ", which link the shades with a nose";
		
part nose-ones

the ones are a plural-named thing in the bucket. the ones are flippable. rgtext of ones is "[rc][rc][rc][rc]". lgth of ones is 4. gpos of ones is 2. rpos of ones is 3.

understand "1s" and "one" as ones.

check taking the ones:
	say "Too many of them. They'd spill all over." instead;

understand "remove [something] from [something]" as a mistake("To keep things simple, I recommend TAKE as a synonym. REMOVE is also okay, but it defaults to carried or worn items, and I know no way around that.")

the nose is a disguise-piece. the elevation of the nose is 1.

to say nose-desc:
	if number of things that are part of nose is 0:
		say "nose";
	otherwise:
		say "nose (with shades attached to it)";

the printed name of the nose is "[nose-desc]".

description of nose is "It would probably slip off your own nose just fine with the proper support. Thankfully, it's not sealed, and it looks like it can be hooked up to something[if nose is part of beard]. It's currently hooked up to a fake beard[else if shades are part of nose]. It's currently hooked up to the shades[end if]."

description of the ones is "They easily pass the smell test for being ones."

part bread-beard

the bread is edible. the bread is flippable.

before taking a disguise-piece:
	if noun is part of a disguise-piece:
		say "No, why break up what you built?" instead;

the beard is a disguise-piece. the beard is edible. the elevation of the beard is 0.

the printed name of the beard is "[beard-desc]".

to say beard-desc:
	if number of things that are part of the beard is 0:
		say "beard";
	otherwise if number of things that are part of beard is 1:
		say "beard (with [a random thing that is part of the beard] hooked to it)";
	otherwise:
		say "full-blown beard disguise";

description of the beard is "[if nose is part of beard and shades are part of beard]Your beard is hooked to a nose and shades[otherwise if shades are part of beard]Your beard is hooked to some shades[otherwise if nose is part of beard]Your beard is hooked to a nose[otherwise]It's a deplorably generic fake beard. It has a couple little bits that hook, one at the center and one at the edges, but they're not in use[end if]."

to decide what thing is the current getup:
	if nose is part of the beard or shades are part of the beard:
		decide on beard;
	if shades are part of nose:
		decide on nose;
	if shades are visible:
		decide on shades;
	if nose is visible:
		decide on nose;
	decide on beard.

the current getup is a thing that varies.

understand "disguise" as the beard.
understand "disguise" as the nose when beard is not visible or shades are part of the nose.
understand "disguise" as the shades when nose is not visible and beard is not visible.

instead of eating bread:
	say "It's not as disgusting as the meat you've seen, but you're not hungry now. There must be some other use.";
	
instead of eating beard:
	say "It's less appetizing as a beard than as bread. And if there WERE anything edible in it, ... eww.";

part smilies-missile

description of smilies is "On reflection, the smilies are the centerpiece of the poem. They're not purely happy, but not sarcastic or violent, either. They've had enough and are willing to launch, or be launched, into some great cause, exploding in the face of elitism and excess.". legth of smilies is 7. gpos of smilies is 2. rpos of smilies is 6. rgtext of smilies is "[rc][rc][rc][rc][gc][rc][rc]".

the missile is a thing.

the missile is pushable between rooms.

instead of taking the missile:
	say "Too heavy. You can drag it where you need to, though. Maybe even push it. It's cylindrical. It can roll.";

description of missile is "It's covered with smiley faces and surprisingly nerfy to the touch. You haven't experienced any ill effects when you're around it, so maybe it isn't all that violent or deadly."

part liches-chisel

Include (-
	has transparent animate
-) when defining liches.

The liches are scenery in coe. The liches are flippable. rgtext of liches is "[rc][rc][rc][rc][gc][rc]". lgth of liches is 6. gpos of liches is 3. rpos of liches is 1.

understand "lich" as liches

after doing something with liches:
	set the pronoun it to liches;
	continue the action;

description of liches is "They seem to be moaning about being reduced to a tool of greater evil."

check listening when player is in coe:
	if liches are visible:
		say "The liches are whining about their former lives. 'Tried to...I rotted...' One was a sculpted abstract art worth far less than his fees, one cheated others for any amount of money he felt, and one was an obsessive bodybuilder. All forgot about their spiritual life[if banshee is visible]. They almost drown out the banshee's screams[end if]." instead;
	if banshee is visible:
		say "The banshee is still moaning. It'd be nice to get rid of him, but he doesn't seem threatening." instead;

The chisel is a thing. it is not fixed in place.

description of the chisel is "It seems like it can PEEL all sorts of things you may not want to handle."

check going north in coe:
	if liches are in coe:
		say "The liches won't let you past." instead;
	if drapes are in coe:
		say "The drapes are too thick." instead;
	if spread is in coe and spread is not cut:
		say "The spread still blocks you[if player does not have sliver]. And don't think the chisel could quite, uh, cut it at cutting it[end if]." instead;
	if player does not have maps and player does not have spam:
		say "You're worried about that notice. What if you get lost?" instead;
	if player does not have shotgun:
		say "The enemies, whatever they are, must be close, now. You're worried you have no way to protect yourself." instead;
	if shotgun is not loaded:
		say "You don't know what's to the north. You remember something about a lupine lineup, but that unloaded shotgun won't be enough[if silver is off-stage], and you don't have anything that works against them[otherwise], but maybe you can refashion what you have into a different sort of weapon[end if]." instead;

book reg-verbs

part going

roomgoing is an action applying to one thing.

understand the command "go [any room]" as something new.

understand "go [any room]" as roomgoing.

carry out roomgoing:
	let myrm be a location;
	if noun is air and map region of location of player is sortie:
		let myrm be roomroom;
	if noun is not a room:
		let myrm be the location of noun;
	otherwise:
		let myrm be noun;
	if myrm is location of player:
		say "You're already here. Or there. The precise grammar's neither here nor there." instead;
	if map region of myrm is not map region of location of player:
		say "That'd be hyperwarping to another region. Sorry." instead;
	if myrm is subsite:
		say "Goodness, no." instead;
	if myrm is dy:
		say "Wouldn't be as fun second time around." instead;
	if myrm is the nick:
		say "No, you don't want to go back there." instead;
	if myrm is unvisited:
		say "You haven't gotten there yet." instead;
	if map region of player is resort:
		if red bull burdell is visible:
			say "He's saying GET OUT just to be obnoxious and intimidating." instead;
		say "Gotta keep going forward.";
	say "Zooomp...[line break]";
	move player to myrm;

part examining

check examining (this is the now-examined rule):
	if noun is a thing:
		now noun is examined;

part putting into gun

loading is an action applying to one thing.

understand the command "load [something]" as something new.
understand "load [something]" as loading.

carry out loading:
	if noun is not shotgun:
		say "That's not something you can load." instead;
	if player does not have silver:
		say "Nothing to load it with." instead;
	try inserting silver into shotgun instead;

understand the command "load [something] in/into/inside [something]" as something new.

understand "load [something] in/into/inside [something]" as inserting it into.

does the player mean inserting the silver into the shotgun: it is likely.

check inserting it into (this is the forest-insert rule):
	if second noun is sandwich:
		if noun is cow liver or noun is chicken liver:
			say "That might only make the sandwich grosser, and besides, the bread's already stuck to the meat." instead;
		otherwise:
			say "The sandwich won't be helped by adding anything." instead;
	if noun is sliver:
		if second noun is shotgun:
			say "It is about the right shape, but it doesn't seem quite molded into the right material." instead;
	if noun is silver:
		if second noun is shotgun:
			say "You hear some melodramatic nonsense in your head about a dreamer rearmed. More practically, you're ready for a firefight.";
			now shotgun is loaded;
			now silver is in lalaland instead;
	if second noun is shotgun:
		if shotgun is loaded:
			say "Already loaded." instead;
		say "You'll need to put a bullet in the shotgun. That's not a bullet." instead;

part cutwithing

cutwithing is an action applying to two things.

understand the command "cut [something] with [something]" as something new.
understand "cut [something] with [something]" as cutwithing.

check cutwithing:
	if second noun is sword:
		if noun is beats or noun is beast:
			try attacking noun instead;
		if noun is mattress:
			if bump is part of mattress:
				say "You don't need to cut the mattress to find what's in there." instead;
			otherwise:
				say "There's not much left in there." instead;
		say "The sword is probably for some big climactic fight. Don't go waving it around." instead;
	if second noun is not sliver:
		say "That's not sharp enough to cut." instead;
	otherwise:
		try cutting noun;
	
part cuting

before cutting:
	if player does not have sliver:
		say "You have nothing sharp enough to cut." instead;
	if noun is drapes:
		say "They're too thick. Even if you were able to cut them top-to-bottom, it'd be tough for them to unfold." instead;
	if noun is spread:
		if spread is cut:
			say "You've done what you need." instead;
		otherwise:
			say "The spread falls off to the side, revealing a doorway.";
			now spread is cut;
			now doorway is in coe instead;
	if noun is wolves:
		say "You're way outnumbered. And you have a better weapon." instead;
	if noun is instructional tag:
		if player has sliver:
			ask-to-cut instead;
	if noun is player:
		say "Don't do that, now." instead;
	say "Useless vandalism." instead;

to ask-to-cut:
	say "You could cut the tag off the gadget with that, but this might break the gadget. Are you sure?";
	if the player consents:
		now printed name of tagged gadget is "a gadget";
		now tagged gadget is broken;
		say "With a sad BOOOOP, the gadget shuts down. The tag flutters off. Hooray for adding to the challenge!";
		now instructional tag is in lalaland;
		now undo-code is 3;
	otherwise:
		say "Good choice.";

the doorway is scenery.

understand "door" as doorway when doorway is visible.

description of doorway is "It's pretty imposing. That spread was pretty big."

instead of opening doorway:
	try going north;
	
instead of entering doorway:
	try going north;
	
instead of going inside in coe:
	try going north;
	
part scoring

to decide which number is player-rank:
	let temp-rank be 3;
	if dy is not visited:
		decide on 1;
	if notices section is not visited:
		decide on 2;
	if trips strip is not visited:
		decide on 3;
	if two-of-three is true:
		if forest-sc is visited or underside is visited or ufc is visited:
			increment temp-rank;
	decrement temp-rank;
	increase temp-rank by number of solved regions;
	if map region of location of player is roster:
		increment temp-rank;
	decide on temp-rank.

table of ranks
my-rank
"me, a nervy everyman"
"reduced deducer"
"stunted student"
"benigner beginner"
"solider soldier"
"a mature amateur"
"new hero now here"
"a smart gamin' anagrammist"

to say decide-rank:
	choose row player-rank in table of ranks;
	say "[my-rank entry]";

check requesting the score for the first time:
	say "Tip, son: points are worthless as Pintos. But since you asked..."

carry out requesting the score:
	if map region of location of player is nothing:
		say "[bug-report]: This location needs a region." instead;
	otherwise:
		say "You currently have [cur-score of map region of location of player] out of [max-score of map region of location of player] total points for the [map region of location of player] region.";
	repeat with Q running through regions:
		if number of visited rooms in Q > 0:
			if map region of location of player is not Q:
				say "You have scored [cur-score of Q] out of [max-score of Q] total points for the [Q] region.";
	if there is a solved region:
		say "[line break]Won now: the [list of solved regions] region[if number of solved regions > 1]s[end if]";
	otherwise:
		say "[line break]You haven't solved any regions yet";
	say ". You have the rank of [bold type][decide-rank][roman type].";
	if debug-state is true:
		say "See the total scores possible?";
		if the player consents:
			repeat with Q running through regions:
				if max-score of Q > 0:
					say "[Q] [max-score of Q][line break]";
	the rule succeeds;

part walkthroughing

walkthroughing is an action out of world.

understand the command "walkthrough" as something new.

understand "walkthrough" as walkthroughing.

carry out walkthroughing:
	say "You will want to visit the walkthrough.txt file that comes with the game. Alternatively, for a context-specific hint, type HINT.";
	the rule succeeds;

part swearing

instead of swearing obscenely:
	say "[one of]No [']scusing on cussing[or]Swearing? [']s wearing[cycling][plus-check]."
	
to say plus-check:
	if notices section is visited:
		say "[one of]. Plus, I checked. Or had someone more vulgar than me check. Nothing[or][stopping]";
	
instead of swearing mildly:
	if the player's command matches the text "drat":
		say "You'd get a dart in the neck for such talk.";
	otherwise:
		try swearing obscenely;

part wearing (silent rut)

the disguise-piece creation rule is listed before the can't put onto what's not a supporter rule in the check putting it on rules.

the disguise-piece conglomeration rule is listed before the can't put what's not held rule in the check putting it on rules.

the tie disguises rule is listed before the block tying rule in the check tying it to rules.

understand the command "hook" as "tie".

check tying it to (this is the tie disguises rule):
	if noun is a disguise-piece and second noun is a disguise-piece:
		ignore the block tying rule;
		try putting the noun on the second noun instead;

[note try all combinations]

[alternate possibility is to do an increment sort of thing]
[then the disguise-piece-piece with the greatest weight, everything is a part of it]
[needed for descripton]

check putting it on(this is the disguise-piece conglomeration rule):
	if noun is a disguise-piece:
		[if noun is part of nose:
			try putting nose on second noun instead;]
		let A be a random thing incorporating noun;
		if A is nothing:
			continue the action;
		otherwise:
			try putting A on second noun instead;
		if player does not have A:
			try taking A;
		do nothing instead;

to say you-int:
	say "You've got an interesting contraption, but it might not be enough to get past the gate."
	
before wearing a disguise-piece:
	if noun is part of the beard:
		say "(Well, the beard.)[line break]";
		try wearing the beard instead;
	if noun is part of the nose:
		try wearing the nose instead;
		
check wearing the shades:
	if shades are not part of anything:
		say "You put them on, briefly feel cool, and take them off again. You can't help feeling they might go better with something else." instead;
	if shades are part of the nose:
		say "[you-int]" instead;
	if nose is not part of the beard:
		say "[you-int]" instead;

check wearing the nose:
	if shades are part of the nose:
		say "[you-int]" instead;
	otherwise:
		say "The plastic nose just slips off yours." instead;

check wearing the beard:
	if shades are not part of the beard:
		say "It would just slip off your face as-is[if nose is part of the beard], even hooked up to the nose[end if]. Now what could hang on your face? And stay there?" instead;
	if nose is not part of the beard:
		say "[you-int]" instead;

report wearing the beard:
	if player wore the beard:
		say "You adjust the disguise you're already wearing." instead;
	say "It's suitably silly to fit in with the folks passing through the turnstile, but different enough you're not conforming.";
	the rule succeeds;
	
check putting it on(this is the disguise-piece creation rule):
	if noun is a disguise-piece and second noun is a disguise-piece:
		if elevation of second noun > elevation of noun:
			try putting second noun on noun instead;
		say "The [noun] hook[if noun is not shades]s[end if] over the [second noun] easily and naturally.";
		now noun is part of second noun;
		if shades are part of nose and nose is part of beard:
			now shades are part of beard;
		if nose is part of beard and shades are part of beard:
			say "[line break]You've made a full disguise now!";
			now beard is wearable;
		otherwise:
			do nothing;
		do nothing instead;
	if noun is a disguise-piece and second noun is not a disguise-piece:
		say "Only one of those would make part of a clever, tricky disguise." instead;
	if second noun is a disguise-piece and noun is not a disguise-piece:
		say "Only one of those would make part of a clever, tricky disguise." instead;

part waiting

to say die-die-die:
	say "Before collapsing from the cold, you fire the shotgun in desperation. Six missiles fly out, and they fall just after you do! Old Man Almond receives news of your death, and the forts are not frozen by the time he gets there. The good news: the next adventurer will be able to use your gadget. The bad? Er...";
	end the story;

instead of waiting:
	if player is in frost forts:
		say "[one of]Looks like you have a Mexican standoff, if Mexico were in Greenland.[or]The wolves get antsy. I mean, they don't turn into ants. You are getting colder.[or]Your blood runs cold as the wolves inch closer. These events aren't totally causally related.[or]If you don't do something cold-blooded, you'll be too cold-blooded to do anything.[or][die-die-die][stopping]" instead;
	if player is in roman manor:
		say "Well, you sort of almost have it, but you want to (not) do something with the exits." instead;
	if player is in dy and goat is not visible:
		say "There's something important to do here. But what?" instead;
	say "[one of]Time passes... [no line break][or][stopping]";
	say "[one of]A mist seeps...[or]I see stamps...[or]Me, sit? Passe![or]Sit, see maps...[at random]";

book senses

part smelling

the block smelling rule is not listed in any rulebook.

before smelling the player:
	say "You remembered your deodorant this morning. Yay, you." instead;

check smelling in Trips Strip:
	say "There's a mixture of smells coming from the shops. You could probably go up to one and examine it." instead;

check smelling in dy:
	if noun is not toga and toga is visible:
		say "Something here smells vaguely like a barn." instead;

check smelling the toga:
	say "Woo. Stinky. You can't quite identify which animal, being a city slicker." instead;

check smelling in sf:
	now smelled-sf is true;
	say "You feel a breeze from the [rdir], so that's the direction the smell must be from." instead;
	
the odor is scenery. it is in sf.

understand "smell" and "breeze" as odor.

instead of taking odor:
	say "Impossible."

instead of examining odor:
	try smelling instead;

check smelling in rf:
	if rut-door is not in rf:
		say "It's an odor of new places and adventure and puzzles and how to get there." instead;
	otherwise:
		say "The door is of some odd wood you don't recognize." instead;

check going nowhere in rf:
	if odor is visible:
		say "That smell seems to be drawing you to new places. If you left it, the trees would not steer you to a reset here.";
	otherwise:
		say "And maybe lose where this door you made is? Bad idea.";
	the rule succeeds;

does the player mean smelling the garbage: it is likely.

instead of entering mattress:
	try sleeping instead;

carry out smelling:
	if noun is clover:
		if player has clover:
			say "Still got that planty smell, though it's more the touch that's important." instead;
		otherwise:
			say "It's too close to the ground, and the smell doesn't matter." instead;
	if noun is cask:
		if cask is closed:
			say "Faintly alcoholic. Might be honey, too. Not very strong, though." instead;
		otherwise:
			if oils are in cask:
				try smelling oils instead;
			otherwise:
				say "The cask smells vaguely of the honey-ish alcohol that was in it." instead;
	if noun is night thing:
		say "The one good thing about its roaring and flailing is, you can't get close enough to REALLY smell it." instead;
	if noun is oils:
		say "The oils smell like nothing else and everything else at the same time. They're sacred, you know." instead;
	if noun is drainage:
		say "It smells like the opposite of a garden. Powerful." instead;
	if noun is gardenia:
		say "Smells wonderful, like cooking." instead;
	if noun is chicken liver or noun is cow liver:
		say "Thankfully, it's been frozen, so it smells vaguely like freezer-burn." instead;
	if noun is taco:
		say "Awesome. You take righteous pride in your culinary ability." instead;
	if noun is perfume:
		say "Eurgh." instead;
	if noun is wolves:
		say "You wouldn't get near enough." instead;
	if noun is hotsauce:
		say "Nostril-clearing." instead;
	if noun is lettuce:
		say "Pleasantly inoffensive." instead;
	if noun is steak:
		say "Wonderful." instead;
	if noun is bread:
		say "Not like fresh bread--more store-bought." instead;
	if noun is sandwich:
		say "Frozen enough not to smell. Thankfully." instead;
	if noun is spam:
		say "It reminds you of long-gone days you liked its smell. Before having a taste." instead;
	if noun is tomato:
		say "Could be worse, like rotting meat. You're worried it might be very rotten on the inside." instead;
	if noun is motto:
		say "Okay, it doesn't PHYSICALLY stink." instead;
	if noun is toh:
		say "You've smelled worse, but the night thing's [if night thing is not visible]lingering stench is [end if]pretty bad." instead;
	if noun is aya:
		try smelling garbage instead;
	if noun is slough:
		say "I suppose it's a smell of nearby evil or something." instead;
	if noun is frost forts:
		say "Too cold to smell, really." instead;
	if location is kitchen:
		say "The smell of delicious fried comfort foods. You'll probably not want to make anything too nutritious here." instead;
	say "Nothing unusual." instead;

part listening

the block listening rule is not listed in any rulebook.

check listening:
	if location of player is spis:
		say "The nerds are nattering about something or other[one of]. Stupid people who don't vote, and who vote stupid when they do[or]. Some dolt who got TOLD[or]. How text adventures are for losers stuck in the eighties[or]. How nothing's stopping dumb people from getting on the internet and building the same soundproof bunker they have[or][stopping]." instead;
	if location of player is Abyss:
		if beats are visible:
			say "The beats are worse than ever[if siren is visible], accompanied by that siren[end if]. You have come to the heart of what has dragged the city to a halt." instead;
		otherwise:
			say "The beats are gone, replaced with a beast's roaring[if siren is visible], which clashes nastily with the siren[end if]." instead;
	if map region of location of player is metros:
		if player is in aya:
			if words are visible:
				if player has heaths:
					say "'That's just words. Shift them to a weapon, and you'll have something!'" instead;
				if player has sheath:
					say "'Talk is cheap! Now, weapons, there's something valuable!'" instead;
				if player has begonias or player has noise bag:
					if words are not in bag:
						say "'Put a bag over it!' you hear." instead;
					otherwise:
						say "The semi-sheltered alley offers some respite from the beats, especially with the words gone. But not enough." instead;
				say "The words drown out the beats for the moment. They're sharp words, attacking words, hard to shield yourself from, but they're local enough they can't be twisted." instead;
		if player is in red camp:
			say "The sharp harps are out of tune but a change of pace from the beats. At least." instead;
		if night thing is visible:
			say "The groans from the Night Thing's organs are blocking out the usual thumping. For now." instead;
		if player is in cordoned red condo:
			say "The conversation changes subject rapidly and randomly. [one of]'Apt, Pat.' 'Mart's Smart!'[or]People sound grateful the condo serves no cod.[or]Some posher hopers think domestics should cost dimes.[or]'It can't be a real toxin. It gives relaxation!'.[or]'Gotwin Towing charges poor people the same as us rich people. And yet they still whine!'.[or]Someone decries a reckless clerkess who is lucky to have a job.[or]Obvious points are repeatedly flayed deafly[or]They discuss prices that make you think CRIPES.[or]Someone agrees with Kramer's remarks entirely, whatever they are.[or]Someone babbles about a new frozen fish that filets itself.[or]A badly dressed fellow deplores the skankiness of snakeskins.[or]Nothing, for a moment. Everyone gives you a look of skilled dislike.[or]You learn they're scared of lunatics that don't cut nails.[or]Parleys about parsley that deserve no replays.[or]The conversationalists decry conservationalists.[or]Someone asks if they can arrest that starer.[or]Apparently next week's party is with Bub and Russ in the suburbs.[or]Couples chatter about not wanting that--NEW GUY up close.[or]Someone complains they've never had a plainer praline than THIS.[or]Someone complains about the rudest duster they ever employed.[or]People complain how tacky it is a restaurant like Ascot Coast now serves tacos.[or]Someone wonders aloud why traders should have to deal with retards.[at random]" instead;
		say "That infernal thumping. Even when you close your ears." instead;
	if player is in Busiest Subsite:
		say "[one of]A passing conversation among the rushing and yelling for others to move aside: 'I bet they take the hotel fees out of our severance.'[paragraph break]'THIS.'[paragraph break]You've always hated when people say that, but it feels nearly profane today.[or]'That lecture'll fix your helpless spells, eh?' someone walking by asks.[or]'Boy, I need to re-care about my career,' someone actually says with a straight face.[cycling]" instead;
	if player is in dy:
		if goat is visible:
			say "The goat snores slightly as he half-naps, but that's it." instead;
		otherwise:
			say "The toga flutters in some wind you can't quite feel." instead;
	if player is in notices:
		if gateman is visible:
			say "'Y'can't have anything worth listening to without anything to ask me about!' says Old Man Almond." instead;
	if player is in sorting area:
		say "The notes stone contains written, not musical, notes." instead;
	if player is in moor:
		if anapest is visible:
			try listening to the poetry instead;
		if peasant is visible:
			say "With less, he whistles, eh?" instead;
	if player is in olw:
		say "You hear something from above the [p-s] to the east, but you'll need to get closer to make out details." instead;
	if player is in kitchen:
		say "[one of]A song with the chorus 'He oft partook of the okra pot.' The tune and the thought of okra make you gag, even when you realize it's a novelty song and not serious.[or]It's actually sort of catchy, if you think about it.[or][no-you]ou hate the stupid okra song, again. You need to get cooking on something else.[stopping]" instead;
	if player is in slough:
		say "[one of]You can't hear the dog any more. You're not sure if you want to[or]You realize you'll be freaked out whether or not you can hear the dog[stopping]." instead;
	if dollhouse is visible:
		if static is visible:
			say "You hear crackling coming from the dollhouse." instead;
		otherwise:
			say "Each attic is tacit now." instead;
	say "[one of]Silent[or]Quite quiet[cycling]." instead;

to say no-you:
	say "[one of]No, y[or]Y[stopping]";

part abouting

abouting is an action out of world.

understand the command "about" as something new.

understand "about" as abouting.

carry out abouting:
	say "This game is about[if cur-score of Intro is 0], well, I don't want to spoil that yet until you get your first point, which I hope isn't too hard. It's a bit different from other text adventures, being a wordplay game, but the hope is that the right sort of (one-word) command to solve certain puzzles is clued in more than one way.[paragraph break]This is much simpler than it sounds, and I think/hope I've made efforts to make it less guess-the-verb. Or verbish.[paragraph break]You can see tester credits with CREDITS[otherwise] a certain sort of wordplay. I think you know it, now that your score is positive?[paragraph break]It's a tribute to Infocom's [italic type]Nord and Bert Couldn't Make Head or Tail of It[roman type] and something they didn't do. Probably not due to lack of creativity--either they thought something like this would be a mess, or maybe it was just due to technical or disk space limitations.[paragraph break]I hope it is the latter, and it's enough to be its own game, and I hope you find so too.[paragraph break]As hard as my testers worked, the rate they turned up bugs made me realize there're probably some lurking around, and in fact, I made some touch-ups after receiving all transcripts that caused bugs in the final release which are all my fault.[paragraph break]Send reports--especially ones with transcripts--to schultz.andrew@sbcglobal.net.[paragraph break]If you've never made a transcript before, type TRANSCRIPT and save to a text file and start any comments with an asterisk. Comments can be 'this is broken' or 'this was hinted poorly.' Authors enjoy getting transcripts--I mean, after we realize what we did wrong, we can learn a lot.[paragraph break]People who found bugs (and they found ~700) before this was released for IFComp--and people who sent transcripts, suggestions, etc. after--are in CREDITS[end if][i-love-transcripts].[paragraph break]Also, you can report issues at http://code.google.com/p/shuffling-around/issues/list.";
	the rule succeeds;

to say i-love-transcripts:
	say "[one of]. Also, I am (still) very interested in transcripts to tweak this game. If you've never made a transcript, type TRANSCRIPT and you can pick the file to save to. Then send the transcript to schultz.andrew@sbcglobal.net with my thanks[or][stopping]";

part creditsing

creditsing is an action out of world.

understand the command "credits" and "credit" as something new.

understand "credits" and "credit" as creditsing.

carry out creditsing:
	if cur-score of Intro is 0:
		say "There is a list of websites I would like to credit. But it might spoil things before you score anything. So I'll just list beta-testers and general help.[paragraph break]";
	say "[if cur-score of Intro > 0]Tester Street residents ('no on tiredness:')[paragraph break][end if]Adri, Anthony Hope, DJ Hastings, Gavin Myers-Leman, Hulk Handsome ([if cur-score of intro > 0]who nicely handles hokum like huge bars and bear hugs in his own IFComp 2012 game[otherwise][italic type]not spoiled til you score a point[roman type][end if],) Joey Jones, John Nitchals, Paul Lee, Robert Patten and Tomie Campf, in alphabetical first-name order. They found 700+ bugs.[paragraph break]Source (or cues) : Heartless Zombie, who found a lot of bugs AND helped tighten up my post-release code so certain classes of bugs comp releases CAN'T happen any more.[paragraph break]Storied Editors (post-release fixes) also include: Jason Orendorff, Matt Weiner. Reviews on the Internet also helped me fix things--Carl Muckenhoupt and Simon Carless discovered an unwinnable state, and I was able to to tighten up initial cluing based on another.[paragraph break]It must be noted that several bugs that popped up in the release version were due to me trying to slip in one more small thing without adequate re-testing. If there is anything obvious (and there was, in the initial release,) it is my fault and not theirs.[paragraph break]John Nitchals made the cover art. Cover image is a derivative of 'LED scrolling nametags' (http://www.flickr.com/photos/clanlife/385380701/) by Phil Campbell, used under a Creative Commons Attribution 3.0 Unported (CC BY 3.0) license: http://creativecommons.org/licenses/by/3.0/[paragraph break]Contact me with suggestions (technical or aesthetic) at schultz.andrew@sbcglobal.net, and you can join these worthy people above.[paragraph break]Also, thanks to the folks at intfiction.org who helped me code things.[paragraph break]Finally, type SITES for a list of sites that helped[if cur-score of intro is 0], which will totally spoil things right now[end if]."

chapter sitesing

sitesing is an action out of world.

understand the command "sites" as something new.

understand "sites" as sitesing.

carry out sitesing:
	if cur-score of intro is 0:
		say "Last chance spoiler warning. Sure you want to look?";
		if the player consents:
			do nothing;
		otherwise:
			say "Ok. Back to the game." instead;
	say "http://wordsmith.org/anagram/ -- Anu Garg's Rearrangement Server at wordsmith.org was invaluable to me. Hard to believe I've known about it for fifteen years. You've probably seen it before, too, but it's the big one for all this nonsense.[line break]http://www.anagrammy.com/anagrams/faq6.html[line break]http://www.english-for-students.com/One-Word-Anagrams.html[line break]http://www.enchantedlearning.com/english/anagram/ was quite nice for common anagrams by subject.[line break]http://www2.vo.lu/homepages/phahn/anagrams/oneword.htm[line break]http://www.ssynth.co.uk/~gay/anagram.html[line break]http://www.wellho.net/resources/ex.php4?item=p669/anagram (basic succinct PERL anagram finder)[line break]http://boards.straightdope.com/sdmb/archive/index.php/t-291149.html for specific words and also ideas how to script more complex stuff[line break]http://www.rinkworks.com/words/oddities.shtml[line break]http://www.sporcle.com/ had many puzzles that helped me determine what was fair and what wasn't[line break]http://jamesgart.com/anagram/[line break]PERL for letting me parse through word-to-word anagrams and also helping me break one promising word into two.[paragraph break]I'd be interested if someone from France can ref these sites to make a similar game, someone from Spain could take pains, or a German manager could do so too. It'd be interesting to see.";
	the rule succeeds; [forgot where I found the names from]

book stubs

section regional manip

to solve-region (b - a region):
	follow the notify score changes rule;
	say "[line break]Congratulations! [b] node: done![line break]";
	now b is solved;
	if b is forest:
		now forest-sc is checkedoff;
	if b is sortie:
		now sortie-x is checkedoff;
	if b is metros:
		now metros-x is checkedoff;
	item-warp;
	wfak;
	move player to trips strip;
	if number of solved regions is 2:
		cabinet-check;

to cabinet-check:
	if player does not have phail phial:
		say "You hear a bouncing and rushing in the distance. The cabinet has somehow followed you. It lurches forward, dropping the phial [if cabinet-bit-me is true]it didn't let you have before[otherwise]you didn't take from it[end if] in your hands and seems almost to bend forward in apology before bouncing off[if player does not have gadget]. The gadget from the cabinet falls and hits your foot, too[end if].";
		if player does not have gadget:
			now player has gadget;
		now player has phail phial;
		continue the action;
	if gadget is in cabinet or gadget is in notices section:
		say "You see the cabinet bouncing off in the distance. You're a bit confused until you se the tagged gadget at your feet. Well, [if number of solved regions is 1]that might help you when you take another shot at things[otherwise]that might make the next bit go quicker[end if].";
		now gadget is in trips strip;

to item-warp:
	let A be the map region of the location of the player;
	d "So far you have [cur-score of A] of [max-score of A] points.[line break]";
	now last-loc of A is location of the player;
	if number of carried not warpable things > 0 or number of worn not warpable things > 0:
		if A is solved:
			if A is intro:
				say "This game just removed an item it should not have. [bug-report]! => ([list of carried not warpable things]) ([list of worn not warpable things]).";
			otherwise: [?? needs testing]
				say "Out of nowhere swoops the [if cabinet is in lalaland]nice bat[otherwise]cabinet[end if]. [run paragraph on][if number of solved regions is 2]You're about to complain, but then you realize it's helping you not carry all that junk around. It sits there, out of reach[else if cabinet is in lalaland]You're still a little spooked, but grateful[otherwise]The cabinet's trying its best to be good and helpful and you wonder if maybe you can do something for it[end if].";
				if cabinet is in notices section:
					now cabinet is in trips strip;
		now all carried not warpable things are in lalaland;
		now all worn not warpable things are in lalaland;
				
[museum muse. Um. Those Extras. Taxers. Tributes I'd distribute]


to say is-it-all:
	if number of solved regions is 3:
		say ". Don't worry about that last one, if you don't want to. Someone apprentice adventurer can clean it up"

to reset-regions:
	repeat through table of lastlocs:
		now the item-list of r entry is {};
		now the worn-list of r entry is {};

section reg-inc

to reg-inc:
	increment the score;
	increment the cur-score of the map region of the location of the player;

part directional cleanup

up-nearby is a truth state that varies.

before going up:
	if player is in olw:
		try going east instead;
	if player is in ett or player is in condo or player is in underside or player is on fuzzy looking wall:
		continue the action;
[	if up-nearby is true:
		say "[if cur-score of Intro is 0]I, uh, lied. There's nowhere you can go up in the game. Or nearby. But that was a clue[otherwise]Well, since you've scored a point, you might guess why 'I'd go nearby or up' is a hint[end if].";
		now up-nearby is false instead;]
	say "You have no boost boots for dances to ascend. And most of the time, if you need to go up, there'll be something clear to climb. Hopefully." instead;
	
before going down in busiest subsite:
	say "You're not waiting for the elevator back down. And the only stairs are fire stairs, which will set off an alarm." instead;

before going down:
	if player is in busiest subsite or player is in cordoned red condo:
		continue the action;
	if player is on fuzzy looking wall:
		try getting off the fuzzy looking wall instead;
	if player is in trips strip:
		if sortie-x is visible or metros-x is visible:
			continue the action;
		otherwise:
			say "You can't see any [if sortie is solved or metros is solved]other [end if]way down[if sortie is not solved or metros is not solved][otherwise] any more[end if]." instead;
	say "You see no clear way down. Usually there'll be stairs, if you need to." instead;

part pause-manipulation

chapter remindersing

remindersing is an action out of world.

understand the command "reminders" as something new.

understand "reminders" as remindersing.

reminders is a truth state that varies. reminders is true.

carry out remindersing:
	if reminders is true:
		now reminders is false;
	otherwise:
		now reminders is true;
	the rule succeeds;

chapter wait for any key

first-wfak is a truth state that varies. first-wfak is false.

to wfak:
	if debug-state is false:
		if first-wfak is false:
			say "(this is the first in-game pause[if numsaves > 0] after a saved game[end if]. Whenever the game pauses without a cursor in the future, you need to push any key to continue.)";
			now first-wfak is true;
		wait for any key;
		
numsaves is a number that varies.

report saving the game:
	increment numsaves;
	if reminders is true:
		now first-wfak is false;
	continue the action;

chapter pause the game

to ptg: [possibly deprecated as wait for any key works better or seems to]
	if debug-state is false:
		pause the game;

book irreg-verbs

chapter biteing

biteing is an action applying to one thing.

understand the command "bite [something]" as something new.

understand "bite [something]" as biteing.

carry out biteing:
	if noun is cabinet:
		say "The cabinet doesn't deserve that. Plus, no good surface." instead;
	if noun is nice bat:
		say "It would bite back harder and germier." instead;
	if noun is toe:
		say "It's be hard to get ahold of with your mouth." instead;
	if noun is saltine:
		say "It'll only take one bite to eat.";
	try eating noun instead;

chapter instructionsing

instructionsing is an action out of world.

understand the command "help/info/instructions" as something new.

understand "help" and "info" and "instructions" as instructionsing.

carry out instructionsing:
	say "[if location of player is busiest subsite]There's not much to do here, but EXAMINEing different things may give you a clue what's going on[else if cur-score of intro is 0]If you look around, there should be one word that suggests itself--but is never said[otherwise]What you did with the goat will clue the rest of the game[end if]. This is a part of all the puzzles in the game, but you'll have to do other things, too[one of].[paragraph break]Note: If you want actual hints, you may wish to try HINT or HINTS[or][stopping].";
	the rule succeeds;

chapter knocking

knocking is an action applying to one thing.

understand the command "knock on [something]" as something new.
understand the command "knock on [something]" as something new.

understand "knock [something]" as knocking.
understand "knock [something]" as knocking.

does the player mean knocking the black door: it is very likely;
does the player mean knocking the cabinet: it is very likely;
does the player mean knocking the rut-door: it is very likely;
does the player mean knocking the metallic: it is very likely;
does the player mean knocking the turnstile: it is very likely;
does the player mean knocking a portal:
	if cabinet is visible:
		it is unlikely;
	it is very likely;

carry out knocking:
	if noun is a portal:
		say "Better just to try to enter." instead;
	if noun is cabinet:
		say "That's potentially an act of violence." instead;
	if noun is black door:
		say "[if black door is part of silo]Nobody answers, unsurprisingly[otherwise]How polite! And ineffective[end if]." instead;
	if noun is rut-door:
		say "[if player has bugle]It's a bit awkward knocking with the bugle in your hand[otherwise]You brush against the bulge as you knock[end if]. You get no response." instead;
	if noun is metallic door:
		say "[if player has tulip]You don't need to go back[else if spis is visited]Nah, just walk in[otherwise]Weird. It doesn't make any noise. Or the noise drowns quickly. You suspect the door's intended to keep people out, though[end if]." instead;
	if noun is turnstile:
		say "A good way to scrape your knuckles, seeing how quickly it turns." instead;
	say "Knock, knock. Who's there? Conkk, conkk." instead;
	the rule succeeds;

chapter grinding

grinding is an action applying to one thing.

understand the command "grind [something]" as something new.

understand "grind" as grinding.

carry out grinding:
	if canister is visible:
		try inserting noun into the canister;
	otherwise:
		say "There appears to be nothing that can grind anything here.";
	the rule succeeds.
	
chapter untieing

untieing is an action applying to one thing.

understand the command "untie [something]" as something new.

understand "untie [something]" as untieing.

carry out untieing:
	if noun is heaths or noun is stems:
		say "They're actually not tied, though they seem like they should be. Hmph. Magic." instead;
	if noun is ropes:
		say "You have better things to do than take apart potentially useful ropes. Like getting up that [p-s] over there." instead;
	if noun is swing:
		say "The knots where it is tied to the tree are too high up." instead;
	say "That doesn't seem tied to anything.";
	the rule succeeds.
	
chapter reading

understand the command "read" as something new.
understand "read [something]" as reading.

reading is an action applying to one thing.

check reading (this is the reading is almost examining rule):
	if noun is great grate:
		try examining the branding instead;
	if noun is toga:
		try examining the toga-writing instead;
	if noun is gadget:
		try examining tag instead;
	if noun is gadget-screen:
		try scaning air instead;
	try examining the noun instead;

chapter walkthruing

walkthruing is an action out of world.

understand the command "walkthru" as something new.
understand the command "walkthrough" as something new.

understand "walkthru" as walkthruing.
understand "walkthrough" as walkthruing.

carry out walkthruing:
	say "This game's gblorb file should come packaged with a file called walkthrough.txt. However, there is a risk that will spoil a lot you won't want, especially since the areas are compartmentalized.[paragraph break]You can ask for hints with the HINT command--they should deal with the puzzle currently restricting you the most and may be doled out slowly to avoid spoilers.";
	the rule succeeds;

chapter following

following is an action applying to one thing.

understand the command "follow [something]" as something new.

understand "follow [something]" as following.

carry out following:
	if noun is masses:
		say "You are swept up with the people entering the lecture.";
		try going east instead;
	if noun is above-sign or noun is passage:
		say "You duck inside the passage.";
		try going inside instead;
	if noun is gy or noun is gy2:
		try going north instead;
	if noun is riot or noun is protest:
		say "They aren't convincing enough to make you hate yourself." instead;
	if noun is potters:
		say "Yours is the way of the adventurer, not the artist." instead;
	if noun is the player:
		say "Tautological." instead;
	if noun is the beats:
		if player is in ett:
			try going east instead;
		say "No telling where they're coming from." instead;
	say "That doesn't seem to be going anywhere.";
	the rule succeeds;

chapter playing

playing is an action applying to one thing.

understand the command "play [something]" as something new.
understand "play [something]" as playing.

understand the command "blow[something]" as something new.
understand "blow [something]" as playing.

does the player mean playing the bugle: it is very likely;

carry out playing:
	if noun is not bugle:
		say "That's not an instrument you can play." instead;
	say "The note from the bugle shatters the door--well, the frame remains--then the bugle itself.";
	now rut-opened is true;
	now bugle is in lalaland;
	reg-inc;
	follow the notify score changes rule;
	the rule succeeds;

chapter gunshoting


gunshoting is an action out of world.

understand the command "gunshot" as something new.

understand "gunshot" as gunshoting.

carry out gunshoting:
	if shotgun is visible:
		say "You need to SHOOT the gun[if silver is not visible], and you don't have a bullet yet anyway[end if]." instead;
	if noughts are visible:
		say "You need something more tangible than that. Something that could make a gunshot, perhaps." instead;
	say "[reject]";
	the rule succeeds;

chapter lifting

lifting is an action applying to one thing.

understand the command "lift [something]" as something new.

understand "lift [something]" as lifting.

carry out lifting:
	if noun is great grate or noun is gateway or noun is getaway:
		try opening the noun instead;
	say "(I'm assuming you meant taking it.)";
	the rule succeeds;

chapter unfolding

unfolding is an action applying to one thing.

understand the command "unfold [something]" as something new.

understand "unfold [something]" as unfolding.

carry out unfolding:
	if noun is poem and poem is folded:
		say "The poem deserves obscurity. Folding it out of view was a good first step." instead;
	if noun is drapes:
		say "They are much too heavy and thick." instead;
	say "You don't need to, or you can't." instead;
	the rule succeeds;

chapter pating

pating is an action applying to one thing.

understand the command "stroke/pet/pat [something]" as something new.

understand "stroke [something]" and "pet [something]" and "pat [something]" as pating.

carry out pating:
	if noun is spout:
		say "You sense that you can fill the cask by placing it under the spout. Having faith and all." instead;
	if noun is goat:
		say "The goat bleats and pushes your hand away with its hoof. It's too tired for that nonsense." instead;
	if noun is shrub:
		say "It's a shrub, not the real thing, and it's a bit pointy. Someone took time to create it." instead;
	if noun is wolves:
		say "Man, you must really like dogs, or something. Or trying to break text adventure games." instead;
	say "It's not couth to touch." instead;
	the rule succeeds;

check kissing:
	if noun is goat:
		say "The goat doesn't seem like it appreciates affection.";

chapter danceing

danceing is an action applying to nothing.

understand the command "dance" as something new.

understand "dance" as danceing.

carry out danceing:
	say "You won't get caned for that, but it isn't constructive." instead;

chapter folding

folding is an action applying to one thing.

understand the command "fold [something]" as something new.

understand "fold [something]" as folding.

poem can be folded. poem is not folded.

the creases are part of the poem. understand "diagonals" as creases.

description of creases is "[one of]One down the middle and two diagonally up from its base--two parallel to the middle crease bisect the diagonals.[paragraph break]Oh, forget it, let's try the non-technical explanation. From what you remember from being ten, it'll fold into a plane. Maybe one that can't fly very far, but a plane.[or]Wow! The peasant's poem is multifaceted! It could be a plane if you fold it, too![stopping]"

understand "folded/ paper/ plane/" as poem when poem is folded.

tried-smilies is a truth state that varies. tried-smilies is false.

carry out folding:
	if noun is not poem:
		say "That won't achieve much." instead;
	if smilies are part of the poem:
		now tried-smilies is true;
		say "You're about to fold the poem up, but those smilies launch into a surprising display of emotion. It's an impressive passive resistance--they do it all without any violence." instead;
	if poem is folded:
		say "It already is." instead;
	now noun is folded;
	say "You fold the paper, and it looks sort of like a paper plane."; [?? smilies clue]
	now printed name of poem is "folded paper plane";
	now creases are in lalaland;
	the rule succeeds;

understand "folded paper plane" as poem when poem is folded.

the panel is a thing.

the wires are part of the panel.

description of wires is "They're not, like, chewed or anything. They seem [if panel is part of silo]not to be breaking anything[otherwise]in decent shape[end if]."

description of panel is "All sorts of capacitors and resistors and such. You [if panel is part of silo]thankfully didn't[otherwise]probably don't[end if] have to understand that to know where to use it."

chapter filling

filling is an action applying to one thing.

understand the command "fill [something]" as something new.

understand "fill [something]" as filling.

the oils are a plural-named thing in sacred cedars. the description of oils is "[if oils are not in cask]You try looking up the spout for a glimpse of the oils, but it doesn't work[otherwise]You can't see through the cask--or its narrow hole very well--but you remember the oils being thick and somewhat golden. You trust they are sacred enough for your job, though[end if].". rgtext of oils is "[rc]??[rc]". legth of oils is 4. gpos of oils is 4. rpos of oils is 1. [change to 3 once...]

understand "oil" as oils when oils are visible.

instead of taking oils:
	say "They'd go through your fingers[if player has cask]. You can pour the oils, or fill the cask, instead[end if].";

rule for printing a locale paragraph about the oils:
	now oils are mentioned;

caskfillings is a number that varies. caskfillings is usually 0.

carry out filling:
	if map region of location of player is not sortie:
		say "You don't need that verb for this region. Though, if it makes sense to use it, let me know via email." instead;
	if noun is scraped wall:
		if player has sack:
			if hay is in sack:
				try inserting hay into scraped wall instead;
			if straw is in sack:
				try inserting straw into scraped wall instead;
		say "You don't have any sort of stuffing to put in the wall." instead;
	if noun is fridge:
		say "You don't need a separate storage space." instead;
	if sack is not visible and cask is not visible:
		say "You don't see any containers." instead;
	if noun is sack:
		if straw is in sack or hay is in sack:
			say "You already have something in there." instead;
		if straw is visible:
			try inserting straw into sack instead;
		if cask is visible:
			try inserting hay into sack instead;
	if location of player is not cedars:
		say "You may have to be more specific. Try PUT x IN y instead." instead;
	if noun is sack:
		say "The oils would leak out of it." instead;
	if noun is not cask:
		say "You need something solid to pour the oils into." instead;
	if silo is in moor and soil is in moor:
		say "A voice. It must be Lois. 'WIN,' she whines[one of]. You won't be getting any more oils[or][stopping]." instead;
	if oils are in cask:
		say "The cask is already full." instead;
	say "Gainly laying the cask under the spout, you pour in some oils.";[check for 3rd time]
	increment caskfillings;
	say "A voice calls SO NOW HALLS SHALL SWOON. You're pretty sure it has a particular one in mind. You hope you've gotten what you need.";
	now oils are in cask;
	the rule succeeds;

part odd verbs/actions

chapter badabooming

to badaboom:
	d "Badabooming.";
	if the player's command matches the regular expression " ":
		say "Those words have no power here.";
	otherwise:
		say "That word has no power here.";
	reject the player's command;

chapter sorrying

instead of saying sorry:
	say "I can't do anything with that apology."

chapter xyzzying

xyzzying is an action applying to nothing.

understand the command "xyzzy" as something new.

understand "xyzzy" as xyzzying.

carry out xyzzying:
	say "[one of]What a mess! Hard to do anything with that[if goat is off-stage]. You'll see why[end if].[or]This sort of game, I can only make lame meta-jokes with that[if cur-score of Intro > 0]...you should've figured that out[else], and if you knew why, that'd be sort of a spoiler[end if].[or]Testing all 29 possibilities...just to make sure...nope.[or]Testing the 119 possibilities...including duplicates...[10-wait]nope. You happy now?[or]If you're looking for jokes, that's the best I can do.[stopping]";
	the rule succeeds;
	
to say 10-wait:
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;
	wfak;

chapter shooting

shooting is an action applying to one thing.

understand the command "shoot/fire at/ [something]" as something new.

understand "shoot at [something]" as shooting.
understand "shoot [something]" as shooting.
understand "fire at [something]" as shooting.
understand "fire [something]" as shooting.

does the player mean shooting the wolves when the wolves are visible: it is very likely.

carry out shooting:
	if noun is shotgun:
		say "Mention what you want to shoot with the gun." instead;
	if player does not have shotgun:
		say "No gun." instead;
	if noun is drapes:
		say "[if silver is off-stage]If you had ammo, y[otherwise]Y[end if]ou'd probably want to shoot something more dangerous than the drapes." instead;
	if noun is the player:
		say "Oh dear. Did the puzzles really lead you to that? I'm so sorry[if shotgun is not loaded], and you didn't even get to load the gun[end if]!" instead;
	if shotgun is not loaded:
		say "Your shotgun is not loaded[if wolves are not visible], and there's nothing threatening enough around, yet[end if]." instead;
	if noun is dice:
		say "Gambling? At a dramatic point in the narrative like THIS?!" instead;
	if noun is wolves:
		say "Blam! Some part of the shotgun you couldn't see carves the silver bullet into sixths. Each flies off in a slightly different direction--taking out the wolves. The poison works quickly on them, and the frost forts and iced dice begin to melt quickly. Behind them, you see the strip. You walk back there.[paragraph break]The lupine lineup is no more, thanks to your adept hit in the death pit!";
		reg-inc;
		solve-region Forest;
		now wolves are off-stage instead;
	if noun is vowels:
		say "For someone who I'm guessing likes text adventures, that'd be kind of out of character.[paragraph break]Plus, they aren't really attacking you. Yet." instead;
	say "That'd be wasting firepower.";
	the rule succeeds;

chapter peeling

the dont-peel-taken-chicken rule is listed before the use-cur-liv rule in the does the player mean rulebook.
the dont-peel-taken-cow rule is listed before the use-cur-liv rule in the does the player mean rulebook.

test pe with "purloin chisel/gonear chicken/rules/peel liver/peel liver"

peeling it with is an action applying to two things.

understand the command "pry/scrape/peel [thing] with [thing]" as something new.

understand "peel [thing] with [thing]" as peeling it with.
understand "scrape [thing] with [thing]" as peeling it with.
understand "pry [thing] with [thing]" as peeling it with.

understand "livers" as the plural of glopmeats.

does the player mean doing something with livers when a glopmeat is visible: it is very unlikely.

rule for supplying a missing second noun when peeling: now the second noun is chisel.

does the player mean peeling the chicken liver with when the player has the chicken liver (this is the dont-peel-taken-chicken rule): it is very unlikely.

does the player mean peeling the cur-liv with when the player does not have the cur-liv (this is the peel-untaken-liv rule): it is very likely.

does the player mean peeling the cow liver with when the player has the cow liver (this is the dont-peel-taken-cow rule): it is very unlikely.

does the player mean peeling the cur-liv with when the player has the cur-liv: it is very unlikely.

does the player mean peeling with the chisel: it is likely.

does the player mean doing something with the cur-liv (this is the use-cur-liv rule):
	if the current action is peelxing:
		if cur-liv is not stuck:
			it is very unlikely;
	if the current action is peeling with:
		if cur-liv is not stuck:
			it is very unlikely;
	it is likely.

to decide which glopmeat is the non-cur-liver:
	if cur-liv is chicken liver:
		decide on cow liver;
	decide on chicken liver.

does the player mean inserting into cur-liv: it is unlikely;
does the player mean putting on cur-liv: it is unlikely;
does the player mean tying to cur-liv: it is unlikely;

[does the player mean tying noun to itself: it is very unlikely.
does the player mean putting noun on itself: it is very unlikely.]

check putting a glopmeat on a glopmeat:
	try tying noun to second noun instead;

check inserting a glopmeat into a glopmeat:
	try tying noun to second noun instead;

check tying cow liver to chicken liver:
	try tying chicken liver to cow liver instead;

check tying chicken liver to cow liver:
	say "You can't blend them that thoroughly by yourself. Some device might be more handy." instead;

carry out peeling it with:
	if second noun is not chisel:
		say "Wrong instrument." instead;
	if noun is instructional tag:
		say "That'd be one way to remove it. But it would probably disable the gadget.";
		ask-to-cut instead;
	if noun is not a glopmeat:
		say "That doesn't need to be scraped. Try something stuck to something." instead;
	if noun is not stuck:
		say "You already peeled it off." instead;
	say "Cr-r-r-r-k. The chisel successfully leverages the freezer-burned [noun] from the wall.";
	now noun is not stuck;
	now player has noun;
	the rule succeeds;

peelxing is an action applying to one thing.

understand the command "peel/pry/scrape [thing]" as something new.

understand "peel [thing]" as peelxing.
understand "pry [thing]" as peelxing.
understand "scrape [thing]" as peelxing.

carry out peelxing:
	if noun is barcode:
		say "[dont-peel]" instead;
	if player does not have chisel:
		say "You have nothing to peel anything with." instead;
	try peeling noun with chisel instead;

chapter existing

existing is an action applying to nothing.

understand the command "exist" as something new.

understand "exist" as existing.

carry out existing:
	if player is in Roman Manor:
		say "That is all that's left to do. It's time to become an adventurer emeritus. Perhaps one day you will resume it, but for now, the allures of your laurels are too great. Doing without the outwith, you begin your being binge, mentally writing the happiest epitaphs to your vaunted, er, adventure. Where IS that notepad?";
		wfak;
		say "[line break]I...what? End a pad with a deadpan: TREMENDOUS = ME (ENDS TOUR).";
		reg-inc;
		now resort is solved;
		follow the notify score changes rule;
		now right hand status line is "[cur-score of map region of location of player]/[max-score of map region of location of player]";
		say "[line break]Final rank: a smart gamin['] anagrammist.";
		end the story finally;
	otherwise:
		say "You continue to exist. Perhaps this ability will come in handy some day." instead;

chapter itemdefaulting

itemdefaulting is an action applying to one thing.

understand the command "[something]" as something new.

understand "[something]" as itemdefaulting.

carry out itemdefaulting:
	if noun is a portal and noun is examined:
		try entering noun instead;
	try examining noun instead;

chapter propfliptoing

section basic

propfliptoing is an action applying to one visible thing.

to decide which thing is new form of (a - thing):
	d "Deciding...[list of things which oneway a] - [list of things which anagram a].";
	repeat with Q running through things which oneway a:
		if Q is not nothing:
			if Q is visible:
				decide on Q;
	repeat with Q running through things which anagram a:
		if Q is not nothing:
			if Q is visible:
				decide on Q;
	decide on a random thing which anagrams a;

a thing can be flipped-yet. a thing is usually not flipped-yet.

carry out propfliptoing:
	let Q be new form of noun;
	if Q is nothing:
		badaboom instead;
	if Q is not flipped-yet:
		if Q is not flippable:
			say "You can't seem to reverse what you did." instead;
		now Q is flipped-yet;
		reg-inc;
	d "Ok, you typed [noun] with the thing you wanted to change presumably being [Q].[line break]";
	if Q is a sto:
		if location is Trips Strip:
			if Q is not visible:
				try entering noun instead;
	if noun is visible:
		try examining noun instead;
	if nothing is Q:
		if noun is visible:
			try examining noun instead;
			[say "[if noun is plural-named]They seem[otherwise]It seems[end if] unaffected." instead;]
		otherwise:
			reject the player's command;
	if Q is visible:
		d "Check if [Q] is flippable.[line break]";
		check-reject Q;
		if Q is flippable:
			let flipcheck be false;
			repeat through the table of anagrams:
				if the-from entry is Q and the-to entry is noun:
					say "[from-msg entry][line break]";
					if force-take entry is true:
						now player carries the-to entry;
					now flipcheck is true;
			if flipcheck is false:
				say "Bam! The [Q] changes into the [printed name of noun]. [bug-report]";
			check-special Q;
			if player has the Q:
				now player carries the noun;
			otherwise if player does not have the noun and noun is not visible:
				move noun to location of player; [special case for slippery sword]
				if noun is rentals or noun is shoes:
					move noun to lalaland;
			if Q is cask:
				now Q is flipped-yet;
			if Q is shoot button or Q is steer button:
				now Q is part of panel;
			[check-hint-tables;]
			if Q is a backdrop:
				move Q to LLL instead;
			if Q is oils:
				move Q to sacred cedars instead;
			move Q to Lalaland instead;
		otherwise:
			say "You can't seem to reverse what you did." instead;
	say "[reject]";
	the rule succeeds;

section special cases

to check-special (a - a thing):
	if a is liches:
		now drapes are in coe;

to check-reject (a - a thing):
	if a is silver:
		if drapes are not cut:
			say "The sliver wobbles a little but doesn't do anything. Perhaps you haven't done what you should with it.";

check propfliptoing panel:
	if poem is not folded:
		say "[reject]" instead;

check propfliptoing missile:
	if smilies are visible:
		if player is not in moor:
			say "Bad idea. There'd be more room for that outside where you found the poem." instead;

check propfliptoing gateman:
	if gateman is visible:
		try examining the gateman instead;
	if the player's command matches the regular expression "gate ?man", case insensitively:
		do nothing;
	otherwise:
		say "[reject]" instead;
	if notices section is unvisited:
		say "The nametag seems to try to rip itself from your clothes for a moment, but it settles back down. Maybe the time isn't right." instead;
	if nametag is not visible and gateman is not visible:
		badaboom instead;
	if location of player is not notices section and notices section is visited:
		say "Your nametag twitches. Very odd." instead;
	if gateway is not visible:
		say "No use summoning a gateman with no gate around." instead;
	if player wears the nametag:
		say "If you really CAN change the nametag, you don't want a gateman in your face--so you chuck the nametag away. Not really believing...";
	now nametag is in location of player;

check propfliptoing beast:
	if beats are visible:
		if player is not in Bassy Abyss:
 			say "That might unleash a beast on innocent citizens. If there is one, you must face it in its own lair." instead;

chapter rooming

rooming is an action applying to nothing.

understand the command "room" as something new.

understand "room" as rooming.

carry out rooming:
	if woeful pat is visible:
		say "[one of]As you pop back to the room, Woeful Pat looks visibly shocked. You have left him speechless, which is good news, but he is reaching for his pen, which is bad news for some poor soul in the future[or]Woeful Pat is less impressed this time, sniffing that it's been done[stopping].";
	if player is not in moor:
		try examining the air instead;
	move player to roomroom;

chapter mooring

mooring is an action applying to nothing.

understand the command "moor" as something new.

understand "moor" as mooring.

moored is a truth state that varies. moored is usually false.

carry out mooring:
	if map region of location of player is not sortie:
		say "[reject]" instead;
	if location of player is moor:
		say "Yes, it is a moor. At least, right now, it is." instead;
	if moor is visited:
		if location of player is cedars and caskfillings is 2:
			say "You hear a crash as you teleport. Maybe you'll find what it was about when you return.";
		otherwise:
			say "Oh, hey, teleporting's easier with experience[if player was not wearing coat]. Of course, you know what to expect, and you put your coat back on first[end if][if player has been in moor and woeful pat is in moor][one of]. Woeful Pat seems hurt that you did not return with a larger audience[or][stopping][end if].";
		now player is in moor instead;
	if player is in roomroom:
		if shoes are not in lalaland:
			say "You see a flash and get a glimpse of the moor, but your movement's gummed up quickly by the ooze below. Best to find something better to put on your feet." instead;
		if player carries coat and player is not wearing coat:
			now player wears the coat;
		if player is not wearing coat:
			now moored is true;
			say "You see a flash and get a glimpse of a moor, but it is just too cold. You blink and find yourself back in the room[if coat is off-stage]. You'll need to pierce the recipe to build something warm that can cover the rest of your body than just the shoes[otherwise]. That coat you made would be handy[end if]." instead;
		[if cask is in roomroom or cask is in ufc or sack is in roomroom or sack is in ufc:
			say "You might want to take a souvenir along." instead;]
		if moor is unvisited:
			say "There you go! You're outside, now. Your coat keeps you warm[if player was not wearing coat]--you managed to slip it on as the scenery changed and the temperature dropped[end if]. (N.B.: you can say MOOR from anywhere in this area in the future.)";
			reg-inc;
			follow the notify score changes rule;
			now moored is false;
			now player is in moor;
		otherwise:
			say "Oh, hey, teleporting's easier with experience[if player was not wearing coat]. Of course, you know what to expect, and you put your coat back on first[end if][if player has been in moor and woeful pat is in moor][one of]. Woeful Pat seems hurt that you did not return with a larger audience[or][stopping][end if].";
			now player is in moor;
		do nothing instead;
	say "[reject]" instead;

book endgame

chapter amusing

table of big yucks
yux	reg
"Examinining store c? Several times?"	--
"Eating the hot sauce or the taco?"	sortie
"XYZZY several times?"	--
"Waiting and listening to the nerds?"	metros
"Waiting and listening to the protesters?"	--
"Waiting and listening to the condo chatter?"	metros
"Trying to explore the condo before you cleared it?"	metros
"Saying GOD or GOB in the ghouls['] slough?"	forest
"Changing the toga to a gato?"	--
"Putting the barcode on the tomato?"	metros
"Scanning the deadbeat?"	metros
"Asking the beats anything?"	metros
"Praying in Sacred Cedars?"	sortie
"Reading the sidewalk in the trips strip a bunch of times?"	--
"The following anagrams in the forest: steam, padres, rasped, spader?"	forest
"The following anagrams in the metros: sap,asp,smitten,stalker,snores,senors,tar,liar(s)?"	metros
"The following anagrams in the sortie: stake, takes, teaks, pears, limes, thicken, lacy, mope or tap (by Pat), similes, sever, serve,"	sortie
"Converting the sprig to grips in the woodland?"	--
"Any anagrams I missed? Let me know. I'l credit you."	--

Rule for amusing a victorious player:
	say "Have you tried:[paragraph break]";
	repeat through table of big yucks:
		if there is no reg entry or reg entry is solved:
			say "--[yux entry][line break]";

book rules

part clarifying the parser's choice

chapter avoiding spoilers

[this section quashes annoying or spoilering disambiguation]

Rule for clarifying the parser's choice of musical note: do nothing.
Rule for clarifying the parser's choice of merchandise: do nothing.
Rule for clarifying the parser's choice of arena dig flyer: do nothing.
Rule for clarifying the parser's choice of go rest flyer: do nothing.
Rule for clarifying the parser's choice of metros-x: do nothing.
Rule for clarifying the parser's choice of the poetry: do nothing.
Rule for clarifying the parser's choice of the swing when the current action is swinging: do nothing.
Rule for clarifying the parser's choice of air when the current action is listening to: do nothing.
Rule for clarifying the parser's choice of nametag when swinging: do nothing.
Rule for clarifying the parser's choice of gateman: do nothing.
Rule for clarifying the parser's choice of gateway: do nothing.
Rule for clarifying the parser's choice of getaway: do nothing.
Rule for clarifying the parser's choice of taco: do nothing.
Rule for clarifying the parser's choice of tortilla: do nothing.
Rule for clarifying the parser's choice of chicken liver when liver-disambig-yet is false: do nothing.
Rule for clarifying the parser's choice of rut-door: do nothing.
Rule for clarifying the parser's choice of store-a: do nothing.
Rule for clarifying the parser's choice of cow liver when cow liver is not visible: do nothing.
Rule for clarifying the parser's choice of chicken liver when chicken liver is not visible: do nothing.
Rule for clarifying the parser's choice of the player when scaning: do nothing.

book tables

part coolnames

table of full names
anname	f	reserved
"Alonso Solano"	--	false
"Alonzo Lozano"
"Alyce Lacey"	true
"Angel Lange"	true
"Arline Lanier"
"Arnold Roland"
"Chas Cash"
"Cheri Reich"	true
"Chris Risch"
"Claus Lucas"
"Clay Lacy"
"Cleo Cole"
"Conrad Conard"
"Damien Medina"
"Dandre Darden"
"Darby Brady"
"Darell Allred"
"Debra Beard"	true
"Dereck Decker"
"Dolly Lloyd"	true
"Edna Dean"	true
"Ella Leal"	true
"Elroy Oyler"
"Elmira Mailer"	true
"Emery Meyer"
"Emory Moyer"
"Enoch Cohen"
"Enola O'Neal"	true
"Eric Rice"
"Erich Reich"
"Erland Randle"
"Ernst Stern"
"Esther Hester"	true
"Etta Tate"	true
"Galen Nagel"
"Gary Gray"
"Gena Egan"	true
"Gerda Edgar"	true
"Gerri Grier"	true
"Gisele Siegel"	true
"Hank Khan"
"Hans Nash"
"Herma Maher"	true
"Jamie Mejia"
"Kaleb Blake"
"Kavon Novak"
"Kory York"
"Lacy Clay"	true
"Lady Daly"	true
"Lawton Walton"
"Leah Hale"	true
"Leda Dale"	true
"Lena Neal"	true
"Lenard Randle"
"Leona Oneal"	true
"Lesli Ellis"	true
"Lewis Wiles"
"Lida Dial"	true
"Lionel Oneill"
"Lisle Ellis"
"Lonna Nolan"	true
"Mandie Medina"	true
["Lyman Manly"]
"Marie Maier"	true
"Marion Marino"
"Marni Marin"	true
"Marsha Sharma"	true
"Melany Manley"	true
"Merlin Milner"
"Milton Tomlin"
"Minna Inman"	true
"Montel Melton"
"Morgan Gorman"
"Nasir Rains"
"Neal Lane"
"Nella Allen"	true
"Norma Roman"	true
"Ramon Moran"
"Omar Mora"
"Reba Baer"	true
"Reggie Geiger"
"Rhea Hare"	true
"Ronald Orland"
"Romeo Moore"
"Royce Corey"
"Ruth Hurt"	true
"Sadie Desai"	true
"Serina Raines"	true
"Shan Nash"	true
"Shanon Hanson"	true
"Shay Hays"
"Shayne Haynes"
"Shenna Hansen"	true
"Sheyla Ashley"	true
"Shonna Hanson"	true
"Skye Keys"	true
"Somer Morse-Moser"	true
"Soren Rosen"
"Steve Teves"
"Telly Lytle"
"Teresa Easter"	true
"Thor Roth"
"Tomas Matos"
"Torrey Troyer"
"Tyrel Tyler"
"Vanda Davan"
"Waldo Woald"
"Warren Warner"
"Wash Shaw"
"Zaid Diaz"
"Zuri Ruiz"	true

part anagrams

to say drop-straw:
	say ", which is too heavy to carry. You drop it";
	choose row with short of "warts" in table of pad-stuff;
	if known entry is true:
		say ". You run a big line through your entry on warts in your dope tan notepad";
		now known entry is false;
	move straw to location of the player;
	
to say trap-check:
	now tories are in lalaland;
	if td is visible:
		say ", collapsing the trap door";
	otherwise:
		say ", collapsing a trap door you hadn't noticed otherwise";
	now td is in lalaland;

to say give-sword:
	if stickyhanded is true:
		say "leaps into your hand";
		now player has the sword;
	otherwise:
		say "jumps and slides out of your hand";
		now sword is in Abyss;

to say which-roar:
	now noise bag is in lalaland;
	say "[if beats are visible]loud beats[otherwise]beast's roaring[end if]"
	
to say process-sandwich:
	if spam is part of the sandwich:
		if player has sandwich:
			say ". You juggle the spam but don't quite drop it";
			now player has spam;
		otherwise:
			now spam is in location of the player;
		now sandwich is in lalaland;

to say check-sandwich-on-spam:
	if bread is part of the sandwich:
		if player has sandwich:
			say ". You juggle the bread but don't quite drop it";
			now player has bread;
		otherwise:
			now bread is in location of player;
		now sandwich is in lalaland;
	
to say move-toeholds:
	if toeholds are part of the poles:
		now toeholds are part of the slope;
		
to say holds-1:
	reg-inc;
	if poles are visible:
		say "poles";
		now toeholds are part of the poles;
	otherwise:
		say "slope";
		now toeholds are part of the slope;

to say now-chain:
	now chain links are in Potshot Hotspot;

to say rope-worn:
	now player has the ropes;

to say pat-off-stage:
	now woeful pat is in lalaland;
	now poetry is in lalaland;

to say silo-plus:
	reg-inc;

to say beast-beats:
	say "[if beast is visible]beats[otherwise]beats";

after propfliptoing resin:
	if resin is visible:
		now resin is in lalaland;
		now stickyhanded is true;
		continue the action;

to say tool-clue:
	if grips are visible and ropes are visible:
		say ". The combination of grips and ropes is probably enough to make climbing gear[if toolshed is visible]. You can probably try to mess with the toolshed, but it doesn't seem like you need to[end if]";

to say ropes-grips-lying:
	if ropes are visible and grips are visible:
		say ". You probably don't need those tools you made";
	otherwise if ropes are visible:
		say ". The ropes won't be necessary";
	otherwise if grips are visible:
		say ". The grips won't be necessary";

to say wear-sword:
	now player wears the sheath;

to say what-about-gate:
	if gateway is visible:
		say ". Actually, the attics are a lot tougher to figure out than the gateway, especially with [if player has gadget]that tool you have[otherwise]those thingamabobs in the cabinet[end if]";
	otherwise:
		say ". [if player has gadget]Not much left to do for you here[otherwise]You got the attics, but other stuff'll be tougher. Have a look in that cabinet, maybe[end if]"

to say check-plur:
	if player's command does not include "attics":
		say "(Okay, no need to be pedantic about plurals.) [run paragraph on]";

table of anagrams
the-from	the-to	reversible	exact-text (topic)	text-back (topic)	from-msg	force-take	hashkey
toga	goat	0	"goat"	"toga"	"The dingy toga shudders. It seems to rip, make legs, and twist around, like one of those balloon animals you were never good at. All this twisting has left him with an appetite, and he walks over to the delicious thorns and brambles.[paragraph break]They're not much, though, so he mauls the shrub a bit before making a dessert of the brush that [if brush is unexamined]you just now notice [end if]spelled GATO. Exhausted, he turns around three times and falls asleep.[paragraph break]Wait. Did you just do that? Change a toga into a goat? You aren't sure if you were REALLY thinking about doing so, but you come to the realization you didn't rationalize back there, and you may have a latent talent."	false
nametag	gateman	0	"gateman" or "gate man"	"nametag"	"Whoah! The nametag pulses and pops in directions you didn't think something that flat could. You hear a gish, then a sigh. A tall, grouchy old man in sober robes so aged you almost say 'Egad' cries 'The eyes! They see!' He grumbles how he shoulda been a portal king in the parking lot, he's such a talking pro. Then he notices you.[paragraph break]'Ehh? Heh. Smug gums. But our only hope. Yorpwald's been shuffled. Almost f-flushed. I'm Old Man Almond. What d'you want? This isn't some RPG where you can ask everyone on the way for help. I'm pretty much it.'"	false	0
static	attics	1	"attics" or "attic"	--	"[check-plur]The static cuts off and seems to grow opaque. Then it forms into a small box with a cupola, pyramid, and other shapes. They fit with a click on top of the dollhouse[if gateman is visible]. Old Man Almond golf-claps. 'Good work[what-about-gate].'[otherwise]. Too bad nobody was around to see it![end if]"	false
attics	static	1	"static"	--	"You undo your artistic work for perhaps more practical considerations like learning how to use the new toys from the cabinet."	false
gateway	getaway	0	"getaway" or "get away" or "get a way"	"gateway"	"[if gateman is visible]The scenery behind the gate rumbles to grass, hills, and even a few thatched huts. 'If you have any more questions...now's a good time to ask,' says Old Man Almond.[otherwise]But you can't get away with those iron crossings in the way.[end if]"	false	[end introduction anagrams]
store f	forest-sc	0	"forest"	"store f"	"The greens and browns of Store F coagulate and pull apart into an actual forest."	false	[start trips anagrams]
store i	sortie-x	0	"sortie"	"store i"	"The store rumbles, destroying the portraits of famous Tories (this is not a political statement) and revealing the small sortie down[trap-check]. A stairway down remains, but that's about it."	false
store m	metros-x	0	"metros"	"store m"	"The store rumbles, with the collections of small-scale cities disappearing. You see an escalator leading--well, somewhere populated."	false
store r	r-p	0	"resort"	"store r"	"Store R rumbles and reforms into something far posher. A huge resort! You see, in the distance, a manor that seems made for you. 'I know what you're looking at!' calls some random well-wisher. 'You've earned it! For defeating Red Bull Burdell!'[paragraph break]Before you reply you haven't, he's already run behind store 7, yelling 'Go! Rest!'"	false
cabinet	nice bat	0	"nice bat"	"cabinet"	"The cabinet seems to expand like an amoeba, then, POP! It becomes a rather large bat, which jumps up and down excitedly. It's clearly grateful it has become more than just something to store things in."	false	[end trips strip anagrams]
odor	rut-door	0	"door"	"odor"	"The odor becomes thick and choking, then a wood you've never smelled before but know it's wood. The odor swirls into a door--plain except for a bulge. Like the gateway, nothing seems to be behind it. You'll probably need to figure how to open it."	false	[start forest anagrams]
bulge	bugle	0	"bugle"	"bulge"	"The ovular shape on the door rumbles then falls off. You see that extra bit is a horn--yes, you've definitely found a bugle. The door looks as impervious to force as ever, but maybe...?"	true
ones	nose	0	"nose"	"ones"	"The ones line up next to each other in a pair of not quite v-shaped semicircles. Then they melt into a nose. You take it."	true
dashes	shades	0	"shades"	"dashes"	"Most of the dashes reform into a pair of eyeglass rims, and the remaining one fill in as lenses. They lighten a little, and you have fully functioning shades."	true
noughts	shotgun	0	"shotgun"	"noughts"	"Bam! The noughts slink together to form a gun barrel, then a whole big powerful gun."	true
bread	beard	0	"beard"	"bread"	"The bread seems to grow a fuzzy dark mold, then narrows into a sickle. It's now a beard, though not a particularly beautiful one. As you pick it up, you shake it for crumbs, and the hairs don't crumble into mold[process-sandwich]."	true
livers	sliver	0	"sliver"	"livers"	"The livers narrow a bit and become more pointy at the end. They are a nice long sliver now[if player does not have sliver], which you pick up[end if]."	true
sliver	silver	0	"silver"	"silver"	"The sliver changes and grows even more metallic. It becomes a row of six bullets, tied together like plastic twist-off toys, but obviously more lethal."	true
liches	chisel	0	"chisel"	"liches"	"As the final shade heads to Hades, a chisel clatters behind. And the liches were hiding some drapes along the north wall, it appears. The chisel doesn't seem to vibrate scarily or anything, so you take it."	true
banshee	hasbeen	0	"hasbeen" or "has been"	"banshee"	"The banshee quits moaning and starts screaming as he swirls into a small dot. He is in--not a better place, but a better suited place. Where the fates feast. As not just a has-been but a has-has-been, his moaning will be more appropriate punishment for the souls who have done much worse than he has."	false
drapes	spread	0	"spread"	"drapes"	"Shazam! The drapes unfold, and they become a large horizontal spread. It's wide enough, you can't detect the message that was there before."	false
spam	maps	0	"maps"	"spam"	"With a sickening SCHLURP, the [spam] curdles and flattens to form several maps[check-sandwich-on-spam]."	false
vowels	wolves	0	"wolves"	"vowels"	"Well, you've done it now. The imposing vowels become werewolves--but they notice your shotgun and stand back. The first forward gets shot--or so they imagine."	false	[end forest anagrams]
warts	straw	0	"straw"	"warts"	"The warts quickly peel off and lengthen into straw[drop-straw]."	false	[start sortie anagrams]
skate	steak	0	"steak"	"skate"	"The skate turns reddish, and the blade retracts and becomes the main connecting bone in a slab of steak."	false
cult tee	lettuce	0	"lettuce"	"lettuce"	"The t-shirt crumples and then shreds before turning into a light green head of lettuce."	false
spearman	parmesan	0	"parmesan"	"spearman"	"The spearman transforms into something cheesier--parmesan cheese! Unfortunately, it doesn't have one of those cute plastic spears sticking from it, but you can't have everything."	false
cathouse	hotsauce	0	"hotsauce" or "hot sauce"	"cathouse" or "cat house"	"The CATHOUSE perfume turns into a packet of equally over-capitalized and under-spaced hot sauce."	false
grist	grits	0	"grits"	"grist"	"With a squelch, the gristin the fridge refines itself into something more edible. Relatively."	false
cake pan	pancake	0	"pancake"	"cake pan" or "cakepan"	"You hear a clatter, and then you poke back into the fridge to see the cake pan is a pancake."
tall trio	tortilla	0	"tortilla"	"tall trio"	"Poof! The doughy fellows maintain their Mexican-ness and, err, breadiness, but they flatten out into a tortilla."	false
taco	coat	0	"coat"	"taco"	"The hot-to-your-tongue taco becomes a warm-to-your-body coat."	false
cask	sack	1	"sack"	--	"The cask retains its color but looks visibly frayed as its wood turns to burlap. The sack it has become collapses in a heap on the floor. You pick it up."	true
sack	cask	1	"cask"	--	"The sack stiffens, rises and becomes less blobby. It's the cask again, nice and upright[if straw was in sack]. The straw falls out[end if][if hay was in sack]. The hay falls out[end if]."	true
hoses	shoes	0	"shoes"	"hoses"	"The pair of rubber hoses bends and opens and become a comfortable pair of shoes that swallows your old shoes--you'd forgotten how ratty they were. A few steps show walking's much smooth. So smooth, you forget you're wearing them. And you didn't even have to pay a fortune."	false
anapest	peasant	0	"peasant"	"anapest"	"Nothing happens. You worry your magic powers have failed, until a peasant strides out from the edge of the moor, carrying a bale of hay and singing a cheery song about nothing in particular. Distracted, you look over and smile.[paragraph break]'Oh, does THAT resonate with your stone ear?' whines Woeful Pat.[paragraph break]You notice his papers have crumbled (but don't worry, he has PLENTY of written drafts.) He storms off, claiming you will make a perfect arch-villain in his new socially significant blank-verse epic. Or another poor henchman who deserves but one line before a horrible fate--or a mega-rip of an epigram[pat-off-stage]!"	false
roadblock	black door	0	"black door" or "blackdoor"	"roadblock" or "road block"	"Bam! The fissure in the roadblock covers up, and a black door appears where it was. It's light but bulky--you can probably put or push it where it needs to go[if pat is visible]. Woeful Pat shows commendable concentration ignoring all this[end if]."	false
poem	panel	0	"panel"	"plane"	"Poof! The paper plane becomes a panel. It's light enough to carry, you suppose[if player has poem or player has panel]. Well, it doesn't cause an immediate hernia[end if]."	false
smilies	missile	0	"missile"	"smilies"	"The smilies seem to waver. But they sever from the verse, grow, and clump together to form a big yellow missile! You reread the poem. You didn't think it would change much without the smilies, but it now seems pointed, warlike, sinister--a restrained, plain-language indictment of not just bad poetry but nastiness in general!"	false
oils	soil	0	"soil"	--	"The oils seep into the moor and make it less squishy and sandy."	false
oils	silo	0	"silo"	--	"The oils seep into the moor, and you back up as a huge silo several times your size takes shape.[silo-plus]"	false
hoots button	shoot button	0	"shoot button/--"	"hoots button/--"	"The hoots button glows--you touch it (so hot,) and it changes to a shoot button as its letters shift."	false
trees button	steer button	0	"steer button/--"	"trees button/--"	"The smell of a chemical ester pervades the air as the trees button rewords to a steer button."	false	[end sortie anagrams]
drainage	gardenia	0	"gardenia"	"drainage" or "aged rain" or "arena dig" or "dear inga"	"In a fit of ecological, aesthetic magic-slinging, you transform the drainage and all the flotsam inside it into a much prettier gardenia. The whole underside is still pretty dingy, but hey, free flower. You take it."	true	[start metros flips]
clover	velcro	0	"velcro"	"clover"	"Pop! The clover expands and grows scratchier and darker. It is now a long strip of velcro. The leaves appear to have expanded into something resembling mittens[if player does not have velcro]. You pick them up[end if]."	true
Motto	tomato	0	"tomato"	"motto" or "a motto"	"The motto--that is, both words and paper--curls up and bundles into a small sphere, and you almost [if motto is in alley]hesitate to pick it up[otherwise]drop it[end if]. Amazingly, it's not icky-soft or anything, and it doesn't smell TOO bad."	true
lost corn	controls	0	"controls"	"lost corn"	"The different-colored kernels only flatten out as the lost corn collapses into a circuit board. The controls you possess--for whatever--seem powerful[if deadbeat is visible]. 'Whoah! DUDE!' moans the deadbeat, half staggering to his feet before he collapses from the effort[end if]."	true
heaths	sheath	0	"sheath"	"heaths"	"The heath grows, tangles around itself, and lumps into a passable sheath that could carry a decent-sized sword, which you wear[wear-sword]."
begonias	noise bag	0	"noise bag"	"begonias"	"The begonias twirl and create a bag. One with weird triangular spikes on the inside. You exclaim surprise at it--the bag whooshes every time you say something."
brocade	barcode	0	"barcode" or "bar code"	"brocade"	"The brocade schlurps and reorganizes into something flatter, but with the same pattern[if player does not have brocade and player does not have barcode]. You pick it up. [end if]."	true
dry cake	keycard	0	"keycard" or "key card"	"dry cake"	"The dry metallic cake turns into a dry metallic keycard[if player does not have dry cake and player does not have keycard], which you take for yourself. Hooray, technology[end if]."	true
antlers	rentals	0	"rentals"	"antlers"	"Suddenly, a man walks in, points to his watch, takes the antlers down with a twist of his screwdriver, and walks off with them[if condo-evac is true]. You smile and nod, but he has already left[otherwise]. People mumble but stay. The free food and alcohol overcome moral considerations[end if]."	false
neon pig	op	0	"opening"	"neon pig"	"[if player is on fuzzy looking wall]Remembering basic electric safety rules and common sense, you move the side before trying your magic. This saves your bacon. [run paragraph on][end if]The neon pig goes on the fritz, sparking first, then making a big POP. It appears you've created an opening. Now you're bakin[']. Uh, cookin[']. (Okay, no more hamming it up.)"	false
siren	resin	0	"resin"	"siren"	"The siren makes some even more spastic noises before grinding into resin powder. You scoop it up and put some on your hands, which feel sticky now. You feel your chances of defeating the [beast-beats] has risen, unless you rinse."	false
beats	beast	0	"beast"	"beats"	"The beats turn into a beast which also yells, but at least it's not that constant electronic thumping. The thing has to take time to breathe. It's an improvement. Well, for your chance to best a beast."	false
words	sword	0	"sword"	"words"	"[if player has sheath]The words go slippery, claiming they didn't mean themselves, giving passive-aggressive threats. But it is too late--they fade, and a sword appears materially.[paragraph break]You sensibly put the sword--as slippery as the final words--into your sheath[in-sheath][otherwise]The words explode from the noise bag, which is blown away by the [which-roar]. In their place is a shining sword that [give-sword][end if]."	false	[end metros flips]
rock	cork	0	"cork"	"rock"	"The rock grows a few holes and immediately becomes lighter. You can probably pick it up easily now. You do."	true	[start resort flips]
swing	wings	0	"wings"	"swing"	"The old swing's ropes defray and the seat somehow transmogrifies. You see a pair of wings. They don't look like they'll last too long, but they're better than nothing. You take them."	true
sprig	grips	0	"grips"	"sprig"	"The fragile sprig pops into a pair of suction grips, which you take[tool-clue]."	true
spore	ropes	0	"ropes"	"spore"	"The spore grows more quickly than you could imagine, into a couple of long ropes tied together. You wind one around your waist[rope-worn][tool-clue]."	false
poles	slope	0	"slope"	"poles"	"The poles rumble and slide. You run away, fearing they may go vertically--then you're a bit disappointed to see they're still at a forty-five degree angle[move-toeholds]."	false
toolshed	toeholds	0	"toeholds"	"toolshed"	"That'll do it! The toolshed crumbles, leaving only the small outcroppings--which somehow fly into the [holds-1].They're plentiful and stable. You could probably climb up now[if poles are visible] if the poles weren't so vertical[end if][ropes-grips-lying]."	false
riot	protest	0	"trio"	"riot"	"Your word seems to have no effect. But that's just because the least enthusiastic people in the back leave first. Others follow--one of them even drops some chain links[now-chain]. And soon, just three people remain. The most energetic and dedicated, of course, but altering that triangle can't be too hard. While it's still a protest, they don't seem fully into it."	false
protest	potters	0	"potters"	"protest" or "riot"	"The trio grows even more lethargic. A fellow in a smock mocks the whole charade before throwing the smock in a corner. He talks with his two friends[if kilns are visible]. Shortly, they see the kilns nearby. They go to work happily[otherwise]. They still aren't happy, though. They'd sort of like something to do that'll help them forget their protesting phase[end if]."	false
links	kilns	0	"kilns"	"links"	"The links burst and swell into luxury kilns[if potters are visible]. 'Ohmigod! Him! Good!' yells one of the potters. 'This is the Mark 9000 brand with wheels underneath for easy mobility! Sir, I--I don't know why we ever protested you! We really must've had nothing better to do! Hm, the clay's scaly, but that's not your fault!'[otherwise]. The yelling's loud as ever but increasingly directed at the kilns and not you. The protesters seem distracted.[end if]"	false	[end resort flips]

carry out propfliptoing goat:
	if player has toga:
		now toga is in location of player;
	if toga is visible:
		now thorn is off-stage;

the toga is a thing in dy. "A toga is lying here. It doesn't look particularly festive or clean.". rgtext of toga is "[rc][gc][rc][rc]". legth of toga is 4. gpos of toga is 3. rpos of toga is 1.

report taking the toga:
	say "From the smell[if toga is examined] and the writing on it[end if], you doubt wearing it will get you anywhere cool in its current form, but you take it anyway.";
	the rule succeeds;
	
the toga-writing is privately-named. The toga-writing is part of the toga. understand "writing" and "message" and "toga writing" as toga-writing when player is in dy.

description of toga-writing is "It's a dingy grey and says 'You GOT A believe!'[one of][no line break] You'd expect that sort of excruciating grammar on a sports team's t-shirt, but not a toga. Hmm.[or][stopping]"

description of the toga is "On it you see a four-legged animal with a sort of beard and horns. In one corner you see some writing. The toga also smells vaguely of barnyards."

check wearing the toga:
	say "If the barnyard smell weren't unpleasant enough, you feel pricked by hairs--like a beard--and maybe even horns. You remove the toga and flip it inside-out, but you don't see anything. Odd." instead;

the big gigantic thorn is scenery in dy. "It's plenty twisty, a plant-kingdom comb-over with prickles all up and down it. You can't tell where it starts or ends, but no way you'll sneak through it without getting cut up pushing it aside to go, well, whatever way it's blocking you from going."

understand "combover" as big gigantic thorn.

the prickles are part of the big gigantic thorn. description of the prickles is "Prickly."

instead of doing something with the prickles:
	if the current action is examining:
		say "They help the big thorn seem to block up even more space than it already does." instead;
	say "You don't need to do anything to, or with, the prickles.";

understand "tinier/tiny/small thorns/thorn/" as prickles.

does the player mean doing something with the prickles: it is unlikely.

check taking big thorn:
	say "You'd just get scratched up." instead;

check taking tiny thorns:
	say "You'd just get scratched up." instead;

description of tiny thorns is "They're kind of lashed onto the big thorn, and while they're not particularly imposing by themselves, they'd scratch you a lot if you tried to walk by."

thor-north is a truth state that varies. thor-north is usually false.

check going north in dy:
	if big gigantic thorn is visible:
		say "You walk into the big gigantic thorn. Ouch! A four-letter word comes to mind, but not one that'll help you get on with things, here." instead;
	otherwise:
		if the player's command matches the regular expression "\bn\b", case insensitively:
			now thor-north is true;
			say "[one of]A vision of the Norse god Thor wielding a copy of 'The Pedant's Guide to Grammar, Spelling and Excuses to Loathe Abbreviations' scares you into not going that way--for the moment[or]It's Thor again, feeling left out. Guess he's trying guilt instead of fear[stopping][one of]. But yes, you're on the right track[or]. Since you knew what to do with the goat, I'm guessing you'll know how to rephrase that last command[or]. Yes, you have to go NORTH and not N[stopping].";
			pad-rec "Thor" instead;
		reg-inc;
		say "With the big thorn gone, why, of course you can go north[if thor-north is true]. The heck with Thor--you can't imagine him appearing again[end if]. Though the goat didn't clear out all the thorns. One rips your nametag slightly. But beyond that, there's a path that leads to an open field.";
		now player is in Notices Section instead;

the goat is an animal. "The goat is half-sleeping here. He's probably best left that way."

understand "animal" as goat when goat is visible.

instead of waking the goat:
	say "If he's having a bad dream, he'll kick at you in his sleep. If you interrupt a good dream, he'll kick whoever woke him.";
	
check taking the goat:
	say "Trying to get the goat would more likely get the goat's goat." instead;

description of goat is "It seems ill-tempered[if thorn is not visible] despite, or perhaps because of, its eclectic recent meal[end if]."

before propfliptoing (this is the tortilla check rule):
	if noun is taco and tortilla is visible:
		if ingredients-in-tort < 4:
			say "[one of]The tortilla's not ENOUGH of a taco yet.[or][stopping]";
			try examining tortilla instead;

before propfliptoing:
	if noun is visible:
		if noun is portal and noun is enter-clued:
			say "(entering instead)";
			try entering noun instead;
		if noun is lois:
			say "(praying instead)";
		otherwise:
			say "(examining instead)";
		try examining noun instead;

after examining a portal (this is the clue entering with the name of a portal rule):
	if noun is not enter-clued:
		say "(You can say the name of the [noun] now to [if noun is gateway]try and [end if]enter it. There's nothing more hidden in its description.)";
		now noun is enter-clued;
	continue the action;
	
check propfliptoing (this is the should we bother flipping rule):
	if noun is not visible:
		if number of things that oneway noun > 0:
			repeat with prosp running through things that oneway noun:
				if prosp is visible:
					continue the action;
		repeat with prosp running through things that anagram noun:
			d "Checking [prosp].[line break]";
			if prosp is visible:
				continue the action;
		if noun is sword and noise bag contains words and location of player is Abyss:
			now noise bag is open;
			continue the action;
		d "[noun] can't seem to be flipped.[line break]";
		say "You can't see anything here like that[if toga is not in dy], or changeable into that[end if]." instead;

check propfliptoing silver:
	if livers are visible:
		say "The livers pulse for a moment, but...nothing happens. Perhaps they have another use, first." instead;
	if sliver is visible:
		if doorway is not in coe:
			say "The sliver discolors for a moment--but it seems it is not ready yet." instead;

check propfliptoing (this is the easy store flip rule):
	let Q be new form of noun;

check propfliptoing soil:
	if oils are not in cask:
		if location of player is sacred cedars:
			say "[if soil is in moor]You've already poured the soil, and bragging does not impress Lois[otherwise]That would really clog up the tap, changing the oils in it to soil[end if]." instead;
		if oils are visible:
			say "You've found a bug, somehow. I'm sorry about this. It's not game-breaking, but it's a bug. Please type UNDO and send a transcript of what you did.";
		continue the action;
	if player is not in moor:
		say "This is not a good place to put soil. Somewhere more wide open?" instead;

check propfliptoing silo:
	if oils are not in cask:
		if location of player is sacred cedars:
			say "[if silo is in moor]You've already built the silo, and bragging does not impress Lois[otherwise]That would really clog up the tap, changing the oils in it to a silo[end if]." instead;
		if oils are visible:
			say "You've found a bug, somehow. I'm sorry about this. It's not game-breaking, but it's a bug. Please type UNDO and send a transcript of what you did.";
		continue the action;
	if player is not in moor:
		say "This is not a good place to put a silo. Somewhere more wide open?" instead;
	if soil is not in moor:
		say "The moor's ground, as is, is not stable enough. You need a foundation first." instead;
		
the soil is scenery.

description of soil is "It's a rich foundation for [if silo is visible]that silo you built[otherwise]a big, big place, you're sure[end if]. Stronger and denser than the usual mush on this moor."

the silo is a transparent container. it is fixed in place. "That silo you poured from the cask of oils towers above you here[if black door is part of the silo]. You see the outline of the black door on it, but you probably don't want to go in[end if][if missile is in the silo]. You know the missile's in there, too[end if][if panel is part of the silo]. There's also a panel attached to the silo[end if]."

does the player mean shooting the missile when the missile is visible: it is likely;

[?? L/SHOOT MISSILE at lots of points in the walkthrough]

check shooting the missile:
	if silo is not visible:
		say "Not on your own you won't shoot the missile." instead;
	if the missile is not in the silo:
		say "You're not strong enough to launch it by yourself, but there's a handy silo nearby you can put it in." instead;
	if missile is in silo:
		if player has panel:
			say "There are no controls attached to the silo at the moment. You happen to have some handy, though." instead;
		if panel is not part of silo:
			say "The missile's in the silo, but there's no button to push to launch it. Or thing that has lots of buttons to push. " instead;
		say "Maybe you should explicitly try a button and see what happens." instead;

the dashed boundary is part of the silo. description of dashed boundary is "It seems to have writing, one letter per dash: PUT CORRECTLY COLORED DOOR ON SILO[if black door is not visible]. You wonder what you could change into a door[otherwise]. Hm, you doubt there's a red or blue door you have to find[end if]."

the dotted rectangle is part of the silo. description of dotted rectangle is "[if player has panel]Curiously, it's about the size of the panel[otherwise]It's as if you need to put something on there, but you wonder what[end if]."

description of silo is "It's thirty feet tall and dark and forbidding, except for [if black door is part of silo]the door you slapped on it[otherwise]a six-by-eight-foot dashed boundary that rises from the ground[end if] and [if panel is part of the silo]that panel you installed[otherwise]a smaller dotted rectangle, chest-high, to the side[end if]."

instead of entering the silo:
	say "[if dashed is part of the silo]Bam! That outline almost looked like a door, but it wasn't[door-clue][otherwise]A quick peek in shows the silo's empty inside. You're not even sure if you could open the door from the inside, [if-missile][end if].";

to say if-missile:
	say "[if missile is in silo]and the missile wouldn't make good company[otherwise]but you could put something that belongs through the door[end if]";

to say door-clue:
	if black door is visible:
		say ". But there's a door nearby";
	otherwise:
		say ". Maybe something can be changed into a door";
		
book region division

a region can be unsolved, bypassed or solved. a region is usually unsolved.

a region has a number called max-score. the max-score of a region is usually zero.

a region has a number called cur-score. the cur-score of a region is usually zero.

Intro is a region. dy, Notices Section, and Busiest Subsite are in Intro.

max-score of Intro is 5.

Stores is a region. Trips Strip is in Stores.

max-score of Stores is 5.

Forest is a region. sf, rf, Silent Rut, Sorting Area, Storing Area, coe, Ghouls' Slough, Frost Forts are in Forest.

max-score of Forest is 17.

Sortie is a region. ufc, roomroom, The Nick, Kitchen, ne-room, Sacred Cedars and Moor are in Sortie.

max-score of Sortie is 27.

Metros is a region. aya, toh, Cordoned Red Condo, Cramped Red Camp, Elf Row's Flowers, Undesired Underside, spis, ett, Bassy Abyss are in Metros.

the beats are in metros.

max-score of Metros is 18.

Resort is a region. tiles, islet, olw, Potshot Hotspot, Roman Manor are in Resort.

max-score of Resort is 13.

LLL is a region. Lalaland is in LLL. [LLL is only a region for backdrops.]

book undo tags

undo-code is a number that varies. undo-code is usually 0.

chapter hashcodes

Table of Hashcodes
Letter(indexed text)	Code
"a"	2187818
"b"	18418905
"c"	19005585
"d"	21029089
"e"	127806109
"f"	26514896
"g"	32599702
"h"	37282299
"i"	44992846
"j"	48960525
"k"	52933178
"l"	53813839
"m"	64075153
"n"	68907508
"o"	74352577
"p"	81465959
"q"	84405617
"r"	85323803
"s"	96273966
"t"	103110018
"u"	105105807
"v"	107164820
"w"	107934773
"x"	112768081
"y"	122359252
"z"	122969618

to decide what indexed text is the filtered name of (t - a value of kind K):
	let s be t in lower case;
	replace the regular expression "<^abcdefghijklmnopqrstuvwxyz>" in s with "";	[ a-z would include accented characters]
	decide on s;

to decide what number is the hash of (t - a value of kind K):
	let s be the filtered name of t;
	let hash be 0;
	repeat with c running from 1 to the number of characters in s:
		increase hash by the Code corresponding to a Letter of character number c in s in the Table of Hashcodes;
	decide on hash;

when play begins:	[compute hashkeys]
	let wnum be 0;
	repeat through the table of odd-lgths:
		now wordnum of odd-ana entry is which-word entry;
	repeat through the table of anagrams:
		if wordnum of the-from entry > 0:
			now wnum is wordnum of the-from entry;
			d "word # [wnum] [word number wnum in the printed name of the-from entry] ~ [the-from entry][line break]";
			now hashkey entry is the hash of word number wnum in the printed name of the-from entry;
		otherwise:
			now hashkey entry is the hash of the printed name of the-from entry;
		d "The hash of [the-from entry] is [hashkey entry].[line break]";

volume 2 which entails various cheats and should be NFRd

description of main-window is "bug"

chapter angle-glean-cheat

chapter hahaing

hahaing is an action out of world.

understand the command "haha" as something new.

understand "haha" as hahaing.

carry out hahaing:
	if haha is false:
		now helpdebugflag is true;
		now haha is true;
	otherwise:
		now haha is false;
	the rule succeeds;

haha is a truth state that varies.

every turn (this is the haha-debug rule):
	if haha is true:
		if helpdebugflag is true:
			let temp-haha be just-print;
			now just-print is false;
			all-say "DEBUG: GLEAN = [run paragraph on]";
			try gleaning;
			all-say "DEBUG: ANGLE = [run paragraph on]";
			try angleing;
			now just-print is temp-haha;

chapter hint debugs

every turn:
	if hintfull is true or helpdebugflag is true:
		now just-print is false;
		try mainhelping;
		now just-print is true;


chapter ufcing

ufcing is an action applying to nothing.

understand the command "ufc" as something new.

understand "ufc" as ufcing.

carry out ufcing:
	now centrifuge-stopped is false;
	decrement the cur-score of sortie;
	d "Resetting centrifuge.";
	now the numset of the dial is 0;
	the rule succeeds;

chapter askthruing

askthruing is an action applying to nothing.

understand the command "askthru" as something new.

understand "askthru" as askthruing.

carry out askthruing:
	repeat with the-thingamabob running through the list of visible things:
		say "Asking about [the-thingamabob].";
		try objasking gateman about the-thingamabob;
	repeat with the-thingamabob running through the list of things in subsite:
		say "Asking about [the-thingamabob].";
		try objasking gateman about the-thingamabob;
	repeat with the-thingamabob running through the list of things in dy:
		say "Asking about [the-thingamabob].";
		try objasking gateman about the-thingamabob;
	the rule succeeds;


section wpb - not for release

when play begins (this is the please remove before release rule):
	say "This version of the game involves special tricks for the player to warp through the game. It is volume 2 in the source and should be marked as NOT FOR RELEASE before release. However, it's okay now. [paragraph break]Type [bold type]dc[roman type] for all debug commands, but the one most helpful to me is [bold type]hd[roman type], which tracks the hints you receive. Shufhints.glkdata is a text file created and appended every move that you can send to me to make sure hints are valid. It's 2 dirs up in Windows and in /home in Mac."

chapter xraying

xraying is an action applying to nothing.

understand the command "xray" as something new.

understand "xray" as xraying.

carry out xraying:
	repeat with ABC running through visible things:
		say "Xraying [ABC]:";
		try saltining ABC;
	the rule succeeds;

chapter sc1ing

sc1ing is an action applying to nothing.

understand the command "sc1" as something new.

understand "sc1" as sc1ing.

carry out sc1ing:
	if player does not have tagged gadget:
		say "Giving you the gadget." instead;
		now player has tagged gadget;
	repeat with ABC running through visible things:
		say "Scanning [ABC]:[line break]";
		try scaning ABC;
	the rule succeeds;

test sfs with "sc1/switch gadget/sc1/switch gadget"

chapter xalling

xalling is an action applying to nothing.

understand the command "xall" as something new.

understand "xall" as xalling.

carry out xalling:
	repeat with this-item running through visible things:
		say "[this-item]: [no line break]";
		try examining this-item;
	the rule succeeds;

chapter dcing

dcing is an action applying to nothing.

understand the command "dc" as something new.

understand "dc" as dcing.

carry out dcing:
	say "[paragraph break]rj jumps to various regions, 1-4, 4 being the final bit.[line break]sk 1, sk 2 or sk 3 solves a region[line break]disas disassembles the disguise if you want to try to break that.[line break]ff switches between your 2 gadgets.[line break]j skips the intro and gets you near the stores.[line break]hf shows full hints every move.[line break][bold type]hd toggles printing full hints to a debug file called shufhints, or shufhints.glkdata. This is bolded annoyingly to stress that any hint transcript is helpful to me.[roman type] Shufhints(.glkdata) is a text file that appears in c:\program files\inform 7 (or 2 directories up) on Windows and in your home directory on the Mac.[line break]hn turns hints off."

chapter hfing

hfing is an action applying to nothing.

understand the command "hf" as something new.

understand "hf" as hfing.

carry out hfing:
	if hintfull is false:
		say "Now showing full hints with the game. Not recommended for testers--use hd instead.";
		now hintfull is true;
	otherwise:
		say "Now hiding hints.";
		now hintfull is false;
	the rule succeeds;

chapter hding

hding is an action applying to nothing.

understand the command "hd" as something new.

understand "hd" as hding.

carry out hding:
	if helpdebugflag is false:
		say "Hint debugging on, exported to shufhints.glksave[if hintfull is false]. Type HF to get full hints each move[end if].";
		append "==================starting hint debug session[paragraph break]" to the file of shufhints;
		now helpdebugflag is true;
	otherwise:
		say "Hint debugging off.";
		append "==================ending hint debug session[paragraph break]" to the file of shufhints;
		now helpdebugflag is false;
	the rule succeeds;

chapter jing

jing is an action applying to nothing.

understand the command "j" as something new.

understand "j" as jing.

carry out jing:
	if trips strip is visited:
		say "Already beat intro." instead;
	increase the cur-score of intro by 4;
	say "** Checking/pretending to visit the areas before the Trips Strip **";
	if dy is unvisited:
		move player to dy, without printing a room description;
	if notices section is unvisited:
		move player to notices section, without printing a room description;
	solve-region intro;
	if player has gadget:
		say "You already have a special item. Jumping.";
	otherwise:
		say "I'm giving you the handy tagged gadget from the Notices Section.";
		now player has tagged gadget;
	the rule succeeds;

chapter j2ing

j2ing is an action applying to nothing.

understand the command "j2" as something new.

understand "j2" as j2ing.

carry out j2ing:
	if trips strip is visited:
		say "Already beat intro." instead;
	increase the cur-score of intro by 4;
	solve-region intro;
	say "You're playing without the gadget.";
	now gadget is in lalaland;
	the rule succeeds;

chapter ffing

ffing is an action applying to nothing.

understand the command "ff" as something new.

understand "ff" as ffing.

carry out ffing:
	say "Executing heinous cheat--doing so is a [bug-report] in the final release version. Got gadget now.";
	now player has tagged;
	the rule succeeds;

chapter rjing

rjing is an action applying to one number.

understand the command "rj [number]" as something new.

understand "rj [number]" as rjing.

[the magic numbers below are bad but they work. Basically, -1/1 toggles if forest is solved. -2/2 toggles if sortie is solved. -3/3 toggles if metros is solved.]

carry out rjing:
	let rj be the number understood;
	if rj < 1 or rj > 4:
		say "Region 1-4 please. 1=store 6 2=store 9 3=store 13 4=store 18." instead;
	say "(skipping introductory text with hyperwarp)";
	if rj is 1:
		move player to sf;
	if rj is 2:
		move player to ufc;
	if rj is 3:
		move player to undesired underside;
	if rj is 4:
		move player to tiles;

chapter sking

sking is an action applying to one number.

understand the command "sk [number]" as something new.

understand "sk [number]" as sking.

understand the command "sa [number]" as something new.

understand "sa [number]" as sking.

[the magic numbers below are bad but they work. Basically, -1/1 toggles if forest is solved. -2/2 toggles if sortie is solved. -3/3 toggles if metros is solved.]

carry out sking:
	let A be false;
	let sk be the number understood;
	let my-reg be forest;
	if sk < 0:
		now A is true;
		change sk to 0 - sk;
	if sk > 3:
		say "Only 3 sections to change: forest, sortie, metros." instead;
	if sk is 1:
		now my-reg is forest;
		if forest-sc is available:
			now forest-sc is checkedoff;
		otherwise:
			now forest-sc is available;
	if sk is 2:
		now my-reg is sortie;
		if sortie-x is available:
			now sortie-x is checkedoff;
		otherwise:
			now sortie-x is available;
	if sk is 3:
		now my-reg is metros;
		if metros-x is available:
			now metros-x is checkedoff;
		otherwise:
			now metros-x is available;
	if A is true:
		now my-reg is unsolved;
	otherwise:
		now my-reg is solved;
	say "Now [my-reg] is [if A is true]not [end if]solved.";
	the rule succeeds;

chapter disasing

disasing is an action applying to nothing.

understand the command "disas" as something new.

understand "disas" as disasing.

carry out disasing:
	now shades are in location of player;
	now nose is in location of player;
	now beard is in location of player;
	the rule succeeds;

book tests

[be sure to do stuff when you're not in the right room. This is CRITICAL to testing.]

part areas

chapter player tests

section sortie-specific

test setup with "test intro/gonear poet/abstract poet to lalaland/purloin sack/abstract straw to sack/abstract missile to moor/purloin cask/abstract black door to moor/abstract silo to moor/purloin panel/put panel in sack/put panel on sack/attach panel to sack/put missile on straw/put missile in straw/attach missile to straw/put missile in silo/put missile on silo/attach missile to silo/put door in silo/put door on silo/put missile in silo/put missile on silo/attach missile to silo/" in subsite

test dial with "sortie/d/60/40/10/25/5/20/18/14/15/16" in trips strip

section as

test yack with "in/goat/north/gateman/babble" in subsite

test hinttest with "hint/y/ask me about 1 should not show/hint/hf/ask me about 2 should not show/hint/hd/ask me about 3 should show/hint/hf/ask me about 4 should show/hd/hint"

section big-picture

test win123 with "test intro/test forest/test sortie/test metros/test resort" in subsite.

test q12 with "test intro/sk 1/sk 2/test resort" in subsite.
test q13 with "test intro/sk 1/sk 3/test resort" in subsite.
test q21 with "test intro/sk 2/sk 1/test resort" in subsite.
test q23 with "test intro/sk 2/sk 3/test resort" in subsite.
test q31 with "test intro/sk 3/sk 1/test resort" in subsite.
test q32 with "test intro/sk 3/sk 2/test resort" in subsite.

test recuse-metros with "test intro/secure/test forest/test sortie/recuse/test resort" in subsite.
test recuse-sortie with "test intro/secure/test forest/test metros/recuse/test resort" in subsite.
test recuse-forest with "test intro/secure/test sortie/test metros/recuse/test resort" in subsite.

section talk-testing

test gman with "in/goat/north/gateman/ask about abc/ask about him" in subsite.

test alltalk with "ask about me/ask about xyz/in/goat/ask about me/ask about xyz/north/ask about me/ask about xyz/gateman/ask about me/ask about xyz/gonear deadbeat/ask about me/ask about xyz/gonear faeries/ask about me/ask about xyz/gonear night thing/ask about me/ask about xyz/gonear nerds/ask about me/ask about xyz/gonear woeful pat/ask about me/ask about xyz/ask about xyz/peasant/ask about me/ask about xyz/gonear talkers/ask about me/ask about xyz/gonear tiles/islet/cork/e/grips/ropes/slope/u/ask about me/ask about xyz/trio/ask about me/ask about xyz/potters/kilns/ask about me/ask about xyz/gonear frost forts/ask about me/ask about xyz/gonear frost forts/wolves/ask about me/ask about xyz/gonear turnstile/ask about me/ask about xyz/gonear liches/ask about me/ask about xyz/" in subsite.

test default-talk with "gonear red bull/ask about me/ask about xyz/gonear woeful pat/ask about me/ask about xyz" in subsite.

section rm-testing

test alldirs with "nodir ne/nodir se/nodir sw/nodir nw/rmneglect lalaland/allrooms"

test abouts with "about/credits/in/goat/about/credits"

section tester-scripts

test pl9m with "test intro/metros/d/beast/gardenia/x gardenia/n/velcro/ask deadbeat about himself/ask deadbeat about cake/e/ask faeries about clover/give clover to faeries/ask faeries about heaths/ask faeries about begonias/ask faeries about gardenias/x brocade/x scratchings/barcode/x flowers/w/get lost corn/w/x garbage/smell/smell garbage/smell tomato/beats/sword/e/e/sheath/w/w/e/n/throw tomato/s/s/w/throw tomato/get emitter/e/n/n/switch emitter/get cake/keycard/s/get corn/" in subsite

[u/u/jump/]

test pl9 with "x tag/x room/x air/get sign/get banner/get brazier/in/goat/exit/n/n/n/n/x tag/x room/north/x tag/x room/gateman/getaway/ask about goat/ask about toga/ask about xyzzy/ask about me/ask about him/ask about gate/ask about attics/ask about abcde/ask about gadget/get device/scan static/switch gadget/attics/in/x stores/pad stores/forest/pad stores/sortie/pad stores/metros/pad stores/resort/pad stores/enter sortie/w/n/turn dial/turn dial to 57/g/turn dial to 16/n/x room/x air/x me/kitchen/x animal/x fridge/lettuce/steak/tortilla/x tortilla/parmesan/put parmesan on steak/put steak on tortilla/x tortilla/put parmesan in steak/x tortilla/put steak on lettuce/x tortilla/put tortilla on lettuce/x tortilla/sauce/hotsauce/put sauce on lettuce/eat taco/smell taco/coat/scan me/x warts/scan warts/straw/e/s/x cask/open cask/sack/coat/door/black/black door" in subsite

[x cask/open cask/x hole/
get straw]

test pl8 with "j/sortie/d/e/n/ne/16/straw/get straw/e/sack/w/get straw/cask/put straw in cask/sack/e/n/w/kitchen/tortilla/sauce/hot sauce/lettuce/parmesan/steak/put lettuce on steak/put tortilla in lettuce//put tortilla on lettuce/put tortilla in lettuce/put sauce on lettuce/put steak in lettuce/coat/e/scrape wall/i/pad/s/pad" in subsite

test pl7 with "x brazier/m/f/r/hd/hf/hd/hf/in/x cat/pet cat/pet goat/gato/x gato/man/north/goat/go north/x new land/x gateway/open gateway/gate man/open gateway/in/l/x new land/x gadget/x slider/x flipside/x example/get slider/attack small slider/x device/swap/x device/swap/swap gadget/swap slider/swap slider/swap gadget/in/get a way/talk to man/ask about mumbletypeg/ask about toga/ask about goat/ask about slider/ask about gadget/ask about game/x gateway/attack gateman/attack old man/attack gadget/attack slider/attack almond/scan attic/scan attics/scan static/scan dollhouse/in/x trapdoor/x roster/m/f/r/forest/in/x leaves/m/f/r/smell/door/n/x leaves/x device/door/scan door/x music/x note/x musical note/scan bulge/scan bugle/bugle/play bugle/in/e/x shell/x bubble/x bucket/nose/shades/hook nose to shades/w/w/open sandwich/beard/hook shades to beard/wear beard/e/n/chisel/s/w/get chicken/get cow/put chicken in canister/put cow in canister" in subsite

section wc

chapter for-testers

test hb with "hd/z/l/hf/x me/l/hd/x myself/l/hf/xyzzy/l"

chapter win

test win with "test intro/test metros/test forest/test sortie/test resort" in subsite

chapter gbg

test g-1 with "jump/sing/listen/smell/wave/think/sleep/wake/x me/i/z/"

test g0 with "forest/store f/sortie/store i/metros/store m/resort/store r/tories/roster/1/2/3/x store/x stores/x mall/x malls/x strip"

test g1 with "goat/gateman/gateway/getaway/thorn/soil/silo/oils/lois/plane/panel/attics/static/ones/nose/shades/dashes/shotgun/noughts/wolves/vowels/spam/maps/chisel/liches/lich/beard/bread/x liver/livers/sliver/silver/spread/drapes"

test g2 with "cask/sack/cask/sack/6/9/13/18/19/16/straw/kitchen/the nick/moor/room/missile/smilies/anapest/peasant/haywall/hallway/coat/aco/skate/parmesan/tortilla/hot sauce/hotsauce/cathouse/cat house/lettuce/shoot button/hoots/trees/steer/reset"

test g3 with "drainage/gardenia/brocade/barcode/sheath/heaths/begonias/noise bag/dry cake/keycard/sword/words/tomato/motto/a motto/lost corn/controls/velcro/clover/beats/beast"

test g4 with "cork/rock/swing/wings/toolshed/toeholds/grips/sprig/ropes/spore/trio/riot/potters/protest/kilns/links/tug toe"

chapter intro

test intro with "in/goat/north/gateman/getaway/ask gateman about cabinet/get all/enter gate" in subsite

test inog with "in/goat/north/gateman/getaway/enter gate/enter gate" in subsite

chapter forest

test fo with "forest/forest/smell/n/s/e/w/enter door" in trips strip

test shades with "w/get bread/e/e/dashes/ones/shades/nose/beard/put shades on nose/put nose on beard/disas/put shades on beard/put nose on beard/disas/put nose on beard/put shades on beard/disas/put nose on beard/put shades on nose" in silent rut

test forest with "forest/enter forest/smell/n/e/s/w/door/bugle/play bugle/enter door/e/nose/shades/shotgun/put shades on nose/get nose and shotgun/w/w/get bread/get spam/beard/put nose on beard/wear beard/e/n/chisel/get chisel/s/w/peel chicken/peel cow/put chicken in canister/put cow in canister/spam/e/n/silver/sliver/silver/spread/silver/cut spread/silver/put silver in shotgun/maps/n/read maps/wolves/shoot wolves" in trips strip

chapter metros

test metbasic with "test intro/metros/g/g/gardenia/n/w/tomato/e/s/w/give tomato to thing/x mattress/e/n/n/switch emitter/get cake/keycard/s/velcro/controls" in subsite

test metros with "test metros-s-n1" in trips strip

test metros-s-n1 with "metros/enter metros/gardenia/n/w/tomato/e/s/w/throw tomato at night thing/search mattress/e/n/n/switch emitter/keycard/s/e/give gardenia to faeries/get brocade/get heaths/w/w/sheath/sword/barcode/put barcode on keycard/e/controls/velcro/wear velcro/s/e/ask nerds about darkness/w/s/opening/climb wall/put controls in opening/attack beast/e/resin/beast/sword/attack beast" in trips strip

test metros-s-n2 with "metros/enter metros/gardenia/n/w/tomato/e/s/w/throw tomato at night thing/search mattress/e/n/n/switch emitter/keycard/s/e/give gardenia to faeries/get brocade/get heaths/w/w/sheath/sword/barcode/put barcode on keycard/e/show emitter to deadbeat/controls/velcro/wear velcro/s/e/switch emitter/s/opening/climb wall/put controls in opening/attack beast/e/resin/beast/sword/attack beast" in trips strip

test metros-n-n1 with "metros/enter metros/gardenia/n/w/tomato/e/s/w/throw tomato at night thing/search mattress/e/n/n/switch emitter/keycard/s/e/give gardenia to faeries/get brocade/get begonias/w/w/noise bag/open noise bag/barcode/put barcode on keycard/e/controls/velcro/wear velcro/s/e/ask nerds about darkness/w/s/opening/climb wall/put controls in opening/attack beast/e/resin/beast/sword/attack beast" in trips strip

test metros-n-n2 with "metros/enter metros/gardenia/n/w/tomato/e/s/w/throw tomato at night thing/search mattress/e/n/n/switch emitter/keycard/s/e/give gardenia to faeries/get brocade/get begonias/w/w/noise bag/open noise bag/barcode/put barcode on keycard/e/show emitter to deadbeat/controls/velcro/wear velcro/s/e/switch emitter/w/s/opening/climb wall/put controls in opening/attack beast/e/resin/beast/sword/attack beast" in trips strip

test metros-n-n3 with "metros/enter metros/gardenia/n/w/tomato/e/s/w/throw tomato at night thing/search mattress/e/n/n/switch emitter/keycard/s/e/give gardenia to faeries/get brocade/get begonias/w/w/noise bag/open noise bag/barcode/put barcode on keycard/e/controls/velcro/wear velcro/s/e/open bag/s/opening/climb wall/put controls in opening/attack beast/e/resin/beast/sword/attack beast" in trips strip

[get sword/resin/sword/beast/attack beast]

[resin/sword/attack beast"]

chapter sortie

test sortie-e with "j/sortie/d" in trips strip

test sortie with "sortie/enter sortie/sixteen/n/kitchen/s/e/straw/sack/put straw in sack/get sack/n/w/tortilla/lettuce/steak/parmesan/hotsauce/put hotsauce in tortilla/put steak on tortilla/put lettuce on tortilla/put parmesan on tortilla/coat/wear coat/e/s/shoes/moor/peasant/give straw/room/n/put hay in wall/hallway/e/cask/fill cask/w/s/moor/soil/room/n/e/fill cask/w/s/moor/silo/missile/black door/fold poem/panel/put door on silo/put missile in silo/put panel on silo/shoot button/steer button/push steer button/push shoot button" in trips strip

test sortie-delete with "sortie/enter sortie/sixteen/n/kitchen/s/e/straw/sack/put straw in sack/get sack/n/w/tortilla/lettuce/steak/parmesan/hotsauce/put hotsauce in tortilla/put steak on tortilla/put lettuce on tortilla/put parmesan on tortilla/coat/wear coat/e/s/moor/peasant/give straw/room/n/put hay in wall/hallway/e/cask/fill cask/w/s/moor/soil/room/n/e/fill cask/w/s/moor/silo/missile/black door/fold poem/panel/put panel on silo/" in trips strip

test moor with "sortie/d/sixteen/e/straw/sack/put straw in sack/get sack/n/w/kitchen/tortilla/lettuce/steak/parmesan/hotsauce/put hotsauce in tortilla/put steak on tortilla/put lettuce on tortilla/put parmesan on tortilla/coat/wear coat/e/s/moor/peasant/give straw/room/n/put hay in wall/hallway/e/cask/fill cask/w/s/moor/soil/room/n/e/fill cask/w/s/moor/silo/missile/black door/fold poem/panel" in trips strip

chapter resort

test resort with "resort/enter resort/islet/cork/wings/e/slope/grips/toeholds/e/trio/potters/kilns/tug toe/e/exist" in trips strip

test r1 with "gonear tiles/islet/cork/e/toeholds/e/trio/potters/kilns/tug toe/e" in tiles

test r2 with "gonear tiles/islet/wings/e/slope/grips/e/toeholds/e/trio/kilns/potters/tug toe/e" in tiles

test toolclue with "scan tiles/swap/scan tiles/islet/cork/e/grips/ropes" in tiles

chapter bugs

book debug - not for release

understand "xxyyzz" as goat.


volume 3 - not for release

when play begins (this is the check objects and rooms rule):
	if undescribed-objects is 0:
		say "All objects accounted for and described.";
	if undescribed-rooms is 0:
		say "All rooms accounted for and described.";

chapter yacking

yacking is an action applying to nothing.

understand the command "yack" as something new.

understand "yack" as yacking.

carry out yacking:
	let cur-guy be talkers;
	let cur-line be 0;
	repeat through the table of subject-blather:
		if him-who entry is not visible:
			move him-who entry to location of player;
		say "ASK [him-who entry] ABOUT [person-subj entry]: [him-say entry][line break]";
	repeat through table of default-sub-blather:
		if him-who entry is not visible:
			move him-who entry to location of player;
		say "ASK [him-who entry] ABOUT cork: [him-say entry][line break]";
	repeat through the table of default-gen-blather:
		if default-talker entry is not visible:
			move default-talker entry to location of player;
			say "ASK [default-talker entry] ABOUT mumblety-peg: [gen-blah entry][line break]";
	repeat through the table of general-blather:
		if him-who entry is not visible:
			move him-who entry to location of player;
		if him-who entry is not cur-guy:
			now cur-guy is him-who entry;
			now cur-line is 0;
		increment cur-line;
		say "ASK [him-who entry] ABOUT (topic [cur-line]): [him-say entry][line break]";
	the rule succeeds;

chapter unsolve

to unsolve-metros:
	now drainage is in underside;
	now gardenia is off-stage;
	now brocade is in elf row's flowers;
	now barcode is off-stage;
	now heaths are in elf row's flowers;
	now sheath is off-stage;
	now begonias are in elf row's flowers;
	now noise bag is off-stage;
	now words are in aya;
	now sword is off-stage;
	now motto is in aya;
	now tulip is in spis;
	now night thing is in toh;
	now termite emitter is off-stage;
	now termite emitter is not angstgnatted;
	now tomato are off-stage;
	now lost corn is off-stage;
	now controls are off-stage;
	now red camp is unvisited;
	now resin is off-stage;
	now siren is in Bassy;
	now beats are in metros;
	now beast is off-stage;
	now clover is in camp;
	now velcro is off-stage;
	now metros is unsolved;
	now cur-score of metros is 0;
	
to unsolve-sortie:
	now nick is unvisited;
	now numset of dial is 0;
	now sortie is unsolved;
	now cur-score of sortie is 0;
	
to unsolve-forest:
	now odor is in forest-sc;
	now bulge is part of rut-door;
	now bugle is off-stage;
	now nose is off-stage;
	now beard is off-stage;
	now shades are off-stage;
	now sandwich is in storing area;
	now bread is part of sandwich;
	now spam is part of sandwich;
	now maps are off-stage;
	now liches are in coe;
	now chisel is off-stage;
	now spread is in coe;
	now vowels are in frost forts;
	now wolves are off-stage;
	now livers are off-stage;
	now silver is off-stage;
	now sliver is off-stage;
	now forest is unsolved;
	now cur-score of forest is 0;
	
chapter uaing

uaing is an action applying to one number.

understand the command "ua [number]" as something new.

understand "ua [number]" as uaing.

carry out uaing:
	if trips strip is not visited:
		say "You need to visit the trips strip before using this." instead;
	if the number understood is 3:
		say "Resetting metros.";
		move player to trips strip, without printing a room description;
		unsolve-metros instead;
	if the number understood is 2:
		say "Resetting sortie.";
		move player to trips strip, without printing a room description;
		unsolve-sortie instead;
	if the number understood is 1:
		say "Resetting forest.";
		move player to trips strip, without printing a room description;
		unsolve-forest instead;
	say "Invalid number. 1=forest 2=sortie 3=metros";
	the rule succeeds;



chapter babbleing

babbleing is an action applying to nothing.

understand the command "babble" as something new.

understand "babble" as babbleing.

carry out babbleing: [Any way to pull out the topic?]
	let X be indexed text;
	repeat through table of general-blather:
		say "============[location of him-who entry]: [him-who entry].";
		move player to location of him-who entry, without printing a room description;
		try asking him-who entry about topic entry;
	the rule succeeds;

chapter booting

booting is an action applying to one visible thing.

understand the command "boot [any thing]" as something new.

understand "boot [any thing]" as booting.

carry out booting:
	now noun is in lalaland;
	the rule succeeds;

chapter feating

feating is an action applying to one visible thing.

understand the command "feat [any thing]" as something new.

understand "feat [any thing]" as feating.

carry out feating:
	let Q be the number of things that anagram the noun;
	if Q is 0:
		say "Warning no anagrams.";
	otherwise:
		now all things that anagram noun are in lalaland;
		say "[list of all things that anagram noun] swept away.";
	now noun is in location of player;
	say "[noun] is in the room.";
	the rule succeeds;

chapter pluraling

pluraling is an action applying to nothing.

understand the command "plural" as something new.

understand "plural" as pluraling.

carry out pluraling:
	repeat with Q running through things:
		say "[Q] = [if Q is plural-named]plural[otherwise]singular[end if] and [if Q is fixed in place]fixed[otherwise]takeable[end if].";
	the rule succeeds;

chapter scanalling

scanalling is an action applying to nothing.

understand the command "scanall" as something new.

understand "scanall" as scanalling.

carry out scanalling:
	let yay be 0;
	let boo be 0;
	let Z be indexed text;
	now player has tagged gadget;
	repeat with Q running through not inflexible things:
		if Q is not off-stage and Q is not in lalaland and Q is not a backdrop:
			let Y be the map region of the location of Q;
			if Y is sortie or Y is forest or Y is resort or Y is intro or Y is stores or Y is metros:
				say "[Q]: [rgtext of Q].";
				now Z is "[rgtext of Q]";
				if Z matches the text "R":
					increment yay;
				otherwise:
					increment boo;
					say "Boo, [Q].";
	say "[yay] yay, [boo] boo.";
	the rule succeeds;

chapter debuging

debuging is an action applying to nothing.

understand the command "debug" as something new.

understand "debug" as debuging.

carry out debuging:
	if debug-state is true:
		say "Debugging off.";
		now debug-state is false;
	otherwise:
		say "Debugging on.";
		now debug-state is true;
	the rule succeeds;

book notes to delete before release

[splinter]

[parotic apricot]

[rebuff buffer]

[offed/deffo]

["yez didn't duzzes nuffink boss"]

[leading aligned]

[bug hay cask sack spills out ??]

[post release
A B or A or B
1. d -> d is shorter
]